MKS Vector Calculus

Based on MKS Tutorials Playlist Vector Calculus

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

(%i2) `reset()$kill(all)$`

(%i1) `derivabbrev:true$`

(%i2) `ratprint:false$`

(%i3) `fpprintprec:5$`

(%i4) `load(linearalgebra)$`

(%i5) `if get('draw,'version)=false then load(draw)$`

(%i6) `wxplot_size:[1024,768]$`

(%i7) `if get('drawdf,'version)=false then load(drawdf)$`

(%i8)
```
set_draw_defaults(xtics=1,ytics=1,ztics=1,xyplane=0,nticks=100,
xaxis=true,xaxis_type=solid,xaxis_width=3,
yaxis=true,yaxis_type=solid,yaxis_width=3,
zaxis=true,zaxis_type=solid,zaxis_width=3,
background_color=light_gray)$
```

(%i9) `if get('vect,'version)=false then load(vect)$`

(%i10) `norm(u):=block(ratsimp(radcan(√(u.u))))$`

(%i11) `normalize(v):=block(v/norm(v))$`

(%i12) `angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$`

(%i13) `mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$`

(%i14) `if get('cartan,'version)=false then load(cartan)$`

(%i15) `if get('format,'version)=false then load(format)$`

(%i16) `declare(trigsimp,evfun)$`

1 Gradient of a Vector

Based on MKS Tutorials Video Gradient of a Vector

(%i17) `kill(x,y,z)$`

(%i18) `ζ:[x,y,z]$`

(%i19) `scalefactors(ζ)$`

(%i20) `init_cartan(ζ)$`

Find $\nabla(x^2yz)$

(%i21) `ldisplay(f:x^2*y*z)$`

$$f = x^2yz \tag{\%t21}$$

(%i22) `ldisplay(gradf:ev(express(grad(f)),diff))$`

$$gradf = [2xyz, x^2z, x^2y] \tag{\%t22}$$

(%i23) `ldisplay(df:edit(ext_diff(f)))$`

$$df = x^2y\ dz + x^2z\ dy + 2xyz\ dx \tag{\%t23}$$

2 Directional Derivative

Based on MKS Tutorials Video Directional Derivative

Directional Derivative

$$\frac{\mathrm{d}\phi}{\mathrm{d}s} = \hat{a} \cdot \nabla\phi$$

Divergence

$$\nabla \cdot \vec{F}$$

Curl

$$\nabla \times \vec{F}$$

3 Directional Derivative Problem #1

Based on MKS Tutorials Video Directional Derivative Problem # 1

Find the directional derivative of $\phi = 3x^2yz - 4y^2z^3$ in the direction of the vector $3\hat{\imath} - 4\hat{\jmath} + 2\hat{k}$ at point $(2, -1, 3)$.

(%i24) `ldisplay(ϕ:3*x^2*y*z-4*y^2*z^3)$`

$$\phi = 3x^2yz - 4y^2\,z^3 \tag{\%t24}$$

(%i25) `ldisplay(gradϕ:ev(express(grad(ϕ)),diff))$`

$$grad\phi = [6xyz, 3x^2z - 8y\,z^3, 3x^2y - 12y^2\,z^2] \tag{\%t25}$$

(%i26) `ldisplay(dϕ:edit(ext_diff(ϕ)))$`

$$d\phi = \left(3x^2y - 12y^2\,z^2\right)\;dz + \left(3x^2z - 8y\,z^3\right)\;dy + 6xyz\;dx \tag{\%t26}$$

(%i27) `ldisplay(a:[3,-4,2])$`

$$a = [3, -4, 2] \tag{\%t27}$$

(%i28) `ldisplay(P:[2,-1,3])$`

$$P = [2, -1, 3] \tag{\%t28}$$

(%i29) `ldisplay(D:factor(normalize(a).gradϕ))$`

$$D = \frac{2\left(16y\,z^3 - 12y^2\,z^2 + 9xyz - 6x^2z + 3x^2y\right)}{\sqrt{29}} \tag{\%t29}$$

(%i30) `ldisplay(D:factor(normalize(a)|dϕ))$`

$$D = \frac{2\left(16y\,z^3 - 12y^2\,z^2 + 9xyz - 6x^2z + 3x^2y\right)}{\sqrt{29}} \tag{\%t30}$$

(%i31) `ldisplay(D_p:at(D,map("=",ζ,P)))$`

$$D_p = -\frac{1356}{\sqrt{29}} \tag{\%t31}$$

4 Directional Derivative Problem #2

Based on MKS Tutorials Video Directional Derivative Problem # 2

Find the directional derivative of $\phi = x^2 - 2y^2 + 4z^2$ at the point $(1, 1, -1)$ in the direction of $2\hat{\imath} + \hat{\jmath} - \hat{k}$.

(%i32) `ldisplay(ϕ:x^2-2*y^2+4*z^2)$`

$$\phi = 4z^2 - 2y^2 + x^2 \tag{\%t32}$$

(%i33) `ldisplay(gradϕ:ev(express(grad(ϕ)),diff))$`

$$grad\phi = [2x, -4y, 8z] \tag{\%t33}$$

(%i34) `ldisplay(dϕ:edit(ext_diff(ϕ)))$`

$$d\phi = 8z\;dz - 4y\;dy + 2x\;dx \tag{\%t34}$$

(%i35) `ldisplay(a:[2,1,-1])$`

$$a = [2, 1, -1] \tag{\%t35}$$

(%i36) `ldisplay(P:[1,1,-1])$`

$$P = [1, 1, -1] \tag{\%t36}$$

(%i37) `ldisplay(D:factor(normalize(a).gradϕ))$`

$$D = -\frac{2^{\frac{3}{2}}\;(2z + y - x)}{\sqrt{3}} \tag{\%t37}$$

(%i38) `ldisplay(D:factor(normalize(a)|dϕ))$`

$$D = -\frac{2^{\frac{3}{2}}\;(2z + y - x)}{\sqrt{3}} \tag{\%t38}$$

(%i39) `ldisplay(D_p:at(D,map("=",ζ,P)))$`

$$D_p = \frac{2^{\frac{5}{2}}}{\sqrt{3}} \tag{\%t39}$$

5 Directional Derivative Problem #3

Based on MKS Tutorials Video Directional Derivative Problem # 3

What is the directional derivative of $\phi = xy^2 + yz^3$ at the point $(2, -1, 1)$ in the direction of normal to the surface $x \log z - y^2 = 4$ at $(-1, 2, 1)$.

(%i40) `ldisplay(ϕ:x*y^2+y*z^3)$`

$$\phi = y\,z^3 + x\,y^2 \tag{\%t40}$$

(%i41) `ldisplay(P:[2,-1,1])$`

$$P = [2, -1, 1] \tag{\%t41}$$

(%i42) `ldisplay(gradϕ:ev(express(grad(ϕ)),diff))$`

$$grad\phi = [y^2, z^3 + 2xy, 3y\,z^2] \tag{\%t42}$$

(%i43) `ldisplay(dϕ:edit(ext_diff(ϕ)))$`

$$d\phi = 3y\,z^2\;dz + \left(z^3 + 2xy\right)\;dy + y^2\;dx \tag{\%t43}$$

(%i44) `ldisplay(S:x*log(z)-y^2=4)$`

$$x\,\log(z) - y^2 = 4 \tag{\%t44}$$

(%i45) `ldisplay(Q:[-1,2,1])$`

$$Q = [-1, 2, 1] \tag{\%t45}$$

(%i46) `ldisplay(a:ev(express(grad(lhs(S))),diff))$`

$$a = \left[\log(z), -2y, \frac{x}{z}\right] \tag{\%t46}$$

(%i47) `ldisplay(a_Q:at(a,map("=",ζ,Q)))$`

$$a_Q = [0, -4, -1] \tag{\%t47}$$

(%i48) `ldisplay(D:factor(normalize(a_Q).gradϕ))$`

$$D = -\frac{4z^3 + 3y\,z^2 + 8xy}{\sqrt{17}} \tag{\%t48}$$

(%i49) `ldisplay(D:factor(normalize(a_Q)|dϕ))$`

$$D = -\frac{4z^3 + 3y\,z^2 + 8xy}{\sqrt{17}} \tag{\%t49}$$

(%i50) `ldisplay(D_P:at(D,map("=",ζ,P)))$`

$$D_P = \frac{15}{\sqrt{17}} \tag{\%t50}$$

6 Directional Derivative Problem #4

Based on MKS Tutorials Video Directional Derivative Problem #4

The temperature of point in the space is given by $T = x^2 + y^2 - z$. A mosquito located at $(1, 1, 2)$ desires to fly in such a direction that it will get warm as soon as possible. In what direction should it move?

(%i51) `ldisplay(T:x^2+y^2-z)$`

$$T = -z + y^2 + x^2 \qquad (\%t51)$$

(%i52) `ldisplay(P:[1,1,2])$`

$$P = [1, 1, 2] \qquad (\%t52)$$

(%i53) `ldisplay(gradT:ev(express(grad(T)),diff))$`

$$gradT = [2x, 2y, -1] \qquad (\%t53)$$

(%i54) `ldisplay(dT:edit(ext_diff(T)))$`

$$dT = -dz + 2y\,dy + 2x\,dx \qquad (\%t54)$$

(%i55) `ldisplay(gradT_P:normalize(at(gradT,map("=",ζ,P))))$`

$$gradT_P = \left[\frac{2}{3}, \frac{2}{3}, -\frac{1}{3}\right] \qquad (\%t55)$$

7 Directional Derivative Problem #5

Based on MKS Tutorials Video Directional Derivative Problem # 5

Find the constants a and b so that the surface $ax^2 - byz = (a+2)x$ will be orthogonal to the surface $4x^2y + z^3 = 4$ at the point $(1, -1, 2)$.

(%i56) `kill(labels,a,b)$`

(%i1) `ldisplay(P:[1,-1,2])$`

$$P = [1, -1, 2] \tag{\%t1}$$

(%i2) `ldisplay(f_1:a*x^2-b*y*z=(a+2)*x)$`

$$a\,x^2 - byz = (a+2)\,x \tag{\%t2}$$

(%i3) `ldisplay(f_1:lhs(f_1)-rhs(f_1))$`

$$f_1 = -byz + a\,x^2 - (a+2)\,x \tag{\%t3}$$

(%i4) `ldisplay(gradf_1:ev(express(grad(f_1)),diff))$`

$$gradf_1 = [2ax - a - 2, -bz, -by] \tag{\%t4}$$

(%i5) `ldisplay(df_1:edit(ext_diff(f_1)))$`

$$df_1 = -by\;dz - bz\;dy + (2ax - a - 2)\;dx \tag{\%t5}$$

(%i6) `ldisplay(gradf_P_1:at(gradf_1,map("=",ζ,P)))$`

$$gradf\_P_1 = [a-2, -2b, b] \tag{\%t6}$$

(%i7) `ldisplay(f_2:4*x^2*y+z^3=4)$`

$$z^3 + 4x^2y = 4 \tag{\%t7}$$

(%i8) `ldisplay(f_2:lhs(f_2)-rhs(f_2))$`

$$f_2 = z^3 + 4x^2y - 4 \tag{\%t8}$$

(%i9) `ldisplay(gradf_2:ev(express(grad(f_2)),diff))$`

$$gradf_2 = [8xy, 4x^2, 3z^2] \tag{\%t9}$$

(%i10) `ldisplay(df_2:edit(ext_diff(f_2)))$`

$$df_2 = 3z^2\;dz + 4x^2\;dy + 8xy\;dx \tag{\%t10}$$

(%i11) `ldisplay(gradf_P_2:at(gradf_2,map("=",ζ,P)))$`

$$gradf\_P_2 = [-8, 4, 12] \tag{\%t11}$$

The given surfaces intersect orthogonally at point $(1, -1, 2)$

(%i12) `ldisplay(Eq1:gradf_P_1.gradf_P_2=0)$`

$$4b - 8(a - 2) = 0 \tag{\%t12}$$

Also, the point $(1, -1, 2)$ lies in both sufaces.

(%i13) `ldisplay(Eq2:at(f_1,map("=",ζ,P))=0)$`

$$2b - 2 = 0 \tag{\%t13}$$

(%i14) `linsol:linsolve([Eq1,Eq2],[a,b]);`

$$\left[a = \frac{5}{2}, b = 1\right] \tag{linsol}$$

8 Divergence and Curl Problem #1

Based on MKS Tutorials Video Divergence and Curl Problem # 1

Find the divergence and curl of $\vec{F} = 3x^2\hat{\imath} + 5xy^2\hat{\jmath} + xyz^3\hat{k}$ at point $(1, 2, 3)$

(%i15) `ldisplay(P:[1,2,3])$`

$$P = [1, 2, 3] \tag{\%t15}$$

(%i16) `ldisplay(F:[3*x^2,5*x*y^2,x*y*z^3])$`

$$F = [3x^2, 5x\, y^2, xy\, z^3] \tag{\%t16}$$

$\nabla \times \vec{F}$

(%i17) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [x\, z^3, -y\, z^3, 5y^2] \tag{\%t17}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i18) `ldisplay(α:edit(F.cartan_basis))$`

$$\alpha = xy\, z^3\; dz + 5x\, y^2\; dy + 3x^2\; dx \tag{\%t18}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i19) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = x\, z^3\; dy\; dz + y\, z^3\; dx\; dz + 5y^2\; dx\; dy \tag{\%t19}$$

$\nabla \cdot \vec{F}$

(%i20) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = 3xy\, z^2 + 10xy + 6x \tag{\%t20}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i21)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = 3x^2\; dy\; dz - 5x\, y^2\; dx\; dz + xy\, z^3\; dx\; dy \tag{\%t21}$$

$\mathrm{d}\beta \in \mathcal{A}^3(\mathbb{R}^3)$

(%i22) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = \left(3xy\, z^2 + 10xy + 6x\right)\; dx\; dy\; dz \tag{\%t22}$$

At P

(%i23) `ldisplay(divF_P:at(divF,map("=",ζ,P)))$`

$$divF_P = 80 \tag{\%t23}$$

(%i24) `ldisplay(curlF_P:at(curlF,map("=",ζ,P)))$`

$$curlF_P = [27, -54, 20] \tag{\%t24}$$

9 Vector Calculus Problem #1

Based on MKS Tutorials Video Vector Calculus Problem # 1

Show that $\nabla^2 r^n = n(n+1)r^{n-2}$

(%i25) `scalefactors(spherical)$`

(%i26) `declare(n,integer)$`

(%i27) `ev(express(laplacian(r^n)),diff);`

$$n\,(n+1)\,r^{n-2} \tag{\%o27}$$

(%i28) `diff(x^n,x,2);`

$$(n-1)\,n\,x^{n-2} \tag{\%o28}$$

10 Vector Calculus Problem #2

Based on MKS Tutorials Video Vector Calculus Problem # 2

If $f = (x^2 + y^2 + z^2)^{-n}$, find $\nabla \cdot (\nabla f)$ and determine n if $\nabla \cdot (\nabla f) = 0$.

(%i29) `kill(f)$`

(%i30) `scalefactors(ζ)$`

(%i31) `ldisplay(f:(x^2+y^2+z^2)^(-n))$`

$$f = \frac{1}{(z^2 + y^2 + x^2)^n} \tag{\%t31}$$

(%i32) `ldisplay(lapf:ev(express(laplacian(f)),diff,factor))$`

$$lapf = 2n\,(2n-1)\left(z^2 + y^2 + x^2\right)^{-n-1} \tag{\%t32}$$

(%i33) `solve(lapf,n);`

$$\left[n = \frac{1}{2}, n = 0\right] \tag{\%o33}$$

11 Vector Calculus Problem #3

Based on MKS Tutorials Video Vector Calculus Problem # 3

Show that $\nabla^2 f(r) = f'' + \frac{2}{r} f'(r)$

(%i34) `kill(f)$`

(%i35) `scalefactors(spherical)$`

(%i36) `depends(f,r)$`

(%i37) `declare(f,scalar)$`

(%i38) `ldisplay(lapf:ev(express(laplacian(f)),diff,expand))$`

$$lapf = \frac{2\left(f_r\right)}{r} + f_{rr} \tag{\%t38}$$

12 Line Integrals Problem #1

Based on MKS Tutorials Video Line Integrals Problem # 1

Evaluate $\int_C \vec{F} \cdot \mathrm{d}\vec{r}$ where $\vec{F} = (x^2 - y^2)\hat{\imath} + xy\hat{\jmath}$ and C is the arc of curve $y = x^3$ in the xy plane from $(0,0)$ to $(2,8)$.

(%i39) `kill(labels,x,y)$`

(%i1) `ζ:[x,y]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

(%i4) `ldisplay(F:[x^2-y^2,x*y])$`

$$F = [x^2 - y^2, xy] \tag{\%t4}$$

(%i6) `A:[0,0]$B:[2,8]$`

(%i7) `ldisplay(C:[t,t^3])$`

$$C = [t, t^3] \tag{\%t7}$$

(%i8) `ldisplay(C\':diff(C,t))$`

$$C' = [1, 3t^2] \tag{\%t8}$$

(%i9) `ldisplay(FoC:subst(map("=",ζ,C),F))$`

$$FoC = [t^2 - t^6, t^4] \tag{\%t9}$$

(%i10) `ldisplay(integrand:factor(FoC.C\'))$`

$$integrand = t^2\left(2t^4 + 1\right) \tag{\%t10}$$

(%i11) `I:'integrate(integrand,t,0,2)$`

(%i12) `ldisplay(I=box(ev(I,integrate)))$`

$$\int_0^2 t^2\left(2t^4 + 1\right) dt = \left(\frac{824}{21}\right) \tag{\%t12}$$

13 Line Integrals Problem #2

Based on MKS Tutorials Video Line Integrals Problem # 2

Evaluate $\int_C \vec{F} \cdot d\vec{r}$ where $\vec{F} = (x^2 + y^2)\hat{i} - 2xy\hat{j}$ and C is the rectangle in the xy plane bounded by $y = 0$, $x = a$, $y = b$ and $x = 0$.

(%i17) `kill(labels,x,y,I,a,b)$`

(%i1) `ζ:[x,y]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

(%i4) `ldisplay(F:[x^2+y^2,-2*x*y])$`

$$F = [y^2 + x^2, -2xy] \qquad (\%t4)$$

End points

(%i8) `O:[0,0]$A:[a,0]$B:[a,b]$C:[0,b]$`

(%i16)
```
C_1:expand(t*A+(1-t)*O)$L\'_1:diff(C_1,t)$
C_2:expand(t*B+(1-t)*A)$L\'_2:diff(C_2,t)$
C_3:expand(t*C+(1-t)*B)$L\'_3:diff(C_3,t)$
C_4:expand(t*O+(1-t)*C)$L\'_4:diff(C_4,t)$
```

(%i24)
```
FoC_1:subst(map("=",ζ,C_1),F)$integrand_1:FoC_1.L\'_1$
FoC_2:subst(map("=",ζ,C_2),F)$integrand_2:FoC_2.L\'_2$
FoC_3:subst(map("=",ζ,C_3),F)$integrand_3:FoC_3.L\'_3$
FoC_4:subst(map("=",ζ,C_4),F)$integrand_4:FoC_4.L\'_4$
```

(%i28)
```
I_1:'integrate(integrand_1,t,0,1)$
I_2:'integrate(integrand_2,t,0,1)$
I_3:'integrate(integrand_3,t,0,1)$
I_4:'integrate(integrand_4,t,0,1)$
```

(%i32)
```
ldisplay(I_1=box(ev(I_1,integrate,expand)))$
ldisplay(I_2=box(ev(I_2,integrate,expand)))$
ldisplay(I_3=box(ev(I_3,integrate,expand)))$
ldisplay(I_4=box(ev(I_4,integrate,expand)))$
```

$$a^3 \int_0^1 t^2 dt = \left(\frac{a^3}{3}\right) \qquad (\%t29)$$

$$-2a\, b^2 \int_0^1 t dt = \left(-a\, b^2\right) \qquad (\%t30)$$

$$-a \int_0^1 (a - at)^2 + b^2 dt = \left(-a\, b^2 - \frac{a^3}{3}\right) \qquad (\%t31)$$

$$0 = (0) \qquad (\%t32)$$

(%i33) `ldisplay(I=box(ev(I_1+I_2+I_3+I_4,integrate,expand)))$`

$$I = \left(-2a\, b^2\right) \qquad (\%t33)$$

14 Surface Integrals Problem #1

Based on MKS Tutorials Video Surface Integrals Problem # 1

Evaluate $\iint_S \vec{F} \cdot \hat{n} ds$ where $\vec{F} = 4x\hat{i} - 2y\hat{j} + z^2\hat{k}$ is taken in the region bounded by $x^2 + y^2 = 4$, $z = 0$ and $z = 3$.

(%i31)
```
kill(labels,x,y,z,I,a,b)$
```

Define the space $\mathbb{R}^3$

(%i1)
```
ζ:[x,y,z]$
```

(%i2)
```
scalefactors(ζ)$
```

(%i3)
```
init_cartan(ζ)$
```

Parameters

(%i4)
```
assume(R>0,h>0)$
```

(%i5)
```
declare([R,h],constant)$
```

(%i6)
```
params:[R=2,h=3]$
```

Vector field $F \in \mathbb{R}^3$

(%i7)
```
ldisplay(F:[4*x,-2*y,z^2])$
```

$$F = [4x, -2y, z^2] \tag{\%t7}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i8)
```
ldisplay(α:edit(F.cartan_basis))$
```

$$\alpha = z^2\, dz - 2y\, dy + 4x\, dx \tag{\%t8}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i9)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = 4x\, dy\, dz + 2y\, dx\, dz + z^2\, dx\, dy \tag{\%t9}$$

Parametrized surfaces

```
(%i12) ldisplay(S_1:[r*cos(θ),r*sin(θ),0])$
       ldisplay(S_2:[r*cos(θ),r*sin(θ),h])$
       ldisplay(S_3:[R*cos(θ),R*sin(θ),z])$
```

$$S_1 = [r\cos(\theta), r\sin(\theta), 0] \quad (\%t10)$$

$$S_2 = [r\cos(\theta), r\sin(\theta), h] \quad (\%t11)$$

$$S_3 = [R\cos(\theta), R\sin(\theta), z] \quad (\%t12)$$

Integrand according to vector calculus

```
(%i15) ldisplay(integrand_1:at(F,map("=",ζ,S_1)).mycross(diff(S_1,r),diff(S_1,θ)))$
       ldisplay(integrand_2:trigsimp(at(F,map("=",ζ,S_2)).mycross(diff(S_2,θ),diff(S_2,r))))$
       ldisplay(integrand_3:factor(trigsimp(at(F,map("=",ζ,S_3)).mycross(diff(S_3,θ),diff(S_3,z)))))$
```

$$integrand_1 = 0 \quad (\%t13)$$

$$integrand_2 = -h^2 r \quad (\%t14)$$

$$integrand_3 = -2R^2\left(3\sin(\theta)^2 - 2\right) \quad (\%t15)$$

Integrand according to differential forms

```
(%i18) ldisplay(integrand_1:diff(S_1,θ)|(diff(S_1,r)|at(β,map("=",ζ,S_1))))$
       ldisplay(integrand_2:trigsimp(diff(S_2,r)|(diff(S_2,θ)|at(β,map("=",ζ,S_2)))))$
       ldisplay(integrand_3:factor(trigsimp(diff(S_3,z)|(diff(S_3,θ)|at(β,map("=",ζ,S_3))))))$
```

$$integrand_1 = 0 \quad (\%t16)$$

$$integrand_2 = -h^2 r \quad (\%t17)$$

$$integrand_3 = -2R^2\left(3\sin(\theta)^2 - 2\right) \quad (\%t18)$$

```
(%i21) I_1:'integrate('integrate(integrand_1,r,0,R),θ,0,2*π)$
       I_2:'integrate('integrate(integrand_2,r,0,R),θ,0,2*π)$
       I_3:'integrate('integrate(integrand_3,θ,0,2*π),z,0,h)$
(%i22) ldisplay(I_1=box(ev(I_1,integrate,params)))$
```

$$0 = \boxed{0} \quad (\%t22)$$

```
(%i23) ldisplay(I_2=box(ev(I_2,integrate,params)))$
```

$$-2\pi h^2 \int_0^R r\,dr = \boxed{-36\pi} \quad (\%t23)$$

```
(%i24) ldisplay(I_3=box(ev(I_3,integrate,params)))$
```

$$-2R^2 h \int_0^{2\pi} 3\sin(\theta)^2 - 2\,d\theta = \boxed{24\pi} \quad (\%t24)$$

```
(%i25) ldisplay(I=box(ev(I_1+I_2+I_3,integrate,params)))$
```

$$I = \boxed{-12\pi} \quad (\%t25)$$

15 Green's Theorem Problem #1

Based on MKS Tutorials Video Green's Theorem Problem # 1

If C be the vector closed curve in xy-plane bounding any region R and $f_1(x,y)$ and $f_2(x,y)$ be the continuous partial derivative $\frac{\partial f_1}{\partial x}$ and $\frac{\partial f_2}{\partial x}$ in $\mathbb{R}$, then

$$\oint_C (f_1\,\mathrm{d}x + f_2\,\mathrm{d}y) = \iint_R \left(\frac{\partial f_2}{\partial x} - \frac{\partial f_1}{\partial y}\right)\mathrm{d}x\,\mathrm{d}y$$

Verify Green's theorem for the following integral in xy-plane

$$\oint_C \left[(3x^2 - 8y^2)\,\mathrm{d}x + (4y - 6xy)\,\mathrm{d}y\right]$$

where C in the boundary of region bounded by the parabolas $y = \sqrt{x}$ and $y = x^2$.

(%i26) `kill(labels,t,x,y,I)$`

Define the space $\mathbb{R}^2$

(%i1) `ζ:[x,y]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^2$

(%i5)
```
ldisplay(f_1:3*x^2-8*y^2)$
ldisplay(f_2:4*y-6*x*y)$
```

$$f_1 = 3x^2 - 8y^2 \tag{\%t4}$$

$$f_2 = 4y - 6xy \tag{\%t5}$$

(%i6) `ldisplay(F:[f_1,f_2])$`

$$F = [3x^2 - 8y^2, 4y - 6xy] \tag{\%t6}$$

$\nabla \times \vec{F} \in \mathbb{R}^2$

(%i7) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = 10y \tag{\%t7}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i8) `ldisplay(α:F.cartan_basis)$`

$$\alpha = (4y - 6xy)\ dy + \left(3x^2 - 8y^2\right)\ dx \tag{\%t8}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^2)$

(%i9) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = 10y\, dx\, dy \tag{\%t9}$$

Curves $\vec{C} \in \mathbb{R}^2$

(%i11)
```
ldisplay(C_1:[t,t^2])$
ldisplay(C_2:[t^2,t])$
```

$$C_1 = [t, t^2] \quad (\%t10)$$

$$C_2 = [t^2, t] \quad (\%t11)$$

Integrand according to vector calculus

(%i13)
```
ldisplay(integrand_1:expand(subst(map("=",ζ,C_1),F).diff(C_1,t)))$
ldisplay(integrand_2:expand(subst(map("=",ζ,C_2),F).diff(C_2,t)))$
```

$$integrand_1 = -20t^4 + 8t^3 + 3t^2 \quad (\%t12)$$

$$integrand_2 = 6t^5 - 22t^3 + 4t \quad (\%t13)$$

Integrand according to differential forms

(%i15)
```
ldisplay(integrand_1:diff(C_1,t)|subst(map("=",ζ,C_1),α))$
ldisplay(integrand_2:diff(C_2,t)|subst(map("=",ζ,C_2),α))$
```

$$integrand_1 = -20t^4 + 8t^3 + 3t^2 \quad (\%t14)$$

$$integrand_2 = 6t^5 - 22t^3 + 4t \quad (\%t15)$$

Line integrals

(%i17)
```
I_1:'integrate(integrand_1,t,0,1)$
I_2:'integrate(integrand_2,t,1,0)$
```

(%i18)
```
ldisplay(I_1=box(ev(I_1,integrate)))$
```

$$\int_0^1 -20t^4 + 8t^3 + 3t^2 dt = \boxed{-1} \quad (\%t18)$$

(%i19)
```
ldisplay(I_2=box(ev(I_2,integrate)))$
```

$$-\int_0^1 6t^5 - 22t^3 + 4t dt = \boxed{\frac{5}{2}} \quad (\%t19)$$

(%i20)
```
ldisplay(I=box(ev(I_1+I_2,integrate)))$
```

$$I = \boxed{\frac{3}{2}} \quad (\%t20)$$

Use Green's theorem

Integrand according to vector calculus

(%i21) `integrand:curlF;`

$$10y \tag{integrand}$$

Integrand according to differential forms

(%i22) `integrand:diff(ζ,y)|(diff(ζ,x)|dα);`

$$10y \tag{integrand}$$

Double integral

(%i23) `I:'integrate('integrate(curlF,y,x^2,√(x)),x,0,1)$`

(%i24) `ldisplay(I=box(ev(I,integrate)))$`

$$10\int_0^1 \int_{x^2}^{\sqrt{x}} y\,dy\,dx = \boxed{\frac{3}{2}} \tag{\%t24}$$

(%i25)
```
wxdrawdf(F,[x,0,1],[y,0,1],color=red, line_width=2,field_color=green,
apply(parametric,append(C_1,[t,0,1])), apply(parametric,append(C_2,[t,0,1])),
color=black,font_size=15,font="Helvetica")$
```



(%t25)

16 Green's Theorem Problem #2

Based on MKS Tutorials Video Green's Theorem Problem # 2

Verify Green's theorem in the plane for $\oint_C \left[(xy+y^2)\,\mathrm{d}x + x^2\,\mathrm{d}y\right]$ where C is the closed curve of the region bounded by $y=x$ and $y=x^2$.

(%i26) `kill(labels,t,x,y,I)$`

Define the space $\mathbb{R}^2$

(%i1) `ζ:[x,y]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^2$

(%i5)
```
ldisplay(f_1:x*y+y^2)$
ldisplay(f_2:x^2)$
```

$$f_1 = y^2 + xy \tag{\%t4}$$

$$f_2 = x^2 \tag{\%t5}$$

(%i6) `ldisplay(F:[f_1,f_2])$`

$$F = [y^2 + xy, x^2] \tag{\%t6}$$

$\nabla \times \vec{F} \in \mathbb{R}^2$

(%i7) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = x - 2y \tag{\%t7}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i8) `ldisplay(α:F.cartan_basis)$`

$$\alpha = x^2\ dy + \left(y^2 + xy\right)\ dx \tag{\%t8}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^2)$

(%i9) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = (x - 2y)\ dx\ dy \tag{\%t9}$$

Curves $\vec{C} \in \mathbb{R}^2$

(%i11)
```
ldisplay(C_1:[t,t^2])$
ldisplay(C_2:[t,t])$
```

$$C_1 = [t, t^2] \qquad (\%t10)$$

$$C_2 = [t, t] \qquad (\%t11)$$

Integrand according to vector calculus

(%i13)
```
ldisplay(integrand_1:expand(subst(map("=",ζ,C_1),F).diff(C_1,t)))$
ldisplay(integrand_2:expand(subst(map("=",ζ,C_2),F).diff(C_2,t)))$
```

$$integrand_1 = t^4 + 3t^3 \qquad (\%t12)$$

$$integrand_2 = 3t^2 \qquad (\%t13)$$

Integrand according to differential forms

(%i15)
```
ldisplay(integrand_1:diff(C_1,t)|subst(map("=",ζ,C_1),α))$
ldisplay(integrand_2:diff(C_2,t)|subst(map("=",ζ,C_2),α))$
```

$$integrand_1 = t^4 + 3t^3 \qquad (\%t14)$$

$$integrand_2 = 3t^2 \qquad (\%t15)$$

Line integrals

(%i17)
```
I_1:'integrate(integrand_1,t,0,1)$
I_2:'integrate(integrand_2,t,1,0)$
```

(%i18)
```
ldisplay(I_1=box(ev(I_1,integrate)))$
```

$$\int_0^1 t^4 + 3t^3 dt = \boxed{\frac{19}{20}} \qquad (\%t18)$$

(%i19)
```
ldisplay(I_2=box(ev(I_2,integrate)))$
```

$$-3\int_0^1 t^2 dt = \boxed{-1} \qquad (\%t19)$$

(%i20)
```
ldisplay(I=box(ev(I_1+I_2,integrate)))$
```

$$I = \boxed{-\frac{1}{20}} \qquad (\%t20)$$

Use Green's theorem

Integrand according to vector calculus

(%i21) `integrand:curlF;`

$$x - 2y \tag{integrand}$$

Integrand according to differential forms

(%i22) `integrand:diff(ζ,y)|(diff(ζ,x)|dα);`

$$x - 2y \tag{integrand}$$

Double integral

(%i23) `I:'integrate('integrate(curlF,y,x^2,x),x,0,1)$`

(%i24) `ldisplay(I=box(ev(I,integrate)))$`

$$\int_0^1 \int_{x^2}^{x} x - 2y dy\, dx = \left(-\frac{1}{20}\right) \tag{\%t24}$$

(%i25)

```
wxdrawdf(F,[x,0,1],[y,0,1],color=red,
line_width=2,field_color=green,
apply(parametric,append(C_1,[t,0,1])),
apply(parametric,append(C_2,[t,0,1])),
color=black,font_size=15,font="Helvetica")$
```



(%t25)

17 Gauss's Divergence Theorem Problem #1

Based on MKS Tutorials Video Gauss's Divergence Theorem Problem # 1

Relation between surface and volume integrals

$$\iint_S \vec{F} \cdot \hat{n} \, \mathrm{d}s = \iiint_V \nabla \cdot \vec{F} \, \mathrm{d}V$$

From Gauss's Divergence theorem, find $\iint_S \vec{F} \cdot \hat{n} \, \mathrm{d}s$ where $\vec{F} = 4x\hat{i} - 2y^2\hat{j} + z^2\hat{k}$ is taken in the region bounded by $x^2 + y^2 = 4$, $z = 0$ and $z = 3$.

(%i26) `kill(labels,x,y,z,I,R,h,r,θ)$`

Define the space $\mathbb{R}^2$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Parameters

(%i4) `assume(R>0,h>0)$`

(%i5) `declare([R,h],constant)$`

(%i6) `params:[R=2,h=3]$`

Vector field $\vec{F} \in \mathbb{R}^2$

(%i7)
```
ldisplay(F:[4*x,-2*y^2,z^2])$
```

$$F = [4x, -2y^2, z^2] \tag{\%t7}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i8)
```
ldisplay(divF:ev(express(div(F)),diff))$
```

$$divF = 2z - 4y + 4 \tag{\%t8}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i9)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = 4x\ dy\ dz + 2y^2\ dx\ dz + z^2\ dx\ dy \tag{\%t9}$$

$\mathrm{d}\beta \in \mathcal{A}^3(\mathbb{R}^3)$

(%i10)
```
ldisplay(dβ:edit(ext_diff(β)))$
```

$$d\beta = (2z - 4y + 4)\ dx\ dy\ dz \tag{\%t10}$$

Polarcylindrical coordinates

(%i12)
```
assume(0≤r)$
assume(0≤θ,θ≤2*π)$
```

(%i13)
```
ξ:[r,θ,z]$
```

(%i14)
```
ldisplay(Tr:[r*cos(θ),r*sin(θ),z])$
```

$$Tr = [r\ \cos(\theta), r\ \sin(\theta), z] \tag{\%t14}$$

Jacobian of the transformation

(%i15)
```
ldisplay(J:trigsimp(determinant(jacobian(Tr,ξ))))$
```

$$J = r \tag{\%t15}$$

Surfaces $\vec{S} \in \mathbb{R}^3$

(%i18)
```
ldisplay(S_1:[r*cos(θ),r*sin(θ),0]) /* Bottom */$
ldisplay(S_2:[r*cos(θ),r*sin(θ),h]) /* Top */$
ldisplay(S_3:[R*cos(θ),R*sin(θ),z]) /* Walls */$
```

$$S_1 = [r\,\cos(\theta), r\,\sin(\theta), 0] \quad (\%t16)$$

$$S_2 = [r\,\cos(\theta), r\,\sin(\theta), h] \quad (\%t17)$$

$$S_3 = [R\,\cos(\theta), R\,\sin(\theta), z] \quad (\%t18)$$

Integrand according to vector calculus

(%i21)
```
ldisplay(integrand_1:trigsimp(subst(map("=",ζ,S_1),F).mycross(diff(S_1,r),diff(S_1,θ))))$
ldisplay(integrand_2:trigsimp(subst(map("=",ζ,S_2),F).mycross(diff(S_2,θ),diff(S_2,r))))$
ldisplay(integrand_3:trigsimp(subst(map("=",ζ,S_3),F).mycross(diff(S_3,θ),diff(S_3,z))))$
```

$$integrand_1 = 0 \quad (\%t19)$$

$$integrand_2 = -h^2 r \quad (\%t20)$$

$$integrand_3 = 4R^2\cos(\theta)^2 - 2R^3\sin(\theta)^3 \quad (\%t21)$$

Integrand according to differential forms

(%i24)
```
ldisplay(integrand_1:trigsimp(diff(S_1,θ)|(diff(S_1,r)|subst(map("=",ζ,S_1),β))))$
ldisplay(integrand_2:trigsimp(diff(S_2,r)|(diff(S_2,θ)|subst(map("=",ζ,S_2),β))))$
ldisplay(integrand_3:trigsimp(diff(S_3,z)|(diff(S_3,θ)|subst(map("=",ζ,S_3),β))))$
```

$$integrand_1 = 0 \quad (\%t22)$$

$$integrand_2 = -h^2 r \quad (\%t23)$$

$$integrand_3 = 4R^2\cos(\theta)^2 - 2R^3\sin(\theta)^3 \quad (\%t24)$$

Surface integrals

(%i27)
```
I_1:'integrate('integrate(integrand_1,r,0,R),θ,0,2*π)$
I_2:'integrate('integrate(integrand_2,θ,2*π,0),r,0,R)$
I_3:'integrate('integrate(integrand_3,θ,0,2*π),z,0,h)$
```

(%i30)
```
ldisplay(I_1=box(ev(I_1,integrate,params)))$
ldisplay(I_2=box(ev(I_2,integrate,params)))$
ldisplay(I_3=box(ev(I_3,integrate,params)))$
```

$$0 = (0) \quad (\%t28)$$

$$2\pi h^2 \int_0^R r\,dr = (36\pi) \quad (\%t29)$$

$$h \int_0^{2\pi} 4R^2\cos(\theta)^2 - 2R^3\sin(\theta)^3 d\theta = (48\pi) \quad (\%t30)$$

(%i31)
```
ldisplay(I=box(ev(I_1+I_2+I_3,integrate,params)))$
```

$$I = (84\pi) \quad (\%t31)$$

Using Gauss's Divergence Theorem

Integrand according to vector calculus

(%i32) `ldisplay(integrand:trigsimp(subst(map("=",ζ,Tr),divF)*J))$`

$$integrand = -4r^2 \sin(\theta) + 2rz + 4r \qquad (\%t32)$$

Integrand according to differential forms

(%i33) `ldisplay(integrand:trigsimp(diff(Tr,z)|(diff(Tr,θ)|(diff(Tr,r)|subst(map("=",ζ,Tr),dβ)))))$`

$$integrand = -4r^2 \sin(\theta) + 2rz + 4r \qquad (\%t33)$$

Volume integral

(%i34) `I:'integrate('integrate('integrate(integrand,r,0,R),θ,0,2*π),z,0,h)$`

(%i35) `ldisplay(I=box(ev(I,integrate,params)))$`

$$\int_0^h \int_0^{2\pi} \int_0^R -4r^2 \sin(\theta) + 2rz + 4r\,dr\,d\theta\,dz = (84\pi) \qquad (\%t35)$$

Clean up

(%i38)
```
forget(R>0,h>0)$
forget(0≤r)$
forget(0≤θ,θ≤2*π)$
```

3D Direction field

(%i40)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,-3,3))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i42)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F/15))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i43)
```
wxdraw3d(proportional_axes=xy,xu_grid=100,yv_grid=100,
xrange=[-R,R],yrange=[-R,R],zrange=[0,h],
color=cyan,apply(parametric_surface,append(S_3,[θ,0,2*π,z,0,h])),
color=green,apply(parametric_surface,append(S_2,[r,0,R,θ,0,2*π])),
color=blue,apply(parametric_surface,append(S_1,[r,0,R,θ,0,2*π])),
head_length=0.05,head_type='nofilled,line_width=2,color=red,vect3),params$
```



(%t43)

18 Gauss's Divergence Theorem Problem #2

Based on MKS Tutorials Video Gauss's Divergence Theorem Problem # 2

Use Gauss's Divergence theorem for $\vec{F} = (x^2 - yz)\hat{\imath} + (y^2 - zx)\hat{\jmath} + (z^2 - xy)\hat{k}$ taken over the rectangular parallelepiped $0 \le x \le a$, $0 \le y \le b$ and $0 \le z \le c$.

(%i44) `kill(labels,x,y,z,I,a,b,c,u,v)$`

Define the space $\mathbb{R}^2$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Parameters

(%i4) `assume(a>0,b>0,c>0)$`

(%i5) `declare([a,b,c],constant)$`

(%i6) `params:[a=5,b=6,c=4]$`

Vector field $\vec{F} \in \mathbb{R}^3$

(%i7) `ldisplay(F:[x^2-y*z,y^2-z*x,z^2-x*y])$`

$$F = [x^2 - yz, y^2 - xz, z^2 - xy] \tag{\%t7}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i8) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = 2z + 2y + 2x \tag{\%t8}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i9)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = \left(x^2 - yz\right)\ dy\,dz - \left(y^2 - xz\right)\ dx\,dz + \left(z^2 - xy\right)\ dx\,dy \tag{\%t9}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i10) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = (2z + 2y + 2x)\ dx\,dy\,dz \tag{\%t10}$$

Surfaces $\vec{S} \in \mathbb{R}^3$

```
(%i16) ldisplay(S_1:[0,u,v]) /* Left side */$
       ldisplay(S_2:[a,u,v]) /* Right side */$
       ldisplay(S_3:[u,0,v]) /* Front side */$
       ldisplay(S_4:[u,b,v]) /* Back side */$
       ldisplay(S_5:[u,v,0]) /* Bottom side */$
       ldisplay(S_6:[u,v,c]) /* Top side */$
```

$$S_1 = [0, u, v] \qquad (\%t11)$$

$$S_2 = [a, u, v] \qquad (\%t12)$$

$$S_3 = [u, 0, v] \qquad (\%t13)$$

$$S_4 = [u, b, v] \qquad (\%t14)$$

$$S_5 = [u, v, 0] \qquad (\%t15)$$

$$S_6 = [u, v, c] \qquad (\%t16)$$

Integrand according to vector calculus

```
(%i22) ldisplay(integrand_1:ratsimp(subst(map("=",ζ,S_1),F).mycross(diff(S_1,v),diff(S_1,u))))$
       ldisplay(integrand_2:ratsimp(subst(map("=",ζ,S_2),F).mycross(diff(S_2,u),diff(S_2,v))))$
       ldisplay(integrand_3:ratsimp(subst(map("=",ζ,S_3),F).mycross(diff(S_3,u),diff(S_3,v))))$
       ldisplay(integrand_4:ratsimp(subst(map("=",ζ,S_4),F).mycross(diff(S_4,v),diff(S_4,u))))$
       ldisplay(integrand_5:ratsimp(subst(map("=",ζ,S_5),F).mycross(diff(S_5,v),diff(S_5,u))))$
       ldisplay(integrand_6:ratsimp(subst(map("=",ζ,S_6),F).mycross(diff(S_6,u),diff(S_6,v))))$
```

$$integrand_1 = uv \qquad (\%t17)$$

$$integrand_2 = a^2 - uv \qquad (\%t18)$$

$$integrand_3 = uv \qquad (\%t19)$$

$$integrand_4 = b^2 - uv \qquad (\%t20)$$

$$integrand_5 = uv \qquad (\%t21)$$

$$integrand_6 = c^2 - uv \qquad (\%t22)$$

Integrand according to differential forms

```
(%i28) ldisplay(integrand_1:ratsimp(diff(S_1,u)|(diff(S_1,v)|subst(map("=",ζ,S_1),β))))$
       ldisplay(integrand_2:ratsimp(diff(S_2,v)|(diff(S_2,u)|subst(map("=",ζ,S_2),β))))$
       ldisplay(integrand_3:ratsimp(diff(S_3,v)|(diff(S_3,u)|subst(map("=",ζ,S_3),β))))$
       ldisplay(integrand_4:ratsimp(diff(S_4,u)|(diff(S_4,v)|subst(map("=",ζ,S_4),β))))$
       ldisplay(integrand_5:ratsimp(diff(S_5,u)|(diff(S_5,v)|subst(map("=",ζ,S_5),β))))$
       ldisplay(integrand_6:ratsimp(diff(S_6,v)|(diff(S_6,u)|subst(map("=",ζ,S_6),β))))$
```

$$integrand_1 = uv \qquad (\%t23)$$

$$integrand_2 = a^2 - uv \qquad (\%t24)$$

$$integrand_3 = uv \qquad (\%t25)$$

$$integrand_4 = b^2 - uv \qquad (\%t26)$$

$$integrand_5 = uv \tag{\%t27}$$

$$integrand_6 = c^2 - uv \tag{\%t28}$$

Surface integrals

(%i34)
```
I_1:'integrate('integrate(integrand_1,u,0,b),v,0,c)$
I_2:'integrate('integrate(integrand_2,u,0,b),v,0,c)$
I_3:'integrate('integrate(integrand_3,u,0,a),v,0,c)$
I_4:'integrate('integrate(integrand_4,u,0,a),v,0,c)$
I_5:'integrate('integrate(integrand_5,u,0,a),v,0,b)$
I_6:'integrate('integrate(integrand_6,u,0,a),v,0,b)$
```

(%i40)
```
ldisplay(I_1=box(ev(I_1,integrate)))$
ldisplay(I_2=box(ev(I_2,integrate)))$
ldisplay(I_3=box(ev(I_3,integrate)))$
ldisplay(I_4=box(ev(I_4,integrate)))$
ldisplay(I_5=box(ev(I_5,integrate)))$
ldisplay(I_6=box(ev(I_6,integrate)))$
```

$$\int_0^b u du \int_0^c v dv = \left(\frac{b^2\,c^2}{4}\right) \tag{\%t35}$$

$$\int_0^c \int_0^b a^2 - uv du\, dv = \left(-\frac{b^2\,c^2 - 4a^2 bc}{4}\right) \tag{\%t36}$$

$$\int_0^a u du \int_0^c v dv = \left(\frac{a^2\,c^2}{4}\right) \tag{\%t37}$$

$$\int_0^c \int_0^a b^2 - uv du\, dv = \left(-\frac{a^2\,c^2 - 4a\,b^2 c}{4}\right) \tag{\%t38}$$

$$\int_0^a u du \int_0^b v dv = \left(\frac{a^2\,b^2}{4}\right) \tag{\%t39}$$

$$\int_0^b \int_0^a c^2 - uv du\, dv = \left(\frac{4ab\,c^2 - a^2\,b^2}{4}\right) \tag{\%t40}$$

(%i41)
```
ldisplay(I=box(ev(I_1+I_2+I_3+I_4+I_5+I_6,integrate,factor)))$
```

$$I = (abc\,(c + b + a)) \tag{\%t41}$$

Using Gauss's Divergence Theorem

Integrand according to vector calculus

(%i42) `ldisplay(integrand:divF)$`

$$integrand = 2z + 2y + 2x \tag{\%t42}$$

Integrand according to differential forms

(%i43) `ldisplay(integrand:diff(ζ,z)|(diff(ζ,y)|(diff(ζ,x)|dβ)))$`

$$integrand = 2z + 2y + 2x \tag{\%t43}$$

Volume integral

(%i44) `I:'integrate('integrate('integrate(integrand,x,0,a),y,0,b),z,0,c)$`

(%i45) `ldisplay(I=box(ev(I,integrate,factor)))$`

$$\int_0^c \int_0^b \int_0^a 2z + 2y + 2x dx\, dy\, dz = (abc\,(c + b + a)) \tag{\%t45}$$

Clean up

(%i46) `forget(a>0,b>0,c>0)$`

3D Direction field

(%i48)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,0,8))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i50)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F/10))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i51)
```
wxdraw3d(proportional_axes=xy,xu_grid=100,yv_grid=100,view=[65,30],
xrange=[0,a],yrange=[0,b],zrange=[0,c],font_size=20,font="Helvetica",
color=green,apply(parametric_surface,append(S_1,[u,0,b,v,0,c])),
apply(parametric_surface,append(S_2,[u,0,b,v,0,c])),
color=black,label(["S_1",0,b/2,c/2],["S_2",a,b/2,c/2]),
color=blue, apply(parametric_surface,append(S_3,[u,0,a,v,0,c])),
apply(parametric_surface,append(S_4,[u,0,a,v,0,c])),
color=black,label(["S_3",a/2,0,c/2],["S_4",a/2,b,c/2]),
color=cyan, apply(parametric_surface,append(S_5,[u,0,a,v,0,b])),
apply(parametric_surface,append(S_6,[u,0,a,v,0,b])),
color=black,label(["S_5",a/2,b/2,0],["S_6",a/2,b/2,c]),
head_length=0.1,head_type='nofilled,line_width=2,color=red,vect3),params$
```



(%t51)

19 Stokes Theorem Problem #1

Based on MKS Tutorials Video Stokes Theorem Problem # 1

Relation between line integral and surface integral

$$\oint_C \vec{F} \cdot \mathrm{d}\vec{r} = \iint_s (\nabla \times \vec{F}) \cdot \hat{n} \, \mathrm{d}s$$

Verify Stokes theorem for $\vec{F} = (x^2 + y^2)\hat{\imath} - 2xy\hat{\jmath}$ taken around the rectangle bounded by $x = \pm a$ and $y = 0$ to $y = b$.

(%i52)
```
kill(labels,x,y,I,a,b)$
```

Define the space $\mathbb{R}^2$

(%i1)
```
ζ:[x,y]$
```

(%i2)
```
scalefactors(ζ)$
```

(%i3)
```
init_cartan(ζ)$
```

Parameters

(%i4)
```
assume(a>0,b>0)$
```

(%i5)
```
declare([a,b],constant)$
```

(%i6)
```
params:[a=2,b=1]$
```

Vector field $\vec{F} \in \mathbb{R}^2$

(%i7)
```
ldisplay(F:[x^2+y^2,-2*x*y])$
```

$$F = [y^2 + x^2, -2xy] \tag{\%t7}$$

$\nabla \times \vec{F} \in \mathbb{R}^2$

(%i8)
```
ldisplay(curlF:ev(express(curl(F)),diff))$
```

$$curlF = -4y \tag{\%t8}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i9)
```
ldisplay(α:F.cartan_basis)$
```

$$\alpha = \left(y^2 + x^2\right)\, dx - 2xy\, dy \tag{\%t9}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^2)$

(%i10)
```
ldisplay(dα:edit(ext_diff(α)))$
```

$$d\alpha = -4y\, dx\, dy \tag{\%t10}$$

Curves $C \in \mathbb{R}^2$

(%i14)
```
ldisplay(C_1:[t,0])$
ldisplay(C_2:[a,t])$
ldisplay(C_3:[t,b])$
ldisplay(C_4:[-a,t])$
```

$$C_1 = [t, 0] \tag{\%t11}$$

$$C_2 = [a, t] \tag{\%t12}$$

$$C_3 = [t, b] \tag{\%t13}$$

$$C_4 = [-a, t] \tag{\%t14}$$

Integrands according to vector calculus

(%i18)
```
ldisplay(integrand_1:subst(map("=",ζ,C_1),F).diff(C_1,t))$
ldisplay(integrand_2:subst(map("=",ζ,C_2),F).diff(C_2,t))$
ldisplay(integrand_3:subst(map("=",ζ,C_3),F).diff(C_3,t))$
ldisplay(integrand_4:subst(map("=",ζ,C_4),F).diff(C_4,t))$
```

$$integrand_1 = t^2 \tag{\%t15}$$

$$integrand_2 = -2at \tag{\%t16}$$

$$integrand_3 = t^2 + b^2 \tag{\%t17}$$

$$integrand_4 = 2at \tag{\%t18}$$

Integrands according to differential forms

(%i22)
```
ldisplay(integrand_1:diff(C_1,t)|subst(map("=",ζ,C_1),α))$
ldisplay(integrand_2:diff(C_2,t)|subst(map("=",ζ,C_2),α))$
ldisplay(integrand_3:diff(C_3,t)|subst(map("=",ζ,C_3),α))$
ldisplay(integrand_4:diff(C_4,t)|subst(map("=",ζ,C_4),α))$
```

$$integrand_1 = t^2 \tag{\%t19}$$

$$integrand_2 = -2at \tag{\%t20}$$

$$integrand_3 = t^2 + b^2 \tag{\%t21}$$

$$integrand_4 = 2at \tag{\%t22}$$

Line integrals

(%i26)
```
I_1:'integrate(integrand_1,t,-a,a)$
I_2:'integrate(integrand_2,t,0,b)$
I_3:'integrate(integrand_3,t,a,-a)$
I_4:'integrate(integrand_4,t,b,0)$
```

(%i30)
```
ldisplay(I_1=box(ev(I_1,integrate)))$
ldisplay(I_2=box(ev(I_2,integrate)))$
ldisplay(I_3=box(ev(I_3,integrate)))$
ldisplay(I_4=box(ev(I_4,integrate)))$
```

$$\int_{-a}^{a} t^2 dt = \left(\frac{2a^3}{3}\right) \tag{\%t27}$$

$$-2a\int_{0}^{b}t\,dt=\left(-a\,b^2\right) \qquad (\%t28)$$

$$-\int_{-a}^{a}t^2+b^2dt=\left(-\frac{2\left(3a\,b^2+a^3\right)}{3}\right) \qquad (\%t29)$$

$$-2a\int_{0}^{b}t\,dt=\left(-a\,b^2\right) \qquad (\%t30)$$

(%i31) `ldisplay(I=box(ev(I_1+I_2+I_3+I_4,integrate,ratsimp)))$`

$$I=\left(-4a\,b^2\right) \qquad (\%t31)$$

Using Stokes Theorem

Integrand according to vector calculus

(%i32) `ldisplay(integrand:curlF)$`

$$integrand=-4y \qquad (\%t32)$$

Integrand according to differential forms

(%i33) `ldisplay(integrand:diff(ζ,y)|(diff(ζ,x)|dα))$`

$$integrand=-4y \qquad (\%t33)$$

Surface integral

(%i34) `I:'integrate('integrate(integrand,x,-a,a),y,0,b)$`

(%i35) `ldisplay(I=box(ev(I,integrate)))$`

$$-8a\int_{0}^{b}y\,dy=\left(-4a\,b^2\right) \qquad (\%t35)$$

Clean up

(%i36) `forget(a>0,b>0)$`

2D Direction field

(%i37)
```
wxdrawdf(F,[x,-a-1/2,a+1/2],[y,-1/2,b+1/2],
color=blue,line_width=3,field_color=magenta,
apply(parametric,append(C_1,[t,-a,a])),
apply(parametric,append(C_2,[t,0,b])),
apply(parametric,append(C_3,[t,-a,a])),
apply(parametric,append(C_4,[t,0,b])),
color=green,line_width=2,
vector([a,0],[0,b]),vector([a,b],[-2*a,0]),
vector([-a,b],[0,-b]),vector([-a,0],[2*a,0]),
color=black,font_size=20,font="Helvetica",
label(["C_1",0.2,-0.1],["C_2",a+0.2,b/2]),
label(["C_3",0.2,b+0.1],["C_4",-a-0.2,b/2])),params$
```



(%t37)

20 Stokes Theorem Problem #2

Based on MKS Tutorials Video Stokes Theorem Problem # 2

Verify Stokes theorem for the field $\vec{F} = (2x - y)\hat{i} - yz^2\hat{j} - y^2z\hat{k}$ over the upper half surface of $x^2 + y^2 + z^2 = 1$ bounded by its projection on the xy-plane and C is the boundary.

(%i38) `kill(labels,x,y,z,r,θ,ϕ,I,R)$`

Define the space $\mathbb{R}^3$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Parameters

(%i4) `assume(R>0)$`

(%i5) `declare(R,constant)$`

(%i6) `params:[R=5]$`

Vector field $\vec{F} \in \mathbb{R}^3$

(%i7) `ldisplay(F:[2*x-y,-y*z^2,-y^2*z])$`

$$F = [2x - y, -y\,z^2, -y^2 z] \qquad (\%t7)$$

$\nabla \times \vec{F} \in \mathbb{R}^3$

(%i8) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [0, 0, 1] \qquad (\%t8)$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i9) `ldisplay(α:F.cartan_basis)$`

$$\alpha = -y^2 z\; dz - y\,z^2\; dy + (2x - y)\; dx \qquad (\%t9)$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i10) `ldisplay(dα:ext_diff(α))$`

$$d\alpha = dx\; dy \qquad (\%t10)$$

Surface $\vec{S}, \vec{D} \in \mathbb{R}^3$

(%i11) `ldisplay(S:[R*sin(θ)*cos(ϕ),R*sin(θ)*sin(ϕ),R*cos(θ)])$`

$$S = [R\,\sin(\theta)\,\cos(\phi), R\,\sin(\theta)\,\sin(\phi), R\,\cos(\theta)] \qquad (\%t11)$$

Integrand according to vector calculus

(%i12) `ldisplay(integrand:trigsimp(curlF.mycross(diff(S,θ),diff(S,ϕ))))$`

$$integrand = R^2\,\cos(\theta)\,\sin(\theta) \qquad (\%t12)$$

Integrand according to differential forms

(%i13) `ldisplay(integrand:trigsimp(diff(S,ϕ)|(diff(S,θ)|dα)))$`

$$integrand = R^2\,\cos(\theta)\,\sin(\theta) \qquad (\%t13)$$

Surface integral

(%i14) `I:'integrate('integrate(integrand,θ,0,1/2*π),ϕ,0,2*π)$`

(%i15) `ldisplay(I=box(ev(I,integrate)))$`

$$2\pi R^2 \int_0^{\frac{\pi}{2}} \cos(\theta)\,\sin(\theta)d\theta = \boxed{\pi R^2} \qquad (\%t15)$$

Curve $\vec{C} \in \mathbb{R}^3$

(%i16) `ldisplay(C:at(S,[θ=`$\frac{1}{2}$`*π]))$`

$$C = [R\,\cos(\phi), R\,\sin(\phi), 0] \tag{\%t16}$$

Integrand according to vector calculus

(%i17) `ldisplay(integrand:factor(subst(map("=",ζ,C),F).diff(C,ϕ)))$`

$$integrand = R^2\,\sin(\phi)\,(\sin(\phi) - 2\cos(\phi)) \tag{\%t17}$$

Integrand according to differential forms

(%i18) `ldisplay(integrand:factor(diff(C,ϕ)|subst(map("=",ζ,C),α)))$`

$$integrand = R^2\,\sin(\phi)\,(\sin(\phi) - 2\cos(\phi)) \tag{\%t18}$$

Line integral

(%i19) `I:'integrate(integrand,ϕ,0,2*π)$`

(%i20) `ldisplay(I=box(ev(I,integrate)))$`

$$R^2 \int_0^{2\pi} \sin(\phi)\,(\sin(\phi) - 2\cos(\phi))\,d\phi = \left(\pi R^2\right) \tag{\%t20}$$

Clean up

(%i21) `forget(R>0)$`

3D Direction field

(%i23)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,-5,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i25)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F/10))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i26)
```
wxdraw3d(proportional_axes=xy,xu_grid=100,yv_grid=100,
xrange=[-R,R],yrange=[-R,R],zrange=[0,R],font_size=20,font="Helvetica",
color=blue,apply(parametric_surface,append(S,[θ,0,1/2*π,ϕ,0,2*π])),
color=red,line_width=3,apply(parametric,append(C,[ϕ,0,2*π])),
head_length=0.1,color=magenta,line_width=1,head_angle=25,unit_vectors=true,vect3),params$
```



(%t26)