# Connection forms

Lecture Notes for Differential Geometry
James S. Cook
Liberty University
Department of Mathematics
Summer 2015

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

```
(%i2) reset()$kill(all)$
(%i1) derivabbrev:true$
(%i2) ratprint:false$
(%i3) fpprintprec:5$
(%i4) load(linearalgebra)$
(%i5) if get('draw,'version)=false then load(draw)$
(%i6) wxplot_size:[1024,768]$
(%i7) if get('drawdf,'version)=false then load(drawdf)$
(%i8) set_draw_defaults(xtics=1,ytics=1,ztics=1,xyplane=0,nticks=100,
      xaxis=true,xaxis_type=solid,xaxis_width=3,
      yaxis=true,yaxis_type=solid,yaxis_width=3,
      zaxis=true,zaxis_type=solid,zaxis_width=3,
      background_color=light_gray)$
(%i9) if get('vect,'version)=false then load(vect)$
(%i10) norm(u):=block(ratsimp(radcan(√(u.u))))$
(%i11) normalize(v):=block(v/norm(v))$
(%i12) angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$
(%i13) mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$
(%i14) if get('cartan,'version)=false then load(cartan)$
(%i15) declare(trigsimp,evfun)$
```

# 1 Cylindrical coordinate frame

Based on Examples 2.2.9, 2.2.15, 2.6.8, and 2.7.6.

(%i16) `kill(labels,x,y,z,r,θ)$`

(%i1) `ζ:[x,y,z]$`

(%i3)
```
assume(0≤r)$
assume(0≤θ,θ≤2*π)$
```

(%i4) `ξ:[r,θ,z]$`

**Cartesian frame**

(%i5) `U:[U[1],U[2],U[3]]$`

**Initialize cartan package**

(%i6) `init_cartan(ξ)$`

(%i7) `cartan_basis;`

$$[dr, d\theta, dz] \tag{\%o7}$$

(%i8) `cartan_coords;`

$$[r, \theta, z] \tag{\%o8}$$

(%i9) `cartan_dim;`

$$3 \tag{\%o9}$$

(%i10) `extdim;`

$$3 \tag{\%o10}$$

**Transformation formulas**

(%i11) `ldisplay(Tr:[r*cos(θ),r*sin(θ),z])$`

$$Tr = [r\,\cos(\theta), r\,\sin(\theta), z] \tag{\%t11}$$

**Jacobian matrix**

(%i12) `ldisplay(J:jacobian(Tr,ξ))$`

$$J = \begin{pmatrix} \cos(\theta) & -r\,\sin(\theta) & 0 \\ \sin(\theta) & r\,\cos(\theta) & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t12}$$

### Covariant metric tensor

(%i13) `ldisplay(lg:trigsimp(transpose(J).J))$`

$$lg = \begin{pmatrix} 1 & 0 & 0 \\ 0 & r^2 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t13}$$

### Jacobian

(%i14) `ldisplay(Jdet:trigsimp(determinant(J)))$`

$$Jdet = r \tag{\%t14}$$

### Covariant Basis as Matrix

(%i15) `ldisplay(lb:transpose(J))$`

$$lb = \begin{pmatrix} \cos(\theta) & \sin(\theta) & 0 \\ -r\,\sin(\theta) & r\,\cos(\theta) & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t15}$$

### Covariant Basis Vectors

(%i16) `for i thru 3 do ldisplay(le[ξ[i]]:lb[i])$`

$$le_r = [\cos(\theta), \sin(\theta), 0] \tag{\%t16}$$

$$le_\theta = [-r\,\sin(\theta), r\,\cos(\theta), 0] \tag{\%t17}$$

$$le_z = [0, 0, 1] \tag{\%t18}$$

### Line element

(%i19) `ldisplay(ds`$^2$`=diff(ξ).lg.transpose(diff(ξ)))$`

$$ds^2 = r^2\,\mathrm{del}(\theta)^2 + \mathrm{del}(z)^2 + \mathrm{del}(r)^2 \tag{\%t19}$$

### Coframe

(%i20) `ldisplay(coframe:√(lg))$`

$$coframe = \begin{pmatrix} 1 & 0 & 0 \\ 0 & r & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t20}$$

### Contravariant Metric Tensor

(%i21) `ldisplay(ug:trigsimp(invert(lg)))$`

$$ug = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t21}$$

### Contravariant Basis as Matrix

(%i22) `ldisplay(ub:trigsimp(ug.lb))$`

$$ub = \begin{pmatrix} \cos(\theta) & \sin(\theta) & 0 \\ -\frac{\sin(\theta)}{r} & \frac{\cos(\theta)}{r} & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t22}$$

### Contravariant Basis Vectors

(%i23) `for i thru 3 do ldisplay(ue[ξ[i]]:ub[i])$`

$$ue_r = [\cos(\theta), \sin(\theta), 0] \tag{\%t23}$$

$$ue_\theta = \left[-\frac{\sin(\theta)}{r}, \frac{\cos(\theta)}{r}, 0\right] \tag{\%t24}$$

$$ue_z = [0, 0, 1] \tag{\%t25}$$

### Initialize vect package

(%i26) `scalefactors(append([Tr],ξ))$`

(%i27) `sf;`

$$[1, r, 1] \tag{\%o27}$$

(%i28) `sfprod;`

$$r \tag{\%o28}$$

### Volume

(%i29) `[dx,dy,dz]:list_matrix_entries(J.cartan_basis)$`

(%i30) `dv:trigsimp(dx~dy~dz);`

$$r\, dr\, dz\, d\theta \tag{dv}$$

(%i31) `diff(ξ,z)|(diff(ξ,θ)|(diff(ξ,r)|dv));`

$$r \tag{\%o31}$$

(%i32) `ldisplay(dζ:trigsimp(ext_diff(at(ζ,map("=",ζ,Tr)))))$`

$$d\zeta = [dr\, \cos(\theta) - r\, d\theta\, \sin(\theta), dr\, \sin(\theta) + r\, d\theta\, \cos(\theta), dz] \tag{\%t32}$$

### Attitude matrix

(%i33) `ldisplay(A:apply('matrix,makelist(trigsimp(normalize(k)),k,args(transpose(J)))))$`

$$A = \begin{pmatrix} \cos(\theta) & \sin(\theta) & 0 \\ -\sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t33}$$

**Frame**

**(%i34)** `E:[E[1],E[2],E[3]]:trigsimp(list_matrix_entries(A.U))$`

**(%i35)** `ldisplay(E[1],E[2],E[3])$`

$$E_1 = U_2 \sin(\theta) + U_1 \cos(\theta) \tag{\%t35}$$

$$E_2 = U_2 \cos(\theta) - U_1 \sin(\theta) \tag{\%t36}$$

$$E_3 = U_3 \tag{\%t37}$$

**Coframe**

**(%i38)** `ldisplay(Θ:[θ[1],θ[2],θ[3]]:list_matrix_entries(trigsimp(A.[dx,dy,dz])))$`

$$\Theta = [dr, r\, d\theta, dz] \tag{\%t38}$$

**(%i39)** `ldisplay(Θ:list_matrix_entries(trigsimp(A.dζ)))$`

$$\Theta = [dr, r\, d\theta, dz] \tag{\%t39}$$

**(%i40)** `ldisplay(Θ:sf*cartan_basis)$`

$$\Theta = [dr, r\, d\theta, dz] \tag{\%t40}$$

d$A$

**(%i41)** `ldisplay(dA:ext_diff(A))$`

$$dA = \begin{pmatrix} -d\theta \sin(\theta) & d\theta \cos(\theta) & 0 \\ -d\theta \cos(\theta) & -d\theta \sin(\theta) & 0 \\ 0 & 0 & 0 \end{pmatrix} \tag{\%t41}$$

**Change matrix multiplication operator**

**(%i42)** `matrix_element_mult:"~"$`

**Connection form** $\omega \in \mathcal{A}^1(\mathbb{R}^3)$

**(%i43)** `ldisplay(ω:trigsimp(dA.transpose(A)))$`

$$\omega = \begin{pmatrix} 0 & d\theta & 0 \\ -d\theta & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix} \tag{\%t43}$$

**(%i44)** `ldisplay(dω:ext_diff(ω))$`

$$d\omega = \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix} \tag{\%t44}$$

(%i45) `trigsimp(ω.ω);`

$$\begin{pmatrix}0 & 0 & 0\\0 & 0 & 0\\0 & 0 & 0\end{pmatrix} \quad (\%o45)$$

**First structure equation of Cartan**

(%i46) `ldisplay(dΘ:ext_diff(Θ))$`

$$d\Theta = [0, dr\ d\theta, 0] \quad (\%t46)$$

(%i47) `list_matrix_entries(ω.Θ);`

$$[0, dr\ d\theta, 0] \quad (\%o47)$$

**Second structure equation of Cartan**

(%i48) `ldisplay(dω:ext_diff(ω))$`

$$d\omega = \begin{pmatrix}0 & 0 & 0\\0 & 0 & 0\\0 & 0 & 0\end{pmatrix} \quad (\%t48)$$

(%i49) `trigsimp(ω.ω);`

$$\begin{pmatrix}0 & 0 & 0\\0 & 0 & 0\\0 & 0 & 0\end{pmatrix} \quad (\%o49)$$

**Restore matrix multiplication operator**

(%i50) `matrix_element_mult:"*"$`

## 2 Spherical coordinate frame

Based on Examples 2.2.9, 2.2.15, 2.6.8 and 2.7.7.

(%i51)
```
kill(labels,x,y,z,r,θ)$
```

(%i1)
```
ζ:[x,y,z]$
```

(%i5)
```
assume(0≤r)$
assume(0≤θ,θ≤π)$
assume(0≤sin(θ))$
assume(0≤φ,φ≤2*π)$
```

(%i6)
```
ξ:[r,θ,φ]$
```

**Cartesian frame**

(%i7)
```
U:[U[1],U[2],U[3]]$
```

**Initialize cartan package**

(%i8)
```
init_cartan(ξ)$
```

(%i9)
```
cartan_basis;
```

$$[dr, d\theta, d\phi] \tag{\%o9}$$

(%i10)
```
cartan_coords;
```

$$[r, \theta, \phi] \tag{\%o10}$$

(%i11)
```
cartan_dim;
```

$$3 \tag{\%o11}$$

(%i12)
```
extdim;
```

$$3 \tag{\%o12}$$

**Transformation formulas**

(%i13)
```
ldisplay(Tr:[r*sin(θ)*cos(φ),r*sin(θ)*sin(φ),r*cos(θ)])$
```

$$Tr = [r\,\sin(\theta)\,\cos(\phi), r\,\sin(\theta)\,\sin(\phi), r\,\cos(\theta)] \tag{\%t13}$$

**Jacobian matrix**

(%i14)
```
ldisplay(J:jacobian(Tr,ξ))$
```

$$J = \begin{pmatrix} \sin(\theta)\,\cos(\phi) & r\,\cos(\theta)\,\cos(\phi) & -r\,\sin(\theta)\,\sin(\phi) \\ \sin(\theta)\,\sin(\phi) & r\,\cos(\theta)\,\sin(\phi) & r\,\sin(\theta)\,\cos(\phi) \\ \cos(\theta) & -r\,\sin(\theta) & 0 \end{pmatrix} \tag{\%t14}$$

**Covariant metric tensor**

**(%i15)** `ldisplay(lg:trigsimp(transpose(J).J))$`

$$lg = \begin{pmatrix} 1 & 0 & 0 \\ 0 & r^2 & 0 \\ 0 & 0 & r^2 \sin(\theta)^2 \end{pmatrix} \tag{\%t15}$$

**Jacobian**

**(%i16)** `ldisplay(Jdet:trigsimp(determinant(J)))$`

$$Jdet = r^2 \sin(\theta) \tag{\%t16}$$

**Covariant Basis as Matrix**

**(%i17)** `ldisplay(lb:transpose(J))$`

$$lb = \begin{pmatrix} \sin(\theta) \cos(\phi) & \sin(\theta) \sin(\phi) & \cos(\theta) \\ r \cos(\theta) \cos(\phi) & r \cos(\theta) \sin(\phi) & -r \sin(\theta) \\ -r \sin(\theta) \sin(\phi) & r \sin(\theta) \cos(\phi) & 0 \end{pmatrix} \tag{\%t17}$$

**Covariant Basis Vectors**

**(%i18)** `for i thru 3 do ldisplay(le[ξ[i]]:lb[i])$`

$$le_r = [\sin(\theta) \cos(\phi), \sin(\theta) \sin(\phi), \cos(\theta)] \tag{\%t18}$$

$$le_\theta = [r \cos(\theta) \cos(\phi), r \cos(\theta) \sin(\phi), -r \sin(\theta)] \tag{\%t19}$$

$$le_\phi = [-r \sin(\theta) \sin(\phi), r \sin(\theta) \cos(\phi), 0] \tag{\%t20}$$

**Line element**

**(%i21)** `ldisplay(ds`$^2$`=diff(ξ).lg.transpose(diff(ξ)))$`

$$ds^2 = r^2 \sin(\theta)^2 \operatorname{del}(\phi)^2 + r^2 \operatorname{del}(\theta)^2 + \operatorname{del}(r)^2 \tag{\%t21}$$

**Coframe**

**(%i22)** `ldisplay(coframe:`$\sqrt{}$`(lg))$`

$$coframe = \begin{pmatrix} 1 & 0 & 0 \\ 0 & r & 0 \\ 0 & 0 & r \sin(\theta) \end{pmatrix} \tag{\%t22}$$

**Contravariant Metric Tensor**

**(%i23)** `ldisplay(ug:trigsimp(invert(lg)))$`

$$ug = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & \frac{1}{r^2 \sin(\theta)^2} \end{pmatrix} \tag{\%t23}$$

**Contravariant Basis as Matrix**

**(%i24)** `ldisplay(ub:trigsimp(ug.lb))$`

$$ub = \begin{pmatrix} \sin(\theta)\,\cos(\phi) & \sin(\theta)\,\sin(\phi) & \cos(\theta) \\ \frac{\cos(\theta)\,\cos(\phi)}{r} & \frac{\cos(\theta)\,\sin(\phi)}{r} & -\frac{\sin(\theta)}{r} \\ -\frac{\sin(\phi)}{r\,\sin(\theta)} & \frac{\cos(\phi)}{r\,\sin(\theta)} & 0 \end{pmatrix} \quad (\%t24)$$

**Contravariant Basis Vectors**

**(%i25)** `for i thru 3 do ldisplay(ue[ξ[i]]:ub[i])$`

$$ue_r = [\sin(\theta)\,\cos(\phi), \sin(\theta)\,\sin(\phi), \cos(\theta)] \quad (\%t25)$$

$$ue_\theta = \left[\frac{\cos(\theta)\,\cos(\phi)}{r}, \frac{\cos(\theta)\,\sin(\phi)}{r}, -\frac{\sin(\theta)}{r}\right] \quad (\%t26)$$

$$ue_\phi = \left[-\frac{\sin(\phi)}{r\,\sin(\theta)}, \frac{\cos(\phi)}{r\,\sin(\theta)}, 0\right] \quad (\%t27)$$

**Initialize vect package**

**(%i28)** `scalefactors(append([Tr],ξ))$`

**(%i29)** `sf;`

$$[1, r, r\,\sin(\theta)] \quad (\%o29)$$

**(%i30)** `sfprod;`

$$r^2\,\sin(\theta) \quad (\%o30)$$

**Volume**

**(%i31)** `[dx,dy,dz]:list_matrix_entries(J.cartan_basis)$`

**(%i32)** `dv:trigsimp(dx~dy~dz);`

$$r^2\,dr\,d\theta\,d\phi\,\sin(\theta) \quad (dv)$$

**(%i33)** `diff(ξ,ϕ)|(diff(ξ,θ)|(diff(ξ,r)|dv));`

$$r^2\,\sin(\theta) \quad (\%o33)$$

**(%i34)** `ldisplay(dζ:trigsimp(ext_diff(at(ζ,map("=",ζ,Tr)))))$`

$$d\zeta = [-r\,d\phi\,\sin(\theta)\,\sin(\phi) + dr\,\sin(\theta)\,\cos(\phi) + r\,d\theta\,\cos(\theta)\,\cos(\phi), dr\,\sin(\theta)\,\sin(\phi) + r\,d\theta\,\cos(\theta)\,\sin(\phi) + r\,d\phi\,\sin(\theta)\,\cos(\phi), \ldots \quad (\%t34)$$

### Attitude matrix

**(%i35)** `ldisplay(A:apply('matrix,makelist(trigsimp(normalize(k)),k,args(transpose(J)))))$`

$$A = \begin{pmatrix} \sin(\theta)\,\cos(\phi) & \sin(\theta)\,\sin(\phi) & \cos(\theta) \\ \cos(\theta)\,\cos(\phi) & \cos(\theta)\,\sin(\phi) & -\sin(\theta) \\ -\sin(\phi) & \cos(\phi) & 0 \end{pmatrix} \tag{\%t35}$$

### Frame

**(%i36)** `E:[E[1],E[2],E[3]]:trigsimp(list_matrix_entries(A.U))$`

**(%i37)** `ldisplay(E[1],E[2],E[3])$`

$$E_1 = U_2\,\sin(\theta)\,\sin(\phi) + U_1\,\sin(\theta)\,\cos(\phi) + U_3\,\cos(\theta) \tag{\%t37}$$

$$E_2 = U_2\,\cos(\theta)\,\sin(\phi) + U_1\,\cos(\theta)\,\cos(\phi) - U_3\,\sin(\theta) \tag{\%t38}$$

$$E_3 = U_2\,\cos(\phi) - U_1\,\sin(\phi) \tag{\%t39}$$

### Coframe

**(%i40)** `ldisplay(Θ:[θ[1],θ[2],θ[3]]:list_matrix_entries(trigsimp(A.[dx,dy,dz])))$`

$$\Theta = [dr, r\,d\theta, r\,d\phi\,\sin(\theta)] \tag{\%t40}$$

**(%i41)** `ldisplay(Θ:list_matrix_entries(trigsimp(A.dζ)))$`

$$\Theta = [dr, r\,d\theta, r\,d\phi\,\sin(\theta)] \tag{\%t41}$$

**(%i42)** `ldisplay(Θ:sf*cartan_basis)$`

$$\Theta = [dr, r\,d\theta, r\,d\phi\,\sin(\theta)] \tag{\%t42}$$

d$A$

**(%i43)** `ldisplay(dA:ext_diff(A))$`

$$dA = \begin{pmatrix} d\theta\,\cos(\theta)\,\cos(\phi) - d\phi\,\sin(\theta)\,\sin(\phi) & d\theta\,\cos(\theta)\,\sin(\phi) + d\phi\,\sin(\theta)\,\cos(\phi) & -d\theta\,\sin(\theta) \\ -d\phi\,\cos(\theta)\,\sin(\phi) - d\theta\,\sin(\theta)\,\cos(\phi) & d\phi\,\cos(\theta)\,\cos(\phi) - d\theta\,\sin(\theta)\,\sin(\phi) & -d\theta\,\cos(\theta) \\ -d\phi\,\cos(\phi) & -d\phi\,\sin(\phi) & 0 \end{pmatrix} \tag{\%t43}$$

### Change matrix multiplication operator

**(%i44)** `matrix_element_mult:"~"$`

### Connection form $\omega \in \mathcal{A}^1(\mathbb{R}^3)$

**(%i45)** `ldisplay(ω:trigsimp(dA.transpose(A)))$`

$$\omega = \begin{pmatrix} 0 & d\theta & d\phi\,\sin(\theta) \\ -d\theta & 0 & d\phi\,\cos(\theta) \\ -d\phi\,\sin(\theta) & -d\phi\,\cos(\theta) & 0 \end{pmatrix} \tag{\%t45}$$

### First structure equation of Cartan

(%i46) `ldisplay(dΘ:ext_diff(Θ))$`

$$d\Theta = [0, dr\, d\theta, dr\, d\phi\, \sin(\theta) + r\, d\theta\, d\phi\, \cos(\theta)] \tag{\%t46}$$

(%i47) `list_matrix_entries(ω.Θ);`

$$[0, dr\, d\theta, dr\, d\phi\, \sin(\theta) + r\, d\theta\, d\phi\, \cos(\theta)] \tag{\%o47}$$

### Second structure equation of Cartan

(%i48) `ldisplay(dω:ext_diff(ω))$`

$$d\omega = \begin{pmatrix} 0 & 0 & d\theta\, d\phi\, \cos(\theta) \\ 0 & 0 & -d\theta\, d\phi\, \sin(\theta) \\ -d\theta\, d\phi\, \cos(\theta) & d\theta\, d\phi\, \sin(\theta) & 0 \end{pmatrix} \tag{\%t48}$$

(%i49) `trigsimp(ω.ω);`

$$\begin{pmatrix} 0 & 0 & d\theta\, d\phi\, \cos(\theta) \\ 0 & 0 & -d\theta\, d\phi\, \sin(\theta) \\ -d\theta\, d\phi\, \cos(\theta) & d\theta\, d\phi\, \sin(\theta) & 0 \end{pmatrix} \tag{\%o49}$$

### Restore matrix multiplication operator

(%i50) `matrix_element_mult:"*"$`