Trial metric

Based on Trin Tragula: General relativity, step by step

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

(%i2) `reset()$kill(all)$`

(%i1) `derivabbrev:true$`

(%i2) `if get('itensor,'version)=false then load(itensor)$`

(%i3) `imetric(g)$`

(%i4) `if get('ctensor,'version)=false then load(ctensor)$`

(%i5) `declare(trigsimp,evfun)$`

(%i9)
```
assume(0≤r)$
assume(0≤θ,θ≤π)$
assume(0≤sin(θ))$
assume(0≤φ,φ≤2*π)$
```

(%i10) `ξ:ct_coords:[t,r,θ,φ]$`

(%i11) `dim:length(ct_coords)$`

Covariant metric tensor

(%i12) `depends(a,t,q,r)$`

(%i14)
```
lg:matrix([-1,0,0,0],[0,a^2,0,0],[0,0,a^2*q^2,0], [0,0,0,a^2*q^2*sin(θ)^2])$
ishow(g([μ,ν],[])=lg)$
```

$$g_{\mu\nu} = \begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & a^2 & 0 & 0 \\ 0 & 0 & a^2\,q^2 & 0 \\ 0 & 0 & 0 & a^2\,q^2\sin\left(\theta\right)^2 \end{pmatrix} \tag{\%t14}$$

(%i15) `diagmatrixp(lg,dim);`

$$\mathrm{true} \tag{\%o15}$$

Sets up the package for further calculations

(%i16) `cmetric()$`

Contravariant metric tensor

(%i17) `ishow(g([],[μ,ν])=ug)$`

$$g^{\mu\nu} = \begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & \frac{1}{a^2} & 0 & 0 \\ 0 & 0 & \frac{1}{a^2\,q^2} & 0 \\ 0 & 0 & 0 & \frac{1}{a^2\,q^2\sin\left(\theta\right)^2} \end{pmatrix} \tag{\%t17}$$

(%i18) `diagmatrixp(ug,dim);`

$$\text{true} \tag{\%o18}$$

Line element

(%i19) `ldisplay(ds^2=diff(ct_coords).lg.transpose(diff(ct_coords)))$`

$$ds^2 = a^2\,q^2\sin(\theta)^2\,\mathrm{del}(\phi)^2 + a^2\,q^2\,\mathrm{del}(\theta)^2 - \mathrm{del}(t)^2 + a^2\,\mathrm{del}(r)^2 \tag{\%t19}$$

Christoffel symbols

(%i21)
```
christof(false)$
for i thru dim do for j:i thru dim do for k thru dim do
if mcs[i,j,k]≠0 then
ishow(Γ([ξ[i],ξ[j]],[ξ[k]])=mcs[i,j,k])$
```

$$\Gamma_{tr}^{r} = \frac{\dot{a}}{a} \tag{\%t21}$$

$$\Gamma_{t\theta}^{\theta} = \frac{\dot{a}}{a} \tag{\%t21}$$

$$\Gamma_{t\phi}^{\phi} = \frac{\dot{a}}{a} \tag{\%t21}$$

$$\Gamma_{rr}^{t} = a\;(\dot{a}) \tag{\%t21}$$

$$\Gamma_{r\theta}^{\theta} = \frac{q'}{q} \tag{\%t21}$$

$$\Gamma_{r\phi}^{\phi} = \frac{q'}{q} \tag{\%t21}$$

$$\Gamma_{\theta\theta}^{t} = a\;(\dot{a})\;q^2 \tag{\%t21}$$

$$\Gamma_{\theta\theta}^{r} = -q\;(q') \tag{\%t21}$$

$$\Gamma_{\theta\phi}^{\phi} = \frac{\cos(\theta)}{\sin(\theta)} \tag{\%t21}$$

$$\Gamma_{\phi\phi}^{t} = a\;(\dot{a})\;q^2\sin(\theta)^2 \tag{\%t21}$$

$$\Gamma_{\phi\phi}^{r} = -q\;(q')\;\sin(\theta)^2 \tag{\%t21}$$

$$\Gamma_{\phi\phi}^{\theta} = -\cos(\theta)\;\sin(\theta) \tag{\%t21}$$

(%i22) `cdisplay(mcs,2)$`

$$mcs_2 = \begin{pmatrix} 0 & \frac{\dot{a}}{a} & 0 & 0 \\ a\;(\dot{a}) & 0 & 0 & 0 \\ 0 & 0 & \frac{q'}{q} & 0 \\ 0 & 0 & 0 & \frac{q'}{q} \end{pmatrix}$$

(%i23) `matrixp(%);`

$$\text{false} \tag{\%o23}$$

(%i24) ldisplay(Γ_2:genmatrix(lambda([i,j],mcs[2,i,j]),dim,dim))$

$$\mathit{\Gamma}_2 = \begin{pmatrix} 0 & \frac{\dot{a}}{a} & 0 & 0 \\ a\ (\dot{a}) & 0 & 0 & 0 \\ 0 & 0 & \frac{q'}{q} & 0 \\ 0 & 0 & 0 & \frac{q'}{q} \end{pmatrix} \tag{\%t24}$$

(%i25) matrixp(Γ_2);

$$\text{true} \tag{\%o25}$$

Riemann Tensor

(%i29)
```
riemann(false)$
lriemann(false)$
uriemann(false)$
for a thru dim do for b thru dim do
for c thru (if symmetricp(lg,dim) then b else dim) do
for d thru (if symmetricp(lg,dim) then a else dim) do
if riem[a,b,c,d]≠0 then
ishow(R([" ",ξ[a],ξ[b],ξ[c]],[ξ[d]])=riem[a,b,c,d])$
```

$$\mathbf{R}^{t}_{rrt} = a\ (\ddot{a}) \tag{\%t29}$$

$$\mathbf{R}^{t}_{\theta\theta t} = a\ (\ddot{a})\ q^2 \tag{\%t29}$$

$$\mathbf{R}^{r}_{\theta\theta r} = (\dot{a})^2\, q^2 - q\ (q'') \tag{\%t29}$$

$$\mathbf{R}^{t}_{\phi\phi t} = a\ (\ddot{a})\ q^2 \sin(\theta)^2 \tag{\%t29}$$

$$\mathbf{R}^{r}_{\phi\phi r} = \left((\dot{a})^2\, q^2 - q\ (q'')\right)\ \sin(\theta)^2 \tag{\%t29}$$

$$\mathbf{R}^{\theta}_{\phi\phi\theta} = \left(-(q')^2 + (\dot{a})^2\, q^2 + 1\right)\ \sin(\theta)^2 \tag{\%t29}$$

Ricci Tensor

(%i32)
```
ric:zeromatrix(dim,dim)$
ricci(false)$
for a thru dim do for b:a thru dim do
if ric[a,b]≠0 then
ishow(R([ξ[a],ξ[b]])=ric[a,b])$
```

$$\mathbf{R}_{tt} = -\frac{3\,(\ddot{a})}{a} \tag{\%t32}$$

$$\mathbf{R}_{rr} = -\frac{2\,(q'')}{q} + a\ (\ddot{a}) + 2(\dot{a})^2 \tag{\%t32}$$

$$\mathbf{R}_{\theta\theta} = -q\ (q'') - (q')^2 + a\ (\ddot{a})\ q^2 + 2(\dot{a})^2\, q^2 + 1 \tag{\%t32}$$

$$\mathbf{R}_{\phi\phi} = -q\ (q'')\ \sin(\theta)^2 - (q')^2 \sin(\theta)^2 + a\ (\ddot{a})\ q^2 \sin(\theta)^2 + 2(\dot{a})^2\, q^2 \sin(\theta)^2 + \sin(\theta)^2 \tag{\%t32}$$

(%i33) `diagmatrixp(ric,dim);`

$$\text{true} \tag{\%o33}$$

(%i36)
```
uric:zeromatrix(dim,dim)$
uricci(false)$
for a thru dim do for b:a thru dim do
if uric[a,b]≠0 then
ishow(R([ξ[a]],[ξ[b]])=uric[a,b])$
```

$$\mathbf{R}_t^t = \frac{3\,(\ddot{a})}{a} \tag{\%t36}$$

$$\mathbf{R}_r^r = -\frac{2\,(q'') + \left(-a\ (\ddot{a}) - 2(\dot{a})^2\right) q}{a^2 q} \tag{\%t36}$$

$$\mathbf{R}_\theta^\theta = -\frac{q\ (q'') + (q')^2 + \left(-a\ (\ddot{a}) - 2(\dot{a})^2\right)\ q^2 - 1}{a^2\, q^2} \tag{\%t36}$$

$$\mathbf{R}_\phi^\phi = -\frac{q\ (q'') + (q')^2 + \left(-a\ (\ddot{a}) - 2(\dot{a})^2\right)\ q^2 - 1}{a^2\, q^2} \tag{\%t36}$$

(%i37) `diagmatrixp(uric,dim);`

$$\text{true} \tag{\%o37}$$

Vacuum Einstein field equations

(%i38) `map(ldisp,efe:findde(ric,2))$`

$$\ddot{a} \tag{\%t38}$$

$$2\,(q'') - a\ (\ddot{a})\, q - 2(\dot{a})^2 q \tag{\%t39}$$

$$q\ (q'') + (q')^2 - a\ (\ddot{a})\ q^2 - 2(\dot{a})^2\, q^2 - 1 \tag{\%t40}$$

(%i41) `deindex;`

$$[[1,1],[2,2],[3,3]] \tag{\%o41}$$

(%i42) `map(ldisp,efe:findde(uric,2))$`

$$\ddot{a} \tag{\%t42}$$

$$2\,(q'') - a\ (\ddot{a})\, q - 2(\dot{a})^2 q \tag{\%t43}$$

$$q\ (q'') + (q')^2 - a\ (\ddot{a})\ q^2 - 2(\dot{a})^2\, q^2 - 1 \tag{\%t44}$$

(%i45) `deindex;`

$$[[1,1],[2,2],[3,3]] \tag{\%o45}$$

Scalar curvature

(%i46) `factor(radcan(scurvature()));`

$$-\frac{2\left(2q\ (q'')+(q')^2-3a\ (\ddot{a})\ q^2-3(\dot{a})^2\,q^2-1\right)}{a^2\,q^2} \tag{\%o46}$$

Kretschmann invariant

(%i47) `factor(radcan(rinvariant()));`

$$(4(2q^2\,(q'')^2-4(\dot{a})^2\,q^3\ (q'')+(q')^4-2(\dot{a})^2\,q^2\,(q')^2-2(q')^2+3a^2\,(\ddot{a})^2\,q^4+3(\dot{a})^4\,q^4+2(\dot{a})^2\,q^2+1))/(a^4\,q^4) \tag{\%o47}$$

Einstein Tensor

(%i50)
```
ein:zeromatrix(dim,dim)$
einstein(false)$
for i thru dim do for j:i thru dim do
if ein[i,j]≠0 then
ishow(G([ξ[i]],[ξ[j]])=ein[i,j])$
```

$$\mathbf{G}_t^t=\frac{2q\ (q'')+(q')^2-3(\dot{a})^2\,q^2-1}{a^2\,q^2} \tag{\%t50}$$

$$\mathbf{G}_r^r=\frac{(q')^2+\left(-2a\ (\ddot{a})-(\dot{a})^2\right)\ q^2-1}{a^2\,q^2} \tag{\%t50}$$

$$\mathbf{G}_\theta^\theta=\frac{q''+\left(-2a\ (\ddot{a})-(\dot{a})^2\right)q}{a^2 q} \tag{\%t50}$$

$$\mathbf{G}_\phi^\phi=\frac{q''+\left(-2a\ (\ddot{a})-(\dot{a})^2\right)q}{a^2 q} \tag{\%t50}$$

(%i51) `diagmatrixp(ein,dim);`

$$\text{true} \tag{\%o51}$$

(%i54)
```
lein:zeromatrix(dim,dim)$
leinstein(false)$
for i thru dim do for j:i thru dim do
if lein[i,j]≠0 then
ishow(G([ξ[i],ξ[j]],[])=lein[i,j])$
```

$$\mathbf{G}_{tt}=-\frac{2q\ (q'')+(q')^2-3(\dot{a})^2\,q^2-1}{a^2\,q^2} \tag{\%t54}$$

$$\mathbf{G}_{rr}=\frac{(q')^2+\left(-2a\ (\ddot{a})-(\dot{a})^2\right)\ q^2-1}{q^2} \tag{\%t54}$$

$$\mathbf{G}_{\theta\theta}=q\ \left(q''+\left(-2a\ (\ddot{a})-(\dot{a})^2\right)q\right) \tag{\%t54}$$

$$\mathbf{G}_{\phi\phi} = q\,\left(q'' + \left(-2a\,\left(\ddot{a}\right) - \left(\dot{a}\right)^2\right)q\right)\,\sin\left(\theta\right)^2 \tag{\%t54}$$

(%i55) `diagmatrixp(lein,dim);`

$$\text{true} \tag{\%o55}$$

Vacuum Einstein field equations

(%i56) `map(ldisp,findde(ein,2))$`

$$2q\,\left(q''\right) + \left(q'\right)^2 - 3\left(\dot{a}\right)^2 q^2 - 1 \tag{\%t56}$$

$$\left(q'\right)^2 - 2a\,\left(\ddot{a}\right)\,q^2 - \left(\dot{a}\right)^2 q^2 - 1 \tag{\%t57}$$

$$q'' - 2a\,\left(\ddot{a}\right) q - \left(\dot{a}\right)^2 q \tag{\%t58}$$

(%i59) `deindex;`

$$[[1,1],[2,2],[3,3]] \tag{\%o59}$$

(%i60) `map(ldisp,findde(lein,2))$`

$$2q\,\left(q''\right) + \left(q'\right)^2 - 3\left(\dot{a}\right)^2 q^2 - 1 \tag{\%t60}$$

$$\left(q'\right)^2 - 2a\,\left(\ddot{a}\right)\,q^2 - \left(\dot{a}\right)^2 q^2 - 1 \tag{\%t61}$$

$$q'' - 2a\,\left(\ddot{a}\right) q - \left(\dot{a}\right)^2 q \tag{\%t62}$$

(%i63) `deindex;`

$$[[1,1],[2,2],[3,3]] \tag{\%o63}$$

Computes the Weyl conformal tensor

(%i65)
```
weyl(false)$
for i thru dim do
for j from (if symmetricp(lg,dim) then i+1 else 1) thru dim do
for k from (if symmetricp(lg,dim) then i else 1) thru dim do
for l from (if symmetricp(lg,dim) then k+1 else 1) thru (if (symmetricp(lg,dim) and k=i)
then j else dim) do
if weyl[i,j,k,l]≠0 then
ishow(W([ξ[i],ξ[j],ξ[k],ξ[l]],[])=weyl[i,j,k,l])$
```

$$\mathbf{W}_{trtr} = \frac{q\,\left(q''\right) - \left(q'\right)^2 + 1}{3q^2} \tag{\%t65}$$

$$\mathbf{W}_{t\theta t\theta} = -\frac{q\,\left(q''\right) - \left(q'\right)^2 + 1}{6} \tag{\%t65}$$

$$\mathbf{W}_{t\phi t\phi} = -\frac{\left(q\,\left(q''\right) - \left(q'\right)^2 + 1\right)\,\sin\left(\theta\right)^2}{6} \tag{\%t65}$$

$$\mathbf{W}_{r\theta r\theta} = \frac{a^2 q\ (q'') - a^2\ (q')^2 + a^2}{6} \tag{\%t65}$$

$$\mathbf{W}_{r\phi r\phi} = \frac{\left(a^2 q\ (q'') - a^2\ (q')^2 + a^2\right)\ \sin(\theta)^2}{6} \tag{\%t65}$$

$$\mathbf{W}_{\theta\phi\theta\phi} = -\frac{\left(a^2\ q^3\ (q'') - a^2\ q^2\ (q')^2 + a^2\ q^2\right)\ \sin(\theta)^2}{3} \tag{\%t65}$$

Computes the Geodesics

(%i66) `cgeodesic(false)$`

Solve for second derivative of coordinates

(%i67) `linsol:linsolve(listarray(geod),diff(ξ,s,2))$`

(%i68) `map(ldisp,fullratsimp(linsol))$`

$$t_{ss} = -a\ (\dot{a})\ q^2 \sin(\theta)^2\ (\phi_s)^2 - a\ (\dot{a})\ q^2\ (\theta_s)^2 - a\ (\dot{a})\ (r_s)^2 \tag{\%t68}$$

$$r_{ss} = \frac{aq\ (q')\ \sin(\theta)^2\ (\phi_s)^2 + aq\ (q')\ (\theta_s)^2 - 2\,(\dot{a})\ (r_s)\ (t_s)}{a} \tag{\%t69}$$

$$\theta_{ss} = \frac{aq\ \cos(\theta)\ \sin(\theta)\,(\phi_s)^2 + (-2\,(\dot{a})\,q\ (t_s) - 2a\ (q')\ (r_s))\ (\theta_s)}{aq} \tag{\%t70}$$

$$\phi_{ss} = -\frac{(2aq\ \cos(\theta)\ (\theta_s) + (2\,(\dot{a})\,q\ (t_s) + 2a\ (q')\ (r_s))\ \sin(\theta))\ (\phi_s)}{aq\ \sin(\theta)} \tag{\%t71}$$

Formula to raise one index of a $(0,2)$ tensor

```
(%i72) kill(labels)$
(%i1) ishow(Raise:B([ν],[α])=g([],[μ,α])*A([μ,ν],[]))$
```

$$B_{\nu}^{\alpha} = g^{\mu\alpha}\, A_{\mu\nu} \tag{\%t1}$$

```
(%i2) Raise:ic_convert(Raise)$
```

Formula to lower one index of a $(2,0)$ tensor

```
(%i3) ishow(Lower:D([α],[ν])=g([μ,α],[])*C([],[μ,ν]))$
```

$$D_{\alpha}^{\nu} = C^{\mu\nu}\, g_{\mu\alpha} \tag{\%t3}$$

```
(%i4) Lower:ic_convert(Lower)$
```

Einstein field equation formula

```
(%i5) ishow(EFE:A([μ,ν])=κ*B([μ,ν])-Λ*g([μ,ν]))$
```

$$A_{\mu\nu} = \kappa\, B_{\mu\nu} - \Lambda\, g_{\mu\nu} \tag{\%t5}$$

```
(%i6) EFE:ic_convert(EFE)$
```

Energy-Momentum tensor formula

```
(%i7) depends([ρ_0,p_0],t);
```

$$[\rho_0(t), p_0(t)] \tag{\%o7}$$

```
(%i8) ishow(EMT:S([],[μ,ν])=(ρ_0+p_0)*u([],[μ])*u([],[ν])+p_0*g([],[μ,ν]))$
```

$$S^{\mu\nu} = u^{\mu}\, u^{\nu}\, (\rho_0 + p_0) + g^{\mu\nu}\, p_0 \tag{\%t8}$$

```
(%i9) EMT:ic_convert(EMT)$
```

Covariant derivative formulas

```
(%i10) ishow(CD1:Y([β,γ],[α])=subst([%1=σ],rename(covdiff(X([β],[α]),γ))))$
```

$$Y_{\beta\gamma}^{\alpha} = X_{\beta}^{\sigma}\, \Gamma_{\sigma\gamma}^{\alpha} - \Gamma_{\beta\gamma}^{\sigma}\, X_{\sigma}^{\alpha} + X_{\beta,\gamma}^{\alpha} \tag{\%t10}$$

```
(%i11) CD1:ic_convert(CD1)$
(%i12) ishow(CD2:Z([β],[])=subst([%1=σ,%2=λ],rename(covdiff(X([β],[α]),α))))$
```

$$Z_{\beta} = X_{\beta}^{\lambda}\, \Gamma_{\lambda\sigma}^{\sigma} - \Gamma_{\beta\sigma}^{\lambda}\, X_{\lambda}^{\sigma} + X_{\beta,\sigma}^{\sigma} \tag{\%t12}$$

```
(%i13) CD2:ic_convert(CD2)$
```

Energy-Momentum tensor

(%i15)
```
S:zeromatrix(dim,dim)$
u:[1,0,0,0]$
```
(%i16)
```
ev(EMT)$
```
(%i17)
```
ishow(T([],[μ,ν])=S)$
```

$$\mathbf{T}^{\mu\nu} = \begin{pmatrix} \rho_0 & 0 & 0 & 0 \\ 0 & \frac{p_0}{a^2} & 0 & 0 \\ 0 & 0 & \frac{p_0}{a^2\,q^2} & 0 \\ 0 & 0 & 0 & \frac{p_0}{a^2\,q^2\,\sin(\theta)^2} \end{pmatrix} \tag{\%t17}$$

(%i18)
```
diagmatrixp(S,dim);
```

$$\text{true} \tag{\%o18}$$

Lower one index from the Energy-Momentum tensor

(%i20)
```
C:S$
D:zeromatrix(dim,dim)$
```
(%i21)
```
ev(Lower)$
```
(%i22)
```
ishow(T([ν],[μ])=D)$
```

$$\mathbf{T}_{\nu}^{\mu} = \begin{pmatrix} -\rho_0 & 0 & 0 & 0 \\ 0 & p_0 & 0 & 0 \\ 0 & 0 & p_0 & 0 \\ 0 & 0 & 0 & p_0 \end{pmatrix} \tag{\%t22}$$

(%i23)
```
diagmatrixp(D,dim);
```

$$\text{true} \tag{\%o23}$$

Covariant derivative of mixed Energy-Momentum tensor

(%i25)
```
X:D$
Z:[0,0,0,0]$
```
(%i26)
```
ev(CD2)$
```
(%i27)
```
ldisplay(CDT:expand(-a^3*Z))$
```

$$CDT = [a^3\ (\rho_{0\,t}) + 3a^2\ (\dot{a})\ \rho_0 + 3a^2\ (\dot{a})\ p_0, 0, 0, 0] \tag{\%t27}$$

(%i28)
```
is(first(CDT)=diff(a^3*ρ_0,t)+p_0*diff(a^3,t));
```

$$\text{true} \tag{\%o28}$$

Lower second index from the Energy-Momentum tensor

(%i30)
```
C:D$
D:zeromatrix(dim,dim)$
```

(%i31) `ev(Lower)$`

(%i32) `ishow(T([μ,ν],[])=D)$`

$$\mathbf{T}_{\mu\nu} = \begin{pmatrix} \rho_0 & 0 & 0 & 0 \\ 0 & a^2\,p_0 & 0 & 0 \\ 0 & 0 & a^2\,p_0 q^2 & 0 \\ 0 & 0 & 0 & a^2\,p_0 q^2 \sin(\theta)^2 \end{pmatrix} \tag{\%t32}$$

(%i33) `diagmatrixp(D,dim);`

$$\text{true} \tag{\%o33}$$

Einstein field equation

(%i35) `A:zeromatrix(dim,dim)$`
`B:D$`

(%i36) `ev(EFE)$`

(%i37) `ishow(G([μ,ν],[])=factor(A))$`

$$\mathbf{G}_{\mu\nu} = \begin{pmatrix} \kappa\,\rho_0 + \Lambda & 0 & 0 & 0 \\ 0 & a^2\,(p_0\kappa - \Lambda) & 0 & 0 \\ 0 & 0 & a^2\,q^2\,(p_0\kappa - \Lambda) & 0 \\ 0 & 0 & 0 & a^2\,q^2 \sin(\theta)^2\,(p_0\kappa - \Lambda) \end{pmatrix} \tag{\%t37}$$

(%i38) `EFE:makelist(expand(lein[i,i])=factor(A[i,i]),i,dim-1)$`

(%i39) `maplist(ldisp,EFE)$`

$$-\frac{2\,(q'')}{a^2 q} - \frac{(q')^2}{a^2\,q^2} + \frac{1}{a^2\,q^2} + \frac{3(\dot{a})^2}{a^2} = \kappa\,\rho_0 + \Lambda \tag{\%t39}$$

$$\frac{(q')^2}{q^2} - \frac{1}{q^2} - 2a\,(\ddot{a}) - (\dot{a})^2 = a^2\,(p_0\kappa - \Lambda) \tag{\%t40}$$

$$q\,(q'') - 2a\,(\ddot{a})\,q^2 - (\dot{a})^2\,q^2 = a^2\,q^2\,(p_0\kappa - \Lambda) \tag{\%t41}$$

Autonomous equation

(%i42) `kill(labels)$`

(%i1) `Eq1:expand(EFE[3]-q^2*EFE[2]);`

$$q\,(q'') - (q')^2 + 1 = 0 \tag{Eq1}$$

Transform equation Cases of Reduction of Order

(%i2) `declare([K,K_1,K_2,K_3,K_4,K_5,K_6],constant)$`

(%i3) `Eq2:subst(['diff(q,r)=p,'diff(q,r,2)=p*'diff(p,q)],Eq1);`

$$p\,(p_q)\,q - p^2 + 1 = 0 \tag{Eq2}$$

(%i4) `sol:2*logcontract(ode2(Eq2,p,q));`

$$\log\left(\left(p-1\right) \,\left(p+1\right) \right) =2\,\left(\log\left(q\right) +\%c\right) \tag{sol}$$

(%i5) `sol:subst([%c=K],sol);`

$$\log\left(\left(p-1\right) \,\left(p+1\right) \right) =2\,\left(\log\left(q\right) +K\right) \tag{sol}$$

(%i6) `method;`

$$\mathit{separable} \tag{\%o6}$$

(%i7) `sol:expand(exp(lhs(sol))=exp(rhs(sol)));`

$$p^2-1={e}^{2\,K}\,q^2 \tag{sol}$$

First case $p^2-1>0$

(%i8) `sola:solve(sol,p);`

$$\left[p=-\sqrt{{e}^{2\,K}\,q^2+1},p=\sqrt{{e}^{2\,K}\,q^2+1}\right] \tag{sola}$$

(%i9) `sola:subst(p='diff(q,r),sola);`

$$\left[q'=-\sqrt{{e}^{2\,K}\,q^2+1},q'=\sqrt{{e}^{2\,K}\,q^2+1}\right] \tag{sola}$$

(%i10) `solve(ode2(sola[1],q,r),q);`

$$\left[q=-{e}^{-K}\,\sinh\left({e}^{K}\,r+{e}^{K}\,\%c\right) \right] \tag{\%o10}$$

(%i11) `method;`

$$\mathit{separable} \tag{\%o11}$$

(%i12) `Qa1:subst([exp(K)=1/r_0,%c=K_1],%th(2));`

$$\left[q=-\sinh\left(\frac{r}{r_0}+\frac{K_1}{r_0}\right) \,r_0\right] \tag{Qa1}$$

(%i13) `solve(ode2(sola[2],q,r),q);`

$$\left[q={e}^{-K}\,\sinh\left({e}^{K}\,r+{e}^{K}\,\%c\right) \right] \tag{\%o13}$$

(%i14) `method;`

$$\mathit{separable} \tag{\%o14}$$

(%i15) `Qa2:subst([exp(K)=1/r_0,%c=K_2],%th(2));`

$$\left[q=\sinh\left(\frac{r}{r_0}+\frac{K_2}{r_0}\right)r_0\right] \tag{Qa2}$$

Second case $p^2-1<0$

(%i16) `solb:solve(-lhs(sol)=rhs(sol),p);`

$$\left[p=-\sqrt{1-e^{2K}\,q^2},p=\sqrt{1-e^{2K}\,q^2}\right] \tag{solb}$$

(%i17) `solb:subst(p='diff(q,r),solb);`

$$\left[q'=-\sqrt{1-e^{2K}\,q^2},q'=\sqrt{1-e^{2K}\,q^2}\right] \tag{solb}$$

(%i18) `solve(ode2(solb[1],q,r),q);`

$$\left[q=-e^{-K}\,\sin\left(e^K r+e^K\,\%c\right)\right] \tag{\%o18}$$

(%i19) `method;`

$$\mathit{separable} \tag{\%o19}$$

(%i20) `Qb1:subst([exp(K)=1/r_0,%c=K_3],%th(2));`

$$\left[q=-\sin\left(\frac{r}{r_0}+\frac{K_3}{r_0}\right)r_0\right] \tag{Qb1}$$

(%i21) `solve(ode2(solb[2],q,r),q);`

$$\left[q=e^{-K}\,\sin\left(e^K r+e^K\,\%c\right)\right] \tag{\%o21}$$

(%i22) `method;`

$$\mathit{separable} \tag{\%o22}$$

(%i23) `Qb2:subst([exp(K)=1/r_0,%c=K_4],%th(2));`

$$\left[q=\sin\left(\frac{r}{r_0}+\frac{K_4}{r_0}\right)r_0\right] \tag{Qb2}$$

Third case $p^2-1=0$

(%i24) `solc:solve(p^2=1,p);`

$$[p=-1,p=1] \tag{solc}$$

(%i25) `solc:subst(p='diff(q,r),solc);`

$$[q' = -1, q' = 1] \tag{solc}$$

(%i26) `solve(ode2(solc[1],q,r),q)$`

(%i27) `method;`

$$linear \tag{\%o27}$$

(%i28) `Qc1:subst([%c=K_5],%th(2));`

$$[q = K_5 - r] \tag{Qc1}$$

(%i29) `solve(ode2(solc[2],q,r),q)$`

(%i30) `method;`

$$linear \tag{\%o30}$$

(%i31) `Qc2:subst([%c=K_6],%th(2));`

$$[q = r + K_6] \tag{Qc2}$$

Verify solution

(%i32) `is(ev(Eq1,Qa1,diff,trigsimp));`

$$\text{true} \tag{\%o32}$$

(%i33) `is(ev(Eq1,Qa2,diff,trigsimp));`

$$\text{true} \tag{\%o33}$$

(%i34) `is(ev(Eq1,Qb1,diff,trigsimp));`

$$\text{true} \tag{\%o34}$$

(%i35) `is(ev(Eq1,Qb2,diff,trigsimp));`

$$\text{true} \tag{\%o35}$$

(%i36) `is(ev(Eq1,Qc1,diff,trigsimp));`

$$\text{true} \tag{\%o36}$$

(%i37) `is(ev(Eq1,Qc2,diff,trigsimp));`

$$\text{true} \tag{\%o37}$$

Limits

(%i38) `limit(r_0*sin(r/r_0),r_0,∞);`

$$r \tag{\%o38}$$

(%i39) `limit(r_0*sinh(r/r_0),r_0,∞);`

$$r \tag{\%o39}$$

(%i40) `ev(limit(q,r_0,∞),Qa1);`

$$-r-K_1 \tag{\%o40}$$

(%i41) `ev(limit(q,r_0,∞),Qa2);`

$$r+K_2 \tag{\%o41}$$

(%i42) `ev(limit(q,r_0,∞),Qb1);`

$$-r-K_3 \tag{\%o42}$$

(%i43) `ev(limit(q,r_0,∞),Qb2);`

$$r+K_4 \tag{\%o43}$$

(%i44) `ev(limit(q,r_0,∞),Qc1);`

$$K_5-r \tag{\%o44}$$

(%i45) `ev(limit(q,r_0,∞),Qc2);`

$$r+K_6 \tag{\%o45}$$