Killing vectors

Based on Start Somewhere Playlist General Relativity

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

(%i2) `reset()$kill(all)$`

(%i1) `derivabbrev:false$`

(%i4) `load(itensor)$load(ctensor)$load(cartan)$`

(%i5) `imetric(g)$`

(%i11)
```
ctrgsimp:true$
ratchristof:true$
ratriemann:true$
rateinstein:true$
ratweyl:true$
ratfac:true$
```

Killing equation

(%i12) `ishow(subst([%1=σ],rename(liediff(ξ,g([μ,ν])))))$`

$$\xi^{\sigma}_{,\mu}\, g_{\sigma\nu} + \xi^{\sigma}_{,\nu}\, g_{\mu\sigma} + \xi^{\sigma}\, g_{\mu\nu,\sigma} \tag{\%t12}$$

(%i13) `indices(%);`

$$[[\mu,\nu],[\sigma]] \tag{\%o13}$$

(%i14) `ishow(Eq:B([μ,ν])=subst([%1=σ],rename(liediff(ξ,g([μ,ν])))))$`

$$B_{\mu\nu} = \xi^{\sigma}_{,\mu}\, g_{\sigma\nu} + \xi^{\sigma}_{,\nu}\, g_{\mu\sigma} + \xi^{\sigma}\, g_{\mu\nu,\sigma} \tag{\%t14}$$

(%i15) `Eq:ic_convert(Eq)$`

1 Killing Vector for Polar Coordinates

(%i16)
```
kill(labels,r,θ)$
```

(%i1)
```
ζ:[r,θ]$
```

(%i2)
```
Tr:[r*cos(θ),r*sin(θ)]$
```

(%i3)
```
ct_coordsys(append(Tr,[ζ]),all)$
```

(%i5)
```
lg:trigsimp(lg)$
ug:trigsimp(ug)$
```

(%i10)
```
ξ:[ξ_1,ξ_2]$
depends(ξ_1,ζ)$
depends(ξ_2,ζ)$
ξ:ev(ξ)$
B:zeromatrix(dim,dim)$
```

(%i11)
```
ev(Eq)$
```

(%i12)
```
ldisplay(B)$
```

$$B = \begin{pmatrix} 2\left(\frac{d}{dr}\xi_1\right) & r^2\left(\frac{d}{dr}\xi_2\right) + \frac{d}{d\theta}\xi_1 \\ r^2\left(\frac{d}{dr}\xi_2\right) + \frac{d}{d\theta}\xi_1 & 2r^2\left(\frac{d}{d\theta}\xi_2\right) + 2r\,\xi_1 \end{pmatrix} \tag{\%t12}$$

(%i13)
```
map(ldisp,list_matrix_entries(B))$
```

$$2\left(\frac{d}{dr}\xi_1\right) \tag{\%t13}$$

$$r^2\left(\frac{d}{dr}\xi_2\right) + \frac{d}{d\theta}\xi_1 \tag{\%t14}$$

$$r^2\left(\frac{d}{dr}\xi_2\right) + \frac{d}{d\theta}\xi_1 \tag{\%t15}$$

$$2r^2\left(\frac{d}{d\theta}\xi_2\right) + 2r\,\xi_1 \tag{\%t16}$$

(%i19)
```
declare([C_1,C_2,C_3],constant)$
ξ:[C_2*cos(θ)+C_3*sin(θ),-C_2*sin(θ)/r+C_3*cos(θ)/r+C_1]$
B:zeromatrix(dim,dim)$
```

(%i20)
```
ev(Eq)$
```

(%i21)
```
ldisplay(B:trigsimp(B))$
```

$$B = \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix} \tag{\%t21}$$

2

(%i22) `kill(labels,t,r,θ,ϕ,f,g)$`

(%i1) `init_ctensor()$`

(%i5)
```
assume(0≤r)$
assume(0≤θ,θ≤π)$
assume(0≤sin(θ))$
assume(0≤ϕ,ϕ≤2*π)$
```

(%i6) `ζ:ct_coords:[t,r,θ,ϕ]$`

(%i7) `dim:length(ζ)$`

(%i8) `depends([f,g],r)$`

(%i13)
```
lg:zeromatrix(dim,dim)$
lg[1,1]:-exp(f)$
lg[2,2]:exp(g)$
lg[3,3]:r^2$
lg[4,4]:r^2*sin(θ)$
```

(%i14) `cmetric()$`

Covariant metric tensor

(%i15) `ishow(g([μ,ν])=lg)$`

$$g_{\mu\nu} = \begin{pmatrix} -e^{f} & 0 & 0 & 0 \\ 0 & e^{g} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2\,\sin(\theta) \end{pmatrix} \tag{\%t15}$$

Contravariant metric tensor

(%i16) `ishow(g([],[μ,ν])=ug)$`

$$g^{\mu\nu} = \begin{pmatrix} -e^{-f} & 0 & 0 & 0 \\ 0 & e^{-g} & 0 & 0 \\ 0 & 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & 0 & \frac{1}{r^2\,\sin(\theta)} \end{pmatrix} \tag{\%t16}$$

Line element

(%i17) `ldisplay(ds^2=line_element:diff(ζ).lg.transpose(diff(ζ)))$`

$$ds^2 = r^2\,\sin(\theta)\,\mathrm{del}(\phi)^2 + r^2\,\mathrm{del}(\theta)^2 - e^{f}\,\mathrm{del}(t)^2 + e^{g}\,\mathrm{del}(r)^2 \tag{\%t17}$$

(%i18) `christof(false)$`

Christoffel symbols of the first kind

(%i19)
```
for i thru dim do for j:i thru dim do for k thru dim do
if lcs[i,j,k]≠0 then
ishow(Γ([ζ[i],ζ[j],ζ[k]])=lcs[i,j,k])$
```

$$\Gamma_{ttr} = \frac{e^{f}\ (f_r)}{2} \tag{\%t19}$$

$$\Gamma_{trt} = -\frac{e^{f}\ (f_r)}{2} \tag{\%t19}$$

$$\Gamma_{rrr} = \frac{e^{g}\ (g_r)}{2} \tag{\%t19}$$

$$\Gamma_{r\theta\theta} = r \tag{\%t19}$$

$$\Gamma_{r\phi\phi} = r\ \sin(\theta) \tag{\%t19}$$

$$\Gamma_{\theta\theta r} = -r \tag{\%t19}$$

$$\Gamma_{\theta\phi\phi} = \frac{r^2\ \cos(\theta)}{2} \tag{\%t19}$$

$$\Gamma_{\phi\phi r} = -r\ \sin(\theta) \tag{\%t19}$$

$$\Gamma_{\phi\phi\theta} = -\frac{r^2\ \cos(\theta)}{2} \tag{\%t19}$$

Christoffel symbols of the second kind

(%i20)
```
for i thru dim do for j:i thru dim do for k thru dim do
if mcs[i,j,k]≠0 then
ishow(Γ([ζ[i],ζ[j]],[ζ[k]])=mcs[i,j,k])$
```

$$\Gamma_{tt}^{r} = \frac{(f_r)\ e^{f-g}}{2} \tag{\%t20}$$

$$\Gamma_{tr}^{t} = \frac{f_r}{2} \tag{\%t20}$$

$$\Gamma_{rr}^{r} = \frac{g_r}{2} \tag{\%t20}$$

$$\Gamma_{r\theta}^{\theta} = \frac{1}{r} \tag{\%t20}$$

$$\Gamma_{r\phi}^{\phi} = \frac{1}{r} \tag{\%t20}$$

$$\Gamma_{\theta\theta}^{r} = -e^{-g} r \tag{\%t20}$$

$$\Gamma_{\theta\phi}^{\phi} = \frac{\cos(\theta)}{2\sin(\theta)} \tag{\%t20}$$

$$\Gamma_{\phi\phi}^{r} = -e^{-g} r\ \sin(\theta) \tag{\%t20}$$

$$\Gamma_{\phi\phi}^{\theta} = -\frac{\cos(\theta)}{2} \tag{\%t20}$$

Riemann Tensor

(%i24)
```
riemann(false)$
lriemann(false)$
uriemann(false)$
for a thru dim do for b thru dim do
for c thru (if symmetricp(lg,dim) then b else dim) do
for d thru (if symmetricp(lg,dim) then a else dim) do
if riem[a,b,c,d]≠0 then
ishow(R([" ",ζ[a],ζ[b],ζ[c]],[ζ[d]])=trigsimp(riem[a,b,c,d]))$
```

$$R^{t}_{rrt} = \frac{(f_r)\ (g_r) - 2\,(f_{rr}) - (f_r)^2}{4} \tag{\%t24}$$

$$R^{t}_{\theta\theta t} = -\frac{(f_r)\ e^{-g} r}{2} \tag{\%t24}$$

$$R^{r}_{\theta\theta r} = \frac{e^{-g}\ (g_r)\,r}{2} \tag{\%t24}$$

$$R^{t}_{\phi\phi t} = -\frac{(f_r)\ e^{-g} r\ \sin(\theta)}{2} \tag{\%t24}$$

$$R^{r}_{\phi\phi r} = \frac{e^{-g}\ (g_r)\,r\ \sin(\theta)}{2} \tag{\%t24}$$

$$R^{\theta}_{\phi\phi\theta} = \frac{e^{-g}\ \left((e^g - 4)\ \sin(\theta)^2 + e^g\right)}{4\sin(\theta)} \tag{\%t24}$$

Ricci Tensor

(%i28)
```
ric:zeromatrix(dim,dim)$
ricci(false)$
uricci(false)$
for i thru dim do for j:i thru dim do
if ric[i,j]≠0 then
ishow(R([ζ[i],ζ[j]])=trigsimp(ric[i,j]))$
```

$$R_{tt} = -\frac{e^{f-g}\ \left(\left((f_r)\ (g_r) - 2\,(f_{rr}) - (f_r)^2\right) r - 4\,(f_r)\right)}{4r} \tag{\%t28}$$

$$R_{rr} = \frac{\left((f_r)\ (g_r) - 2\,(f_{rr}) - (f_r)^2\right) r + 4\,(g_r)}{4r} \tag{\%t28}$$

$$R_{\theta\theta} = \frac{e^{-g}\ \left(((2\,(g_r) - 2\,(f_r))\,r + e^g - 4)\ \sin(\theta)^2 + e^g\right)}{4\sin(\theta)^2} \tag{\%t28}$$

$$R_{\phi\phi} = \frac{e^{-g}\ \left(((2\,(g_r) - 2\,(f_r))\,r + e^g - 4)\ \sin(\theta)^2 + e^g\right)}{4\sin(\theta)} \tag{\%t28}$$

Scalar curvature

```
(%i29) trigsimp(scurvature());
```

$$\frac{e^{-g}\left(\left(\left((f_r)\;(g_r)-2\,(f_{rr})-(f_r)^2\right)\,r^2+\left(4\,(g_r)-4\,(f_r)\right)r+e^g-4\right)\,\sin(\theta)^2+e^g\right)}{2r^2\sin(\theta)^2} \tag{\%o29}$$

Kretchmann invariant

```
(%i30) trigsimp(rinvariant());
```

$$(e^{-2g}((((f_r)^2\,(g_r)^2+\left(-4\,(f_r)\;(f_{rr})-2(f_r)^3\right)\;(g_r)+4(f_{rr})^2+4(f_r)^2\;(f_{rr})+(f_r)^4)\,r^4+\left(8(g_r)^2+8(f_r)^2\right)\,r^2+e^{2g}-8e^g+16)\sin(\theta \tag{\%o30}$$

Unique Differential Equations

```
(%i31) map(ldisp,findde(ric,2))$
```

$$(f_r)\;(g_r)\,r-2\,(f_{rr})\,r-(f_r)^2r-4\,(f_r) \tag{\%t31}$$

$$(f_r)\;(g_r)\,r-2\,(f_{rr})\,r-(f_r)^2r+4\,(g_r) \tag{\%t32}$$

$$2\,(g_r)\,r\sin(\theta)^2-2\,(f_r)\,r\sin(\theta)^2+2e^g\sin(\theta)^2-4\sin(\theta)^2+e^g\cos(\theta)^2 \tag{\%t33}$$

```
(%i34) deindex;
```

$$[[1,1],[2,2],[3,3]] \tag{\%o34}$$

Einstein Tensor

```
(%i38) lein:zeromatrix(dim,dim)$
       einstein(false)$
       leinstein(false)$
       for i thru dim do for j:i thru dim do
       if lein[i,j]≠0 then
       ishow(G([ζ[i],ζ[j]])=expand(lein[i,j]))$
```

$$G_{tt}=\frac{e^f\cos(\theta)^2}{4r^2\sin(\theta)^2}+\frac{e^{f-g}\;(g_r)}{r}-\frac{e^{f-g}}{r^2}+\frac{e^f}{2r^2} \tag{\%t38}$$

$$G_{rr}=-\frac{e^g\cos(\theta)^2}{4r^2\sin(\theta)^2}+\frac{f_r}{r}-\frac{e^g}{2r^2}+\frac{1}{r^2} \tag{\%t38}$$

$$G_{\theta\theta}=-\frac{(f_r)\;e^{-g}\;(g_r)\;r^2}{4}+\frac{(f_{rr})\;e^{-g}\,r^2}{2}+\frac{(f_r)^2\,e^{-g}\,r^2}{4}-\frac{e^{-g}\;(g_r)\,r}{2}+\frac{(f_r)\;e^{-g}r}{2} \tag{\%t38}$$

$$G_{\phi\phi}=-\frac{(f_r)\;e^{-g}\;(g_r)\;r^2\;\sin(\theta)}{4}+\frac{(f_{rr})\;e^{-g}\,r^2\;\sin(\theta)}{2}+\frac{(f_r)^2\,e^{-g}\,r^2\;\sin(\theta)}{4}-\frac{e^{-g}\;(g_r)\,r\;\sin(\theta)}{2}+\frac{(f_r)\;e^{-g}r\;\sin(\theta)}{2} \tag{\%t38}$$

Computes the Geodesics

(%i39) `cgeodesic(false)$`

Solve for second derivative of coordinates

(%i40) `linsol:linsolve(listarray(geod),diff(ct_coords,s,2))$`

(%i41) `map(ldisp,linsol:expand(linsol))$`

$$t_{ss} = -(f_r)\ (r_s)\ (t_s) \tag{\%t41}$$

$$r_{ss} = e^{-g} r\ \sin(\theta)\,(\phi_s)^2 + e^{-g} r\,(\theta_s)^2 - \frac{(f_r)\ e^{f-g}\,(t_s)^2}{2} - \frac{(g_r)\ (r_s)^2}{2} \tag{\%t42}$$

$$\theta_{ss} = \frac{\cos(\theta)\,(\phi_s)^2}{2} - \frac{2\,(r_s)\ (\theta_s)}{r} \tag{\%t43}$$

$$\phi_{ss} = -\frac{\cos(\theta)\ (\theta_s)\ (\phi_s)}{\sin(\theta)} - \frac{2\,(r_s)\ (\phi_s)}{r} \tag{\%t44}$$

Reduce order

(%i45) `ξ:[T,R,Θ,Φ]$`

(%i46) `depends(ξ,s)$`

(%i50)
```
gradef(t,s,T)$
gradef(r,s,R)$
gradef(θ,s,Θ)$
gradef(ϕ,s,Φ)$
```

Compute the Geodesics

(%i51) `cgeodesic(false)$`

Solve for second derivative of coordinates

(%i52) `linsol:linsolve(listarray(geod),diff(ζ,s,2))$`

(%i53) `map(ldisp,linsol:expand(linsol))$`

$$T_s = -RT\ (f_r) \tag{\%t53}$$

$$R_s = e^{-g} r\ \Phi^2\ \sin(\theta) + e^{-g} r\ \Theta^2 - \frac{R^2\ (g_r)}{2} - \frac{T^2\ (f_r)\ e^{f-g}}{2} \tag{\%t54}$$

$$\Theta_s = \frac{\Phi^2\ \cos(\theta)}{2} - \frac{2R\Theta}{r} \tag{\%t55}$$

$$\Phi_s = -\frac{\Theta\Phi\ \cos(\theta)}{\sin(\theta)} - \frac{2R\Phi}{r} \tag{\%t56}$$

3

(%i57) `kill(labels,x_1,x_2,x_3)$`

(%i1) `init_ctensor()$`

(%i2) `ζ:ct_coords:[x_1,x_2,x_3]$`

(%i3) `dim:length(ζ)$`

(%i7)
```
lg:zeromatrix(dim,dim)$
lg[1,1]:1$
lg[2,2]:2*x_1$
lg[3,3]:2*x_2$
```

(%i8) `cmetric()$`

Covariant metric tensor

(%i9) `ishow(g([μ,ν])=lg)$`

$$g_{\mu\nu} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 2x_1 & 0 \\ 0 & 0 & 2x_2 \end{pmatrix} \qquad (\%t9)$$

Contravariant metric tensor

(%i10) `ishow(g([],[μ,ν])=ug)$`

$$g^{\mu\nu} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \frac{1}{2x_1} & 0 \\ 0 & 0 & \frac{1}{2x_2} \end{pmatrix} \qquad (\%t10)$$

Line element

(%i11) `ldisplay(ds^2=line_element:diff(ζ).lg.transpose(diff(ζ)))$`

$$ds^2 = 2x_2 \mathrm{del}\left(x_3\right)^2 + 2x_1 \mathrm{del}\left(x_2\right)^2 + \mathrm{del}\left(x_1\right)^2 \qquad (\%t11)$$

Determinant of the metric tensor

(%i12) `gdet;`

$$4x_1x_2 \qquad (\%o12)$$

Computes the Christoffel symbols of both kinds

(%i13) `christof(false)$`

Christoffel symbols of the first kind

(%i14)
```
for i thru dim do for j:i thru dim do for k thru dim do
if lcs[i,j,k]≠0 then
ishow(Γ([ζ[i],ζ[j],ζ[k]])=lcs[i,j,k])$
```

$$\Gamma_{x\_1 x\_2 x_2} = 1 \qquad (\%t14)$$

$$\mathit{\Gamma}_{x\_2x\_2x_1} = -1 \tag{\%t14}$$

$$\mathit{\Gamma}_{x\_2x\_3x_3} = 1 \tag{\%t14}$$

$$\mathit{\Gamma}_{x\_3x\_3x_2} = -1 \tag{\%t14}$$

Christoffel symbols of the second kind

(%i15)
```
for i thru dim do for j:i thru dim do for k thru dim do
if mcs[i,j,k]≠0 then
ishow(Γ([ζ[i],ζ[j]],[ζ[k]])=mcs[i,j,k])$
```

$$\mathit{\Gamma}^{x_2}_{x\_1x_2} = \frac{1}{2x_1} \tag{\%t15}$$

$$\mathit{\Gamma}^{x_1}_{x\_2x_2} = -1 \tag{\%t15}$$

$$\mathit{\Gamma}^{x_3}_{x\_2x_3} = \frac{1}{2x_2} \tag{\%t15}$$

$$\mathit{\Gamma}^{x_2}_{x\_3x_3} = -\frac{1}{2x_1} \tag{\%t15}$$

Riemann Tensor

(%i19)
```
riemann(false)$
lriemann(false)$
uriemann(false)$
for a thru dim do for b thru dim do
for c thru (if symmetricp(lg,dim) then b else dim) do
for d thru (if symmetricp(lg,dim) then a else dim) do
if riem[a,b,c,d]≠0 then
ishow(R([" ",ζ[a],ζ[b],ζ[c]],[ζ[d]])=riem[a,b,c,d])$
```

$$R^{x_1}_{x\_2x\_2x_1} = \frac{1}{2x_1} \tag{\%t19}$$

$$R^{x_2}_{x\_3x\_3x_1} = \frac{1}{4{x_1}^2} \tag{\%t19}$$

$$R^{x_1}_{x\_3x\_3x_2} = \frac{1}{2x_1} \tag{\%t19}$$

$$R^{x_2}_{x\_3x\_3x_2} = \frac{1}{4x_1x_2} \tag{\%t19}$$

Ricci Tensor

(%i23)
```
ric:zeromatrix(dim,dim)$
ricci(false)$
uricci(false)$
for i thru dim do for j:i thru dim do
if ric[i,j]≠0 then
ishow(R([ζ[i],ζ[j]])=ric[i,j])$
```

$$R_{x\_1x_1} = \frac{1}{4{x_1}^2} \tag{\%t23}$$

$$R_{x\_1 x_2} = \frac{1}{4 x_1 x_2} \tag{\%t23}$$

$$R_{x\_2 x_2} = \frac{2 {x_2}^2 + x_1}{4 x_1 {x_2}^2} \tag{\%t23}$$

$$R_{x\_3 x_3} = \frac{1}{4 x_1 x_2} \tag{\%t23}$$

Scalar curvature

(%i24) `scurvature();`

$$\frac{2 {x_2}^2 + x_1}{4 {x_1}^2 {x_2}^2} \tag{\%o24}$$

Kretchmann invariant

(%i25) `rinvariant();`

$$\frac{1}{4 {x_1}^3 {x_2}^2} + \frac{1}{16 {x_1}^2 {x_2}^4} + \frac{1}{4 {x_1}^4} \tag{\%o25}$$

Einstein Tensor

(%i29)
```
lein:zeromatrix(dim,dim)$
einstein(false)$
leinstein(false)$
for i thru dim do for j:i thru dim do
if lein[i,j]≠0 then
ishow(G([ζ[i],ζ[j]])=lein[i,j])$
```

$$G_{x\_1 x_1} = -\frac{1}{8 x_1 {x_2}^2} \tag{\%t29}$$

$$G_{x\_1 x_2} = \frac{1}{4 x_1 x_2} \tag{\%t29}$$

$$G_{x\_3 x_3} = -\frac{x_2}{2 {x_1}^2} \tag{\%t29}$$

Compute the Geodesics

(%i30) `cgeodesic(false)$`

Solve for second derivative of coordinates

(%i31) `linsol:linsolve(listarray(geod),diff(ζ,s,2))$`

(%i32) `map(ldisp,factor(linsol))$`

$$x_{1ss} = \left(x_{2s}\right)^2 \tag{\%t32}$$

$$x_{2ss} = \frac{\left(x_{3s}\right)^2 - 2\left(x_{1s}\right)\left(x_{2s}\right)}{2 x_1} \tag{\%t33}$$

$$x_{3ss} = -\frac{(x_{2s})\,(x_{3s})}{x_2} \tag{\%t34}$$

Reduce order

(%i35) `ξ:[X_1,X_2,X_3]$`

(%i36) `depends(ξ,s)$`

(%i39)
```
gradef(x_1,s,X_1)$
gradef(x_2,s,X_2)$
gradef(x_3,s,X_3)$
```

Compute the Geodesics

(%i40) `cgeodesic(false)$`

Solve for second derivative of coordinates

(%i41) `linsol:linsolve(listarray(geod),diff(ζ,s,2))$`

(%i42) `map(ldisp,linsol)$`

$$X_{1s} = {X_2}^2 \tag{\%t42}$$

$$X_{2s} = -\frac{2X_1X_2 - {X_3}^2}{2x_1} \tag{\%t43}$$

$$X_{3s} = -\frac{X_2X_3}{x_2} \tag{\%t44}$$

4 Killing Vector for Cylindrical Coordinates

(%i45) `kill(labels,r,θ,z)$`

(%i1) `ζ:[r,θ,z]$`

(%i2) `Tr:[r*cos(θ),r*sin(θ),z]$`

(%i3) `ct_coordsys(append(Tr,[ζ]),all)$`

(%i5)
```
lg:trigsimp(lg)$
ug:trigsimp(ug)$
```

4.1

(%i4)
```
kill(labels,ξ_1,ξ_2,ξ_3)$
ξ:[ξ_1,ξ_2,ξ_3]$
depends(ξ,ζ)$
ldisplay(ξ:ev(ξ))$
B:zeromatrix(dim,dim)$
```

$$\xi = [\xi_1, \xi_2, \xi_3] \tag{\%t3}$$

(%i5) `ev(Eq)$`

(%i6) `ldisplay(B)$`

$$B = \begin{pmatrix} 2\left(\xi_{1r}\right) & r^2\ \left(\xi_{2r}\right) + \xi_{1\theta} & \xi_{3r} + \xi_{1z} \\ r^2\ \left(\xi_{2r}\right) + \xi_{1\theta} & 2r^2\ \left(\xi_{2\theta}\right) + 2r\,\xi_1 & \xi_{3\theta} + r^2\ \left(\xi_{2z}\right) \\ \xi_{3r} + \xi_{1z} & \xi_{3\theta} + r^2\ \left(\xi_{2z}\right) & 2\left(\xi_{3z}\right) \end{pmatrix} \tag{\%t6}$$

(%i7) `symmetricp(B,dim);`

$$\text{true} \tag{\%o7}$$

(%i8) `map(ldisp,list_matrix_entries(B))$`

$$2\left(\xi_{1r}\right) \tag{\%t8}$$

$$r^2\ \left(\xi_{2r}\right) + \xi_{1\theta} \tag{\%t9}$$

$$\xi_{3r} + \xi_{1z} \tag{\%t10}$$

$$r^2\ \left(\xi_{2r}\right) + \xi_{1\theta} \tag{\%t11}$$

$$2r^2\ \left(\xi_{2\theta}\right) + 2r\,\xi_1 \tag{\%t12}$$

$$\xi_{3\theta} + r^2\ \left(\xi_{2z}\right) \tag{\%t13}$$

$$\xi_{3r} + \xi_{1z} \tag{\%t14}$$

$$\xi_{3\theta} + r^2\ \left(\xi_{2z}\right) \tag{\%t15}$$

$$2\left(\xi_{3z}\right) \tag{\%t16}$$

4.2

(%i6)
```
kill(labels,ξ_1,ξ_2,ξ_3)$
ξ:[ξ_1,ξ_2,ξ_3]$
depends(ξ_1,[θ,z])$
depends(ξ_2,ζ)$
depends(ξ_3,[r,θ])$
ldisplay(ξ:ev(ξ))$
B:zeromatrix(dim,dim)$
```

$$\xi = [\xi_1, \xi_2, \xi_3] \tag{\%t5}$$

(%i7)
```
ev(Eq)$
```

(%i8)
```
ldisplay(B)$
```

$$B = \begin{pmatrix} 0 & r^2\,(\xi_{2r}) + \xi_{1\theta} & \xi_{3r} + \xi_{1z} \\ r^2\,(\xi_{2r}) + \xi_{1\theta} & 2r^2\,(\xi_{2\theta}) + 2r\,\xi_1 & \xi_{3\theta} + r^2\,(\xi_{2z}) \\ \xi_{3r} + \xi_{1z} & \xi_{3\theta} + r^2\,(\xi_{2z}) & 0 \end{pmatrix} \tag{\%t8}$$

(%i9)
```
symmetricp(B,dim);
```

$$\text{true} \tag{\%o9}$$

(%i10)
```
map(ldisp,list_matrix_entries(B))$
```

$$0 \tag{\%t10}$$

$$r^2\,(\xi_{2r}) + \xi_{1\theta} \tag{\%t11}$$

$$\xi_{3r} + \xi_{1z} \tag{\%t12}$$

$$r^2\,(\xi_{2r}) + \xi_{1\theta} \tag{\%t13}$$

$$2r^2\,(\xi_{2\theta}) + 2r\,\xi_1 \tag{\%t14}$$

$$\xi_{3\theta} + r^2\,(\xi_{2z}) \tag{\%t15}$$

$$\xi_{3r} + \xi_{1z} \tag{\%t16}$$

$$\xi_{3\theta} + r^2\,(\xi_{2z}) \tag{\%t17}$$

$$0 \tag{\%t18}$$

4.3

(%i7)
```
kill(labels,ξ_1,ξ_2,ξ_3)$
ξ:[ξ_1,ξ_2,ξ_3]$
depends(f,[θ,z])$ξ_1:f$
depends(ξ_2,ζ)$
depends(ξ_3,[r,θ])$
ldisplay(ξ:ev(ξ))$
B:zeromatrix(dim,dim)$
```

$$\xi = [f, \xi_2, \xi_3] \tag{\%t6}$$

(%i8) `ev(Eq)$`

(%i9) `ldisplay(B)$`

$$B = \begin{pmatrix} 0 & r^2\ (\xi_{2r}) + f_\theta & \xi_{3r} + f_z \\ r^2\ (\xi_{2r}) + f_\theta & 2r^2\ (\xi_{2\theta}) + 2fr & \xi_{3\theta} + r^2\ (\xi_{2z}) \\ \xi_{3r} + f_z & \xi_{3\theta} + r^2\ (\xi_{2z}) & 0 \end{pmatrix} \tag{\%t9}$$

(%i10) `map(ldisp,list_matrix_entries(B))$`

$$0 \tag{\%t10}$$

$$r^2\ (\xi_{2r}) + f_\theta \tag{\%t11}$$

$$\xi_{3r} + f_z \tag{\%t12}$$

$$r^2\ (\xi_{2r}) + f_\theta \tag{\%t13}$$

$$2r^2\ (\xi_{2\theta}) + 2fr \tag{\%t14}$$

$$\xi_{3\theta} + r^2\ (\xi_{2z}) \tag{\%t15}$$

$$\xi_{3r} + f_z \tag{\%t16}$$

$$\xi_{3\theta} + r^2\ (\xi_{2z}) \tag{\%t17}$$

$$0 \tag{\%t18}$$

4.4

(%i19) `integrate(solve(B[1,2],diff(ξ_2,r)),r);`

$$\left[\int \xi_{2r} dr = \frac{f_\theta}{r} + \%c1\right] \tag{\%o19}$$

(%i8)
```
kill(labels,ξ_1,ξ_2,ξ_3)$
ξ:[ξ_1,ξ_2,ξ_3]$
depends(f,[θ,z])$ξ_1:f$
depends(g,[θ,z])$ξ_2:diff(f,θ)/r+g$
depends(ξ_3,[r,θ])$
ldisplay(ξ:ev(ξ))$
B:zeromatrix(dim,dim)$
```

$$\xi = \left[f, \frac{f_\theta}{r} + g, \xi_3\right] \tag{\%t7}$$

(%i9) `ev(Eq)$`

(%i10) `ldisplay(B)$`

$$B = \begin{pmatrix} 0 & 0 & \xi_{3r} + f_z \\ 0 & 2\left(\frac{f_{\theta\theta}}{r} + g_\theta\right) r^2 + 2fr & \xi_{3\theta} + \left(\frac{f_{z\theta}}{r} + g_z\right) r^2 \\ \xi_{3r} + f_z & \xi_{3\theta} + \left(\frac{f_{z\theta}}{r} + g_z\right) r^2 & 0 \end{pmatrix} \tag{\%t10}$$

(%i11) `map(ldisp,list_matrix_entries(B))$`

$$0 \tag{\%t11}$$

$$0 \tag{\%t12}$$

$$\xi_{3r} + f_z \tag{\%t13}$$

$$0 \tag{\%t14}$$

$$2\left(\frac{f_{\theta\theta}}{r} + g_\theta\right) r^2 + 2fr \tag{\%t15}$$

$$\xi_{3\theta} + \left(\frac{f_{z\theta}}{r} + g_z\right) r^2 \tag{\%t16}$$

$$\xi_{3r} + f_z \tag{\%t17}$$

$$\xi_{3\theta} + \left(\frac{f_{z\theta}}{r} + g_z\right) r^2 \tag{\%t18}$$

$$0 \tag{\%t19}$$

4.5

(%i20) `integrate(solve(B[1,3],diff(ξ_3,r)),r);`

$$\left[\int \xi_{3r} dr = \%c2 - (f_z)\, r\right] \tag{\%o20}$$

(%i9)
```
kill(labels,ξ_1,ξ_2,ξ_3)$
ξ:[ξ_1,ξ_2,ξ_3]$
depends(f,[θ,z])$ξ_1:f$
depends(g,[θ,z])$ξ_2:diff(f,θ)/r+g$
depends(h,[θ,z])$ξ_3:-r*diff(f,z)+h$
ldisplay(ξ:ev(ξ))$
B:zeromatrix(dim,dim)$
```

$$\xi = \left[f, \frac{f_\theta}{r} + g, h - (f_z)\, r\right] \tag{\%t8}$$

(%i10) `ev(Eq)$`

(%i11) `ldisplay(B)$`

$$B = \begin{pmatrix} 0 & 0 & 0 \\ 0 & 2\left(\frac{f_{\theta\theta}}{r} + g_\theta\right) r^2 + 2fr & \left(\frac{f_{z\theta}}{r} + g_z\right) r^2 - (f_{z\theta})\, r + h_\theta \\ 0 & \left(\frac{f_{z\theta}}{r} + g_z\right) r^2 - (f_{z\theta})\, r + h_\theta & 2\,(h_z) - 2\,(f_{zz})\, r \end{pmatrix} \tag{\%t11}$$

(%i12) `map(ldisp,list_matrix_entries(B))$`

$$0 \tag{\%t12}$$

$$0 \tag{\%t13}$$

$$0 \tag{\%t14}$$

$$0 \tag{\%t15}$$

$$2\left(\frac{f_{\theta\theta}}{r}+g_\theta\right)r^2+2fr \tag{\%t16}$$

$$\left(\frac{f_{z\theta}}{r}+g_z\right)r^2-(f_{z\theta})\,r+h_\theta \tag{\%t17}$$

$$0 \tag{\%t18}$$

$$\left(\frac{f_{z\theta}}{r}+g_z\right)r^2-(f_{z\theta})\,r+h_\theta \tag{\%t19}$$

$$2\,(h_z)-2\,(f_{zz})\,r \tag{\%t20}$$

4.6

(%i9)
```
kill(labels,ξ_1,ξ_2,ξ_3)$
ξ:[ξ_1,ξ_2,ξ_3]$
depends(f,[θ,z])$ξ_1:f$
depends(g,[θ])$ξ_2:diff(f,θ)/r+g$
depends(h,[z])$ξ_3:-r*diff(f,z)+h$
ldisplay(ξ:ev(ξ))$
B:zeromatrix(dim,dim)$
```

$$\xi=\left[f,\frac{f_\theta}{r}+g,h-(f_z)\,r\right] \tag{\%t8}$$

(%i10)
```
ev(Eq)$
```

(%i11)
```
ldisplay(B)$
```

$$B=\begin{pmatrix}0 & 0 & 0\\ 0 & 2\left(\frac{f_{\theta\theta}}{r}+g_\theta\right)r^2+2fr & 0\\ 0 & 0 & 2\,(h_z)-2\,(f_{zz})\,r\end{pmatrix} \tag{\%t11}$$

(%i12)
```
map(ldisp,list_matrix_entries(B))$
```

$$0 \tag{\%t12}$$

$$0 \tag{\%t13}$$

$$0 \tag{\%t14}$$

$$0 \tag{\%t15}$$

$$2\left(\frac{f_{\theta\theta}}{r}+g_\theta\right)r^2+2fr \tag{\%t16}$$

$$0 \tag{\%t17}$$

$$0 \tag{\%t18}$$

$$0 \tag{\%t19}$$

$$2\,(h_z)-2\,(f_{zz})\,r \tag{\%t20}$$

4.7

(%i21) `expand(B[2,2]);`

$$2\left(g_{\theta}\right)\, r^2 + 2\left(f_{\theta\theta}\right) r + 2 f r \qquad (\%o21)$$

(%i22)
```
Eq1:f*coeff(expand(B[2,2]),f)+
diff(f,θ,1)*coeff(expand(B[2,2]),diff(f,θ,1))+
diff(f,θ,2)*coeff(expand(B[2,2]),diff(f,θ,2));
```

$$2\left(f_{\theta\theta}\right) r + 2 f r \qquad (\mathrm{Eq1})$$

(%i23) `sol1:ode2(Eq1,f,θ);`

$$f = \%k1\ \sin(\theta) + \%k2\ \cos(\theta) \qquad (\mathrm{sol1})$$

(%i24)
```
Eq2:g*coeff(expand(B[2,2]),g)+
diff(g,θ,1)*coeff(expand(B[2,2]),diff(g,θ,1))+
diff(g,θ,2)*coeff(expand(B[2,2]),diff(g,θ,2));
```

$$2\left(g_{\theta}\right)\, r^2 \qquad (\mathrm{Eq2})$$

(%i25) `sol2:ode2(Eq2,g,θ);`

$$g = \%c \qquad (\mathrm{sol2})$$

(%i7)
```
kill(labels,ξ_1,ξ_2,ξ_3)$
ξ:[ξ_1,ξ_2,ξ_3]$
declare([C_1,C_2,C_3,C_4],constant)$
ξ_1:C_2*sin(θ)+C_3*cos(θ)$
ξ_2:diff(ξ_1,θ)/r+C_1$
ξ_3:C_4$
ldisplay(ξ:ev(ξ))$
B:zeromatrix(dim,dim)$
```

$$\xi = \left[C_2 \sin(\theta) + C_3 \cos(\theta), \frac{C_2 \cos(\theta) - C_3 \sin(\theta)}{r} + C_1, C_4\right] \qquad (\%t6)$$

(%i8) `ev(Eq)$`

(%i9) `ldisplay(B:trigsimp(B))$`

$$B = \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix} \qquad (\%t9)$$

4.8

(%i7)
```
kill(labels,ξ_1,ξ_2,ξ_3)$
ξ:[ξ_1,ξ_2,ξ_3]$
declare([C_1,C_2,C_3,C_4],constant)$
ξ_1:C_2+C_3*exp(-θ)$
ξ_2:C_1-C_3*exp(-θ)/r$
ξ_3:r*exp(-θ)+C_4$
ldisplay(ξ:ev(ξ))$
B:zeromatrix(dim,dim)$
```

$$\xi = \left[C_3 e^{-\theta} + C_2, C_1 - \frac{C_3 e^{-\theta}}{r}, r\,e^{-\theta} + C_4\right] \tag{\%t6}$$

(%i8)
```
ev(Eq)$
```

(%i9)
```
ldisplay(B:expand(B))$
```

$$B = \begin{pmatrix} 0 & 0 & e^{-\theta} \\ 0 & 4C_3 r\,e^{-\theta} + 2C_2 r & -r\,e^{-\theta} \\ e^{-\theta} & -r\,e^{-\theta} & 0 \end{pmatrix} \tag{\%t9}$$

(%i10)
```
map(ldisp,list_matrix_entries(B))$
```

$$0 \tag{\%t10}$$

$$0 \tag{\%t11}$$

$$e^{-\theta} \tag{\%t12}$$

$$0 \tag{\%t13}$$

$$4C_3 r\,e^{-\theta} + 2C_2 r \tag{\%t14}$$

$$-r\,e^{-\theta} \tag{\%t15}$$

$$e^{-\theta} \tag{\%t16}$$

$$-r\,e^{-\theta} \tag{\%t17}$$

$$0 \tag{\%t18}$$

5 Geodesics for Polar Coordinates

(%i19) `kill(labels,r,θ)$`

(%i1) `init_ctensor()$`

(%i3) `assume(0≤r)$`
`assume(0≤θ,θ≤2*π)$`

(%i4) `ζ:ct_coords:[r,θ]$`

(%i5) `dim:length(ζ)$`

(%i8) `lg:zeromatrix(dim,dim)$`
`lg[1,1]:1$`
`lg[2,2]:r^2$`

(%i9) `cmetric()$`

Covariant metric tensor

(%i10) `ishow(g([μ,ν])=lg)$`

$$g_{\mu\nu} = \begin{pmatrix} 1 & 0 \\ 0 & r^2 \end{pmatrix} \tag{\%t10}$$

Contravariant metric tensor

(%i11) `ishow(g([],[μ,ν])=ug)$`

$$g^{\mu\nu} = \begin{pmatrix} 1 & 0 \\ 0 & \frac{1}{r^2} \end{pmatrix} \tag{\%t11}$$

Line element

(%i12) `ldisplay(ds^2=line_element:diff(ζ).lg.transpose(diff(ζ)))$`

$$ds^2 = r^2 \operatorname{del}(\theta)^2 + \operatorname{del}(r)^2 \tag{\%t12}$$

(%i13) `christof(false)$`

Christoffel symbols of the first kind

(%i14) `for i thru dim do for j:i thru dim do for k thru dim do`
`if lcs[i,j,k]≠0 then`
`ishow(Γ([ζ[i],ζ[j],ζ[k]])=lcs[i,j,k])$`

$$\Gamma_{r\theta\theta} = r \tag{\%t14}$$

$$\Gamma_{\theta\theta r} = -r \tag{\%t14}$$

Christoffel symbols of the second kind

(%i15)
```
for i thru dim do for j:i thru dim do for k thru dim do
if mcs[i,j,k]≠0 then
ishow(Γ([ζ[i],ζ[j]],[ζ[k]])=mcs[i,j,k])$
```

$$\Gamma_{r\theta}^{\theta} = \frac{1}{r} \tag{\%t15}$$

$$\Gamma_{\theta\theta}^{r} = -r \tag{\%t15}$$

Compute the Geodesics

(%i16)
```
cgeodesic(false)$
```

Solve for second derivative of coordinates

(%i17)
```
linsol:linsolve(listarray(geod),diff(ct_coords,s,2))$
```

(%i18)
```
map(ldisp,linsol)$
```

$$r_{ss} = r\left(\theta_s\right)^2 \tag{\%t18}$$

$$\theta_{ss} = -\frac{2\left(r_s\right)\,\left(\theta_s\right)}{r} \tag{\%t19}$$

(%i20)
```
eliminate(listarray(geod),[diff(θ,s)]);
```

$$\left[4(r_s)^2\,\left(r_{ss}\right) - r^3\left(\theta_{ss}\right)^2\right] \tag{\%o20}$$

6 Geodesics for Cylindrical Coordinates

(%i21) `kill(labels,r,θ,z)$`

(%i1) `init_ctensor()$`

(%i3) `assume(0≤r)$`
`assume(0≤θ,θ≤2*π)$`

(%i4) `ζ:ct_coords:[r,θ,z]$`

(%i5) `dim:length(ζ)$`

(%i9) `lg:zeromatrix(dim,dim)$`
`lg[1,1]:1$`
`lg[2,2]:r^2$`
`lg[3,3]:1$`

(%i10) `cmetric()$`

Covariant metric tensor

(%i11) `ishow(g([μ,ν])=lg)$`

$$g_{\mu\nu} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & r^2 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t11}$$

Contravariant metric tensor

(%i12) `ishow(g([],[μ,ν])=ug)$`

$$g^{\mu\nu} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t12}$$

Line element

(%i13) `ldisplay(ds^2=line_element:diff(ζ).lg.transpose(diff(ζ)))$`

$$ds^2 = r^2 \operatorname{del}(\theta)^2 + \operatorname{del}(z)^2 + \operatorname{del}(r)^2 \tag{\%t13}$$

(%i14) `christof(false)$`

Christoffel symbols of the first kind

(%i15) `for i thru dim do for j:i thru dim do for k thru dim do`
`if lcs[i,j,k]≠0 then`
`ishow(Γ([ζ[i],ζ[j],ζ[k]])=lcs[i,j,k])$`

$$\Gamma_{r\theta\theta} = r \tag{\%t15}$$

$$\Gamma_{\theta\theta r} = -r \tag{\%t15}$$

Christoffel symbols of the second kind

(%i16)
```
for i thru dim do for j:i thru dim do for k thru dim do
if mcs[i,j,k]≠0 then
ishow(Γ([ζ[i],ζ[j]],[ζ[k]])=mcs[i,j,k])$
```

$$\Gamma^{\theta}_{r\theta} = \frac{1}{r} \tag{\%t16}$$

$$\Gamma^{r}_{\theta\theta} = -r \tag{\%t16}$$

Computes the Geodesics

(%i17)
```
cgeodesic(false)$
```

Solve for second derivative of coordinates

(%i18)
```
linsol:linsolve(listarray(geod),diff(ct_coords,s,2))$
```

(%i19)
```
map(ldisp,linsol)$
```

$$r_{ss} = r\,(\theta_s)^2 \tag{\%t19}$$

$$\theta_{ss} = -\frac{2\,(r_s)\,(\theta_s)}{r} \tag{\%t20}$$

$$z_{ss} = 0 \tag{\%t21}$$

(%i22)
```
eliminate(listarray(geod),[diff(θ,s)]);
```

$$\left[(z_{ss})^2, 4(r_s)^2\,(r_{ss}) - r^3\,(\theta_{ss})^2\right] \tag{\%o22}$$