CHAPTER 1 INTRODUCTION

Lecture Notes for Differential Geometry
James S. Cook
Liberty University
Department of Mathematics
Summer 2015

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

(%i2) `reset()$kill(all)$`

(%i1) `derivabbrev:true$`

(%i2) `ratprint:false$`

(%i3) `fpprintprec:5$`

(%i4) `load(linearalgebra)$`

(%i5) `if get('draw,'version)=false then load(draw)$`

(%i6) `wxplot_size:[1024,768]$`

(%i7) `if get('drawdf,'version)=false then load(drawdf)$`

(%i8)
```
set_draw_defaults(xtics=1,ytics=1,ztics=1,xyplane=0,nticks=100,
xaxis=true,xaxis_type=solid,xaxis_width=3,
yaxis=true,yaxis_type=solid,yaxis_width=3,
zaxis=true,zaxis_type=solid,zaxis_width=3,
background_color=light_gray)$
```

(%i9) `if get('vect,'version)=false then load(vect)$`

(%i10) `norm(u):=block(ratsimp(radcan(√(u.u))))$`

(%i11) `normalize(v):=block(v/norm(v))$`

(%i12) `angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$`

(%i13) `mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$`

(%i14) `if get('cartan,'version)=false then load(cartan)$`

(%i15) `declare(trigsimp,evfun)$`

1 points and vectors

2 on tangent and cotangent spaces and bundles

2.1 Example 1.2.2.

Define $f = 2\,x^1 - 3\,x^4 + (x_n)^2$ then $f(p) = \left(2\,x^1 - 3\,x^4 + (x_n)^2\right)(p)$ and by the usual addition of functions $f(p) = 2\,p^1 - 3p^4 + (p_n)^2$.

2.2 Example 1.2.3.

Define $f = x + y^2 + z^3$. Observe $f(a, b, c) = a + b^2 + c^3$.

(%i16) `ldisplay(f:x+y^2+z^3)$`

$$f = z^3 + y^2 + x \qquad (\%t16)$$

(%i17) `at(f,[x=a,y=b,z=c]);`

$$c^3 + b^2 + a \qquad (\%o17)$$

2.3 Example 1.2.13.

Let $f = \sqrt{(x^1)^2 + \cdots + (x^n)^2}$ clearly $f^2 = (x^1)^2 + \cdots + (x^n)^2$ and thus $\mathrm{d}f^2 = 2f\,\mathrm{d}f$ and $\mathrm{d}(f^2) = 2x^1\,\mathrm{d}x^1 + \cdots + 2x^n\,\mathrm{d}x^n$ thus

$$\mathrm{d}f = \frac{x^1\,\mathrm{d}x^1 + \cdots + x^n\,\mathrm{d}x^n}{\sqrt{(x^1)^2 + \cdots + (x^n)^2}}$$

3 the wedge product and differential forms

3.1 Example 1.3.4.

A force $\vec{F}$ is conservative if there exists f such that $\vec{F} = -\nabla\phi$. In the language of differential forms, this means the one-form $\omega_{\vec{F}}$ represents a conservative force if $\omega_{\vec{F}} = \omega_{-\nabla\phi} = -\mathrm{d}\phi$. Observe, $\omega_{\vec{F}} = -\mathrm{d}\phi$ implies $\mathrm{d}\omega_{\vec{F}} = -\mathrm{d}^2\phi = 0$. As an application, consider $\omega_{\vec{F}} = -y\,\mathrm{d}x + x\,\mathrm{d}y + \mathrm{d}z$, is $\vec{F}$ conservative?
Calculate

$$\mathrm{d}\omega_{\vec{F}} = -\mathrm{d}y \wedge \mathrm{d}x + \mathrm{d}x \wedge \mathrm{d}y + \mathrm{d}(\mathrm{d}z) = 2\,\mathrm{d}x \wedge \mathrm{d}y \neq 0$$

```
(%i18) kill(labels,x,y,z)$
(%i1)  ζ:[x,y,z]$
(%i2)  init_cartan(ζ)$
(%i3)  ldisplay(ω_F:-y*dx+x*dy+dz)$
```

$$\omega_F = dz + x\,dy - y\,dx \tag{\%t3}$$

```
(%i4)  ldisplay(dω_F:edit(ext_diff(ω_F)))$
```

$$d\omega_F = 2dx\,dy \tag{\%t4}$$

thus $\vec{F}$ is not conservative.

3.2 Example 1.3.5.

It turns out that Maxwell's equations can be expressed in terms of the exterior derivative of the potential one-form A. The one-form contains both the voltage function and the magnetic vector-potential from which the time-derivative and gradient derive the electric and magnetic Fields. In spacetime the relation between the potentials and fields is simply $\vec{F} = \mathrm{d}A$. The choice of A is far from unique. There is a **gauge freedom**. In particular, we can add an exterior derivative function of spacetime λ and create $A' = A + \mathrm{d}\lambda$. Note, $\mathrm{d}A' = \mathrm{d}A + \mathrm{d}^2\lambda$ hence A and A' generate the same electric and magnetic fields (which make up the Faraday tensor F)

3.3 Definition 1.3.6.

Work and Flux form correspondance

```
(%i5)  ζ:[x,y,z]$
(%i6)  init_cartan(ζ)$
(%i7)  F:[P,Q,R]$
(%i8)  ldisplay(ω_F:F.cartan_basis)$
```

$$\omega_F = R\,dz + Q\,dy + P\,dx \tag{\%t8}$$

```
(%i10) cartan_flow:makelist((dx~dy~dz)/k,k,cartan_basis)$
       cartan_sign:makelist((-1)^(k+1),k,cartan_dim)$
(%i11) ldisplay(Φ_F:F.(cartan_flow*cartan_sign))$
```

$$\Phi_F = P\,dy\,dz - Q\,dx\,dz + R\,dx\,dy \tag{\%t11}$$

3.4 Proposition 1.3.7.

Vector algebra in Differential form.

(%i12)
```
kill(t,x,y,z)$
```

(%i13)
```
ζ:[x,y,z]$
```

(%i14)
```
init_cartan(ζ)$
```

(%i17)
```
A:[A_1,A_2,A_3]$
B:[B_1,B_2,B_3]$
C:[C_1,C_2,C_3]$
```

(%i20)
```
ldisplay(ω_A:A.cartan_basis)$
ldisplay(ω_B:B.cartan_basis)$
ldisplay(ω_C:C.cartan_basis)$
```

$$\omega_A = A_3\,dz + A_2\,dy + A_1\,dx \tag{\%t18}$$

$$\omega_B = B_3\,dz + B_2\,dy + B_1\,dx \tag{\%t19}$$

$$\omega_C = C_3\,dz + C_2\,dy + C_1\,dx \tag{\%t20}$$

(%i22)
```
cartan_flow:makelist((dx~dy~dz)/k,k,cartan_basis)$
cartan_sign:makelist((-1)^(k+1),k,cartan_dim)$
```

(%i25)
```
ldisplay(Φ_A:A.(cartan_flow*cartan_sign))$
ldisplay(Φ_B:B.(cartan_flow*cartan_sign))$
ldisplay(Φ_C:C.(cartan_flow*cartan_sign))$
```

$$\mathit{\Phi}_A = A_1\,dy\ dz - A_2\,dx\ dz + A_3\,dx\ dy \tag{\%t23}$$

$$\mathit{\Phi}_B = B_1\,dy\ dz - B_2\,dx\ dz + B_3\,dx\ dy \tag{\%t24}$$

$$\mathit{\Phi}_C = C_1\,dy\ dz - C_2\,dx\ dz + C_3\,dx\ dy \tag{\%t25}$$

$\omega_A \wedge \omega_B = \Phi_{A\times B}$

(%i26)
```
edit(ω_A~ω_B);
```

$$(A_2B_3 - A_3B_2)\ dy\ dz + (A_1B_3 - A_3B_1)\ dx\ dz + (A_1B_2 - A_2B_1)\ dx\ dy \tag{\%o26}$$

(%i27)
```
edit(mycross(A,B).(cartan_flow*cartan_sign));
```

$$(A_2B_3 - A_3B_2)\ dy\ dz + (A_1B_3 - A_3B_1)\ dx\ dz + (A_1B_2 - A_2B_1)\ dx\ dy \tag{\%o27}$$

(%i28)
```
is(%=%th(2));
```

$$\text{true} \tag{\%o28}$$

$\omega_A \wedge \omega_B \wedge \omega_C = A \cdot (B \times C)\,\mathrm{d}x \wedge\ \mathrm{d}y \wedge\ \mathrm{d}z$

(%i29)
```
edit(ω_A~ω_B~ω_C);
```

$$(A_1B_2C_3 - A_2B_1C_3 - A_1B_3C_2 + A_3B_1C_2 + A_2B_3C_1 - A_3B_2C_1)\ dx\ dy\ dz \tag{\%o29}$$

(%i30) `edit(A.(mycross(B,C))*(dx~dy~dz));`

$$(A_1B_2C_3 - A_2B_1C_3 - A_1B_3C_2 + A_3B_1C_2 + A_2B_3C_1 - A_3B_2C_1)\,dx\,dy\,dz \tag{\%o30}$$

(%i31) `is(%=%th(2));`

$$\text{true} \tag{\%o31}$$

$\omega_A \wedge \Phi_B = (A \cdot B)\,\mathrm{d}x \wedge \mathrm{d}y \wedge \mathrm{d}z$

(%i32) `edit(ω_A~Φ_B);`

$$(A_3B_3 + A_2B_2 + A_1B_1)\,dx\,dy\,dz \tag{\%o32}$$

(%i33) `edit((A.B)*(dx~dy~dz));`

$$(A_3B_3 + A_2B_2 + A_1B_1)\,dx\,dy\,dz \tag{\%o33}$$

(%i34) `is(%=%th(2));`

$$\text{true} \tag{\%o34}$$

Differential vector calculus in differential form.

3.5 Proposition 1.3.8.

(%i35) `kill(t,x,y,z,f)$`

(%i36) `ζ:[x,y,z]$`

(%i37) `init_cartan(ζ)$`

(%i39)
```
F:[F_1,F_2,F_3]$
G:[G_1,G_2,G_3]$
```

(%i42)
```
depends(f,ζ)$
depends(F,ζ)$
depends(G,ζ)$
```

(%i44)
```
ldisplay(ω_F:F.cartan_basis)$
ldisplay(ω_G:G.cartan_basis)$
```

$$\omega_F = F_3\,dz + F_2\,dy + F_1\,dx \tag{\%t43}$$

$$\omega_G = G_3\,dz + G_2\,dy + G_1\,dx \tag{\%t44}$$

(%i46)
```
cartan_flow:makelist((dx~dy~dz)/k,k,cartan_basis)$
cartan_sign:makelist((-1)^(k+1),k,cartan_dim)$
```

(%i48)
```
ldisplay(Φ_F:F.(cartan_flow*cartan_sign))$
ldisplay(Φ_G:G.(cartan_flow*cartan_sign))$
```

$$\Phi_F = F_1\,dy\,dz - F_2\,dx\,dz + F_3\,dx\,dy \tag{\%t47}$$

$$\Phi_G = G_1 dy\, dz - G_2 dx\, dz + G_3 dx\, dy \tag{\%t48}$$

$\mathrm{d}f = \omega_{\nabla f}$

(%i49) `ext_diff(f);`

$$(f_z)\ dz + (f_y)\ dy + (f_x)\ dx \tag{\%o49}$$

(%i50) `ev(express(grad(f)),diff).cartan_basis;`

$$(f_z)\ dz + (f_y)\ dy + (f_x)\ dx \tag{\%o50}$$

(%i51) `is(%=%th(2));`

$$\text{true} \tag{\%o51}$$

$\mathrm{d}\omega_F = \Phi_{\nabla\times F}$

(%i52) `edit(ext_diff(ω_F));`

$$\left(F_{3y} - F_{2z}\right) dy\, dz + \left(F_{3x} - F_{1z}\right) dx\, dz + \left(F_{2x} - F_{1y}\right) dx\, dy \tag{\%o52}$$

(%i53) `edit(ev(express(curl(F)),diff).(cartan_flow*cartan_sign));`

$$\left(F_{3y} - F_{2z}\right) dy\, dz + \left(F_{3x} - F_{1z}\right) dx\, dz + \left(F_{2x} - F_{1y}\right) dx\, dy \tag{\%o53}$$

(%i54) `is(%=%th(2));`

$$\text{true} \tag{\%o54}$$

$\mathrm{d}\Phi_G = (\nabla\cdot G)\,\mathrm{d}x\wedge\,\mathrm{d}y\wedge\,\mathrm{d}z$

(%i55) `edit(ext_diff(Φ_G));`

$$\left(G_{3z} + G_{2y} + G_{1x}\right) dx\, dy\, dz \tag{\%o55}$$

(%i56) `ev(express(div(G)),diff)*(dx~dy~dz);`

$$\left(G_{3z} + G_{2y} + G_{1x}\right) dx\, dy\, dz \tag{\%o56}$$

(%i57) `is(%=%th(2));`

$$\text{true} \tag{\%o57}$$

4 paths and curves

(%i58) `kill(labels)$`

4.1 Example 1.4.4.

Let $\alpha(t) = p + t(q - p)$ for a given pair of distinct points $p, q \in \mathbb{R}^n$. You should identify α as the line connecting point $p = \alpha(0)$ and $q = \alpha(1)$. If we define $v = q - p$ then the velocity of α is given by:

$$\alpha'(t) = v^1 \frac{\partial}{\partial x^1}\bigg|_{\alpha(t)} + v^2 \frac{\partial}{\partial x^2}\bigg|_{\alpha(t)} + \cdots + v^n \frac{\partial}{\partial x^n}\bigg|_{\alpha(t)}$$

Specializing to $n = 2$ and $v = \langle a, b \rangle$ we have $\alpha(t) = (p^1 + ta, p^2 + tb)$ and

$$\alpha'(t) = a \frac{\partial}{\partial x}\bigg|_{\alpha(t)} + b \frac{\partial}{\partial y}\bigg|_{\alpha(t)}$$

As an easy to check case, take $p = (0, 0)$ hence $p^1 = 0$ and $p^2 = 0$ hence $\alpha'(t)[f] = 2t(a^2 + b^2)$. For $t > 0$ we see f is increasing as we travel away from the origin along the line $\alpha(t)$. But, f is just the distance from the origin squared so the rate of change is quite reasonable. If we were to impose $a^2 + b^2 = 1$ then t represents the distance from the origin and the result reduces to $\alpha'[f] = 2t$ which makes sense as $f(\alpha(t)) = (ta)^2 + (tb)^2 = t^2(a^2 + b^2) = t^2$

Notice that $\alpha'(t)[f]$ gives the usual third-semester-American calculus directional derivative in the direction of $\alpha'(t)[f]$ only if we choose a parameter t for which $\|\alpha'(t)\| = 1$. This choice of parametrization is known as the *arclength* or *unit-speed* parametrization.

(%i1) `kill(t,x,y,z)$`

(%i2) `ζ:[x,y]$`

(%i3) `scalefactors(ζ)$`

(%i4) `init_cartan(ζ)$`

(%i5) `declare([a,b,p_1,p_2],constant)$`

(%i6) `v:[a,b]$`

(%i7) `P:[p_1,p_2]$`

(%i8) `ldisplay(α:P+t*v)$`

$$\alpha = [at + p_1, bt + p_2] \tag{\%t8}$$

(%i9) `ldisplay(α\':diff(α,t))$`

$$\alpha' = [a, b] \tag{\%t9}$$

(%i10) `norm(α\');`

$$\sqrt{b^2 + a^2} \tag{\%o10}$$

(%i11) `f:x^2+y^2$`

(%i12) `ldisplay(gradf:ev(express(grad(f)),diff))$`

$$gradf = [2x, 2y] \tag{\%t12}$$

(%i13) `at(α\'.gradf,map("=",ζ,α));`

$$2b\,(bt + p_2) + 2a\,(at + p_1) \tag{\%o13}$$

4.2 Example 1.4.5.

Let $R, m > 0$ be constants and $\alpha(t) = \langle R\cos(t), R\sin(t), mt\rangle$ for $t \in \mathbb{R}$. We say α is a helix with slope m and radius R. Notice $\alpha(t)$ falls on the cylinder $x^2 + y^2 = R^2$. Of course, we could define helices around other circular cylinders. The velocity vector field for α is given by:

$$\alpha' = \left(-R\sin(t)\frac{\partial}{\partial x} + R\cos(t)\frac{\partial}{\partial y} + m\frac{\partial}{\partial z}\right)\Big|_{\alpha(t)}$$

Then, $f(x, y, z) = x^2 + y^2$ has

$$\alpha'(t)[f] = (-2xR\sin(t) + 2yR\cos(t))|_{\alpha(t)} = -2R^2\cos(t)\sin t + 2R^2\sin(t)\cos(t) = 0.$$

(%i14) `kill(t,x,y,z,R,m)$`

(%i15) `ζ:[x,y,z]$`

(%i16) `scalefactors(ζ)$`

(%i17) `init_cartan(ζ)$`

(%i18) `orderless(m,R)$`

(%i19) `declare([R,m],constant)$`

(%i20) `assume(R>0,m>0)$`

(%i21) `ldisplay(α:[R*cos(t),R*sin(t),m*t])$`

$$\alpha = [R\,\cos(t), R\,\sin(t), mt] \tag{\%t21}$$

(%i22) `ldisplay(α\':diff(α,t))$`

$$\alpha' = [-R\,\sin(t), R\,\cos(t), m] \tag{\%t22}$$

(%i23) `trigsimp(norm(α\'));`

$$\sqrt{R^2 + m^2} \tag{\%o23}$$

(%i24) `f:x^2+y^2$`

(%i25) `ldisplay(gradf:ev(express(grad(f)),diff))$`

$$gradf = [2x, 2y, 0] \tag{\%t25}$$

(%i26) `at(α\'.gradf,map("=",ζ,α));`

$$0 \tag{\%o26}$$

4.3 Example 1.4.8.

Consider the helix defined by $R, m > 0$ and $\alpha(t) = (R\cos(t), R\sin(t), m\,t)$ for $t \in \mathbb{R}$. The speed of this helix is simply $\|\alpha'(t)\| = \sqrt{R^2+m^2}$. Let $h(s) = \frac{s}{\sqrt{R^2+m^2}}$ then if β is α reparametrized by h we calculate by Preposition 1.4.7

(%i27) `ldisplay(h:s/√(R²+m²))$`

$$h = \frac{s}{\sqrt{R^2+m^2}} \tag{\%t27}$$

(%i28) `ldisplay(β\':diff(h,s)*at(α\',t=h))$`

$$\beta' = \left[-\frac{R\,\sin\left(\frac{s}{\sqrt{R^2+m^2}}\right)}{\sqrt{R^2+m^2}}, \frac{R\,\cos\left(\frac{s}{\sqrt{R^2+m^2}}\right)}{\sqrt{R^2+m^2}}, \frac{m}{\sqrt{R^2+m^2}} \right] \tag{\%t28}$$

(%i29) `trigsimp(norm(β\'));`

$$1 \tag{\%o29}$$

(%i30) `g:z$`

(%i31) `ldisplay(gradg:ev(express(grad(g)),diff))$`

$$gradg = [0,0,1] \tag{\%t31}$$

(%i32) `at(β\'.gradg,map("=",ζ,α));`

$$\frac{m}{\sqrt{R^2+m^2}} \tag{\%o32}$$

(%i33) `unorder()$`

5 the push-forward or differential of a map

5.1 Example 1.5.1.

(%i34) `kill(t,x_1,x_2,a,b)$`

Let $F(x^1,x^2) = \left(x^1+x^2, x^1-x^2\right)$. Consider a parametrized curve $\alpha(t) = (a(t), b(t))$. The image of α under F is:

$$(F \circ \alpha(t)) = F\left(a(t), b(t)\right) = (a(t)+b(t), a(t)-b(t))$$

(%i35) `ζ:[x_1,x_2]$`

(%i36) `depends(ζ,t)$`

(%i37) `F:[x_1+x_2,x_1-x_2]$`

(%i38) `ldisplay(α:[a,b])$`

$$\alpha = [a, b] \qquad (\%t38)$$

(%i39) `depends(α,t)$`

(%i40) `ldisplay(α\':diff(α,t))$`

$$\alpha' = [a_t, b_t] \qquad (\%t40)$$

(%i41) `at(F,map("=",ζ,α));`

$$[b+a, a-b] \qquad (\%o41)$$

(%i42) `diff(%,t);`

$$[b_t+a_t, a_t-b_t] \qquad (\%o42)$$

(%i43) `list_matrix_entries(jacobian(F,ζ).α\');`

$$[b_t+a_t, a_t-b_t] \qquad (\%o43)$$

(%i44) `is(%=%th(2));`

$$\text{true} \qquad (\%o44)$$

5.2 Example 1.5.2.

Another example, $F(x^1, x^2) = \left(e^{x^1+x^2}, \sin(x^2), \cos(x^2)\right)$. Once more, consider the curve $\alpha = (a, b)$ hence $\alpha' = \langle a', b' \rangle$ and

$$(F \circ \alpha)' = \langle e^{a+b}\,(a') + b'), \cos(b)\,b', (-\sin(b))\,b' \rangle$$

5.3 Example 1.5.4.

(%i45) `kill(t,x,y)$`

Let $F(x, y) = x^2 + y^2$ then $F(x, y) = R^2$ is a circle and

(%i46) `ζ:[x,y]$`

(%i47) `scalefactors(ζ)$`

(%i48) `F:x^2+y^2$`

(%i49) `ldisplay(J_F:ev(express(grad(F)),diff))$`

$$J_F = [2x, 2y] \tag{\%t49}$$

We see $y \neq 0$ implies the last column is nonzero hence we may solve for y near such points. In this case, $G(x) = \pm\sqrt{R^2 - x^2}$ where we choose $\pm$ appropriate to the location of the local solution.

5.4 Example 1.5.5.

(%i50) `kill(t,x,y,z)$`

Let $F(x, y, z) = cos(x) + y + z^2$ then

(%i51) `ζ:[x,y,z]$`

(%i52) `scalefactors(ζ)$`

(%i53) `F:cos(x)+y+z^2$`

(%i54) `ldisplay(J_F:ev(express(grad(F)),diff))$`

$$J_F = [-\sin(x), 1, 2z] \tag{\%t54}$$

this tells me I can solve for $z = z(x, y)$ when $z \neq 0$, or I can solve for $y = y(x, z)$ anywhere on $F(x, y, z) = c$, or I can solve for $x = x(y, z)$ when $x \neq n\pi$ for $n \in \mathbb{Z}$. Notice we can rearrange coordinates to put x or y as the last coordinate.