Routh's procedure

Based on Goldstein Classical Mechanics Book 8.3 Routh's procedure

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2)
```
info:build_info()$info@version;
```

(%o2)

5.38.1

(%i2)
```
reset()$kill(all)$
```

(%i1)
```
derivabbrev:true$
```

(%i2)
```
ratprint:false$
```

(%i3)
```
fpprintprec:5$
```

(%i4)
```
if get('draw,'version)=false then load(draw)$
```

(%i5)
```
wxplot_size:[1024,768]$
```

(%i6)
```
if get('optvar,'version)=false then load(optvar)$
```

(%i7)
```
if get('rkf45,'version)=false then load(rkf45)$
```

(%i8)
```
declare(t,mainvar)$
```

(%i9)
```
declare(trigsimp,evfun)$
```

1 Single particle moving in a plane

Single particle moving in a plane under the influence of the inverse-square central force $f(r)$ derived from the potential $V(r) = -k/r^n$. (Page 349)

(%i10) `ξ:[r,θ]$`

(%i11) `depends(ξ,t)$`

(%i12) `dim:length(ξ)$`

Lagrangian

(%i13) `declare(K,constant)$`

(%i14) `ldisplay(L:1/2*m*(diff(r,t)^2+r^2*diff(θ,t)^2)+K/r^2)$`

$$L = \frac{m\left(r^2\left(\dot{\theta}\right)^2 + (\dot{r})^2\right)}{2} + \frac{K}{r^2} \tag{\%t14}$$

Momentum Conjugate

(%i15) `ldisplay(P_r:ev(diff(L,'diff(r,t))))$`

$$P_r = m\,(\dot{r}) \tag{\%t15}$$

(%i16) `linsolve(p_r=P_r,diff(r,t)),factor;`

$$\left[\dot{r} = \frac{p_r}{m}\right] \tag{\%o16}$$

(%i17) `ldisplay(P_θ:ev(diff(L,'diff(θ,t))))$`

$$P_\theta = m\,r^2\left(\dot{\theta}\right) \tag{\%t17}$$

(%i18) `linsolve(p_θ=P_θ,diff(θ,t)),factor;`

$$\left[\dot{\theta} = \frac{p_\theta}{m\,r^2}\right] \tag{\%o18}$$

Generalized Forces

(%i19) `ldisplay(F_r:factor(expand(diff(L,r))))$`

$$F_r = \frac{m\,r^4\left(\dot{\theta}\right)^2 - 2K}{r^3} \tag{\%t19}$$

(%i20) `ldisplay(F_θ:factor(expand(diff(L,θ))))$`

$$F_\theta = 0 \tag{\%t20}$$

Euler-Lagrange Equations

(%i21) `aa:el(L,ξ,t)$`

(%i25) `bb:ev(aa,eval,diff)$`

(%i26) `declare([E,J],constant)$`

(%i28)
```
bb[1]:subst([k[0]=-E],-bb[1])$
bb[4]:subst([k[2]=J],bb[4])$
```

(%i32)
```
bb[1]:rhs(bb[1])=lhs(bb[1])$
bb[2]:lhs(bb[2])-rhs(bb[2])=0$
bb[3]:lhs(bb[3])-rhs(bb[3])=0$
bb[4]:rhs(bb[4])=lhs(bb[4])$
```

Conservation Laws

(%i33) `map(ldisp,part(bb,[1,4]))$`

$$E=-\frac{m\left(r^2\left(\dot{\theta}\right)^2+\left(\dot{r}\right)^2\right)}{2}+m\,r^2\left(\dot{\theta}\right)^2+m\left(\dot{r}\right)^2-\frac{K}{r^2} \tag{\%t33}$$

$$J=m\,r^2\left(\dot{\theta}\right) \tag{\%t34}$$

Energy as a function of Angular momentum

(%i35) `expand(solve(eliminate(part(bb,[1,4]),[diff(θ,t)]),E));`

$$\left[E=\frac{m\left(\dot{r}\right)^2}{2}+\frac{J^2}{2m\,r^2}-\frac{K}{r^2}\right] \tag{\%o35}$$

Equations of Motion

(%i36) `map(ldisp,expand(part(bb,[2,3])))$`

$$-mr\left(\dot{\theta}\right)^2+m\left(\ddot{r}\right)+\frac{2K}{r^3}=0 \tag{\%t36}$$

$$m\,r^2\left(\ddot{\theta}\right)+2mr\left(\dot{r}\right)\left(\dot{\theta}\right)=0 \tag{\%t37}$$

Solve for second derivative of coordinates

(%i38) `linsol:linsolve(part(bb,[2,3]),diff(ξ,t,2))$`

(%i39) `map(ldisp,expand(linsol))$`

$$\ddot{r}=r\left(\dot{\theta}\right)^2-\frac{2K}{m\,r^3} \tag{\%t39}$$

$$\ddot{\theta}=-\frac{2\left(\dot{r}\right)\left(\dot{\theta}\right)}{r} \tag{\%t40}$$

Check Conservation of Energy

(%i41) `bb[1];`

$$E = -\frac{m\left(r^2\left(\dot{\theta}\right)^2 + (\dot{r})^2\right)}{2} + m\,r^2\left(\dot{\theta}\right)^2 + m\,(\dot{r})^2 - \frac{K}{r^2} \qquad (\%o41)$$

(%i42) `subst(linsol,diff(rhs(bb[1]),t)),fullratsimp;`

$$0 \qquad (\%o42)$$

Check Conservation of Angular momentum

(%i43) `bb[4];`

$$J = m\,r^2\left(\dot{\theta}\right) \qquad (\%o43)$$

(%i44) `subst(linsol,diff(rhs(bb[4]),t)),fullratsimp;`

$$0 \qquad (\%o44)$$

Legendre Transformation

(%i45) `kill(labels)$`

(%i1) `Legendre:linsolve([p_r=P_r,p_θ=P_θ],['diff(r,t),'diff(θ,t)])$`

(%i2) `map(ldisp,Legendre)$`

$$\dot{r} = \frac{p_r}{m} \qquad (\%t2)$$

$$\dot{\theta} = \frac{p_\theta}{m\,r^2} \qquad (\%t3)$$

Hamiltonian

(%i4) `ldisplay(H:ev(p_r*'diff(r,t)+p_θ*'diff(θ,t)-L,Legendre,factor))$`

$$H = \frac{{p_r}^2 r^2 + {p_\theta}^2 - 2Km}{2m\,r^2} \qquad (\%t4)$$

Equations of Motion

(%i5) `Hq:makelist(Hq[i],i,1,2*dim)$`

(%i9)
```
Hq[1]:'diff(r,t)=diff(H,p_r)$
Hq[2]:'diff(θ,t)=diff(H,p_θ)$
Hq[3]:'diff(p_r,t)=-diff(H,r)$
Hq[4]:'diff(p_θ,t)=-diff(H,θ)$
```

(%i10) `map(ldisp,Hq:scanmap(fullratsimp,Hq))$`

$$\dot{r} = \frac{p_r}{m} \qquad (\%t10)$$

$$\dot{\theta} = \frac{p_\theta}{m\,r^2} \tag{\%t11}$$

$$\dot{p}_r = \frac{{p_\theta}^2 - 2Km}{m\,r^3} \tag{\%t12}$$

$$\dot{p}_\theta = 0 \tag{\%t13}$$

Analytical solution of θ, p_θ

(%i16)
```
atvalue(θ(t),[t=0],θ_0)$
atvalue(p_θ(t),[t=0],J)$
desol:desolve(convert(part(Hq,[2,4]),[θ,p_θ],t), convert([θ,p_θ],[θ,p_θ],t));
```

$$\left[\theta(t) = \frac{J t}{m\,r^2} + \theta_0, p_\theta(t) = J\right] \tag{desol}$$

Check Conservation of Energy

(%i17)
```
depends([p_r,p_θ],t)$
```

(%i18)
```
subst(Hq,diff(H,t)),fullratsimp;
```

$$0 \tag{\%o18}$$

Routhian Transformation

(%i19)
```
ldisplay(Routhian:linsolve(bb[4],'diff(θ,t)))$
```

$$Routhian = \left[\dot{\theta} = \frac{J}{m\,r^2}\right] \tag{\%t19}$$

(%i20)
```
ldisplay(R:ev(L-p_θ*'diff(θ,t),[p_θ=J],Routhian,expand))$
```

$$R = \frac{m\,(\dot{r})^2}{2} - \frac{J^2}{2m\,r^2} + \frac{K}{r^2} \tag{\%t20}$$

Momentum Conjugate

(%i21)
```
ldisplay(P_r:ev(diff(R,'diff(r,t))))$
```

$$P_r = m\,(\dot{r}) \tag{\%t21}$$

(%i22)
```
linsolve(p_r=P_r,diff(r,t)),factor;
```

$$\left[\dot{r} = \frac{p_r}{m}\right] \tag{\%o22}$$

Generalized Forces

(%i23)
```
ldisplay(F_r:expand(diff(R,r)))$
```

$$F_r = \frac{J^2}{m\,r^3} - \frac{2K}{r^3} \tag{\%t23}$$

Euler-Lagrange Equations

(%i24) `aa:el(R,r,t)$`

(%i26) `cc:ev(aa,eval,diff)$`

(%i27) `cc[1]:subst([k[0]=-E],-cc[1])$`

(%i29) `cc[1]:rhs(cc[1])=lhs(cc[1])$`
`cc[2]:lhs(cc[2])-rhs(cc[2])=0$`

Conservation Laws

(%i30) `cc[1];`

$$E = \frac{m\left(\dot{r}\right)^2}{2} + \frac{J^2}{2m\,r^2} - \frac{K}{r^2} \tag{\%o30}$$

Equations of Motion

(%i31) `cc[2];`

$$m\,\left(\ddot{r}\right) - \frac{J^2}{m\,r^3} + \frac{2K}{r^3} = 0 \tag{\%o31}$$

Solve for second derivative of coordinates

(%i32) `linsol:linsolve(cc[2],diff(r,t,2))$`

(%i33) `map(ldisp,expand(linsol))$`

$$\ddot{r} = \frac{J^2}{m^2\,r^3} - \frac{2K}{m\,r^3} \tag{\%t33}$$

Check Conservation of Energy

(%i34) `cc[1];`

$$E = \frac{m\left(\dot{r}\right)^2}{2} + \frac{J^2}{2m\,r^2} - \frac{K}{r^2} \tag{\%o34}$$

(%i35) `subst(linsol,diff(rhs(cc[1]),t)),fullratsimp;`

$$0 \tag{\%o35}$$

Legendre Transformation

(%i36) `kill(labels)$`

(%i1) `Legendre:linsolve([p_r=P_r],['diff(r,t)])$`

(%i2) `map(ldisp,Legendre)$`

$$\dot{r} = \frac{p_r}{m} \tag{\%t2}$$

Hamiltonian

(%i3) `ldisplay(H:ev(p_r*'diff(r,t)-L,Legendre,factor))$`

$$H = -\frac{m^2\,r^4\,\left(\dot{\theta}\right)^2 - {p_r}^2 r^2 + 2Km}{2m\,r^2} \tag{\%t3}$$

Equations of Motion

(%i4) `Rq:makelist(Rq[i],i,1,2)$`

(%i6) `Rq[1]:'diff(r,t)=diff(H,p_r)$`
`Rq[2]:'diff(p_r,t)=-diff(H,r)$`

(%i7) `map(ldisp,Rq:scanmap(fullratsimp,Rq))$`

$$\dot{r} = \frac{p_r}{m} \tag{\%t7}$$

$$\dot{p}_r = \frac{m\,r^4\,\left(\dot{\theta}\right)^2 - 2K}{r^3} \tag{\%t8}$$

Check Conservation of Energy

(%i9) `subst(Rq,diff(H,t)),fullratsimp;`

$$-m\,r^2\,\left(\dot{\theta}\right)\,\left(\ddot{\theta}\right) \tag{\%o9}$$

Compare

r, P_r

(%i10) `part(Hq,[1,3]),fullratsimp;`

$$\left[\dot{r} = \frac{p_r}{m}, \dot{p}_r = \frac{{p_\theta}^2 - 2Km}{m\,r^3}\right] \tag{\%o10}$$

(%i11) `part(Rq,[1,2]),fullratsimp;`

$$\left[\dot{r} = \frac{p_r}{m}, \dot{p}_r = \frac{m\,r^4\,\left(\dot{\theta}\right)^2 - 2K}{r^3}\right] \tag{\%o11}$$

(%i12) `subst(part(Hq,[2,4]),%),fullratsimp;`

$$\left[\dot{r} = \frac{p_r}{m}, \dot{p}_r = \frac{{p_\theta}^2 - 2Km}{m\,r^3}\right] \tag{\%o12}$$

(%i13) `is(%=%th(3));`

$$\text{true} \tag{\%o13}$$

2 Central potential in spherical coordinates

(%i14) `kill(labels,L,H,Hq,R,Rq,V,t,r,θ,ϕ,P_r,P_θ,P_ϕ)$`

Based on Wikipedia Article Routhian mechanics

(%i1) `ξ:[r,θ,ϕ]$`

(%i2) `depends(ξ,t)$`

(%i3) `depends(V,r)$`

(%i4) `dim:length(ξ)$`

Lagrangian

(%i5) `declare(m,constant)$`

(%i6) `ldisplay(L:1/2*m*(diff(r,t)^2+r^2*diff(θ,t)^2+r^2*sin(θ)^2*diff(ϕ,t)^2)-V)$`

$$L = \frac{m\left(r^2 \sin(\theta)^2 \left(\dot{\phi}\right)^2 + r^2 \left(\dot{\theta}\right)^2 + (\dot{r})^2\right)}{2} - V \tag{\%t6}$$

Momentum Conjugate

(%i7) `ldisplay(P_r:ev(diff(L,'diff(r,t))))$`

$$P_r = m\,(\dot{r}) \tag{\%t7}$$

(%i8) `linsolve(p_r=P_r,diff(r,t)),factor;`

$$\left[\dot{r} = \frac{p_r}{m}\right] \tag{\%o8}$$

(%i9) `ldisplay(P_θ:ev(diff(L,'diff(θ,t))))$`

$$P_\theta = m\,r^2 \left(\dot{\theta}\right) \tag{\%t9}$$

(%i10) `linsolve(p_θ=P_θ,diff(θ,t)),factor;`

$$\left[\dot{\theta} = \frac{p_\theta}{m\,r^2}\right] \tag{\%o10}$$

(%i11) `ldisplay(P_ϕ:ev(diff(L,'diff(ϕ,t))))$`

$$P_\phi = m\,r^2 \sin(\theta)^2 \left(\dot{\phi}\right) \tag{\%t11}$$

(%i12) `linsolve(p_ϕ=P_ϕ,diff(ϕ,t)),factor;`

$$\left[\dot{\phi} = \frac{p_\phi}{m\,r^2 \sin(\theta)^2}\right] \tag{\%o12}$$

Generalized Forces

(%i13) `ldisplay(F_r:factor(expand(diff(L,r))))$`

$$F_r = mr\sin(\theta)^2 \left(\dot{\phi}\right)^2 + mr\left(\dot{\theta}\right)^2 - V_r \qquad (\%t13)$$

(%i14) `ldisplay(F_θ:factor(expand(diff(L,θ))))$`

$$F_\theta = m\,r^2\,\cos(\theta)\,\sin(\theta)\left(\dot{\phi}\right)^2 \qquad (\%t14)$$

(%i15) `ldisplay(F_ϕ:factor(expand(diff(L,ϕ))))$`

$$F_\phi = 0 \qquad (\%t15)$$

Euler-Lagrange Equations

(%i16) `aa:el(L,ξ,t)$`

(%i21) `bb:ev(aa,eval,diff)$`

(%i22) `declare([E,J],constant)$`

(%i24)
```
bb[1]:subst([k[0]=-E],-bb[1])$
bb[5]:subst([k[3]=J],bb[5])$
```

(%i29)
```
bb[1]:rhs(bb[1])=lhs(bb[1])$
bb[2]:lhs(bb[2])-rhs(bb[2])=0$
bb[3]:lhs(bb[3])-rhs(bb[3])=0$
bb[4]:lhs(bb[4])-rhs(bb[4])=0$
bb[5]:rhs(bb[5])=lhs(bb[5])$
```

Conservation Laws

(%i30) `map(ldisp,part(bb,[1,5]))$`

$$E = -\frac{m\left(r^2\sin(\theta)^2\left(\dot{\phi}\right)^2 + r^2\left(\dot{\theta}\right)^2 + (\dot{r})^2\right)}{2} + m\,r^2\sin(\theta)^2\left(\dot{\phi}\right)^2 + m\,r^2\left(\dot{\theta}\right)^2 + m\,(\dot{r})^2 + V \qquad (\%t30)$$

$$J = m\,r^2\sin(\theta)^2\left(\dot{\phi}\right) \qquad (\%t31)$$

Energy as a function of Angular momentum

(%i32) `expand(solve(eliminate(part(bb,[1,5]),[diff(ϕ,t)]),E));`

$$\left[E = \frac{m\,r^2\left(\dot{\theta}\right)^2}{2} + \frac{J^2}{2m\,r^2\sin(\theta)^2} + \frac{m\,(\dot{r})^2}{2} + V\right] \qquad (\%o32)$$

Equations of Motion

(%i33) `map(ldisp,expand(part(bb,[2,3,4])))$`

$$-mr\sin\left(\theta\right)^{2}\left(\dot{\phi}\right)^{2}-mr\left(\dot{\theta}\right)^{2}+m\left(\ddot{r}\right)+V_{r}=0 \tag{\%t33}$$

$$-m\,r^{2}\cos\left(\theta\right)\sin\left(\theta\right)\left(\dot{\phi}\right)^{2}+m\,r^{2}\left(\ddot{\theta}\right)+2mr\left(\dot{r}\right)\left(\dot{\theta}\right)=0 \tag{\%t34}$$

$$m\,r^{2}\sin\left(\theta\right)^{2}\left(\ddot{\phi}\right)+2m\,r^{2}\cos\left(\theta\right)\sin\left(\theta\right)\left(\dot{\theta}\right)\left(\dot{\phi}\right)+2mr\left(\dot{r}\right)\sin\left(\theta\right)^{2}\left(\dot{\phi}\right)=0 \tag{\%t35}$$

Solve for second derivative of coordinates

(%i36) `linsol:linsolve(part(bb,[2,3,4]),diff(ξ,t,2))$`

(%i37) `map(ldisp,expand(linsol))$`

$$\ddot{r}=r\sin\left(\theta\right)^{2}\left(\dot{\phi}\right)^{2}+r\left(\dot{\theta}\right)^{2}-\frac{V_{r}}{m} \tag{\%t37}$$

$$\ddot{\theta}=\cos\left(\theta\right)\sin\left(\theta\right)\left(\dot{\phi}\right)^{2}-\frac{2\left(\dot{r}\right)\left(\dot{\theta}\right)}{r} \tag{\%t38}$$

$$\ddot{\phi}=-\frac{2\cos\left(\theta\right)\left(\dot{\theta}\right)\left(\dot{\phi}\right)}{\sin\left(\theta\right)}-\frac{2\left(\dot{r}\right)\left(\dot{\phi}\right)}{r} \tag{\%t39}$$

Check Conservation of Energy

(%i40) `bb[1];`

$$E=-\frac{m\left(r^{2}\sin\left(\theta\right)^{2}\left(\dot{\phi}\right)^{2}+r^{2}\left(\dot{\theta}\right)^{2}+\left(\dot{r}\right)^{2}\right)}{2}+m\,r^{2}\sin\left(\theta\right)^{2}\left(\dot{\phi}\right)^{2}+m\,r^{2}\left(\dot{\theta}\right)^{2}+m\left(\dot{r}\right)^{2}+V \tag{\%o40}$$

(%i41) `subst(linsol,diff(rhs(bb[1]),t)),fullratsimp;`

$$0 \tag{\%o41}$$

Check Conservation of Angular momentum

(%i42) `bb[5];`

$$J=m\,r^{2}\sin\left(\theta\right)^{2}\left(\dot{\phi}\right) \tag{\%o42}$$

(%i43) `subst(linsol,diff(rhs(bb[5]),t)),fullratsimp;`

$$0 \tag{\%o43}$$

Legendre Transformation

(%i44) `kill(labels)$`

```
(%i1) Legendre:linsolve([p_r=P_r,p_θ=P_θ,p_ϕ=P_ϕ], ['diff(r,t),'diff(θ,t),'diff(ϕ,t)])$
(%i2) map(ldisp,Legendre)$
```

$$\dot{r} = \frac{p_r}{m} \tag{\%t2}$$

$$\dot{\theta} = \frac{p_\theta}{m\,r^2} \tag{\%t3}$$

$$\dot{\phi} = \frac{p_\phi}{m\,r^2\sin(\theta)^2} \tag{\%t4}$$

Hamiltonian

```
(%i5) ldisplay(H:ev(p_r*'diff(r,t)+p_θ*'diff(θ,t)+p_ϕ*'diff(ϕ,t)- L,Legendre,fullratsimp))$
```

$$H = \frac{\left(\left({p_r}^2 + 2mV\right) r^2 + {p_\theta}^2\right)\sin(\theta)^2 + {p_\phi}^2}{2m\,r^2\sin(\theta)^2} \tag{\%t5}$$

Equations of Motion

```
(%i6) Hq:makelist(Hq[i],i,1,2*dim)$
(%i12) Hq[1]:'diff(r,t)=diff(H,p_r)$
       Hq[2]:'diff(θ,t)=diff(H,p_θ)$
       Hq[3]:'diff(ϕ,t)=diff(H,p_ϕ)$
       Hq[4]:'diff(p_r,t)=-diff(H,r)$
       Hq[5]:'diff(p_θ,t)=-diff(H,θ)$
       Hq[6]:'diff(p_ϕ,t)=-diff(H,ϕ)$
(%i13) map(ldisp,Hq:scanmap(fullratsimp,Hq))$
```

$$\dot{r} = \frac{p_r}{m} \tag{\%t13}$$

$$\dot{\theta} = \frac{p_\theta}{m\,r^2} \tag{\%t14}$$

$$\dot{\phi} = \frac{p_\phi}{m\,r^2\sin(\theta)^2} \tag{\%t15}$$

$$\dot{p}_r = -\frac{\left(m\,(V_r)\,r^3 - {p_\theta}^2\right)\sin(\theta)^2 - {p_\phi}^2}{m\,r^3\sin(\theta)^2} \tag{\%t16}$$

$$\dot{p}_\theta = \frac{{p_\phi}^2\cos(\theta)}{m\,r^2\sin(\theta)^3} \tag{\%t17}$$

$$\dot{p}_\phi = 0 \tag{\%t18}$$

Analytical solution of ϕ, p_ϕ

```
(%i21) atvalue(ϕ(t),[t=0],ϕ_0)$
       atvalue(p_ϕ(t),[t=0],J)$
       desol:desolve(convert(part(Hq,[3,6]),[ϕ,p_ϕ],t), convert([ϕ,p_ϕ],[ϕ,p_ϕ],t));
```

$$\left[\phi(t) = \phi_0 + \frac{J t}{m\,r^2\sin(\theta)^2}, p_\phi(t) = J\right] \tag{desol}$$

Check Conservation of Energy

```
(%i22) depends([p_r,p_θ,p_ϕ],t)$
(%i23) subst(Hq,diff(H,t)),fullratsimp;
```

$$0 \tag{\%o23}$$

Routhian Transformation

```
(%i24) ldisplay(Routhian:linsolve(bb[5],'diff(ϕ,t)))$
```

$$Routhian = \left[\dot{\phi} = \frac{J}{m\,r^2 \sin(\theta)^2}\right] \tag{\%t24}$$

```
(%i25) ldisplay(R:ev(L-p_ϕ*'diff(ϕ,t),[p_ϕ=J],Routhian,expand))$
```

$$R = \frac{m\,r^2 \left(\dot{\theta}\right)^2}{2} - \frac{J^2}{2m\,r^2 \sin(\theta)^2} + \frac{m\,(\dot{r})^2}{2} - V \tag{\%t25}$$

Momentum Conjugate

```
(%i26) ldisplay(P_r:ev(diff(R,'diff(r,t))))$
```

$$P_r = m\ (\dot{r}) \tag{\%t26}$$

```
(%i27) linsolve(p_r=P_r,diff(r,t)),factor;
```

$$\left[\dot{r} = \frac{p_r}{m}\right] \tag{\%o27}$$

```
(%i28) ldisplay(P_θ:ev(diff(R,'diff(θ,t))))$
```

$$P_\theta = m\,r^2\ \left(\dot{\theta}\right) \tag{\%t28}$$

```
(%i29) linsolve(p_θ=P_θ,diff(θ,t)),factor;
```

$$\left[\dot{\theta} = \frac{p_\theta}{m\,r^2}\right] \tag{\%o29}$$

Generalized Forces

```
(%i30) ldisplay(F_r:expand(diff(R,r)))$
```

$$F_r = mr \left(\dot{\theta}\right)^2 + \frac{J^2}{m\,r^3 \sin(\theta)^2} - V_r \tag{\%t30}$$

```
(%i31) ldisplay(F_θ:expand(diff(R,θ)))$
```

$$F_\theta = \frac{J^2\ \cos(\theta)}{m\,r^2 \sin(\theta)^3} \tag{\%t31}$$

Euler-Lagrange Equations

(%i32) `ξ:[r,θ]$`

(%i33) `dim:length(ξ)$`

(%i34) `aa:el(R,ξ,t)$`

(%i37) `cc:ev(aa,eval,diff)$`

(%i38) `cc[1]:subst([k[0]=-E],-cc[1])$`

(%i41)
```
cc[1]:rhs(cc[1])=lhs(cc[1])$
cc[2]:lhs(cc[2])-rhs(cc[2])=0$
cc[3]:lhs(cc[3])-rhs(cc[3])=0$
```

Conservation Laws

(%i42) `cc[1];`

$$E = \frac{m\,r^2\left(\dot{\theta}\right)^2}{2} + \frac{J^2}{2m\,r^2\sin\left(\theta\right)^2} + \frac{m\left(\dot{r}\right)^2}{2} + V \qquad (\%o42)$$

Equations of Motion

(%i43) `map(ldisp,expand(part(cc,[2,3])))$`

$$-mr\left(\dot{\theta}\right)^2 - \frac{J^2}{m\,r^3\sin\left(\theta\right)^2} + m\,\left(\ddot{r}\right) + V_r = 0 \qquad (\%t43)$$

$$m\,r^2\,\left(\ddot{\theta}\right) + 2mr\,\left(\dot{r}\right)\,\left(\dot{\theta}\right) - \frac{J^2\,\cos\left(\theta\right)}{m\,r^2\sin\left(\theta\right)^3} = 0 \qquad (\%t44)$$

Solve for second derivative of coordinates

(%i45) `linsol:linsolve(part(cc,[2,3]),diff(ξ,t,2))$`

(%i46) `map(ldisp,expand(linsol))$`

$$\ddot{r} = r\left(\dot{\theta}\right)^2 + \frac{J^2}{m^2\,r^3\sin\left(\theta\right)^2} - \frac{V_r}{m} \qquad (\%t46)$$

$$\ddot{\theta} = \frac{J^2\,\cos\left(\theta\right)}{m^2\,r^4\sin\left(\theta\right)^3} - \frac{2\left(\dot{r}\right)\,\left(\dot{\theta}\right)}{r} \qquad (\%t47)$$

Check Conservation of Energy

(%i48) `cc[1];`

$$E = \frac{m\,r^2\left(\dot{\theta}\right)^2}{2} + \frac{J^2}{2m\,r^2\sin\left(\theta\right)^2} + \frac{m\left(\dot{r}\right)^2}{2} + V \qquad (\%o48)$$

(%i49) `subst(linsol,diff(rhs(cc[1]),t)),fullratsimp;`

$$0 \qquad (\%o49)$$

Legendre Transformation

(%i50) `kill(labels)$`

(%i1) `Legendre:linsolve([p_r=P_r,p_θ=P_θ], ['diff(r,t),'diff(θ,t)])$`

(%i2) `map(ldisp,Legendre)$`

$$\dot{r} = \frac{p_r}{m} \tag{\%t2}$$

$$\dot{\theta} = \frac{p_\theta}{m\,r^2} \tag{\%t3}$$

Hamiltonian

(%i4) `ldisplay(H:ev(p_r*'diff(r,t)+p_θ*'diff(θ,t)- L,Legendre,fullratsimp))$`

$$H = -\frac{m^2\,r^4\,\sin(\theta)^2\left(\dot{\phi}\right)^2 + \left(-{p_r}^2 - 2mV\right)r^2 - {p_\theta}^2}{2m\,r^2} \tag{\%t4}$$

Equations of Motion

(%i5) `Rq:makelist(Rq[i],i,1,2*dim)$`

(%i9)
```
Rq[1]:'diff(r,t)=diff(H,p_r)$
Rq[2]:'diff(θ,t)=diff(H,p_θ)$
Rq[3]:'diff(p_r,t)=-diff(H,r)$
Rq[4]:'diff(p_θ,t)=-diff(H,θ)$
```

(%i10) `map(ldisp,Rq:scanmap(fullratsimp,Rq))$`

$$\dot{r} = \frac{p_r}{m} \tag{\%t10}$$

$$\dot{\theta} = \frac{p_\theta}{m\,r^2} \tag{\%t11}$$

$$\dot{p}_r = \frac{m^2\,r^4\,\sin(\theta)^2\left(\dot{\phi}\right)^2 - m\,\left(V_r\right)\,r^3 + {p_\theta}^2}{m\,r^3} \tag{\%t12}$$

$$\dot{p}_\theta = m\,r^2\,\cos(\theta)\,\sin(\theta)\left(\dot{\phi}\right)^2 \tag{\%t13}$$

Check Conservation of Energy

(%i14) `subst(Hq,diff(H,t)),fullratsimp;`

$$-p_\phi\left(\ddot{\phi}\right) \tag{\%o14}$$

(%i15) `subst(Rq,diff(H,t)),fullratsimp;`

$$-m\,r^2\,\sin(\theta)^2\left(\dot{\phi}\right)\left(\ddot{\phi}\right) \tag{\%o15}$$

Compare

r, p_r

```
(%i16) part(Hq,[1,4]),fullratsimp;
```

$$\left[\dot{r}=\frac{p_r}{m}, \dot{p}_r=-\frac{\left(m\ (V_r)\ r^3-{p_\theta}^2\right)\sin(\theta)^2-{p_\phi}^2}{m\,r^3\sin(\theta)^2}\right] \tag{\%o16}$$

```
(%i17) part(Rq,[1,3]),fullratsimp;
```

$$\left[\dot{r}=\frac{p_r}{m}, \dot{p}_r=\frac{m^2\,r^4\sin(\theta)^2\left(\dot{\phi}\right)^2-m\ (V_r)\ r^3+{p_\theta}^2}{m\,r^3}\right] \tag{\%o17}$$

```
(%i18) subst(part(Hq,[3,6]),%),fullratsimp;
```

$$\left[\dot{r}=\frac{p_r}{m}, \dot{p}_r=-\frac{\left(m\ (V_r)\ r^3-{p_\theta}^2\right)\sin(\theta)^2-{p_\phi}^2}{m\,r^3\sin(\theta)^2}\right] \tag{\%o18}$$

```
(%i19) is(%=%th(3));
```

$$\text{true} \tag{\%o19}$$

θ, p_θ

```
(%i20) part(Hq,[2,5]),fullratsimp;
```

$$\left[\dot{\theta}=\frac{p_\theta}{m\,r^2}, \dot{p}_\theta=\frac{{p_\phi}^2\cos(\theta)}{m\,r^2\sin(\theta)^3}\right] \tag{\%o20}$$

```
(%i21) part(Rq,[2,4]),fullratsimp;
```

$$\left[\dot{\theta}=\frac{p_\theta}{m\,r^2}, \dot{p}_\theta=m\,r^2\ \cos(\theta)\ \sin(\theta)\left(\dot{\phi}\right)^2\right] \tag{\%o21}$$

```
(%i22) subst(part(Hq,[3,6]),%),fullratsimp;
```

$$\left[\dot{\theta}=\frac{p_\theta}{m\,r^2}, \dot{p}_\theta=\frac{{p_\phi}^2\cos(\theta)}{m\,r^2\sin(\theta)^3}\right] \tag{\%o22}$$

```
(%i23) is(%=%th(3));
```

$$\text{true} \tag{\%o23}$$