Stoke's Theorem

Reference Wikipedia article Stokes' theorem

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2)
```
info:build_info()$info@version;
```

(%o2)

5.38.1

(%i2)
```
reset()$kill(all)$
```

(%i1)
```
derivabbrev:true$
```

(%i2)
```
ratprint:false$
```

(%i3)
```
fpprintprec:5$
```

(%i4)
```
load(linearalgebra)$
```

(%i5)
```
if get('draw,'version)=false then load(draw)$
```

(%i6)
```
wxplot_size:[1024,768]$
```

(%i7)
```
if get('drawdf,'version)=false then load(drawdf)$
```

(%i8)
```
set_draw_defaults(xtics=1,ytics=1,ztics=1,xyplane=0,nticks=100,
xaxis=true,xaxis_type=solid,xaxis_width=3,
yaxis=true,yaxis_type=solid,yaxis_width=3,
zaxis=true,zaxis_type=solid,zaxis_width=3)$
```

(%i9)
```
if get('vect,'version)=false then load(vect)$
```

(%i10)
```
norm(u):=block(ratsimp(radcan(√(u.u))))$
```

(%i11)
```
normalize(v):=block(v/norm(v))$
```

(%i12)
```
angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$
```

(%i13)
```
mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$
```

(%i14)
```
if get('cartan,'version)=false then load(cartan)$
```

(%i15)
```
declare(trigsimp,evfun)$
```

1 Stoke's Theorem

Based on Michael Penn Video Stoke's Theorem

Let $\vec{S}$ be a piecewise smooth oriented surface with boundary $\vec{C}$ (a simple closed curve with positive orientation). If $\vec{F}$ is a vector field with continuous first partial derivatives on an open region containing $\vec{S}$ then:

$$\oint_C \vec{F} \cdot \mathrm{d}\vec{r} = \iint_S \left(\nabla \times \vec{F}\right) \cdot \mathrm{d}\vec{S}$$

Positive orientation: If you walk around $\vec{C}$ with your head pointing in the direction of $\hat{n}$ then the surface $\vec{S}$ is on your left.

(%i16)
```
kill(labels,t,x,y,z,f,P,Q,R)$
```

Define the space $\mathbb{R}^3$

(%i1)
```
ζ:[x,y,z]$
```

(%i2)
```
scalefactors(ζ)$
```

(%i3)
```
init_cartan(ζ)$
```

Vector field $\vec{F} \in \mathbb{R}^3$

(%i4) `depends([P,Q,R],ζ)$`

(%i5) `ldisplay(F:[P,Q,R])$`

$$\vec{F} = [P, Q, R] \tag{\%t5}$$

$\nabla \times \vec{F} \in \mathbb{R}^3$

(%i6) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [R_y - Q_z, P_z - R_x, Q_x - P_y] \tag{\%t6}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i7) `ldisplay(α:F.cartan_basis)$`

$$\alpha = R\,dz + Q\,dy + P\,dx \tag{\%t7}$$

d$\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i8) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = (R_y - Q_z)\ dy\,dz + (R_x - P_z)\ dx\,dz + (Q_x - P_y)\ dx\,dy \tag{\%t8}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i9) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = R_z + Q_y + P_x \tag{\%t9}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i10)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = P\,dy\,dz - Q\,dx\,dz + R\,dx\,dy \tag{\%t10}$$

d$\beta \in \mathcal{A}^3(\mathbb{R}^3)$

(%i11) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = (R_z + Q_y + P_x)\ dx\,dy\,dz \tag{\%t11}$$

(%i12) `dβ/apply("*",cartan_basis);`

$$R_z + Q_y + P_x \tag{\%o12}$$

Surface $\vec{S} \in \mathbb{R}^3$

```
(%i13) depends(f,[x,y])$
(%i14) ldisplay(S:[x,y,f])$
```

$$\vec{S} = [x, y, f] \tag{\%t14}$$

Normal $\vec{N} \in \mathbb{R}^3$

```
(%i15) ldisplay(N:ratsimp(mycross(diff(S,x),diff(S,y))))$
```

$$\vec{N} = [-f_x, -f_y, 1] \tag{\%t15}$$

```
(%i16) ldisplay(n:scanmap(ratsimp,normalize(N)))$
```

$$\hat{n} = \left[-\frac{f_x}{\sqrt{(f_y)^2 + (f_x)^2 + 1}}, -\frac{f_y}{\sqrt{(f_y)^2 + (f_x)^2 + 1}}, \frac{1}{\sqrt{(f_y)^2 + (f_x)^2 + 1}}\right] \tag{\%t16}$$

Hence $\hat{n} = \frac{1}{\|\vec{N}\|} \langle -f_x, -f_y, 1 \rangle$

Surface integral

$$\iint_S \left(\nabla \times \vec{F}\right) \cdot \mathrm{d}\vec{S} = \iint_D \langle R_y - Q_z, P_z - R_x, Q_y - P_x \rangle \cdot \langle -f_x, -f_y, 1 \rangle \, \mathrm{d}A$$

$\vec{F} \cdot \vec{N} \in \mathbb{R}$

```
(%i17) ldisplay(T:ratsimp(F.N))$
```

$$T = -Q\ (f_y) - P\ (f_x) + R \tag{\%t17}$$

Pullback $\vec{S}^{*}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

```
(%i18) ldisplay(Pb:ratsimp(diff(S,y)|(diff(S,x)|ev(β,map("=",ζ,S)))))$
```

$$Pb = -Q\ (f_y) - P\ (f_x) + R \tag{\%t18}$$

Curve $\vec{r} \in \mathbb{R}^3$

(%i19) `depends([x,y],t)$`

(%i20) `declare([a,b],constant)$`

(%i21) `ldisplay(r:[x,y,f])$`

$$\vec{r} = [x, y, f] \tag{\%t21}$$

Derivative of the curve $\vec{r}$

(%i22) `ldisplay(r\':diff(r,t))$`

$$r' = [\dot{x}, \dot{y}, (f_y)\ (\dot{y}) + (f_x)\ (\dot{x})] \tag{\%t22}$$

Line integral

$$\oint_C \vec{F} \cdot \mathrm{d}\vec{r} = \int_a^b \vec{F}\left(\vec{r}(t)\right) \cdot \vec{r}'\,\mathrm{d}t$$

$\vec{F} \cdot \vec{r}' \in \mathbb{R}$

(%i23) `ldisplay(T:collectterms(expand(F.r\'),diff(x,t),diff(y,t)))$`

$$T = (R\ (f_y) + Q)\ (\dot{y}) + (R\ (f_x) + P)\ (\dot{x}) \tag{\%t23}$$

Pullback $\vec{r}^{*}\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i24) `ldisplay(Pb:collectterms(r\'|subst(map("=",ζ,r),α),diff(x,t),diff(y,t)))$`

$$Pb = (R\ (f_y) + Q)\ (\dot{y}) + (R\ (f_x) + P)\ (\dot{x}) \tag{\%t24}$$

Use Green's theorem

$$\oint_C \vec{F} \cdot \mathrm{d}\vec{r} = \oint_{\partial D} \overbrace{(P + Rf_x)}^{\hat{P}}\,\mathrm{d}x + \overbrace{(Q + Rf_y)}^{\hat{Q}}\,\mathrm{d}y = \iint_D \left(\frac{\partial \hat{Q}}{\partial x} - \frac{\partial \hat{P}}{\partial y}\right)\mathrm{d}A$$

$$\oint_C \vec{F} \cdot \mathrm{d}\vec{r} = \iint_D \left(\frac{\partial}{\partial x}(Q + Rf_y) - \frac{\partial}{\partial y}(P + Rf_x)\right)\mathrm{d}A$$

$$\oint_C \vec{F} \cdot \mathrm{d}\vec{r} = \iint_S \left(\nabla \times \vec{F}\right) \cdot \mathrm{d}\vec{S}$$

2 Verifying Stoke's theorem, example 1

Based on Michael Penn Video Verifying Stoke's theorem, example 1

Verify Stoke's with $\vec{F} = \langle z, x, y \rangle$ and $\vec{S}$ is the top half of $x^2 + y^2 + z^2 = a^2$

(%i25) `kill(labels,a,t,x,y,z,ρ,ϕ,θ)$`

Define the space $\mathbb{R}^3$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Parameters

(%i4) `assume(a>0)$`

(%i5) `declare(a,constant)$`

(%i6) `params:[a=1]$`

Vector field $\vec{F} \in \mathbb{R}^3$

(%i7) `ldisplay(F:[z,x,y])$`

$$\vec{F} = [z, x, y] \tag{\%t7}$$

3D Direction field

(%i9)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,0,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i11)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i12)
```
wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$
```



(%t12)

$\nabla \times \vec{F} \in \mathbb{R}^3$

(%i13) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [1, 1, 1] \tag{\%t13}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i14) `ldisplay(α:F.cartan_basis)$`

$$\alpha = y\, dz + x\, dy + z\, dx \tag{\%t14}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i15) `ldisplay(dα:ext_diff(α))$`

$$d\alpha = dy\, dz - dx\, dz + dx\, dy \tag{\%t15}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i16) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = 0 \tag{\%t16}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i17)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = z\, dy\, dz - x\, dx\, dz + y\, dx\, dy \tag{\%t17}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i18) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = 0 \tag{\%t18}$$

Spherical coordinates

(%i22)
```
assume(0≤ρ)$
assume(0≤φ,φ≤1/2*π)$
assume(sin(φ)≥0)$
assume(0≤θ,θ≤2*π)$
```

(%i23)
```
ξ:[ρ,φ,θ]$
```

(%i24)
```
ldisplay(L:[ρ*cos(θ)*sin(φ),ρ*sin(θ)*sin(φ),ρ*cos(φ)])$
```

$$\vec{L} = [\cos(\theta)\rho\,\sin(\phi), \sin(\theta)\rho\,\sin(\phi), \rho\,\cos(\phi)] \tag{\%t24}$$

(%i25)
```
ldisplay(J:jacobian(L,ξ))$
```

$$J = \begin{pmatrix} \cos(\theta)\,\sin(\phi) & \cos(\theta)\rho\,\cos(\phi) & -\sin(\theta)\rho\,\sin(\phi) \\ \sin(\theta)\,\sin(\phi) & \sin(\theta)\rho\,\cos(\phi) & \cos(\theta)\rho\,\sin(\phi) \\ \cos(\phi) & -\rho\,\sin(\phi) & 0 \end{pmatrix} \tag{\%t25}$$

(%i26)
```
ldisplay(lg:trigsimp(transpose(J).J))$
```

$$lg = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \rho^2 & 0 \\ 0 & 0 & \rho^2\sin(\phi)^2 \end{pmatrix} \tag{\%t26}$$

(%i27)
```
ldisplay(Jdet:trigsimp(determinant(J)))$
```

$$Jdet = \rho^2\,\sin(\phi) \tag{\%t27}$$

Surface $\vec{S} \in \mathbb{R}^3$

(%i28) `ldisplay(S:[a*cos(θ)*sin(φ),a*sin(θ)*sin(φ),a*cos(φ)])$`

$$\vec{S} = [a\,\cos(\theta)\,\sin(\phi), a\,\sin(\theta)\,\sin(\phi), a\,\cos(\phi)] \tag{\%t28}$$

Graphics

(%i29) `wxdraw3d(xu_grid=50,yv_grid=50,proportional_axes=xyz,`
`apply(parametric_surface,append(S,[φ,0,`$\frac{1}{2}$`*π,θ,0,2*π]))),params$`



(%t29)

Normal $\vec{N} \in \mathbb{R}^3$

(%i30) `ldisplay(N:trigsimp(mycross(diff(S,`ϕ`),diff(S,`θ`))))$`

$$\vec{N} = [a^2\,\cos(\theta)\sin(\phi)^2, a^2\,\sin(\theta)\sin(\phi)^2, a^2\,\cos(\phi)\,\sin(\phi)] \tag{\%t30}$$

(%i31) `ldisplay(n:scanmap(trigsimp,normalize(N)))$`

$$\hat{n} = [\cos(\theta)\,\sin(\phi), \sin(\theta)\,\sin(\phi), \cos(\phi)] \tag{\%t31}$$

Hence $\hat{n} = \frac{1}{a}\langle x, y, z\rangle$

$\left(\nabla \times \vec{F}\right) \cdot \vec{N} \in \mathbb{R}$

(%i32) `ldisplay(T:factor(trigsimp(curlF.N)))$`

$$T = a^2\,\sin(\phi)\,(\sin(\theta)\,\sin(\phi) + \cos(\theta)\,\sin(\phi) + \cos(\phi)) \tag{\%t32}$$

Pullback $\vec{S}^*\,\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i33) `ldisplay(P:factor(trigsimp(diff(S,`θ`)|(diff(S,`ϕ`)|subst(map("=",`ζ`,S),d`α`)))))$`

$$P = a^2\,\sin(\phi)\,(\sin(\theta)\,\sin(\phi) + \cos(\theta)\,\sin(\phi) + \cos(\phi)) \tag{\%t33}$$

Flux through $\vec{S}$

(%i34) `I:'integrate('integrate(T,`ϕ`,0,`$\frac{1}{2}$`*`π`),`θ`,0,2*`π`)$`

(%i35) `ldisplay(I=box(ev(I,integrate)))$`

$$a^2\int_0^{2\pi}\int_0^{\frac{\pi}{2}} \sin(\phi)\,(\sin(\theta)\,\sin(\phi) + \cos(\theta)\,\sin(\phi) + \cos(\phi))\,d\phi\,d\theta = \boxed{\pi a^2} \tag{\%t35}$$

Curve $\vec{C} \in \mathbb{R}^3$

```
(%i36) ldisplay(C:[a*cos(t),a*sin(t),0])$
```

$$\vec{C} = [a\,\cos(t), a\,\sin(t), 0] \tag{\%t36}$$

Derivative of the curve $\vec{C}$

```
(%i37) ldisplay(C\':diff(C,t))$
```

$$C' = [-a\,\sin(t), a\,\cos(t), 0] \tag{\%t37}$$

$\vec{F} \circ \vec{C}$

```
(%i38) ldisplay(FoC:subst(map("=",ζ,C),F))$
```

$$FoC = [0, a\,\cos(t), a\,\sin(t)] \tag{\%t38}$$

$\vec{F} \cdot \vec{C}' \in \mathbb{R}$

```
(%i39) ldisplay(T:FoC.C\')$
```

$$T = a^2 \cos(t)^2 \tag{\%t39}$$

Pullback $\vec{C}^{*}\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

```
(%i40) ldisplay(P:C\'|subst(map("=",ζ,C),α))$
```

$$P = a^2 \cos(t)^2 \tag{\%t40}$$

Line integral I

```
(%i41) I:'integrate(T,t,0,2*π)$
(%i42) ldisplay(I=box(ev(I,integrate)))$
```

$$a^2 \int_0^{2\pi} \cos(t)^2 dt = \left(\pi a^2\right) \tag{\%t42}$$

Clean up

```
(%i47) forget(a>0)$
       forget(0≤ρ)$
       forget(0≤φ,φ≤1/2*π)$
       forget(sin(φ)≥0)$
       forget(θ≥0,θ≤2*π)$
```

3 Verifying Stoke's theorem, example 2

Based on Michael Penn Video Verifying Stoke's theorem, example 2

Verify Stoke's with $\vec{F} = \langle z, x*z, x*y \rangle$ and $\vec{S}$ is $z = x^2 + y^2$ (paraboloid) below $z = 4$.

(%i48) `kill(labels,a,t,x,y,z,r,θ,ϕ,ρ)$`

Define the space $\mathbb{R}^3$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Parameters

(%i4) `assume(a>0)$`

(%i5) `declare(a,constant)$`

(%i6) `params:[a=2]$`

Vector field $\vec{F} \in \mathbb{R}^3$

(%i7)
```
ldisplay(F:[z,x*z,x*y])$
```

$$\vec{F} = [z, xz, xy] \tag{\%t7}$$

3D Direction field

(%i9)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,0,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i11)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i12)
```
wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$
```



(%t12)

$\nabla \times \vec{F} \in \mathbb{R}^3$

(%i13) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [0, 1 - y, z] \qquad (\%t13)$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i14) `ldisplay(α:F.cartan_basis)$`

$$\alpha = xy\, dz + xz\, dy + z\, dx \qquad (\%t14)$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i15) `ldisplay(dα:ext_diff(α))$`

$$d\alpha = y\, dx\, dz - dx\, dz + z\, dx\, dy \qquad (\%t15)$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i16) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = 0 \qquad (\%t16)$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i17)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = z\, dy\, dz - xz\, dx\, dz + xy\, dx\, dy \qquad (\%t17)$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i18) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = 0 \qquad (\%t18)$$

Cylindrical coordinates

(%i20)
```
assume(0≤r)$
assume(0≤θ,θ≤2*π)$
```

(%i21)
```
ξ:[r,θ,z]$
```

(%i22)
```
ldisplay(L:[r*cos(θ),r*sin(θ),z])$
```

$$\vec{L} = [r\,\cos(\theta), r\,\sin(\theta), z] \tag{\%t22}$$

(%i23)
```
ldisplay(J:jacobian(L,ξ))$
```

$$J = \begin{pmatrix} \cos(\theta) & -r\,\sin(\theta) & 0 \\ \sin(\theta) & r\,\cos(\theta) & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t23}$$

(%i24)
```
ldisplay(lg:trigsimp(transpose(J).J))$
```

$$lg = \begin{pmatrix} 1 & 0 & 0 \\ 0 & r^2 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t24}$$

(%i25)
```
ldisplay(Jdet:trigsimp(determinant(J)))$
```

$$Jdet = r \tag{\%t25}$$

Surface $\vec{S} \in \mathbb{R}^3$

(%i26) `ldisplay(S:[r*cos(θ),r*sin(θ),r^2])$`

$$\vec{S} = [r\ \cos(\theta), r\ \sin(\theta), r^2] \tag{\%t26}$$

Graphics

(%i27) `wxdraw3d(xu_grid=50,yv_grid=50,proportional_axes=xyz,`
`apply(parametric_surface,append(S,[r,0,a,θ,0,2*π]))),params$`



(%t27)

Normal $\vec{N} \in \mathbb{R}^3$

(%i28) `ldisplay(N:trigsimp(mycross(diff(S,r),diff(S,θ))))$`

$$\vec{N} = [-2r^2 \cos(\theta), -2r^2 \sin(\theta), r] \tag{\%t28}$$

(%i29) `ldisplay(n:scanmap(trigsimp,normalize(N)))$`

$$\hat{n} = \left[-\frac{2r\cos(\theta)}{\sqrt{4r^2+1}}, -\frac{2r\sin(\theta)}{\sqrt{4r^2+1}}, \frac{1}{\sqrt{4r^2+1}}\right] \tag{\%t29}$$

Hence $\hat{n} = \dfrac{1}{\|\vec{N}\|}\langle -2x, -2y, 1\rangle$

$\left(\nabla \times \vec{F}\right) \circ \vec{S} \in \mathbb{R}^3$

(%i30) `ldisplay(curlFoS:trigsimp(subst(map("=",ζ,S),curlF)))$`

$$curlFoS = [0, 1 - r\sin(\theta), r^2] \tag{\%t30}$$

$\left(\nabla \times \vec{F}\right) \cdot \vec{N} \in \mathbb{R}$

(%i31) `ldisplay(T:expand(trigsimp(curlFoS.N)))$`

$$T = 2r^3\sin(\theta)^2 - 2r^2\sin(\theta) + r^3 \tag{\%t31}$$

Pullback $\vec{S}^* \,\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i32) `ldisplay(P:expand(trigsimp(diff(S,θ)|(diff(S,r)|subst(map("=",ζ,S),dα)))))$`

$$P = 2r^3\sin(\theta)^2 - 2r^2\sin(\theta) + r^3 \tag{\%t32}$$

Flux through $\vec{S}$

(%i33) `I:'integrate('integrate(T,r,0,a),θ,0,2*π)$`

(%i34) `ldisplay(I=box(ev(I,integrate)))$`

$$\int_0^{2\pi}\int_0^a 2r^3\sin(\theta)^2 - 2r^2\sin(\theta) + r^3 dr\, d\theta = \left(\pi a^4\right) \tag{\%t34}$$

(%i35) `ldisplay(I=box(ev(I,integrate,params)))$`

$$\int_0^{2\pi}\int_0^a 2r^3\sin(\theta)^2 - 2r^2\sin(\theta) + r^3 dr\, d\theta = (16\pi) \tag{\%t35}$$

Surface $\vec{S} \in \mathbb{R}^3$

(%i36) `ldisplay(S:[x,y,x^2+y^2])$`

$$\vec{S} = [x, y, y^2 + x^2] \tag{\%t36}$$

Graphics

(%i37) `wxdraw3d(xu_grid=50,yv_grid=50,proportional_axes=xyz,zrange=[0,a^2],`
`apply(parametric_surface,append(S,[x,-a,a,y,-a,a]))),params$`



(%t37)

Normal $\vec{N} \in \mathbb{R}^3$

```
(%i38) ldisplay(N:ratsimp(mycross(diff(S,x),diff(S,y))))$
```

$$\vec{N} = [-2x, -2y, 1] \tag{\%t38}$$

```
(%i39) ldisplay(n:scanmap(ratsimp,normalize(N)))$
```

$$\hat{n} = \left[-\frac{2x}{\sqrt{4y^2+4x^2+1}}, -\frac{2y}{\sqrt{4y^2+4x^2+1}}, \frac{1}{\sqrt{4y^2+4x^2+1}}\right] \tag{\%t39}$$

Hence $\hat{n} = \dfrac{1}{\|\vec{N}\|} \langle -2x, -2y, 1\rangle$

$\left(\nabla \times \vec{F}\right) \circ \vec{S} \in \mathbb{R}^3$

```
(%i40) ldisplay(curlFoS:ratsimp(subst(map("=",ζ,S),curlF)))$
```

$$curlFoS = [0, 1-y, y^2+x^2] \tag{\%t40}$$

$\left(\nabla \times \vec{F}\right) \cdot \vec{N} \in \mathbb{R}$

```
(%i41) ldisplay(T:ratsimp(curlFoS.N))$
```

$$T = 3y^2 - 2y + x^2 \tag{\%t41}$$

Pullback $\vec{S}^* \, \mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

```
(%i42) ldisplay(P:ratsimp(diff(S,y)|(diff(S,x)|subst(map("=",ζ,S),dα))))$
```

$$P = 3y^2 - 2y + x^2 \tag{\%t42}$$

Change to Cylindrical coordinates

```
(%i43) ldisplay(T:expand(trigsimp(subst(map("=",ζ,L),T))))$
```

$$T = 2r^2 \sin(\theta)^2 - 2r\,\sin(\theta) + r^2 \tag{\%t43}$$

Flux through $\vec{S}$

```
(%i44) I:'integrate('integrate(expand(T*Jdet),r,0,a),θ,0,2*π)$
(%i45) ldisplay(I=box(ev(I,integrate)))$
```

$$\int_0^{2\pi} \int_0^a 2r^3 \sin(\theta)^2 - 2r^2\,\sin(\theta) + r^3 dr\, d\theta = \left(\pi a^4\right) \tag{\%t45}$$

```
(%i46) ldisplay(I=box(ev(I,integrate,params)))$
```

$$\int_0^{2\pi} \int_0^a 2r^3 \sin(\theta)^2 - 2r^2\,\sin(\theta) + r^3 dr\, d\theta = (16\pi) \tag{\%t46}$$

Curve $\vec{C} \in \mathbb{R}^3$

(%i47) `ldisplay(C:[a*cos(t),a*sin(t),a^2])$`

$$\vec{C} = [a\ \cos(t), a\ \sin(t), a^2] \tag{\%t47}$$

Derivative of the curve $\vec{C}$

(%i48) `ldisplay(C\':diff(C,t))$`

$$C' = [-a\ \sin(t), a\ \cos(t), 0] \tag{\%t48}$$

$\vec{F} \circ \vec{C}$

(%i49) `ldisplay(FoC:subst(map("=",ζ,C),F))$`

$$FoC = [a^2, a^3\ \cos(t), a^2\ \cos(t)\ \sin(t)] \tag{\%t49}$$

$\vec{F} \cdot \vec{C}' \in \mathbb{R}$

(%i50) `ldisplay(T:FoC.C\')$`

$$T = a^4 \cos(t)^2 - a^3\ \sin(t) \tag{\%t50}$$

Pullback $\vec{C}^{*}\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i51) `ldisplay(P:C\'|subst(map("=",ζ,C),α))$`

$$P = a^4 \cos(t)^2 - a^3\ \sin(t) \tag{\%t51}$$

Line integral I

(%i52) `I:'integrate(T,t,0,2*π)$`

(%i53) `ldisplay(I=box(ev(I,integrate)))$`

$$\int_0^{2\pi} a^4 \cos(t)^2 - a^3\ \sin(t) dt = \boxed{\pi a^4} \tag{\%t53}$$

(%i54) `ldisplay(I=box(ev(I,integrate,params)))$`

$$\int_0^{2\pi} a^4 \cos(t)^2 - a^3\ \sin(t) dt = \boxed{16\pi} \tag{\%t54}$$

Clean up

(%i57)
```
forget(a>0)$
forget(0≤r)$
forget(0≤θ,θ≤2*π)$
```

4 Two Stoke's theorem examples

Based on Michael Penn Video Two Stoke's theorem examples

Verify Stoke's with $\vec{F} = \langle 2xy^2z, 2x^2yz, x^2y^2 - 2z\rangle$ and $\vec{C}$ is $\vec{r} = \langle \cos(t), \sin(t), \sin(t)\rangle$ with $t \in [0, 2\pi]$

(%i58) `kill(labels,t,x,y,z)$`

Define the space $\mathbb{R}^3$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^3$

(%i4)
```
ldisplay(F:[2*x*y^2*z,2*x^2*y*z,x^2*y^2-2*z])$
```

$$\vec{F} = [2x\,y^2 z, 2x^2 yz, x^2\,y^2 - 2z] \quad (\%t4)$$

3D Direction field

(%i6)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,0,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i8)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i9)
```
wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$
```



(%t9)

$\nabla \times \vec{F} \in \mathbb{R}^3$

(%i10) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [0, 0, 0] \tag{\%t10}$$

Potential ϕ

(%i11) `ldisplay(`ϕ`:potential(F))$`

$$\phi = x^2\, y^2 z - z^2 \tag{\%t11}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i12) `ldisplay(`α`:F.cartan_basis)$`

$$\alpha = \left(x^2\, y^2 - 2z\right)\; dz + 2x^2 yz\; dy + 2x\, y^2 z\; dx \tag{\%t12}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i13) `ldisplay(d`α`:ext_diff(`α`))$`

$$d\alpha = 0 \tag{\%t13}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i14) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = 2y^2 z + 2x^2 z - 2 \tag{\%t14}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i15) `ldisplay(`β`:F[1]*cartan_basis[2]~cartan_basis[3]+`
`F[2]*cartan_basis[3]~cartan_basis[1]+`
`F[3]*cartan_basis[1]~cartan_basis[2])$`

$$\beta = 2x\, y^2 z\; dy\; dz - 2x^2 yz\; dx\; dz + \left(x^2\, y^2 - 2z\right)\; dx\; dy \tag{\%t15}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i16) `ldisplay(d`β`:edit(ext_diff(`β`)))$`

$$d\beta = \left(2y^2 z + 2x^2 z - 2\right)\; dx\; dy\; dz \tag{\%t16}$$

(%i17) `d`β`/apply("*",cartan_basis);`

$$2y^2 z + 2x^2 z - 2 \tag{\%o17}$$

Curve $\vec{C} \in \mathbb{R}^3$

(%i18) `ldisplay(C:[cos(t),sin(t),sin(t)])$`

$$\vec{C} = [\cos(t), \sin(t), \sin(t)] \tag{\%t18}$$

(%i19) `ldisplay(γ:[cos(t),sin(t),0])$`

$$\gamma = [\cos(t), \sin(t), 0] \tag{\%t19}$$

Graphics

(%i20)
```
wxdraw3d(proportional_axes=xyz,line_width=2,
color=red,apply(parametric,append(C,[t,0,2*π])),
color=blue,apply(parametric,append(γ,[t,0,2*π])))$
```



(%t20)

Derivative of the curve $\vec{C}$

(%i21) `ldisplay(C\':diff(C,t))$`

$$C' = [-\sin(t), \cos(t), \cos(t)] \tag{\%t21}$$

$\vec{F} \circ \vec{C}$

(%i22) `ldisplay(FoC:subst(map("=",ζ,C),F))$`

$$FoC = [2\cos(t)\sin(t)^3, 2\cos(t)^2\sin(t)^2, \cos(t)^2\sin(t)^2 - 2\sin(t)] \tag{\%t22}$$

$\vec{F} \cdot \vec{C}' \in \mathbb{R}$

(%i23) `ldisplay(T:trigsimp(FoC.C\'))$`

$$T = -2\cos(t)\,\sin(t) - 5\cos(t)^5 + 7\cos(t)^3 - 2\cos(t) \tag{\%t23}$$

Pullback $\vec{C}^*\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i24) `ldisplay(P:trigsimp(C\'|subst(map("=",ζ,C),α)))$`

$$P = -2\cos(t)\,\sin(t) - 5\cos(t)^5 + 7\cos(t)^3 - 2\cos(t) \tag{\%t24}$$

Line integral I

(%i25) `I:'integrate(T,t,0,2*π)$`

(%i26) `ldisplay(I=box(ev(I,integrate)))$`

$$\int_0^{2\pi} -2\cos(t)\,\sin(t) - 5\cos(t)^5 + 7\cos(t)^3 - 2\cos(t)dt = (0) \tag{\%t26}$$

Surface $\vec{S} \in \mathbb{R}^3$

(%i27) `ldisplay(S:[x,y,y])$`

$$\vec{S} = [x, y, y] \tag{\%t27}$$

Graphics

(%i28)
```
wxdraw3d(xu_grid=50,yv_grid=50,proportional_axes=xyz,
color=red,line_width=2,apply(parametric,append(C,[t,0,2*π])),
color=blue,line_width=1,apply(parametric_surface,append(S,[x,-1,1,y,-1,1])))$
```



(%t28)

$\nabla \times \vec{F} = \vec{0}$

(%i29) `ldisplay(curlF)$`

$$curlF = [0, 0, 0] \tag{\%t29}$$

Normal $\vec{N} \in \mathbb{R}^3$

(%i30) `ldisplay(N:trigsimp(mycross(diff(S,x),diff(S,y))))$`

$$\vec{N} = [0, -1, 1] \tag{\%t30}$$

(%i31) `ldisplay(n:scanmap(trigsimp,normalize(N)))$`

$$\hat{n} = \left[0, -\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right] \tag{\%t31}$$

Hence $\hat{n} = \dfrac{1}{\sqrt{2}} \langle 0, -1, 1 \rangle$

Surface $x^2 + (1/2)(y^2 + z^2) = 1$

(%i32)
```
wxdraw3d(proportional_axes=xy,view=[70,70],
x_voxel=20,y_voxel=20,z_voxel=20,xu_grid=50,yv_grid=50,
color=green,implicit(x^2+1/2*(y^2+z^2)=1,x,-1.5,1.5,y,-1.5,1.5,z,-1.5,1.5),
color=blue,implicit(y=z,x,-1.5,1.5,y,-1.5,1.5,z,-1.5,1.5),
color=red,line_width=3,apply(parametric,append(C,[t,0,2*π])))$
```



(%t32)

Based on Michael Penn Video Two Stoke's theorem examples.

Calculate $\iint_S \left(\nabla \times \vec{F}\right) \cdot \mathrm{d}\vec{S}$ where S is the part of $z = 1 - x^2 - 2y^2$ with $z \geq 0$ and $\vec{F} = \left\langle x, y^2, xe^{xy} \right\rangle$

Vector field $\vec{F} \in \mathbb{R}^3$

(%i33)
```
ldisplay(F:[x,y^2,z*exp(x*y)])$
```

$$F = [x, y^2, e^{xy} z] \tag{\%t33}$$

3D Direction field

(%i35)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,0,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i37)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i38)
```
wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$
```



(%t38)

$\nabla \times \vec{F} \in \mathbb{R}^3$

(%i39) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [x\, e^{xy} z, -y\, e^{xy} z, 0] \tag{\%t39}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i40) `ldisplay(α:F.cartan_basis)$`

$$\alpha = e^{xy} z\; dz + y^2\; dy + x\; dx \tag{\%t40}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i41) `ldisplay(dα:ext_diff(α))$`

$$d\alpha = x\, e^{xy} z\; dy\; dz + y\, e^{xy} z\; dx\; dz \tag{\%t41}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i42) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = e^{xy} + 2y + 1 \tag{\%t42}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i43)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = x\; dy\; dz - y^2\; dx\; dz + e^{xy} z\; dx\; dy \tag{\%t43}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i44) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = \left(e^{xy} + 2y + 1\right)\; dx\; dy\; dz \tag{\%t44}$$

(%i45) `dβ/apply("*",cartan_basis);`

$$e^{xy} + 2y + 1 \tag{\%o45}$$

Surface $\vec{S} \in \mathbb{R}^3$

(%i46) `ldisplay(S:[x,y,1-x^2-2*y^2])$`

$$\vec{S} = [x, y, -2y^2 - x^2 + 1] \tag{\%t46}$$

Normal $\vec{N} \in \mathbb{R}^3$

(%i47) `ldisplay(N:trigsimp(mycross(diff(S,x),diff(S,y))))$`

$$\vec{N} = [2x, 4y, 1] \tag{\%t47}$$

(%i48) `ldisplay(n:scanmap(trigsimp,normalize(N)))$`

$$\hat{n} = \left[\frac{2x}{\sqrt{16y^2 + 4x^2 + 1}}, \frac{4y}{\sqrt{16y^2 + 4x^2 + 1}}, \frac{1}{\sqrt{16y^2 + 4x^2 + 1}}\right] \tag{\%t48}$$

Hence $\hat{n} = \dfrac{1}{\|\vec{N}\|} \langle 2x, 4y, 1\rangle$

$\left(\nabla \times \vec{F}\right) \circ \vec{S} \in \mathbb{R}^3$

(%i49) `ldisplay(curlFoS:ratsimp(subst(map("=",ζ,S),curlF)))$`

$$curlFoS = \left[\left(-2x\, y^2 - x^3 + x\right)\, e^{xy}, \left(2y^3 + \left(x^2 - 1\right) y\right)\, e^{xy}, 0\right] \tag{\%t49}$$

$\left(\nabla \times \vec{F}\right) \cdot \vec{N} \in \mathbb{R}$

(%i50) `ldisplay(T:ratsimp(curlFoS.N))$`

$$T = \left(8y^4 - 4y^2 - 2x^4 + 2x^2\right)\, e^{xy} \tag{\%t50}$$

Pullback $\vec{S}^*\, \mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i51) `ldisplay(P:ratsimp(diff(S,y)|(diff(S,x)|subst(map("=",ζ,S),dα))))$`

$$P = \left(8y^4 - 4y^2 - 2x^4 + 2x^2\right)\, e^{xy} \tag{\%t51}$$

Curve $\vec{C}$

(%i52) `ldisplay(C:[cos(t),sin(t)/√(2),0])$`

$$\vec{C} = \left[\cos(t), \frac{\sin(t)}{\sqrt{2}}, 0\right] \quad (\%t52)$$

Graphics

(%i53) `wxdraw3d(xu_grid=50,yv_grid=50,proportional_axes=xyz,zrange=[0,1],`
`color=red,line_width=3,apply(parametric,append(C,[t,0,2*π])),`
`color=blue,line_width=1,apply(parametric_surface,append(S,[x,-1,1,y,-1,1])))$`



(%t53)

Derivative of the curve $\vec{C}$

(%i54) `ldisplay(C\':diff(C,t))$`

$$C' = \left[-\sin(t), \frac{\cos(t)}{\sqrt{2}}, 0\right] \tag{\%t54}$$

$\vec{F} \circ \vec{C}$

(%i55) `ldisplay(FoC:subst(map("=",ζ,C),F))$`

$$FoC = \left[\cos(t), \frac{\sin(t)^2}{2}, 0\right] \tag{\%t55}$$

$\vec{F} \cdot \vec{C}' \in \mathbb{R}$

(%i56) `ldisplay(T:trigsimp(FoC.C\'))$`

$$T = \frac{\cos(t)\sin(t)^2 - 2^{\frac{3}{2}}\cos(t)\sin(t)}{2^{\frac{3}{2}}} \tag{\%t56}$$

Pullback $\vec{C}^*\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i57) `ldisplay(P:trigsimp(C\'|subst(map("=",ζ,C),α)))$`

$$P = \frac{\cos(t)\sin(t)^2 - 2^{\frac{3}{2}}\cos(t)\sin(t)}{2^{\frac{3}{2}}} \tag{\%t57}$$

Line integral I

(%i58) `I:'integrate(T,t,0,2*π)$`

(%i59) `changevar(I,u=sin(t),u,t);`

solve: using arc-trig functions to get a solution.Some solutions will be lost.

$$0 \tag{\%o59}$$

(%i60) `ldisplay(I=box(ev(I,integrate)))$`

$$\frac{\int_0^{2\pi} \cos(t)\sin(t)^2 - 2^{\frac{3}{2}}\cos(t)\sin(t)dt}{2^{\frac{3}{2}}} = (0) \tag{\%t60}$$