Surface Integrals

Based on Mathispower4u Playlist Mathispower4u Surface Integrals

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

(%i2) `reset()$kill(all)$`

(%i1) `derivabbrev:true$`

(%i2) `ratprint:false$`

(%i3) `fpprintprec:5$`

(%i4) `load(linearalgebra)$`

(%i5) `if get('draw,'version)=false then load(draw)$`

(%i6) `wxplot_size:[1024,768]$`

(%i7)
```
set_draw_defaults(xtics=1,ytics=1,ztics=1,xyplane=0,
xaxis=true,xaxis_type=solid,xaxis_width=3,
yaxis=true,yaxis_type=solid,yaxis_width=3,
zaxis=true,zaxis_type=solid,zaxis_width=3)$
```

(%i8) `if get('vect,'version)=false then load(vect)$`

(%i9) `if get('cartan,'version)=false then load(cartan)$`

(%i10) `norm(u):=block(ratsimp(radcan(√(express(u.u)))))$`

(%i11) `normalize(v):=block(v/norm(v))$`

(%i12) `angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$`

(%i13) `mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$`

(%i14) `declare(trigsimp,evfun)$`

1 Parameterized Surfaces

Based on Mathispower4u Video Parameterized Surfaces

1.1

Determine a parameterization for the given surface. $2\,x - 3\,y + z = 6$

(%i15)
```
kill(labels,x,y,z,u,v)$
```

(%i2)
```
ζ:[x,y,z]$
ξ:[u,v]$
```

(%i3)
```
Eq:2*x-3*y+z=6;
```

$$z - 3y + 2x = 6 \tag{Eq}$$

(%i4)
```
linsol:linsolve(Eq,z);
```

$$[z = 3y - 2x + 6] \tag{linsol}$$

(%i5)
```
ldisplay(S:subst(append(linsol,[x=u,y=v]),ζ))$
```

$$S = [u, v, 3v - 2u + 6] \tag{\%t5}$$

(%i6)
```
wxdraw3d(title="Plane",view=[60,30],ztics=5,
xu_grid=50,yv_grid=50,proportional_axes=xy,
apply(parametric_surface,append(S,[u,-3,3,v,-3,3])))$
```



(%t6)

1.2

Determine a parameterization for the given surface. $x^2 + (y-2)^2 = 4$

(%i7)
```
kill(labels,x,y,z,u,v)$
```

(%i2)
```
ζ:[x,y,z]$
ξ:[u,v]$
```

(%i3)
```
Eq:x^2+(y-2)^2=4;
```

$$(y-2)^2 + x^2 = 4 \tag{Eq}$$

(%i4)
```
ldisplay(S:[2*cos(u),2*sin(u)+2,v])$
```

$$S = [2\cos(u), 2\sin(u) + 2, v] \tag{\%t4}$$

(%i5)
```
is(trigsimp(subst(map("=",ζ,S),Eq)));
```

$$\text{true} \tag{\%o5}$$

(%i6)
```
wxdraw3d(title="Cylinder",view=[60,30],
xu_grid=50,yv_grid=50,proportional_axes=xy,
apply(parametric_surface,append(S,[u,-5,5,v,-5,5])))$
```



(%t6)

1.3

Determine a parameterization for the given surface. $x^2 + y^2 + z^2 = 9$

(%i7)
```
kill(labels,x,y,z,φ,θ)$
```

(%i2)
```
ζ:[x,y,z]$
ξ:[φ,θ]$
```

(%i3)
```
Eq:x^2+y^2+z^2=9;
```

$$z^2 + y^2 + x^2 = 9 \tag{Eq}$$

(%i4)
```
ldisplay(S:[3*sin(φ)*cos(θ),3*sin(φ)*sin(θ),3*cos(φ)])$
```

$$S = [3\cos(\theta)\,\sin(\phi), 3\sin(\theta)\,\sin(\phi), 3\cos(\phi)] \tag{\%t4}$$

(%i5)
```
is(trigsimp(subst(map("=",ζ,S),Eq)));
```

$$\text{true} \tag{\%o5}$$

(%i6)
```
wxdraw3d(title="Sphere",view=[60,30],
xu_grid=50,yv_grid=50,proportional_axes=xy,
apply(parametric_surface,append(S,[φ,0,π,θ,0,2*π])))$
```



(%t6)

1.4

Determine the rectangular equation given by $\vec{r}(u,v) = \langle u, v, \sqrt{u^2+v^2}\rangle$

(%i7)
```
kill(labels,x,y,z,u,v)$
```

(%i2)
```
ζ:[x,y,z]$
ξ:[u,v]$
```

(%i3)
```
ldisplay(S:[u,v,√(u^2+v^2)])$
```

$$S = [u, v, \sqrt{v^2+u^2}] \tag{\%t3}$$

(%i4)
```
subst([u=x,v=y],S);
```

$$[x, y, \sqrt{y^2+x^2}] \tag{\%o4}$$

(%i5)
```
Eq:z=last(%);
```

$$z = \sqrt{y^2+x^2} \tag{Eq}$$

(%i6) wxdraw3d(title="Cone",view=[60,30],
xu_grid=50,yv_grid=50,proportional_axes=xy,
apply(parametric_surface,append(S,[u,-2,2,v,-2,2])))$



(%t6)

1.5

Determine the rectangular equation given by $\vec{r}(u,v) = \langle 3\,u\,\sin(v), 3\,u\,\cos(v), u^2\rangle$

(%i7)
```
kill(labels,x,y,z,u,v)$
```

(%i2)
```
ζ:[x,y,z]$
ξ:[u,v]$
```

(%i3)
```
ldisplay(S:[3*u*sin(v),3*u*cos(v),u^2])$
```

$$S = [3u\,\sin(v), 3u\,\cos(v), u^2] \qquad (\%t3)$$

(%i4)
```
is(trigsimp(S[1]^2+S[2]^2)=9*S[3]);
```

$$\text{true} \qquad (\%o4)$$

(%i5)
```
wxdraw3d(title="Paraboloid",view=[60,30],
xu_grid=50,yv_grid=50,proportional_axes=xy,
apply(parametric_surface,append(S,[u,-2,2,v,0,2*π])))$
```



(%t5)

2 Write a Parameterized Surface Using Cartesian Coordinates

For each surface $\vec{r}(u,v)$, identify the best description

(%i6)
```
kill(labels,x,y,z,u,v)$
```

(%i2)
```
ζ:[x,y,z]$
ξ:[u,v]$
```

Reference: Quadric

2.1

(%i3)
```
ldisplay(S:[v*cos(u),v,v*sin(u)])$
```

$$S = [\cos(u)v, v, \sin(u)v] \qquad (\%t3)$$

(%i4)
```
trigsimp(S[1]^2+S[3]^2-S[2]^2);
```

$$0 \qquad (\%o4)$$

```
(%i5) wxdraw3d(title="Cone",view=[60,30],
      xu_grid=50,yv_grid=50,proportional_axes=xy,
      enhanced3d=[sin(r*s),r,s],
      apply(parametric_surface,append(S,[u,-2*π,2*π,v,-2*π,2*π])))$
```



(%t5)

2.2

(%i6) `ldisplay(S:[v*cos(u),u,v*sin(u)])$`

$$S = [\cos(u)v, u, \sin(u)v] \tag{\%t6}$$

(%i7) `is(trigreduce(S[3]/S[1])=tan(S[2]));`

true (%o7)

(%i8)
```
wxdraw3d(title="Screw(helicoid)",view=[50,60],
xu_grid=50,yv_grid=50,proportional_axes=xy,
enhanced3d=[sin(r*s),r,s],
apply(parametric_surface,append(S,[u,-2*π,2*π,v,-2*π,2*π])))$
```



(%t8)

2.3

(%i9) `ldisplay(S:[v^2,u,v])$`

$$S = [v^2, u, v] \quad (\%t9)$$

(%i10) `is(S[1]=S[3]^2);`

true (%o10)

(%i11)
```
wxdraw3d(title="Parabolic cylinder",view=[63,56],
xtics=5,ytics=1,ztics=1,
xu_grid=50,yv_grid=50,proportional_axes=xy,
apply(parametric_surface,append(S,[u,-6,6,v,-6,6])))$
```



(%t11)

2.4

(%i12) `ldisplay(S:[u,v,u+v])$`

$$S = [u, v, v + u] \qquad (\%t12)$$

(%i13) `is(S[3]=S[1]+S[2]);`

true (%o13)

(%i14)
```
wxdraw3d(title="Plane",view=[60,30],
xu_grid=50,yv_grid=50,proportional_axes=xy,
apply(parametric_surface,append(S,[u,-2,2,v,-2,2])))$
```



(%t14)

3 Graph Parameterized Surfaces Using 3D Calc Plotter

Reference: CalcPlot3D

3.1

(%i15) `ldisplay(S:[v*cos(u),v^2,v*sin(u)])$`

$$S = [\cos(u)v, v^2, \sin(u)v] \tag{\%t15}$$

(%i16) `is(S[2]=trigsimp(S[1]^2+S[3]^2));`

$$\text{true} \tag{\%o16}$$

(%i17)
```
wxdraw3d(title="Paraboloid",view=[50,50],ytics=5,
xu_grid=50,yv_grid=50,proportional_axes=xyz,
apply(parametric_surface,append(S,[u,-2*π,2*π,v,-6,6])))$
```



(%t17)

4 Area of a Parameterized Surface

Based on Mathispower4u Video Area of a Parameterized Surface

4.1

(%i18) `kill(labels,u,v)$`

Determine the surface of a cylinder given by $\vec{r}(u,v) = \langle 3\,\cos(u), 3\,\sin(u), v\rangle$ with $0 \le u \le 2\,\pi$ and $0 \le v \le 4$

$$\iint_S \mathrm{d}s = \iint_R \|\vec{r}_u \times \vec{r}_v\|\,\mathrm{d}u\,\mathrm{d}v$$

(%i1) `ldisplay(S:[3*cos(u),3*sin(u),v])$`

$$S = [3\cos(u), 3\sin(u), v] \tag{\%t1}$$

(%i2)
```
wxdraw3d(title="Cylinder",view=[65,54],
xu_grid=50,yv_grid=50,proportional_axes=xyz,
apply(parametric_surface,append(S,[u,0,2*π,v,0,4])))$
```



(%t2)

(%i3) `ldisplay(N:mycross(diff(S,u),diff(S,v)))$`

$$N = [3\cos(u), 3\sin(u), 0] \qquad (\%t3)$$

(%i4) `ldisplay(\|N\|:trigsimp(norm(N)))$`

$$|N| = 3 \qquad (\%t4)$$

(%i5) `n:trigsimp(normalize(N));`

$$[\cos(u), \sin(u), 0] \qquad (n)$$

(%i6) `ldisplay(A:box(integrate(integrate(\|N\|,u,0,2*π),v,0,4)))$`

$$A = (24\pi) \qquad (\%t6)$$

4.2

(%i7) `kill(labels,u,v)$`

Determine the surface area of a sphere given by $\vec{r}(u,v) = \langle 2\,\sin(u)\,\cos(v), 2\,\sin(u)\,\sin(v), 2\,\cos(u)\rangle$ with $0 \le u \le 2\,\pi$ and $0 \le v \le 2\,\pi$

$$\iint_S \mathrm{d}s = \iint_R \|\vec{r}_u \times \vec{r}_v\|\,\mathrm{d}u\,\mathrm{d}v$$

(%i1) `ldisplay(S:[2*sin(u)*cos(v),2*sin(u)*sin(v),2*cos(u)])$`

$$S = [2\sin(u)\,\cos(v), 2\sin(u)\,\sin(v), 2\cos(u)] \qquad (\%t1)$$

(%i2)
```
wxdraw3d(title="Sphere",view=[60,30],
xu_grid=50,yv_grid=50,proportional_axes=xyz,
apply(parametric_surface,append(S,[u,0,π,v,0,2*π])))$
```



(%t2)

(%i3) `ldisplay(N:trigsimp(mycross(diff(S,u),diff(S,v))))$`

$$N = [4\sin(u)^2 \cos(v), 4\sin(u)^2 \sin(v), 4\cos(u) \sin(u)] \tag{\%t3}$$

(%i4) `ldisplay(\|N\|:trigsimp(norm(N)))$`

$$|N| = 4\sin(u) \tag{\%t4}$$

(%i5) `n:trigsimp(normalize(N));`

$$[\sin(u) \cos(v), \sin(u) \sin(v), \cos(u)] \tag{n}$$

(%i6) `A:'integrate('integrate(\|N\|,u,0,π),v,0,2*π)$`

(%i7) `ldisplay(A=box(ev(A,integrate)))$`

$$8\pi \int_0^\pi \sin(u)du = \boxed{16\pi} \tag{\%t7}$$

5 Surface Integrals with Explicit Surface

For a surface S given by $z = g(x, y)$ that is continuous and differentiable over a region R in the xy-plane

$$\iint_S f(x,y,z)\,\mathrm{d}s = \iint_R f(x,y,g(x,y))\sqrt{1+(g_x)^2+(g_y)^2}\,\mathrm{d}x\,\mathrm{d}y$$

Notice: if $f(x, y, z) = 1$, we have the surface area as discussed in the previous video.

5.1

Based on Mathispower4u Video Surface Integrals with Explicit Surface Part 1

(%i8) `kill(labels,x,y,z,f,g)$`

A roof is given by the graph of $g(x, y) = 25 + 0.5\,x + 0.5\,y$ over $0 \le x \le 40$, $0 \le y \le 20$. If the density of the roof is given by $f(x, y, z) = 150 - 2\,z$, determine the mass of the roof.

(%i1) `ldisplay(g:25+`$\frac{1}{2}$`*x+`$\frac{1}{2}$`*y)$`

$$g = \frac{y}{2} + \frac{x}{2} + 25 \qquad (\%t1)$$

(%i2) `ldisplay(f:150-2*z)$`

$$f = 150 - 2z \qquad (\%t2)$$

Calculate $f \circ g$

(%i3) `ldisplay(fog:subst([z=g],f))$`

$$fog = 150 - 2\left(\frac{y}{2} + \frac{x}{2} + 25\right) \qquad (\%t3)$$

(%i4) $\sqrt{}$`(1+diff(g,x)`^{2}`+diff(g,y)`^{2}`);`

$$\frac{\sqrt{3}}{\sqrt{2}} \qquad (\%o4)$$

(%i5) `M:'integrate('integrate(fog*%,x,0,40),y,0,20)$`

(%i6) `ldisplay(M=box(ev(M,integrate,numer)))$`

$$\frac{\sqrt{3}\int_0^{20}\int_0^{40} 150 - 2\left(\frac{y}{2} + \frac{x}{2} + 25\right) dx\,dy}{\sqrt{2}} = \left(6.858610^{4}\right) \qquad (\%t6)$$

5.2

Based on Mathispower4u Video Surface Integrals with Explicit Surface Part 2

Integrate $f(x, y, z) = x\,y$ over the surface $z = 4 - 2\,x - 2\,y$ in the first octant.

(%i7) `kill(labels,x,y,z,f,g)$`

$$\iint_S f(x,y,z)\,\mathrm{d}s = \iint_R f(x,y,g(x,y))\sqrt{1+(g_x)^2+(g_y)^2}\,\mathrm{d}x\,\mathrm{d}y$$

(%i1) `ldisplay(g:4-2*x-2*y)$`

$$g = -2y - 2x + 4 \qquad (\%t1)$$

(%i2) `ldisplay(f:x*y)$`

$$f = xy \qquad (\%t2)$$

Calculate $f \circ g$

(%i3) `ldisplay(fog:subst([z=g],f))$`

$$fog = xy \qquad (\%t3)$$

(%i4) $\sqrt{}$`(1+diff(g,x)`^{2}`+diff(g,y)`^{2}`);`

$$3 \qquad (\%o4)$$

(%i5) `M:'integrate('integrate(fog*%,y,0,2-x),x,0,2)$`

(%i6) `ldisplay(M=box(ev(M,integrate)))$`

$$3\int_0^2 x\int_0^{2-x} y\,dy\,dx = (2) \qquad (\%t6)$$

6 Surface Area of a Function of Two Variables

Based on Mathispower4u Video Ex: Surface Area of a Function of Two Variables (Surface Integral)

(%i7)
```
kill(labels,x,y,r,θ)$
```

$$\iint_S \mathrm{d}s = \iint_R \sqrt{1+(g_x)^2+(g_y)^2}\,\mathrm{d}x\,\mathrm{d}y$$

Find the area of the surface of the paraboloid $z = x^2 + y^2$ that lies below the plane $z = 16$.

(%i2)
```
ζ:[x,y]$
ξ:[r,θ]$
```

(%i3)
```
Tr:[r*cos(θ),r*sin(θ)]$
```

(%i4)
```
ldisplay(g:x^2+y^2)$
```

$$g = y^2 + x^2 \tag{\%t4}$$

(%i5)
```
√(1+diff(g,x)^2+diff(g,y)^2);
```

$$\sqrt{4y^2+4x^2+1} \tag{\%o5}$$

(%i6)
```
trigsimp(subst(map("=",ζ,Tr),%));
```

$$\sqrt{4r^2+1} \tag{\%o6}$$

(%i7)
```
M:'integrate('integrate(r*%,r,0,4),θ,0,2*π)$
```

(%i8)
```
ldisplay(M=box(ev(M,integrate,ratsimp)))$
```

$$2\pi \int_0^4 r\sqrt{4r^2+1}dr = \boxed{\frac{\left(65^{\frac{3}{2}}-1\right)\pi}{6}} \tag{\%t8}$$

(%i9)
```
ldisplay(M=box(ev(M,integrate,numer)))$
```

$$2\pi \int_0^4 r\sqrt{4r^2+1}dr = \boxed{273.87} \tag{\%t9}$$

7 Surface Integrals with Parameterized Surface

Based on Mathispower4u Video Surface Integrals with Parameterized Surface

(%i10) `kill(labels,x,y,z,u,v)$`

Given a smooth surface given by $\vec{r}(u,v) = \langle x(u,v), y(u,v), z(u,v)\rangle$ and f is a continuous function

$$\iint_S f(x,y,x)\,\mathrm{d}s = \iint_R f(x(u,v), y(u,v), z(u,v))\,\|\vec{r}_u \times \vec{r}_v\|\,\mathrm{d}A$$

7.1

Evaluate $\iint_S f(x,y,x)\,\mathrm{d}s$ using a parametric surface given by $f(x,y,z) = x\,y$ and S is $x^2+y^2=4$ with $0 \le z \le 8$ in the first octant

(%i2)
```
ζ:[x,y,z]$
ξ:[u,v]$
```

(%i3) `ldisplay(f:x*y)$`

$$f = xy \tag{\%t3}$$

(%i4) `ldisplay(S:[2*cos(u),2*sin(u),v])$`

$$S = [2\cos(u), 2\sin(u), v] \tag{\%t4}$$

(%i5) `ldisplay(foS:subst(map("=",ζ,S),f))$`

$$foS = 4\cos(u)\,\sin(u) \tag{\%t5}$$

(%i6) `ldisplay(N:mycross(diff(S,u),diff(S,v)))$`

$$N = [2\cos(u), 2\sin(u), 0] \tag{\%t6}$$

(%i7) `ldisplay(\|N\|:trigsimp(norm(N)))$`

$$|N| = 2 \tag{\%t7}$$

(%i8) `ldisplay(n:trigsimp(normalize(N)))$`

$$n = [\cos(u), \sin(u), 0] \tag{\%t8}$$

(%i9) `I:'integrate('integrate(\|N\|*foS,v,0,8),u,0,`$\frac{1}{2}$`*`π`)$`

(%i10) `ldisplay(I=box(ev(I,integrate)))$`

$$64\int_0^{\frac{\pi}{2}} \cos(u)\,\sin(u)du = \boxed{32} \tag{\%t10}$$

7.2

Evaluate $\iint_S f(x,y,x)\,\mathrm{d}s$ using a parametric surface given by $f(x,y,z) = x^2 + y^2$ and S is the hemisphere $x^2 + y^2 + z^2 = 1$ above the xy-plane

(%i12) `ζ:[x,y,z]$`
`ξ:[u,v]$`

(%i13) `ldisplay(f:x^2+y^2)$`

$$f = y^2 + x^2 \tag{\%t13}$$

(%i14) `ldisplay(S:[sin(u)*cos(v),sin(u)*sin(v),cos(u)])$`

$$S = [\sin(u)\ \cos(v), \sin(u)\ \sin(v), \cos(u)] \tag{\%t14}$$

(%i15) `ldisplay(foS:trigsimp(subst(map("=",ζ,S),f)))$`

$$foS = \sin(u)^2 \tag{\%t15}$$

(%i16) `ldisplay(N:trigsimp(mycross(diff(S,u),diff(S,v))))$`

$$N = [\sin(u)^2\ \cos(v), \sin(u)^2\ \sin(v), \cos(u)\ \sin(u)] \tag{\%t16}$$

(%i17) `ldisplay(\|N\|:trigsimp(norm(N)))$`

$$|N| = \sin(u) \tag{\%t17}$$

(%i18) `ldisplay(n:trigsimp(normalize(N)))$`

$$n = [\sin(u)\ \cos(v), \sin(u)\ \sin(v), \cos(u)] \tag{\%t18}$$

(%i19) `I:'integrate('integrate(\|N\|*foS,u,0,`$\frac{1}{2}$`*π),v,0,2*π)$`

(%i20) `ldisplay(I=box(ev(I,integrate)))$`

$$2\pi \int_0^{\frac{\pi}{2}} \sin(u)^3 du = \left(\frac{4\pi}{3}\right) \tag{\%t20}$$

8 Surface Area of a Vector Valued Function Over a Region

Based on Mathispower4u Video Double Integrals - Surface Area of a Vector Valued Function Over a Region

Find the surface area of the helicoid (spiral ramp) with vector equation $\vec{r}(u, v) = \langle u \cos v, u \sin v, v \rangle$ over the region $0 \le u \le 1$ and $0 \le v \le \pi$

$$S = \iint_R \|\vec{r}_u \times \vec{r}_v\| \, \mathrm{d}A$$

(%i21) `kill(labels,x,y,z,u,v)$`

(%i1) `ldisplay(S:[u*cos(v),u*sin(v),v])$`

$$S = [u \cos(v), u \sin(v), v] \tag{\%t1}$$

(%i2) `ldisplay(N:trigsimp(mycross(diff(S,u),diff(S,v))))$`

$$N = [\sin(v), -\cos(v), u] \tag{\%t2}$$

(%i3) `ldisplay(\|N\|:trigsimp(norm(N)))$`

$$|N| = \sqrt{u^2+1} \tag{\%t3}$$

(%i4) `ldisplay(n:trigsimp(normalize(N)))$`

$$n = \left[\frac{\sin(v)}{\sqrt{u^2+1}}, -\frac{\cos(v)}{\sqrt{u^2+1}}, \frac{u}{\sqrt{u^2+1}}\right] \tag{\%t4}$$

(%i5) `I:'integrate('integrate(\|N\|,u,0,1),v,0,π)$`

(%i6) `ldisplay(I=box(ev(I,integrate)))$`

$$\pi \int_0^1 \sqrt{u^2+1} du = \left(\frac{\pi\left(\operatorname{asinh}(1)+\sqrt{2}\right)}{2}\right) \tag{\%t6}$$

(%i7) `ldisplay(I=box(ev(I,integrate,numer)))$`

$$\pi \int_0^1 \sqrt{u^2+1} du = (3.6059) \tag{\%t7}$$

9 Surface Area of a Parametric Surface

Based on Mathispower4u Video Ex: Surface Area of a Parametric Surface (Surface Integral)

(%i8) `kill(labels,x,y,z,u,v)$`

Find the area of the surface of the cone with the vector equation $\vec{r}(u,v) = \langle u\,\cos v, u\,\sin v, u\rangle$ with $0 \leq u \leq 2$ and $0 \leq v \leq 2\,\pi$

$$\iint_S \mathrm{d}s = \iint_R \|\vec{r}_u \times \vec{r}_v\|\,\mathrm{d}A$$

(%i1) `ldisplay(S:[u*cos(v),u*sin(v),u])$`

$$S = [u\,\cos(v), u\,\sin(v), u] \tag{\%t1}$$

(%i2) `ldisplay(N:trigsimp(mycross(diff(S,u),diff(S,v))))$`

$$N = [-u\,\cos(v), -u\,\sin(v), u] \tag{\%t2}$$

(%i3) `ldisplay(\|N\|:trigsimp(norm(N)))$`

$$|N| = \sqrt{2}u \tag{\%t3}$$

(%i4) `ldisplay(n:trigsimp(normalize(N)))$`

$$n = \left[-\frac{\cos(v)}{\sqrt{2}}, -\frac{\sin(v)}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right] \tag{\%t4}$$

(%i5) `I:'integrate('integrate(\|N\|,u,0,2),v,0,2*π)$`

(%i6) `ldisplay(I=box(ev(I,integrate)))$`

$$2^{\frac{3}{2}}\pi \int_0^2 u du = \left(2^{\frac{5}{2}}\pi\right) \tag{\%t6}$$

(%i7) `ldisplay(I=box(ev(I,integrate,numer)))$`

$$2^{\frac{3}{2}}\pi \int_0^2 u du = (17.772) \tag{\%t7}$$

10 Surface Integral of a Vector Field

Based on Mathispower4u Video Surface Integral of a Vector Field - Part 1

Based on Mathispower4u Video Surface Integral of a Vector Field - Part 2

(%i8) `kill(labels,x,y,z,r,θ)$`

Oriented upward

$$\iint_S \vec{F}\cdot\vec{n}\,\mathrm{d}s = \iint_R \vec{F}\cdot\langle -g_x(x,y), -g_y(x,y), 1\rangle\,\mathrm{d}A$$

Oriented downward

$$\iint_S \vec{F}\cdot\vec{n}\,\mathrm{d}s = \iint_R \vec{F}\cdot\langle g_x(x,y), g_y(x,y), -1\rangle\,\mathrm{d}A$$

10.1

Determine the flux across the given surface. $\vec{F} = \langle 0, -1, -2\rangle$ across the surface $z = 6 - x - y$ in the first octant. Use a downward orientation.

(%i1) `ldisplay(F:[0,-1,-2])$`

$$F = [0, -1, -2] \tag{\%t1}$$

(%i2) `ldisplay(g:6-x-y)$`

$$g = -y - x + 6 \tag{\%t2}$$

(%i3) `ldisplay(Δ:[diff(g,x),diff(g,y),-1])$`

$$\Delta = [-1, -1, -1] \tag{\%t3}$$

(%i4) `sol:solve(g=0,y);`

$$[y = 6 - x] \tag{sol}$$

(%i5) `I:'integrate('integrate(F.Δ,y,0,6-x),x,0,6)$`

(%i6) `ldisplay(I=box(ev(I,integrate)))$`

$$3\int_0^6 6 - x dx = (54) \tag{\%t6}$$

10.2

Determine the flux across the given surface. $\vec{F} = \langle x, y, z \rangle$ across the surface $z = 9 - x^2 - y^2$ above the xy-plane with an unit normal vector oriented upward.

(%i7) `ldisplay(F:[x,y,z])$`

$$F = [x, y, z] \tag{\%t7}$$

(%i8) `ldisplay(g:9-x^2-y^2)$`

$$g = -y^2 - x^2 + 9 \tag{\%t8}$$

(%i9) `ldisplay(FoS:subst([z=g],F))$`

$$FoS = [x, y, -y^2 - x^2 + 9] \tag{\%t9}$$

(%i10) `ldisplay(Δ:[-diff(g,x),-diff(g,y),1])$`

$$\Delta = [2x, 2y, 1] \tag{\%t10}$$

Calculate xy trace

(%i11) `ζ:[x,y]$`

(%i12) `sol:solve(g=0,y);`

$$[y = -\sqrt{9 - x^2}, y = \sqrt{9 - x^2}] \tag{sol}$$

(%i13) `integrand:FoS.Δ;`

$$y^2 + x^2 + 9 \tag{integrand}$$

(%i14) `I:2*'integrate('integrate(integrand,y,-√(9-x^2),√(9-x^2)),x,0,3)$`

(%i15) `ldisplay(I=box(ev(I,integrate)))$`

$$2\int_0^3 \int_{-\sqrt{9-x^2}}^{\sqrt{9-x^2}} y^2 + x^2 + 9 dy\, dx = \left(\frac{243\pi}{2}\right) \tag{\%t15}$$

Polar coordinates

(%i16) `ξ:[r,θ]$`

(%i17) `Tr:[r*cos(θ),r*sin(θ)]$`

(%i18) `integrand:trigsimp(subst(map("=",ζ,Tr),FoS.Δ));`

$$r^2 + 9 \tag{integrand}$$

(%i19) `I:'integrate('integrate(integrand*r,r,0,3),θ,0,2*π)$`

(%i20) `ldisplay(I=box(ev(I,integrate)))$`

$$2\pi \int_0^3 r\,\left(r^2 + 9\right) dr = \left(\frac{243\pi}{2}\right) \tag{\%t20}$$

11 Evaluate a Surface Integral

Based on Mathispower4u Video Ex: Evaluate a Surface Integral (Basic Explicit Surface - Plane Over Rectangle)

(%i21) `kill(labels,x,y,z)$`

Evaluate $\iint_S x^2\, z\, \mathrm{d}s$ where S is the part of the plane $z = 4 + x + 3\,y$ above the rectangle $[0, 2] \times [0, 3]$

(%i1) `ldisplay(f:x^2*z)$`

$$f = x^2 z \tag{\%t1}$$

(%i2) `ldisplay(g:4+x+3*y)$`

$$g = 3y + x + 4 \tag{\%t2}$$

(%i3) `ldisplay(fog:subst([z=g],f))$`

$$fog = x^2\, (3y + x + 4) \tag{\%t3}$$

(%i4) `√(diff(g,x)^2+diff(g,y)^2+1);`

$$\sqrt{11} \tag{\%o4}$$

(%i5) `I:'integrate('integrate(fog*%,x,0,2),y,0,3)$`

(%i6) `ldisplay(I=box(ev(I,integrate)))$`

$$\sqrt{11} \int_0^3 \int_0^2 x^2\, (3y + x + 4)\, dx\, dy\; = \left(80\sqrt{11}\right) \tag{\%t6}$$

(%i7) `ldisplay(I=box(ev(I,integrate,numer)))$`

$$\sqrt{11} \int_0^3 \int_0^2 x^2\, (3y + x + 4)\, dx\, dy\; = (265.33) \tag{\%t7}$$

12 Evaluate a Surface Integral

Based on Mathispower4u Video Ex: Evaluate a Surface Integral Using Polar Coordinates- Implicit Surface (Cone)

(%i8)
```
kill(labels,x,y,z)$
```

Evaluate $\iint_S x^2 + y^2 - z\,\mathrm{d}s$ where S is the part of the cone $z^2 = x^2 + y^2$ that lies between the planes $z = 2$ and $z = 3$.

(%i1)
```
ζ:[x,y]$
```

(%i2)
```
ldisplay(f:x^2+y^2-z)$
```

$$f = -z + y^2 + x^2 \qquad (\%t2)$$

(%i3)
```
ldisplay(g:√(x^2+y^2))$
```

$$g = \sqrt{y^2 + x^2} \qquad (\%t3)$$

(%i4)
```
ldisplay(fog:subst([z=g],f))$
```

$$fog = -\sqrt{y^2 + x^2} + y^2 + x^2 \qquad (\%t4)$$

(%i5)
```
rootscontract(√(diff(g,x)^2+diff(g,y)^2+1));
```

$$\sqrt{\frac{y^2}{y^2 + x^2} + \frac{x^2}{y^2 + x^2} + 1} \qquad (\%o5)$$

(%i6)
```
factor(fullratsimp(fog*%));
```

$$-\sqrt{2}\left(\sqrt{y^2 + x^2} - y^2 - x^2\right) \qquad (\%o6)$$

Polar coordinates

(%i8)
```
assume(0≤r)$
assume(0≤θ,θ≤2*π)$
```

(%i9)
```
ξ:[r,θ]$
```

(%i10)
```
Tr:[r*cos(θ),r*sin(θ)]$
```

(%i11)
```
integrand:factor(trigsimp(subst(map("=",ζ,Tr),%th(5))));
```

$$\sqrt{2}\,(r-1)\,r \qquad (\text{integrand})$$

(%i12)
```
I:'integrate('integrate(integrand*r,r,2,3),θ,0,2*π)$
```

(%i13)
```
ldisplay(I=box(ev(I,integrate)))$
```

$$2^{\frac{3}{2}}\pi \int_2^3 (r-1)\,r^2 dr = \boxed{\frac{119\pi}{3\sqrt{2}}} \qquad (\%t13)$$

(%i14) `ldisplay(I=box(ev(I,integrate,numer)))$`

$$2^{\frac{3}{2}}\pi\int_{2}^{3}(r-1)\,r^2dr\;=(88.117) \tag{\%t14}$$

(%i15)
```
wxdraw3d(title="Cone",view=[75,26],zrange=[1,4],
proportional_axes=xy,
x_voxel=20,y_voxel=20,z_voxel=20,
implicit(z^2=x^2+y^2,x,-4,4,y,-4,4,z,1,4),
color=red,surface_hide=true,
implicit(z=2,x,-4,4,y,-4,4,z,0,4),
implicit(z=3,x,-4,4,y,-4,4,z,0,4))$
```



(%t15)

13 Evaluate a Flux Integral with Surface Given Explicitly

Based on Mathispower4u Video Ex: Evaluate a Flux Integral with Surface Given Explicitly

(%i16) `kill(labels,x,y,z)$`

Find the flux of the vector field $\vec{F} = \langle y, -z, x\rangle$ across the part of the plane $z = 3+4\,x+y$ above the rectangle $[0,5] \times [0,4]$ with upwards orientation.

(%i1) `ζ:[x,y,z]$`

(%i2) `ldisplay(F:[y,-z,x])$`

$$F = [y, -z, x] \tag{\%t2}$$

3D Direction field

(%i4)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,0,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i6)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i7)
```
wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$
```



(%t7)

(%i8) `ldisplay(g:3+4*x+y)$`

$$g = y + 4x + 3 \tag{\%t8}$$

(%i9) `ldisplay(Δ:[-diff(g,x),-diff(g,y),1])$`

$$\mathit{\Delta} = [-4, -1, 1] \tag{\%t9}$$

Calculate $\iint_S \vec{F} \cdot \mathrm{d}\vec{s} = \iint_S \vec{F} \cdot \vec{N}\, \mathrm{d}s$

(%i10) `ldisplay(Fog:subst([z=g],F))$`

$$Fog = [y, -y - 4x - 3, x] \tag{\%t10}$$

(%i11) `I:'integrate('integrate(Fog.Δ,x,0,5),y,0,4)$`

(%i12) `ldisplay(I=box(ev(I,integrate)))$`

$$\int_0^4 \int_0^5 -3y + 5x + 3 dx\, dy = \boxed{190} \tag{\%t12}$$

14 Evaluate a Flux Integral with Surface Given Parametrically

Based on Mathispower4u Video Ex: Evaluate a Flux Integral with Surface Given Parametrically (helicoid)

(%i13)
```
kill(labels,x,y,z,u,v)$
```

Evaluate $\iint_S \vec{F} \cdot d\vec{s}$ where $\vec{F} = \langle y, -x, z^3 \rangle$ and S is the helicoid with vector equation $\vec{r}(u,v) = \langle u\cos v, u\sin v, v \rangle$ with $0 \le u \le 2$ and $0 \le v \le \pi$ with upward orientation.

(%i1)
```
ζ:[x,y,z]$
```

(%i2)
```
ldisplay(F:[y,-x,z^3])$
```

$$F = [y, -x, z^3] \qquad (\%t2)$$

3D Direction field

(%i4)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,0,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i6)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i7) `wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$`



(%t7)

(%i8) ξ:[u,v]$

(%i9) ldisplay(S:[u*cos(v),u*sin(v),v])$

$$S = [u\ \cos(v), u\ \sin(v), v] \tag{\%t9}$$

(%i10) wxdraw3d(title="Helicoid",view=[60,30],ztics=5,
xu_grid=50,yv_grid=50,proportional_axes=xy,
color=green,
apply(parametric_surface,append(S,[u,0,2,v,0,π])))$



(%t10)

(%i11) wxdraw3d(title="Helicoid",view=[0,0],ztics=5,
xu_grid=100,yv_grid=100,proportional_axes=xy,
color=green,
apply(parametric_surface,append(S,[u,0,2,v,0,π])))$



(%t11)

Calculate $\iint_S \vec{F}\cdot \mathrm{d}\vec{s} = \iint_S \vec{F}\cdot\vec{N}\,\mathrm{d}s$

(%i12) `ldisplay(FoS:subst(map("=",ζ,S),F))$`

$$FoS = [u\,\sin(v), -u\,\cos(v), v^3] \tag{\%t12}$$

(%i13) `ldisplay(N:trigsimp(mycross(diff(S,u),diff(S,v))))$`

$$N = [\sin(v), -\cos(v), u] \tag{\%t13}$$

(%i14) `integrand:trigsimp(FoS.N);`

$$u\,v^3 + u \tag{integrand}$$

(%i15) `I:'integrate('integrate(integrand,u,0,2),v,0,π)$`

(%i16) `ldisplay(I=box(ev(I,integrate,expand)))$`

$$\int_0^{\pi}\int_0^2 u\,v^3 + u\,du\,dv = \boxed{\frac{\pi^4}{2} + 2\pi} \tag{\%t16}$$

(%i17) `ldisplay(I=box(ev(I,integrate,numer)))$`

$$\int_0^{\pi}\int_0^2 u\,v^3 + u\,du\,dv = \boxed{54.988} \tag{\%t17}$$

15 Using a Flux Integral to Determine a Mass Flow Rate

Based on Mathispower4u Video Ex: Using a Flux Integral to Determine a Mass Flow Rate

(%i18)
```
kill(labels,x,y,z,u,v)$
```

A fluid has density $800\, kg/m^3$ and flows with velocity $\vec{v} = x\,\hat{i} + y\,\hat{j} + z\,\hat{k}$ where x, y and z are measured in meters and the components of $\vec{v}$ are measured in meters per second. Find the rate of flow outward through the part of the paraboloid $z = 16 - x^2 - y^2$ that lies above the xy-plane.

(%i1)
```
ζ:[x,y,z]$
```

(%i2)
```
ldisplay(F:[x,y,z])$
```

$$F = [x, y, z] \qquad (\%t2)$$

3D Direction field

(%i4)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,-2,2))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i6)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i7) `wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$`



(%t7)

(%i8) `wxdraw3d(title="Paraboloid",view=[60,30],ztics=5, proportional_axes=xy,surface_hide=true, x_voxel=20,y_voxel=20,enhanced3d=true, implicit(z=16-x^2-y^2,x,-4,4,y,-4,4,z,0,16))$`



(%t8)

(%i9) `wxdraw3d(title="Paraboloid",view=[0,0],ztics=5, proportional_axes=xy,surface_hide=true, x_voxel=20,y_voxel=20,enhanced3d=true, implicit(z=16-x^2-y^2,x,-4,4,y,-4,4,z,0,16))$`



(%t9)

(%i10) `ldisplay(ρ:800)$`

$$\rho = 800 \quad (\%t10)$$

(%i11) `ldisplay(g:16-x^2-y^2)$`

$$g = -y^2 - x^2 + 16 \quad (\%t11)$$

(%i12) `ldisplay(Fog:subst([z=g],F))$`

$$Fog = [x, y, -y^2 - x^2 + 16] \quad (\%t12)$$

(%i13) `ldisplay(Δ:[-diff(g,x),-diff(g,y),1])$`

$$\Delta = [2x, 2y, 1] \quad (\%t13)$$

(%i14) `integrand:ρ*(Fog.Δ);`

$$800\left(y^2 + x^2 + 16\right) \quad (\text{integrand})$$

Polar coordinates

(%i16) `ζ:[x,y]$`
`ξ:[r,θ]$`

(%i17) `Tr:[r*sin(θ),r*cos(θ)]$`

(%i18) `integrand:factor(trigsimp(subst(map("=",ζ,Tr),integrand)));`

$$800\left(r^2 + 16\right) \quad (\text{integrand})$$

(%i19) `I:'integrate('integrate(integrand*r,r,0,4),θ,0,2*π)$`

(%i20) `ldisplay(I=box(ev(I,integrate,expand)))$`

$$1600\pi \int_0^4 r\,\left(r^2 + 16\right) dr = (307200\pi) \quad (\%t20)$$

(%i21) `ldisplay(I=box(ev(I,integrate,numer)))$`

$$1600\pi \int_0^4 r\,\left(r^2 + 16\right) dr = \left(9.65110^5\right) \quad (\%t21)$$