Divergence theorem

Reference Wikipedia article Divergence theorem

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2)
```
info:build_info()$info@version;
```

(%o2)

5.38.1

(%i2)
```
reset()$kill(all)$
```

(%i1)
```
derivabbrev:true$
```

(%i2)
```
ratprint:false$
```

(%i3)
```
fpprintprec:5$
```

(%i4)
```
load(linearalgebra)$
```

(%i5)
```
if get('draw,'version)=false then load(draw)$
```

(%i6)
```
wxplot_size:[1024,768]$
```

(%i7)
```
if get('drawdf,'version)=false then load(drawdf)$
```

(%i8)
```
set_draw_defaults(xtics=1,ytics=1,ztics=1,xyplane=0,nticks=100,
xaxis=true,xaxis_type=solid,xaxis_width=3,
yaxis=true,yaxis_type=solid,yaxis_width=3,
zaxis=true,zaxis_type=solid,zaxis_width=3)$
```

(%i9)
```
if get('vect,'version)=false then load(vect)$
```

(%i10)
```
norm(u):=block(ratsimp(radcan(√(u.u))))$
```

(%i11)
```
normalize(v):=block(v/norm(v))$
```

(%i12)
```
angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$
```

(%i13)
```
mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$
```

(%i14)
```
if get('cartan,'version)=false then load(cartan)$
```

(%i15)
```
declare(trigsimp,evfun)$
```

1 Divergence theorem

Based on Michael Penn Video Divergence theorem

Let $\vec{S}$ be a piecewise smooth surface that encloses a solid $E \subseteq \mathbb{R}^3$ and is orientsd outward. Let $\vec{F}$ be a vector field with continuous partial derivatives on an open region containing E then

$$\iiint_E \left(\nabla \cdot \vec{F}\right) \mathrm{d}V = \iint_S \vec{F} \cdot \mathrm{d}\vec{S}$$

(%i16) `kill(labels,x,y,z,P,Q,R)$`

Define the space $\mathbb{R}^3$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^3$

```
(%i4) depends([P,Q,R],ζ)$
(%i5) ldisplay(F:[P,Q,R])$
```

$$\vec{F} = [P, Q, R] \tag{\%t5}$$

$\nabla \times \vec{F} \in \mathbb{R}^3$

```
(%i6) ldisplay(curlF:ev(express(curl(F)),diff))$
```

$$curlF = [R_y - Q_z, P_z - R_x, Q_x - P_y] \tag{\%t6}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

```
(%i7) ldisplay(α:F.cartan_basis)$
```

$$\alpha = R\,dz + Q\,dy + P\,dx \tag{\%t7}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

```
(%i8) ldisplay(dα:edit(ext_diff(α)))$
```

$$d\alpha = (R_y - Q_z)\;dy\,dz + (R_x - P_z)\;dx\,dz + (Q_x - P_y)\;dx\,dy \tag{\%t8}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

```
(%i9) ldisplay(divF:ev(express(div(F)),diff))$
```

$$divF = R_z + Q_y + P_x \tag{\%t9}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

```
(%i10) ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
       F[2]*cartan_basis[3]~cartan_basis[1]+
       F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = P\,dy\,dz - Q\,dx\,dz + R\,dx\,dy \tag{\%t10}$$

$\mathrm{d}\beta \in \mathcal{A}^3(\mathbb{R}^3)$

```
(%i11) ldisplay(dβ:edit(ext_diff(β)))$
```

$$d\beta = (R_z + Q_y + P_x)\;dx\,dy\,dz \tag{\%t11}$$

```
(%i12) dβ/apply("*",cartan_basis);
```

$$R_z + Q_y + P_x \tag{\%o12}$$

End points

```
(%i13) declare([x_0,x_1,y_0,y_1,z_0,z_1],constant)$
```

Top $\vec{S}_1 \in \mathbb{R}^3$

```
(%i17) ldisplay(S_1:[x,y,z_0])$
       ldisplay(N_1:mycross(diff(S_1,x),diff(S_1,y)))$
       ldisplay(T_1:ratsimp(F.N_1))$
       ldisplay(Pb_1:ratsimp(diff(S_1,y)|(diff(S_1,x)|subst(map("=",ζ,S_1),β))))$
```

$$\vec{S}_1 = [x, y, z_0] \tag{\%t14}$$

$$\vec{N}_1 = [0, 0, 1] \tag{\%t15}$$

$$T_1 = R \tag{\%t16}$$

$$Pb_1 = R \tag{\%t17}$$

```
(%i18) ldisplay(I_1:'integrate('integrate(T_1,x,x_0,x_1),y,y_0,y_1))$
```

$$I_1 = (x_1 - x_0)\,(y_1 - y_0)\,R \tag{\%t18}$$

Bottom $\vec{S}_2 \in \mathbb{R}^3$

```
(%i22) ldisplay(S_2:[x,y,z_1])$
       ldisplay(N_2:mycross(diff(S_2,x),diff(S_2,y)))$
       ldisplay(T_2:ratsimp(F.N_2))$
       ldisplay(Pb_2:ratsimp(diff(S_2,y)|(diff(S_2,x)|subst(map("=",ζ,S_2),β))))$
```

$$\vec{S}_2 = [x, y, z_1] \tag{\%t19}$$

$$\vec{N}_2 = [0, 0, 1] \tag{\%t20}$$

$$T_2 = R \tag{\%t21}$$

$$Pb_2 = R \tag{\%t22}$$

```
(%i23) ldisplay(I_2:'integrate('integrate(T_2,x,x_0,x_1),y,y_0,y_1))$
```

$$I_2 = (x_1 - x_0)\,(y_1 - y_0)\,R \tag{\%t23}$$

2 Verifying the Divergence Theorem

Based on Michael Penn Video Verifying the Divergence Theorem

Verify the Divergence Theorem with $\vec{F} = \langle 2x + y, x + z, y - 3z \rangle$ and $\vec{S}$ is the cone $z = \sqrt{x^2 + y^2}$ and top given by $z = 1$.

(%i24) `kill(labels,r,t,x,y,z,ρ,θ,ϕ)$`

Define the space $\mathbb{R}^3$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^3$

(%i4) `ldisplay(F:[2*x+y,x+z,y-3*z])$`

$$\vec{F} = [y + 2x, z + x, y - 3z] \qquad (\%t4)$$

3D Direction field

(%i6)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,0,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i8)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i9)
```
wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$
```



(%t9)

$\nabla \times \vec{F} \in \mathbb{R}^3$

```
(%i10) ldisplay(curlF:ev(express(curl(F)),diff))$
```

$$curlF = [0, 0, 0] \tag{\%t10}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

```
(%i11) ldisplay(α:F.cartan_basis)$
```

$$\alpha = (y - 3z)\ dz + (z + x)\ dy + (y + 2x)\ dx \tag{\%t11}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

```
(%i12) ldisplay(dα:ext_diff(α))$
```

$$d\alpha = 0 \tag{\%t12}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

```
(%i13) ldisplay(divF:ev(express(div(F)),diff))$
```

$$divF = -1 \tag{\%t13}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

```
(%i14) ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
       F[2]*cartan_basis[3]~cartan_basis[1]+
       F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = (y + 2x)\ dy\, dz - (z + x)\ dx\, dz + (y - 3z)\ dx\, dy \tag{\%t14}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

```
(%i15) ldisplay(dβ:edit(ext_diff(β)))$
```

$$d\beta = -dx\, dy\, dz \tag{\%t15}$$

```
(%i16) dβ/apply("*",cartan_basis);
```

$$-1 \tag{\%o16}$$

Spherical coordinates

(%i20)
```
assume(0≤ρ)$
assume(0≤ϕ,ϕ≤π)$
assume(sin(ϕ)≥0)$
assume(0≤θ,θ≤2*π)$
```

(%i21)
```
ξ:[ρ,ϕ,θ]$
```

(%i22)
```
ldisplay(L:[ρ*cos(θ)*sin(ϕ),ρ*sin(θ)*sin(ϕ),ρ*cos(ϕ)])$
```

$$\vec{L} = [\cos(\theta)\rho\,\sin(\phi), \sin(\theta)\rho\,\sin(\phi), \rho\,\cos(\phi)] \quad (\%t22)$$

(%i23)
```
ldisplay(J:jacobian(L,ξ))$
```

$$J = \begin{pmatrix} \cos(\theta)\,\sin(\phi) & \cos(\theta)\rho\,\cos(\phi) & -\sin(\theta)\rho\,\sin(\phi) \\ \sin(\theta)\,\sin(\phi) & \sin(\theta)\rho\,\cos(\phi) & \cos(\theta)\rho\,\sin(\phi) \\ \cos(\phi) & -\rho\,\sin(\phi) & 0 \end{pmatrix} \quad (\%t23)$$

(%i24)
```
ldisplay(lg:trigsimp(transpose(J).J))$
```

$$lg = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \rho^2 & 0 \\ 0 & 0 & \rho^2\sin(\phi)^2 \end{pmatrix} \quad (\%t24)$$

(%i25)
```
ldisplay(Jdet:trigsimp(determinant(J)))$
```

$$Jdet = \rho^2\,\sin(\phi) \quad (\%t25)$$

(%i29)
```
forget(0≤ρ)$
forget(0≤ϕ,ϕ≤π)$
forget(sin(ϕ)≥0)$
forget(0≤θ,θ≤2*π)$
```

Polar coordinates

(%i30) `assume(0≤r)$`

(%i31) `ξ:[r,θ]$`

(%i32) `ldisplay(L:[r*cos(θ),r*sin(θ)])$`

$$\vec{L} = [r\,\cos(\theta), r\,\sin(\theta)] \tag{\%t32}$$

(%i33) `ldisplay(J:jacobian(L,ξ))$`

$$J = \begin{pmatrix} \cos(\theta) & -r\,\sin(\theta) \\ \sin(\theta) & r\,\cos(\theta) \end{pmatrix} \tag{\%t33}$$

(%i34) `ldisplay(lg:trigsimp(transpose(J).J))$`

$$lg = \begin{pmatrix} 1 & 0 \\ 0 & r^2 \end{pmatrix} \tag{\%t34}$$

(%i35) `ldisplay(Jdet:trigsimp(determinant(J)))$`

$$Jdet = r \tag{\%t35}$$

(%i36) `forget(0≤r)$`

Surface $\vec{S} \in \mathbb{R}^3$

(%i38)
```
assume(0≤ρ)$
assume(0≤r)$
```

(%i39)
```
ldisplay(S_1:[x,y,√(x^2+y^2)])$
```

$$\vec{S}_1 = [x, y, \sqrt{y^2 + x^2}] \quad (\%t39)$$

(%i40)
```
ldisplay(S_1:1/√(2)*[ρ*cos(θ),ρ*sin(θ),ρ])$
```

$$\vec{S}_1 = \left[\frac{\cos(\theta)\rho}{\sqrt{2}}, \frac{\sin(\theta)\rho}{\sqrt{2}}, \frac{\rho}{\sqrt{2}}\right] \quad (\%t40)$$

(%i41)
```
ldisplay(S_2:[r*cos(t),r*sin(t),1])$
```

$$\vec{S}_2 = [r\ \cos(t), r\ \sin(t), 1] \quad (\%t41)$$

(%i42)
```
wxdraw3d(xu_grid=50,yv_grid=50,proportional_axes=xyz,zrange=[0,1],
apply(parametric_surface,append(S_1,[ρ,0,√(2),θ,0,2*π])),
color=red,apply(parametric_surface,append(S_2,[r,0,1,t,0,2*π])))$
```



(%t42)

Normal $\vec{N}_1 \in \mathbb{R}^3$

(%i43) `ldisplay(N_1:trigsimp(mycross(diff(S_1,θ),diff(S_1,ρ))))$`

$$\vec{N}_1 = \left[\frac{\cos(\theta)\rho}{2}, \frac{\sin(\theta)\rho}{2}, -\frac{\rho}{2}\right] \tag{\%t43}$$

(%i44) `ldisplay(n_1:scanmap(trigsimp,normalize(N_1)))$`

$$\hat{n}_1 = \left[\frac{\cos(\theta)}{\sqrt{2}}, \frac{\sin(\theta)}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right] \tag{\%t44}$$

Hence $\hat{n}_1 = \frac{1}{\rho}\langle x, y, -z\rangle$

$\vec{F} \circ \vec{S}_1 \in \mathbb{R}^3$

(%i45) `ldisplay(FoS_1:trigsimp(subst(map("=",ζ,S_1),F)))$`

$$FoS_1 = \left[\frac{(\sin(\theta) + 2\cos(\theta))\,\rho}{\sqrt{2}}, \frac{\left(\sqrt{2}\,\cos(\theta) + \sqrt{2}\right)\rho}{2}, \frac{\left(\sqrt{2}\,\sin(\theta) - 3\sqrt{2}\right)\rho}{2}\right] \tag{\%t45}$$

$\left(\vec{F} \circ \vec{S}_1\right) \cdot \vec{N}_1 \in \mathbb{R}$

(%i46) `ldisplay(T_1:trigsimp(FoS_1.N_1))$`

$$T_1 = \frac{\left(2\cos(\theta)\,\sin(\theta) + 2\cos(\theta)^2 + 3\right)\,\rho^2}{2^{\frac{3}{2}}} \tag{\%t46}$$

Pullback $\vec{S}_1^* \beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i47) `ldisplay(P_1:trigsimp(diff(S_1,ρ)|(diff(S_1,θ)|subst(map("=",ζ,S_1),β))))$`

$$P_1 = \frac{\left(2\cos(\theta)\,\sin(\theta) + 2\cos(\theta)^2 + 3\right)\,\rho^2}{2^{\frac{3}{2}}} \tag{\%t47}$$

Flux through $\vec{S}_1$

(%i48) `I_1:'integrate('integrate(T_1,θ,0,2*π),ρ,0,√(2))$`

(%i49) `ldisplay(I_1=box(ev(I_1,integrate)))$`

$$\frac{\int_0^{2\pi} 2\cos(\theta)\,\sin(\theta) + 2\cos(\theta)^2 + 3 d\theta\ \int_0^{\sqrt{2}} \rho^2 d\rho}{2^{\frac{3}{2}}} = \left(\frac{8\pi}{3}\right) \tag{\%t49}$$

Normal $\vec{N}_2 \in \mathbb{R}^3$

```
(%i50) ldisplay(N_2:trigsimp(mycross(diff(S_2,r),diff(S_2,t))))$
```

$$\vec{N}_2 = [0, 0, r] \tag{\%t50}$$

```
(%i51) ldisplay(n_2:scanmap(trigsimp,normalize(N_2)))$
```

$$\hat{n}_2 = [0, 0, 1] \tag{\%t51}$$

$\vec{F} \circ \vec{S}_2 \in \mathbb{R}^3$

```
(%i52) ldisplay(FoS_2:trigsimp(subst(map("=",ζ,S_2),F)))$
```

$$FoS_2 = [r\,\sin(t) + 2r\,\cos(t), r\,\cos(t) + 1, r\,\sin(t) - 3] \tag{\%t52}$$

$\left(\vec{F} \circ \vec{S}_2\right) \cdot \vec{N}_2 \in \mathbb{R}$

```
(%i53) ldisplay(T_2:trigsimp(FoS_2.N_2))$
```

$$T_2 = r^2\,\sin(t) - 3r \tag{\%t53}$$

Pullback $\vec{S}_2^*\beta \in \mathcal{A}^2(\mathbb{R}^3)$

```
(%i54) ldisplay(P_2:trigsimp(diff(S_2,t)|(diff(S_2,r)|subst(map("=",ζ,S_2),β))))$
```

$$P_2 = r^2\,\sin(t) - 3r \tag{\%t54}$$

Flux through $\vec{S}_2$

```
(%i55) I_2:'integrate('integrate(T_2,r,0,1),t,0,2*π)$
(%i56) ldisplay(I_2=box(ev(I_2,integrate)))$
```

$$\int_0^{2\pi}\int_0^1 r^2\,\sin(t) - 3r\,dr\,dt = \boxed{-3\pi} \tag{\%t56}$$

Total flux through $\vec{S}$

```
(%i57) ldisplay(I_1+I_2=box(ev(I_1+I_2,integrate)))$
```

$$\frac{\int_0^{2\pi} 2\cos(\theta)\,\sin(\theta) + 2\cos(\theta)^2 + 3d\theta\;\int_0^{\sqrt{2}}\rho^2 d\rho}{2^{\frac{3}{2}}} + \int_0^{2\pi}\int_0^1 r^2\,\sin(t) - 3r\,dr\,dt = \boxed{-\frac{\pi}{3}} \tag{\%t57}$$

Use the divergence theorem

Volume of the cone

(%i58) `ldisplay(V:π/3)$`

$$V = \frac{\pi}{3} \qquad (\%t58)$$

Triple integral

(%i59) `box(divF*V);`

$$-\frac{\pi}{3} \qquad (\%o59)$$

Clean up

(%i61)
```
forget(0≤ρ)$
forget(0≤r)$
```

3 When the divergence theorem doesn't apply

Based on Michael Penn Video When the divergence theorem doesn't apply

(%i62) `kill(labels,r,t,x,y,z,ρ,θ,ϕ)$`

Define the space $\mathbb{R}^3$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Parameters

(%i4) `assume(a>0)$`

(%i5) `declare(a,constant)$`

(%i6) `params:[a=2]$`

Vector field $\vec{F} \in \mathbb{R}^3$

(%i7)
```
ldisplay(F:subst([r=√(x^2+y^2+z^2)],(1/r^3)*[x,y,z]))$
```

$$\vec{F} = \left[\frac{x}{(z^2+y^2+x^2)^{\frac{3}{2}}}, \frac{y}{(z^2+y^2+x^2)^{\frac{3}{2}}}, \frac{z}{(z^2+y^2+x^2)^{\frac{3}{2}}}\right] \qquad (\%t7)$$

3D Direction field

(%i9)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,1,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i11)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i12)
```
wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$
```



(%t12)

$\nabla \times \vec{F} \in \mathbb{R}^3$

(%i13) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [0, 0, 0] \tag{\%t13}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i14) `ldisplay(α:F.cartan_basis)$`

$$\alpha = \frac{z\ dz}{(z^2+y^2+x^2)^{\frac{3}{2}}} + \frac{y\ dy}{(z^2+y^2+x^2)^{\frac{3}{2}}} + \frac{x\ dx}{(z^2+y^2+x^2)^{\frac{3}{2}}} \tag{\%t14}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i15) `ldisplay(dα:ext_diff(α))$`

$$d\alpha = 0 \tag{\%t15}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i16) `ldisplay(divF:ev(express(div(F)),diff,ratsimp))$`

$$divF = 0 \tag{\%t16}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i17)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = \frac{x\ dy\ dz}{(z^2+y^2+x^2)^{\frac{3}{2}}} - \frac{y\ dx\ dz}{(z^2+y^2+x^2)^{\frac{3}{2}}} + \frac{z\ dx\ dy}{(z^2+y^2+x^2)^{\frac{3}{2}}} \tag{\%t17}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i18) `ldisplay(dβ:ratsimp(edit(ext_diff(β))))$`

$$d\beta = 0 \tag{\%t18}$$

Case 1: $(0,0,0) \in \vec{S}$ The integral does on converge.

Case 2: $(0,0,0) \notin \vec{S}$ and $(0,0,0) \in \vec{E}$
Important: Divergence theorem applies with $\nabla \cdot \vec{F} = 0$

Case 3: $(0,0,0) \in \vec{E}$ and $(0,0,0) \notin \vec{S}$
Let $a > 0$ be such that $S_a \subseteq E$. Let B_a be the interior of this sphere. The region $E \setminus B_a$ does not contain $(0,0,0)$

$$\iint_S \vec{F} \cdot \mathrm{d}\vec{S} = \iint_{S_a} \vec{F} \cdot \mathrm{d}\vec{S} =$$

Surface $\vec{S}_a \in \mathbb{R}^3$

(%i19) `ldisplay(S_a:[a*cos(θ)*sin(ϕ),a*sin(θ)*sin(ϕ),a*cos(ϕ)])$`

$$\vec{S}_a = [a\,\cos(\theta)\,\sin(\phi), a\,\sin(\theta)\,\sin(\phi), a\,\cos(\phi)] \qquad (\%t19)$$

(%i20) `wxdraw3d(xu_grid=50,yv_grid=50,proportional_axes=xyz,`
`apply(parametric_surface,append(S_a,[ϕ,0,π,θ,0,2*π]))),params$`



(%t20)

Normal $\vec{N}_a \in \mathbb{R}^3$

```
(%i21) ldisplay(N_a:trigsimp(mycross(diff(S_a,φ),diff(S_a,θ))))$
```

$$\vec{N}_a = [a^2 \cos(\theta)\sin(\phi)^2, a^2 \sin(\theta)\sin(\phi)^2, a^2 \cos(\phi) \sin(\phi)] \tag{\%t21}$$

```
(%i22) ldisplay(n_a:scanmap(trigsimp,normalize(N_a)))$
```

$$\hat{n}_a = [\cos(\theta) \sin(\phi), \sin(\theta) \sin(\phi), \cos(\phi)] \tag{\%t22}$$

Hence $\hat{n}_a = \dfrac{1}{a}\langle x, y, z\rangle$

$\vec{F} \circ \vec{S}_a \in \mathbb{R}^3$

```
(%i23) ldisplay(FoS_a:trigsimp(subst(map("=",ζ,S_a),F)))$
```

$$FoS_a = \left[\frac{\cos(\theta) \sin(\phi)}{a^2}, \frac{\sin(\theta) \sin(\phi)}{a^2}, \frac{\cos(\phi)}{a^2}\right] \tag{\%t23}$$

$\left(\vec{F} \circ \vec{S}_a\right) \cdot \vec{N}_a \in \mathbb{R}$

```
(%i24) ldisplay(T_a:trigsimp(FoS_a.N_a))$
```

$$T_a = \sin(\phi) \tag{\%t24}$$

Pullback $\vec{S}_a^{*}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

```
(%i25) ldisplay(P_a:trigsimp(diff(S_a,θ)|(diff(S_a,φ)|subst(map("=",ζ,S_a),β))))$
```

$$P_a = \sin(\phi) \tag{\%t25}$$

Flux through $\vec{S}_a$

```
(%i26) I_a:'integrate('integrate(T_a,φ,0,π),θ,0,2*π)$
(%i27) ldisplay(I_a=box(ev(I_a,integrate)))$
```

$$2\pi \int_0^{\pi} \sin(\phi) d\phi = (4\pi) \tag{\%t27}$$

Clean up

```
(%i28) forget(a>0)$
```

4 Divergence Theorem Example

Based on Michael Penn Video Divergence Theorem Example

(%i29) `kill(labels,x,y,z,r,θ)$`

Define the space $\mathbb{R}^3$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^3$

(%i4)
```
ldisplay(F:[x^3/3+y*z^2,x^2+y^3/3+x*cos(z),z^2])$
```

$$\vec{F} = \left[y\,z^2 + \frac{x^3}{3}, x\,\cos(z) + \frac{y^3}{3} + x^2, z^2\right] \qquad (\%t4)$$

3D Direction field

(%i6)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,0,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i8)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i9)
```
wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$
```



(%t9)

$\nabla \times \vec{F} \in \mathbb{R}^3$

(%i10) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [x\,\sin(z), 2yz, \cos(z) - z^2 + 2x] \tag{\%t10}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i11) `ldisplay(α:F.cartan_basis)$`

$$\alpha = z^2\,dz + \left(x\,\cos(z) + \frac{y^3}{3} + x^2\right)\,dy + \left(y\,z^2 + \frac{x^3}{3}\right)\,dx \tag{\%t11}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i12) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = x\,\sin(z)\,dy\,dz - 2yz\,dx\,dz + \left(\cos(z) - z^2 + 2x\right)\,dx\,dy \tag{\%t12}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i13) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = 2z + y^2 + x^2 \tag{\%t13}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i14)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = \left(y\,z^2 + \frac{x^3}{3}\right)\,dy\,dz - \left(x\,\cos(z) + \frac{y^3}{3} + x^2\right)\,dx\,dz + z^2\,dx\,dy \tag{\%t14}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i15) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = \left(2z + y^2 + x^2\right)\,dx\,dy\,dz \tag{\%t15}$$

(%i16) `dβ/apply("*",cartan_basis);`

$$2z + y^2 + x^2 \tag{\%o16}$$

Surface $\vec{S} \in \mathbb{R}^3$

(%i17) `assume(0≤r)$`

(%i18) `ldisplay(S_1:[r*cos(θ),r*sin(θ),2])$`

$$\vec{S}_1 = [r\ \cos(\theta), r\ \sin(\theta), 2] \tag{\%t18}$$

(%i19) `ldisplay(S_2:[cos(θ),sin(θ),z])$`

$$\vec{S}_2 = [\cos(\theta), \sin(\theta), z] \tag{\%t19}$$

(%i20) `ldisplay(S_3:[r*cos(θ),r*sin(θ),0])$`

$$\vec{S}_3 = [r\ \cos(\theta), r\ \sin(\theta), 0] \tag{\%t20}$$

(%i21)
```
wxdraw3d(xu_grid=50,yv_grid=50,proportional_axes=xyz,
apply(parametric_surface,append(S_2,[θ,0,2*π,z,0,2])),
color=green,apply(parametric_surface,append(S_1,[r,0,1,θ,0,2*π])),
color=cyan,apply(parametric_surface,append(S_3,[r,0,1,θ,0,2*π])))$
```



(%t21)

Normal $\vec{N}_1 \in \mathbb{R}^3$

(%i22) `ldisplay(N_1:trigsimp(mycross(diff(S_1,r),diff(S_1,θ))))$`

$$\vec{N}_1 = [0, 0, r] \tag{\%t22}$$

(%i23) `ldisplay(n_1:scanmap(trigsimp,normalize(N_1)))$`

$$\hat{n}_1 = [0, 0, 1] \tag{\%t23}$$

$\vec{F} \circ \vec{S}_1 \in \mathbb{R}^3$

(%i24) `ldisplay(FoS_1:trigsimp(subst(map("=",ζ,S_1),F)))$`

$$FoS_1 = \left[\frac{12r\,\sin(\theta) + r^3\cos(\theta)^3}{3}, \frac{r^3\sin(\theta)^3 + 3r^2\cos(\theta)^2 + 3\cos(2)r\,\cos(\theta)}{3}, 4\right] \tag{\%t24}$$

$\left(\vec{F} \circ \vec{S}_1\right) \cdot \vec{N}_1 \in \mathbb{R}$

(%i25) `ldisplay(T_1:trigsimp(FoS_1.N_1))$`

$$T_1 = 4r \tag{\%t25}$$

Pullback $\vec{S}_1^*\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i26) `ldisplay(P_1:trigsimp(diff(S_1,θ)|(diff(S_1,r)|subst(map("=",ζ,S_1),β))))$`

$$P_1 = 4r \tag{\%t26}$$

Flux through $\vec{S}_1$

(%i27) `I_1:'integrate('integrate(T_1,r,0,1),θ,0,2*π)$`

(%i28) `ldisplay(I_1=box(ev(I_1,integrate)))$`

$$8\pi \int_0^1 r dr = (4\pi) \tag{\%t28}$$

Normal $\vec{N}_2 \in \mathbb{R}^3$

(%i29) `ldisplay(N_2:trigsimp(mycross(diff(S_2,θ),diff(S_2,z))))$`

$$\vec{N}_2 = [\cos(\theta), \sin(\theta), 0] \tag{\%t29}$$

(%i30) `ldisplay(n_2:scanmap(trigsimp,normalize(N_2)))$`

$$\hat{n}_2 = [\cos(\theta), \sin(\theta), 0] \tag{\%t30}$$

Hence $\hat{n}_2 = \langle x, y, 0\rangle$

$\vec{F} \circ \vec{S}_2 \in \mathbb{R}^3$

(%i31) `ldisplay(FoS_2:trigsimp(subst(map("=",ζ,S_2),F)))$`

$$FoS_2 = \left[\frac{3z^2\,\sin(\theta) + \cos(\theta)^3}{3}, \frac{\sin(\theta)^3 + 3\cos(\theta)^2 + 3\cos(z)\,\cos(\theta)}{3}, z^2\right] \tag{\%t31}$$

$\left(\vec{F} \circ \vec{S}_2\right) \cdot \vec{N}_2 \in \mathbb{R}$

(%i32) `ldisplay(T_2:trigsimp(FoS_2.N_2))$`

$$T_2 = \frac{\sin(\theta)^4 + \left(3\cos(\theta)^2 + \left(3\cos(z) + 3z^2\right)\,\cos(\theta)\right)\,\sin(\theta) + \cos(\theta)^4}{3} \tag{\%t32}$$

Pullback $\vec{S}_2^*\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i33) `ldisplay(P_2:trigsimp(diff(S_2,z)|(diff(S_2,θ)|subst(map("=",ζ,S_2),β))))$`

$$P_2 = \frac{\sin(\theta)^4 + \left(3\cos(\theta)^2 + \left(3\cos(z) + 3z^2\right)\,\cos(\theta)\right)\,\sin(\theta) + \cos(\theta)^4}{3} \tag{\%t33}$$

Flux through $\vec{S}_2$

(%i34) `I_2:'integrate('integrate(T_2,z,0,2),θ,0,2*π)$`

(%i35) `ldisplay(I_2=box(ev(I_2,integrate)))$`

$$\frac{\int_0^{2\pi}\int_0^2 \sin(\theta)^4 + \left(3\cos(\theta)^2 + \left(3\cos(z) + 3z^2\right)\,\cos(\theta)\right)\,\sin(\theta) + \cos(\theta)^4 dz\,d\theta}{3} = \boxed{\pi} \tag{\%t35}$$

Normal $\vec{N}_3 \in \mathbb{R}^3$

(%i36) `ldisplay(N_3:trigsimp(mycross(diff(S_3,r),diff(S_3,θ))))$`

$$\vec{N}_3 = [0, 0, r] \tag{\%t36}$$

(%i37) `ldisplay(n_3:scanmap(trigsimp,normalize(N_3)))$`

$$\hat{n}_3 = [0, 0, 1] \tag{\%t37}$$

$\vec{F} \circ \vec{S}_3 \in \mathbb{R}^3$

(%i38) `ldisplay(FoS_3:trigsimp(subst(map("=",ζ,S_3),F)))$`

$$FoS_3 = \left[\frac{r^3 \cos(\theta)^3}{3}, \frac{r^3 \sin(\theta)^3 + 3r^2 \cos(\theta)^2 + 3r\, \cos(\theta)}{3}, 0\right] \tag{\%t38}$$

$\left(\vec{F} \circ \vec{S}_3\right) \cdot \vec{N}_3 \in \mathbb{R}$

(%i39) `ldisplay(T_3:trigsimp(FoS_3.N_3))$`

$$T_3 = 0 \tag{\%t39}$$

Pullback $\vec{S}_3^* \beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i40) `ldisplay(P_3:trigsimp(diff(S_3,θ)|(diff(S_3,r)|subst(map("=",ζ,S_3),β))))$`

$$P_3 = 0 \tag{\%t40}$$

Total flux through $\vec{S}$

(%i41) `ldisplay(I=box(ev(I_1+I_2,integrate)))$`

$$I = (5\pi) \tag{\%t41}$$

Use the divergence theorem

Volume $\vec{E}$

(%i42) `ldisplay(E:[r*cos(θ),r*sin(θ),z])$`

$$\vec{E} = [r\,\cos(\theta), r\,\sin(\theta), z] \tag{\%t42}$$

$\left(\nabla \cdot \vec{F}\right) \circ \vec{E}$

(%i43) `divFoE:trigsimp(subst(map("=",ζ,E),divF));`

$$2z + r^2 \tag{divFoE}$$

Pullback $\vec{E}^{*}\,\mathrm{d}\beta \in \mathcal{A}^3(\mathbb{R}^3)$

(%i44) `trigsimp(diff(E,z)|(diff(E,θ)|(diff(E,r)|subst(map("=",ζ,E),dβ))));`

$$2rz + r^3 \tag{\%o44}$$

Triple integral

(%i45) `I:'integrate('integrate('integrate(divFoE*r,z,0,2),r,0,1),θ,0,2*π)$`

(%i46) `ldisplay(I=box(ev(I,integrate)))$`

$$2\pi \int_0^1 r \int_0^2 2z + r^2 dz\, dr\; = (5\pi) \tag{\%t46}$$

Clean up

(%i47) `forget(0≤r)$`