Chapter 2 curves and frames

Lecture Notes for Differential Geometry
James S. Cook
Liberty University
Department of Mathematics
Summer 2015

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

```
(%i2) reset()$kill(all)$
(%i1) derivabbrev:true$
(%i2) ratprint:false$
(%i3) fpprintprec:5$
(%i4) load(linearalgebra)$
(%i5) if get('draw,'version)=false then load(draw)$
(%i6) wxplot_size:[1024,768]$
(%i7) if get('drawdf,'version)=false then load(drawdf)$
(%i8) set_draw_defaults(xtics=1,ytics=1,ztics=1,xyplane=0,nticks=100,
      xaxis=true,xaxis_type=solid,xaxis_width=3,
      yaxis=true,yaxis_type=solid,yaxis_width=3,
      zaxis=true,zaxis_type=solid,zaxis_width=3,
      background_color=light_gray)$
(%i9) if get('vect,'version)=false then load(vect)$
(%i10) norm(u):=block(ratsimp(radcan(√(u.u))))$
(%i11) normalize(v):=block(v/norm(v))$
(%i12) angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$
(%i13) mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$
(%i14) if get('cartan,'version)=false then load(cartan)$
(%i15) declare(trigsimp,evfun)$
```

1 on distance in three dimensions

2 vectors and frames in three dimensions

Levi-Civita symbol

(%i16) `ϵ(i,j,k):=`$\frac{1}{2}$`*(i-j)*(j-k)*(k-i)$`

(%i18) `v:[v_1,v_2,v_3]$w:[w_1,w_2,w_3]$`

(%i19) `mycross(v,w);`

$$[v_2w_3 - v_3w_2, v_3w_1 - v_1w_3, v_1w_2 - v_2w_1] \quad (\%o19)$$

(%i20) `makelist(sum(sum(ϵ(i,j,k)*v[i]*w[j],i,1,3),j,1,3),k,1,3);`

$$[v_2w_3 - v_3w_2, v_3w_1 - v_1w_3, v_1w_2 - v_2w_1] \quad (\%o20)$$

(%i21) `is(%=%th(2));`

$$\text{true} \quad (\%o21)$$

(%i22) `v.w;`

$$v_3w_3 + v_2w_2 + v_1w_1 \quad (\%o22)$$

(%i23) `sum(v[i]*w[i],i,1,3);`

$$v_3w_3 + v_2w_2 + v_1w_1 \quad (\%o23)$$

(%i24) `is(%=%th(2));`

$$\text{true} \quad (\%o24)$$

(%i25) `expand(norm(mycross(v,w))`^{2}`);`

$${v_2}^2{w_3}^2 + {v_1}^2{w_3}^2 - 2v_2v_3w_2w_3 - 2v_1v_3w_1w_3 + {v_3}^2{w_2}^2 + {v_1}^2{w_2}^2 - 2v_1v_2w_1w_2 + {v_3}^2{w_1}^2 + {v_2}^2{w_1}^2 \quad (\%o25)$$

(%i26) `expand((v.v)*(w.w)-(v.w)`^{2}`);`

$${v_2}^2{w_3}^2 + {v_1}^2{w_3}^2 - 2v_2v_3w_2w_3 - 2v_1v_3w_1w_3 + {v_3}^2{w_2}^2 + {v_1}^2{w_2}^2 - 2v_1v_2w_1w_2 + {v_3}^2{w_1}^2 + {v_2}^2{w_1}^2 \quad (\%o26)$$

(%i27) `is(%=%th(2));`

$$\text{true} \quad (\%o27)$$

2.1 Example 2.2.7.

Let $p \in \mathbb{R}^3$ then E_1, E_2, E_3 given below form a frame at p

$$E_1 = \frac{1}{\sqrt{3}}\left(\frac{\partial}{\partial x}\bigg|_p + \frac{\partial}{\partial y}\bigg|_p + \frac{\partial}{\partial z}\bigg|_p\right) \quad , \quad E_2 = \frac{1}{\sqrt{2}}\left(\frac{\partial}{\partial x}\bigg|_p - \frac{\partial}{\partial z}\bigg|_p\right) \quad , \quad E_3 = \frac{1}{\sqrt{6}}\left(\frac{\partial}{\partial x}\bigg|_p - 2\,\frac{\partial}{\partial y}\bigg|_p + \frac{\partial}{\partial z}\bigg|_p\right)$$

(%i28) `ldisplay(E_1:1/√(3)*[1,1,1])$`

$$E_1 = \left[\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}\right] \tag{\%t28}$$

(%i29) `ldisplay(E_2:1/√(2)*[1,0,-1])$`

$$E_2 = \left[\frac{1}{\sqrt{2}}, 0, -\frac{1}{\sqrt{2}}\right] \tag{\%t29}$$

(%i30) `ldisplay(E_3:1/√(6)*[1,-2,1])$`

$$E_3 = \left[\frac{1}{\sqrt{6}}, -\frac{2}{\sqrt{6}}, \frac{1}{\sqrt{6}}\right] \tag{\%t30}$$

(%i31) `rootscontract(mycross(E_1,E_2));`

$$\left[-\frac{1}{\sqrt{6}}, \frac{\sqrt{2}}{\sqrt{3}}, -\frac{1}{\sqrt{6}}\right] \tag{\%o31}$$

2.2 Example 2.2.8.

Observe $\{\partial_x, \partial_y, \partial_z\}$ forms the **Cartesian coordinate frame** on $\mathbb{R}^3$. We sometimes denote this frame by the standard notation $\{U_1, U_2, U_3\}$. It is often useful to express a given frame in terms of the **Euclidean frame**. For example, the frame of the preceding example is written as:

(%i32) `U:[U_1,U_2,U_3]$`

(%i33) `ldisplay(E_1:1/√(3)*(U_1+U_2+U_3))$`

$$E_1 = \frac{U_3 + U_2 + U_1}{\sqrt{3}} \tag{\%t33}$$

(%i34) `ldisplay(E_2:1/√(2)*(U_1-U_3))$`

$$E_2 = \frac{U_1 - U_3}{\sqrt{2}} \tag{\%t34}$$

(%i35) `ldisplay(E_3:1/√(6)*(U_1+2*U_2+U_3))$`

$$E_3 = \frac{U_3 + 2U_2 + U_1}{\sqrt{6}} \tag{\%t35}$$

2.3 Example 2.2.9.

The **cylindrical coordinate frame** is given below:
$E_1 = \cos\theta\, U_1 + \sin\theta\, U_2$
$E_2 = -\sin\theta\, U_1 + \cos\theta\, U_2$
$E_3 = U_3$

(%i36) `ldisplay(E_1:cos(θ)*U_1+sin(θ)*U_2)$`

$$E_1 = U_2 \sin(\theta) + U_1 \cos(\theta) \quad (\%t36)$$

(%i37) `ldisplay(E_2:-sin(θ)*U_1+cos(θ)*U_2)$`

$$E_2 = U_2 \cos(\theta) - U_1 \sin(\theta) \quad (\%t37)$$

(%i38) `ldisplay(E_3:U_3)$`

$$E_3 = U_3 \quad (\%t38)$$

I often use the notation $E_1 = \hat{r}$, $E_2 = \hat{\theta}$ and $E_3 = \hat{z}$ in multivariate calculus. This frame is very useful for simplifying calculations with cylindrical symmetry.

2.4 Example 2.2.10.

The **spherical coordinate frame** for the usual spherical coordinates used in third-semester-American calculus is given below:

(%i39) `ldisplay(E_1:cos(θ)*sin(ϕ)*U_1+sin(θ)*sin(ϕ)*U_2+cos(ϕ)*U_3)$`

$$E_1 = U_2 \sin(\theta)\, \sin(\phi) + U_1 \cos(\theta)\, \sin(\phi) + U_3 \cos(\phi) \quad (\%t39)$$

(%i40) `ldisplay(E_2:cos(θ)*cos(ϕ)*U_1+sin(θ)*cos(ϕ)*U_2-sin(ϕ)*U_3)$`

$$E_2 = -U_3 \sin(\phi) + U_2 \sin(\theta)\, \cos(\phi) + U_1 \cos(\theta)\, \cos(\phi) \quad (\%t40)$$

(%i41) `ldisplay(E_3:-sin(θ)*U_1+cos(θ)*U_2)$`

$$E_3 = U_2 \cos(\theta) - U_1 \sin(\theta) \quad (\%t41)$$

I often use the notation $E_1 = \hat{\rho}$, $E_2 = \hat{\phi}$ and $E_3 = \hat{\theta}$ in multivariate calculus. This frame is very useful for simplifying calculations with spherical symmetry.

I should warn the readers of O'neill, he uses a different choice of **spherical coordinates** than we implicitly use in the example above. In fact, the example is based on the formulas:

(%i42) `ξ:[ρ,ϕ,θ]$`

(%i46)
```
assume(0≤ρ)$
assume(0≤ϕ,ϕ≤π)$
assume(0≤sin(ϕ))$
assume(0≤θ,θ≤2*π)$
```

(%i47) `ldisplay(Tr:[ρ*cos(θ)*sin(φ),ρ*sin(θ)*sin(φ),ρ*cos(φ)])$`

$$Tr = [\cos(\theta)\rho\,\sin(\phi), \sin(\theta)\rho\,\sin(\phi), \rho\,\cos(\phi)] \quad (\%t47)$$

(%i48) `scalefactors(append([Tr],ξ))$`

(%i49) `sf;`

$$[1, \rho, \rho\,\sin(\phi)] \quad (\%o49)$$

(%i50) `sfprod;`

$$\rho^2\,\sin(\phi) \quad (\%o50)$$

(%i51) `J:jacobian(Tr,ξ);`

$$\begin{pmatrix} \cos(\theta)\,\sin(\phi) & \cos(\theta)\rho\,\cos(\phi) & -\sin(\theta)\rho\,\sin(\phi) \\ \sin(\theta)\,\sin(\phi) & \sin(\theta)\rho\,\cos(\phi) & \cos(\theta)\rho\,\sin(\phi) \\ \cos(\phi) & -\rho\,\sin(\phi) & 0 \end{pmatrix} \quad (J)$$

(%i52) `lg:trigsimp(transpose(J).J);`

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & \rho^2 & 0 \\ 0 & 0 & \rho^2\sin(\phi)^2 \end{pmatrix} \quad (lg)$$

These coordinates envision ϕ being zero on the positive z-axis then sweeping down to π on the negative z-axis. In contrast, see Figure 2.20 on page 86, O'neill prefers to work with ϕ which is zero on the xy-plane then sweeps up or down to $\pm\pi/2$.

2.5 Definition 2.2.11.

Attitude matrix of a frame

(%i53)
```
A:matrix(
makelist(coeff(E_1,k),k,U),
makelist(coeff(E_2,k),k,U),
makelist(coeff(E_3,k),k,U));
```

$$\begin{pmatrix} \cos(\theta)\,\sin(\phi) & \sin(\theta)\,\sin(\phi) & \cos(\phi) \\ \cos(\theta)\,\cos(\phi) & \sin(\theta)\,\cos(\phi) & -\sin(\phi) \\ -\sin(\theta) & \cos(\theta) & 0 \end{pmatrix} \quad (A)$$

(%i54) `trigsimp(transpose(A).A);`

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \quad (\%o54)$$

(%i55) `is(trigsimp(mycross(A[1],A[2]))=A[3]);`

$$\text{true} \quad (\%o55)$$

2.6 Example 2.2.13.

Following Example 2.2.7.

(%i56) `ldisplay(E_1:1/√(3)*[1,1,1])$`

$$E_1 = \left[\frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}, \frac{1}{\sqrt{3}}\right] \tag{\%t56}$$

(%i57) `ldisplay(E_2:1/√(2)*[1,0,-1])$`

$$E_2 = \left[\frac{1}{\sqrt{2}}, 0, -\frac{1}{\sqrt{2}}\right] \tag{\%t57}$$

(%i58) `ldisplay(E_3:1/√(6)*[1,-2,1])$`

$$E_3 = \left[\frac{1}{\sqrt{6}}, -\frac{2}{\sqrt{6}}, \frac{1}{\sqrt{6}}\right] \tag{\%t58}$$

(%i59) `A:matrix(E_1,E_2,E_3);`

$$\begin{pmatrix} \frac{1}{\sqrt{3}} & \frac{1}{\sqrt{3}} & \frac{1}{\sqrt{3}} \\ \frac{1}{\sqrt{2}} & 0 & -\frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{6}} & -\frac{2}{\sqrt{6}} & \frac{1}{\sqrt{6}} \end{pmatrix} \tag{A}$$

(%i60) `transpose(A).A;`

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%o60}$$

2.7 Example 2.2.14.

Following Example 2.2.8., the attitude of the **Cartesian frame** is the identity matrix:

(%i61) `ldisplay(E_1:[1,0,0])$`

$$E_1 = [1, 0, 0] \tag{\%t61}$$

(%i62) `ldisplay(E_2:[0,1,0])$`

$$E_2 = [0, 1, 0] \tag{\%t62}$$

(%i63) `ldisplay(E_3:[0,0,1])$`

$$E_3 = [0, 0, 1] \tag{\%t63}$$

(%i64) `is(mycross(E_1,E_2)=E_3);`

$$\text{true} \tag{\%o64}$$

(%i65) `A:matrix(E_1,E_2,E_3);`

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{A}$$

(%i66) `transpose(A).A;`

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%o66}$$

(%i67) `is(mycross(A[1],A[2])=A[3]);`

$$\mathrm{true} \tag{\%o67}$$

2.8 Example 2.2.15.

Following Example 2.2.9, the attitude of the **cylindrical coordinate frame** is:

(%i68) `ldisplay(E_1:[cos(θ),sin(θ),0])$`

$$E_1 = [\cos(\theta), \sin(\theta), 0] \tag{\%t68}$$

(%i69) `ldisplay(E_2:[-sin(θ),cos(θ),0])$`

$$E_2 = [-\sin(\theta), \cos(\theta), 0] \tag{\%t69}$$

(%i70) `ldisplay(E_3:[0,0,1])$`

$$E_3 = [0, 0, 1] \tag{\%t70}$$

(%i71) `is(trigsimp(mycross(E_1,E_2)=E_3));`

$$\mathrm{true} \tag{\%o71}$$

(%i72) `A:matrix(E_1,E_2,E_3);`

$$\begin{pmatrix} \cos(\theta) & \sin(\theta) & 0 \\ -\sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{A}$$

(%i73) `trigsimp(transpose(A).A);`

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%o73}$$

(%i74) `is(trigsimp(mycross(A[1],A[2]))=A[3]);`

$$\mathrm{true} \tag{\%o74}$$

2.9 Example 2.2.16.

Following Example 2.2.9 and 2.2.15, the **cylindrical coordinate frame** has attitude matrix:

(%i75) `ldisplay(E_1:[cos(θ)*sin(ϕ),sin(θ)*sin(ϕ),cos(ϕ)])$`

$$E_1 = [\cos(\theta)\ \sin(\phi), \sin(\theta)\ \sin(\phi), \cos(\phi)] \tag{\%t75}$$

(%i76) `ldisplay(E_2:[cos(θ)*cos(ϕ),sin(θ)*cos(ϕ),-sin(ϕ)])$`

$$E_2 = [\cos(\theta)\ \cos(\phi), \sin(\theta)\ \cos(\phi), -\sin(\phi)] \tag{\%t76}$$

(%i77) `ldisplay(E_3:[-sin(θ),cos(θ),0])$`

$$E_3 = [-\sin(\theta), \cos(\theta), 0] \tag{\%t77}$$

(%i78) `is(trigsimp(mycross(E_1,E_2)=E_3));`

$$\text{true} \tag{\%o78}$$

(%i79) `A:matrix(E_1,E_2,E_3);`

$$\begin{pmatrix} \cos(\theta)\ \sin(\phi) & \sin(\theta)\ \sin(\phi) & \cos(\phi) \\ \cos(\theta)\ \cos(\phi) & \sin(\theta)\ \cos(\phi) & -\sin(\phi) \\ -\sin(\theta) & \cos(\theta) & 0 \end{pmatrix} \tag{A}$$

(%i80) `trigsimp(transpose(A).A);`

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%o80}$$

(%i81) `is(trigsimp(mycross(A[1],A[2]))=A[3]);`

$$\text{true} \tag{\%o81}$$

(%i82) `init_cartan(ξ)$`

(%i83) `matrix_element_mult:"~"$`

(%i84) `ldisplay(dA:trigsimp(ext_diff(A)))$`

$$dA = \begin{pmatrix} d\phi\ \cos(\theta)\ \cos(\phi) - d\theta\ \sin(\theta)\ \sin(\phi) & d\theta\ \cos(\theta)\ \sin(\phi) + d\phi\ \sin(\theta)\ \cos(\phi) & -d\phi\ \sin(\phi) \\ -d\phi\ \cos(\theta)\ \sin(\phi) - d\theta\ \sin(\theta)\ \cos(\phi) & d\theta\ \cos(\theta)\ \cos(\phi) - d\phi\ \sin(\theta)\ \sin(\phi) & -d\phi\ \cos(\phi) \\ -d\theta\ \cos(\theta) & -d\theta\ \sin(\theta) & 0 \end{pmatrix} \tag{\%t84}$$

(%i85) `ldisplay(ω:trigsimp(dA.transpose(A)))$`

$$\omega = \begin{pmatrix} 0 & d\phi & d\theta\ \sin(\phi) \\ -d\phi & 0 & d\theta\ \cos(\phi) \\ -d\theta\ \sin(\phi) & -d\theta\ \cos(\phi) & 0 \end{pmatrix} \tag{\%t85}$$

(%i86) `matrix_element_mult:"*"$`

3 calculus of vectors fields along curves

3.1 Example 2.3.2.

Let $\alpha = \left(t, t^2, t^3\right)$ for $t \in \mathbb{R}$ and $Y = x^2\partial_x + (y + sin(z))\partial_z$ then identify we have vector field component functions:

$$Y^1 = x^2 \quad , \quad Y^2 = 0 \quad , \quad Y^3 = y + sin(z)$$

which give parametrized components on $\alpha = \left(t, t^2, t^3\right)$ of

$$\left(Y^1 \circ \alpha\right)(t) = t^2 \quad , \quad \left(Y^2 \circ \alpha\right)(t) = 0 \quad , \quad \left(Y^3 \circ \alpha\right)(t) = t^2 + \sin(t^3)$$

(%i87) `kill(labels,t,x,y,z)$`

(%i1) `ζ:[x,y,z]$`

(%i2) `α:[t,t^2,t^3]$`

(%i3) `Y:[x^2,0,y+sin(z)]$`

(%i4) `ldisplay(Yoα:at(Y,map("=",ζ,α)))$`

$$Yo\alpha = [t^2, 0, \sin\left(t^3\right) + t^2] \tag{\%t4}$$

3.2 Example 2.3.4.

Continuing Example 2.3.2., the vector field along α is given by

$$(Y \circ \alpha)(t) = t^2\, U_1 + \left(t^2 + \sin(t^3)\right)\, U_3$$

thus $Y' = 2\,t\,U_1 + \left(2\,t + 3\,t^2\cos(t^3)\right) \in T_{(t,t^2,t^3)}\mathbb{R}^3$

(%i5) `ldisplay(Yoα\':diff(Yoα,t))$`

$$Yo\alpha' = [2t, 0, 3t^2\,\cos\left(t^3\right) + 2t] \tag{\%t5}$$

3.3 Example 2.3.7.

Let $\alpha = \left(t, t^2, t^3\right)$ for $t \in \mathbb{R}$. Then

$$\alpha'(t) = U_1 + 2\,t\,U_2 + 3\,t^2\,U_3 \quad , \quad \alpha''(t) = 2\,U_2 + 6\,t\,U_3$$

where both α' and α'' are in $T_{\alpha(t)}\mathbb{R}^3$.

(%i6) `ldisplay(α:[t,t^2,t^3])$`

$$\alpha = [t, t^2, t^3] \tag{\%t6}$$

(%i7) `ldisplay(α\':diff(α,t))$`

$$\alpha' = [1, 2t, 3t^2] \tag{\%t7}$$

(%i8) `ldisplay(α\'\':diff(α\',t))$`

$$\alpha'' = [0, 2, 6t] \tag{\%t8}$$

4 Frenet Serret frame of a curve

4.1 Example 2.4.4.

Consider the helix defined by $R, m > 0$ and

$$\alpha(s) = (R\cos(k\,s), R\sin(k\,s), m\,k\,s)$$

for $s \in \mathbb{R}$ and $k = 1/\sqrt{R^2 + m^2}$

```
(%i9)  assume(R>0,m>0)$
(%i10) declare([R,m],constant)$
(%i11) paramk:k=1/√(R²+m²)$
(%i12) ldisplay(α:[R*cos(k*s),R*sin(k*s),m*k*s])$
```

$$\alpha = [R\cos(ks), R\sin(ks), mks] \quad (\%t12)$$

Calculate

```
(%i13) ldisplay(T:α\':diff(α,s))$
```

$$T = [-Rk\sin(ks), Rk\cos(ks), mk] \quad (\%t13)$$

```
(%i14) at(trigsimp(norm(α\')),paramk);
```

$$1 \quad (\%o14)$$

It follows $T = \alpha'$. Differentiate α' to obtain:

```
(%i15) ldisplay(T\':α\'\':diff(α\',s))$
```

$$T' = [-R\,k^2\cos(ks), -R\,k^2\sin(ks), 0] \quad (\%t15)$$

```
(%i16) ldisplay(κ:at(trigsimp(norm(α\'\')),paramk))$
```

$$\kappa = \frac{R}{m^2 + R^2} \quad (\%t16)$$

```
(%i17) trigsimp(norm(T\'));
```

$$R\,k^2 \quad (\%o17)$$

```
(%i18) ldisplay(N:[-cos(k*s),-sin(k*s),0])$
```

$$N = [-\cos(ks), -\sin(ks), 0] \quad (\%t18)$$

```
(%i19) ldisplay(B:trigsimp(mycross(T,N)))$
```

$$B = [mk\sin(ks), -mk\cos(ks), Rk] \quad (\%t19)$$

As a quick check on the calculation, notice $\mathbf{B}\cdot\mathbf{N}=0$ and $\mathbf{B}\cdot\mathbf{T}=0$.

```
(%i20) trigsimp(B.N);
```

$$0 \tag{\%o20}$$

```
(%i21) trigsimp(B.T);
```

$$0 \tag{\%o21}$$

Calculate:

```
(%i22) ldisplay(B\':diff(B,s))$
```

$$B' = [m\,k^2\,\cos(ks), m\,k^2\,\sin(ks), 0] \tag{\%t22}$$

thus:

```
(%i23) ldisplay(τ:-at(trigsimp(B\'.N),paramk))$
```

$$\tau = \frac{m}{m^2+R^2} \tag{\%t23}$$

4.2 Example 2.4.7.

If a curve is on a sphere then it is at least as curved as a great circle on the sphere. To see this, consider $\alpha : I \to \mathbb{R}^3$ a unit-speed regular curve on the sphere with center C and radius R. We are given $\|\alpha(s)-C\| = R$ for all $s \in I$.

4.3 Theorem 2.4.8.

Frenet Serret Equations for non-unit speed curves
Let α be a non-linear regular smooth curve with speed $v = \|\alpha'\|$ and $\mathbf{T},\mathbf{N},\mathbf{B},\kappa$ and τ as defined through the unit-speed reparameterization then:

$$\frac{\mathrm{d}\mathbf{T}}{\mathrm{d}t} = v\kappa\mathbf{N}$$

$$\frac{\mathrm{d}\mathbf{N}}{\mathrm{d}t} = -v\kappa\mathbf{T} + v\tau\mathbf{B}$$

$$\frac{\mathrm{d}\mathbf{B}}{\mathrm{d}t} = -v\tau\mathbf{N}$$

Moreover, $\kappa = \frac{1}{v}\|\mathbf{T}'\|$ and $\tau = -\frac{1}{v}\mathbf{B}'\cdot\mathbf{N}$

4.4 Proposition 2.4.9.

Acceleration in terms of curvature and speed
Let α be a non-linear regular smooth curve with speed $v = \|\alpha'\|$ then $\alpha'' = \frac{\mathrm{d}v}{\mathrm{d}t}\mathbf{T} + \kappa\, v^2\mathbf{N}$

4.5 Example 2.4.10.

Suppose $\alpha(t) = \left(t, t^2, t^3\right)$ Calculate the Frenet apparatus or at least try. Other expressions of the frame

(%i24) `ldisplay(`α`:[t,t`^{2}`,t`^{2}`])$`

$$\alpha = [t, t^2, t^2] \tag{\%t24}$$

(%i25) `ldisplay(`α`\':diff(`α`,t))$`

$$\alpha' = [1, 2t, 2t] \tag{\%t25}$$

(%i26) `ldisplay(`α`\'\':diff(`α`\',t))$`

$$\alpha'' = [0, 2, 2] \tag{\%t26}$$

(%i27) `ldisplay(`α`\'\'\':diff(`α`\'\',t))$`

$$\alpha''' = [0, 0, 0] \tag{\%t27}$$

(%i28) `ldisplay(T:normalize(`α`\'))$`

$$T = \left[\frac{1}{\sqrt{8t^2+1}}, \frac{2t}{\sqrt{8t^2+1}}, \frac{2t}{\sqrt{8t^2+1}}\right] \tag{\%t28}$$

(%i29) `ldisplay(T\':factor(diff(T,t)))$`

$$T' = \left[-\frac{8t}{(8t^2+1)^{\frac{3}{2}}}, \frac{2}{(8t^2+1)^{\frac{3}{2}}}, \frac{2}{(8t^2+1)^{\frac{3}{2}}}\right] \tag{\%t29}$$

(%i30) `ldisplay(N:rootscontract(normalize(T\')))$`

$$N = \left[-t\sqrt{\frac{8}{8t^2+1}}, \frac{1}{\sqrt{16t^2+2}}, \frac{1}{\sqrt{16t^2+2}}\right] \tag{\%t30}$$

(%i31) `ldisplay(B:factor(mycross(T,N)))$`

$$B = \left[0, -\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right] \tag{\%t31}$$

(%i32) `ldisplay(S:mycross(`α`\',`α`\'\'))$`

$$S = [0, -2, 2] \tag{\%t32}$$

(%i33) `ldisplay(`κ`:norm(S)/norm(`α`\')`^{3}`)$`

$$\kappa = \frac{2^{\frac{3}{2}}}{(8t^2+1)^{\frac{3}{2}}} \tag{\%t33}$$

(%i34) `ldisplay(τ:(S.α\'\'\')/norm(S)^2)$`

$$\tau = 0 \tag{\%t34}$$

α'' in the Frenet-Serret frame

(%i35) `ldisplay(α\'\':[(α\'\'.T),(α\'\'.N),(α\'\'.B)])$`

$$\alpha'' = \left[\frac{8t}{\sqrt{8t^2+1}}, \frac{4}{\sqrt{16t^2+2}}, 0\right] \tag{\%t35}$$

4.6 Theorem 2.4.11.

Slick formulas for the Frenet apparatus
Let α be a non-linear regular smooth curve with speed $v = \|\alpha'\|$ and $\mathbf{T}, \mathbf{N}, \mathbf{B}, \kappa$ and τ as defined through the unit-speed reparameterization then:

$$\mathbf{T} = \frac{\alpha'}{\|\alpha'\|}, \quad \mathbf{B} = \frac{\alpha' \times \alpha''}{\|\alpha' \times \alpha''\|}, \quad \mathbf{N} = \mathbf{B} \times \mathbf{T},$$

$$\kappa = \frac{\|\alpha' \times \alpha''\|}{\|\alpha'\|^3}, \quad \tau = \frac{(\alpha' \times \alpha'') \cdot \alpha'''}{\|\alpha' \times \alpha''\|^2}$$

5 covariant derivatives

5.1 Example 2.5.2.

If $W(p) = a\,U_1 + b\,U_2 + c\,U_3$ for constants $a, b, c \in \mathbb{R}$ for all $p \in \mathbb{R}^3$ then $W(p + t\,v) = a\,U_1 + b\,U_2 + c\,U_3$ hence $W'(p + t\,v) = 0$ thus $(\nabla_v W)(p) = 0$ for all $p \in \mathbb{R}^3$ hence $\nabla_v W = 0$ for any choice of $V \in \mathfrak{X}(\mathbb{R}^3)$ as the calculation held for arbitrary v at each p.

(%i36) `kill(t,x,v,z)$`

Define the space $\mathbb{R}^3$

(%i37) `ζ:[x,y,z]$`

(%i38) `scalefactors(ζ)$`

(%i39) `init_cartan(ζ)$`

Define frame $\mathbf{U} \in \mathbb{R}^3$

(%i40) `U:[U_1,U_2,U_3]$`

Define point $P = (1, 2, 3) \in \mathbb{R}^3$

(%i41) `declare([a,b,c],constant)$`

(%i42) `P:[a,b,c]$`

(%i43) `v:[v_1,v_2,v_3]$`

(%i44) `ldisplay(W:x*U_1+y*U_2+z*U_3)$`

$$W = U_3 z + U_2 y + U_1 x \tag{\%t44}$$

(%i45) `ldisplay(W:makelist(coeff(W,i),i,U))$`

$$W = [x, y, z] \tag{\%t45}$$

(%i46) `ldisplay(W_P:at(W,map("=",ζ,P)))$`

$$W_P = [a, b, c] \tag{\%t46}$$

(%i47) `ldisplay(W_t:at(W,map("=",ζ,P+t*v)))$`

$$W_t = [t\,v_1 + a, t\,v_2 + b, t\,v_3 + c] \tag{\%t47}$$

5.2 Example 2.5.3.

What about the change of $W = x^2\,U_1 + y\,U_3$ along $v = 2\,U_2 + U_3$ at $p = (1.2.3)$?
Calculate,

$$W(p + t\,v) = W(1 + 2\,t, 2, 3 + t) = (1 + 2\,t)^2\,U_1 + (3 + t)\,U_3$$

thus,

$$W'(p + t\,v) = 4(1 + 2\,t)\,U_1 + U_3 \Rightarrow W'(p + t\,v)(0) = 4\,U_1 + U_3.$$

Therefore,$(\nabla_v W)(1,2,3) = 4\,U_1 + U_3$

(%i48) `kill(t,x,y,z)$`

Define the space $\mathbb{R}^3$

(%i49) `ζ:[x,y,z]$`

(%i50) `scalefactors(ζ)$`

(%i51) `init_cartan(ζ)$`

Define frame $\mathbf{U} \in \mathbb{R}^3$

(%i52) `U:[U_1,U_2,U_3]$`

Define $\mathbf{W} \in \mathfrak{X}(\mathbb{R}^3)$

(%i53) `ldisplay(W:x^2*U_1-2*z^3*U_2+y*U_3)$`

$$W = -2U_2z^3 + U_3y + U_1x^2 \tag{\%t53}$$

(%i54) `ldisplay(W:makelist(coeff(W,k),k,U))$`

$$W = [x^2, -2z^3, y] \tag{\%t54}$$

Define $\mathbf{v} \in \mathfrak{X}(\mathbb{R}^3)$

(%i55) `ldisplay(v:2*U_2-U_1)$`

$$v = 2U_2 - U_1 \tag{\%t55}$$

(%i56) `ldisplay(v:makelist(coeff(v,k),k,U))$`

$$v = [-1, 2, 0] \tag{\%t56}$$

Define $\mathbf{P} \in \mathbb{R}^3$

(%i57) `ldisplay(P:[1,2,3])$`

$$P = [1, 2, 3] \tag{\%t57}$$

Using definition

Calculate $\mathbf{DW} \in \mathfrak{X}(\mathbb{R}^3)$

(%i58) `ldisplay(DW:at(diff(at(W,map("=",ζ,P+t*v)),t),t=0))$`

$$DW = [-2, 0, 2] \tag{\%t58}$$

(%i59) `ldisplay(DW:DW.U)$`

$$DW = 2U_3 - 2U_1 \tag{\%t59}$$

Using grad

(%i60) `gradW:apply('matrix,ev(express(grad(W)),diff));`

$$\begin{pmatrix} 2x & 0 & 0 \\ 0 & 0 & 1 \\ 0 & -6z^2 & 0 \end{pmatrix} \tag{gradW}$$

Calculate DW1 $\in \mathfrak{X}(\mathbb{R}^3)$

(%i61) `ldisplay(DW1:list_matrix_entries(v.at(gradW,map("=",ζ,P))))$`

$$DW1 = [-2, 0, 2] \tag{\%t61}$$

(%i62) `ldisplay(DW1:DW1.U)$`

$$DW1 = 2U_3 - 2U_1 \tag{\%t62}$$

(%i63) `is(DW1=DW);`

$$\text{true} \tag{\%o63}$$

5.3 Proposition 2.5.4.

Coordinate derivative formula for covariant derivative
Let $\mathbf{V}, \mathbf{W} \in \mathfrak{X}(\mathbb{R}^3)$ then

$$\nabla_V \mathbf{W} = \sum_{j=1}^{3} \mathbf{V}\left[W^j\right] U_j$$

(%i64) `ldisplay(DW2:list_matrix_entries(v.gradW))$`

$$DW2 = [-2x, 0, 2] \tag{\%t64}$$

(%i65) `ldisplay(DW2:DW2.U)$`

$$DW2 = 2U_3 - 2U_1 x \tag{\%t65}$$

5.4 Example 2.5.6.

Let $V = x\,U_1 + y^2\,U_2 + z^3\,U_3$ and $W = y\,z\,U_1 + x\,y\,U_3$. Recall our notation U_1, U_2, U_3 masks the fact that these are derivations; $U_1 = \partial_x$, $U_2 = \partial_y$ and $U_3 = \partial_z$ thus:

(%i66) `kill(t,x,v,z)$`

Define the space $\mathbb{R}^3$

(%i67) `ζ:[x,y,z]$`

(%i68) `scalefactors(ζ)$`

(%i69) `init_cartan(ζ)$`

Define frame U $\in \mathbb{R}^3$

(%i70) `U:[U_1,U_2,U_3]$`

Define W $\in \mathfrak{X}(\mathbb{R}^3)$

(%i71) `ldisplay(W:y*z*U_1+x*y*U_3)$`

$$W = U_1 yz + U_3 xy \tag{\%t71}$$

(%i72) `ldisplay(W:makelist(coeff(W,k),k,U))$`

$$W = [yz, 0, xy] \tag{\%t72}$$

Define v $\in \mathfrak{X}(\mathbb{R}^3)$

(%i73) `ldisplay(v:x*U_1+y^2*U_2+z^3*U_3)$`

$$v = U_3 z^3 + U_2 y^2 + U_1 x \tag{\%t73}$$

(%i74) `ldisplay(v:makelist(coeff(v,k),k,U))$`

$$v = [x, y^2, z^3] \tag{\%t74}$$

Calculate ∇W Matrix

(%i75) `gradW:apply('matrix,ev(express(grad(W)),diff));`

$$\begin{pmatrix} 0 & 0 & y \\ z & 0 & x \\ y & 0 & 0 \end{pmatrix} \tag{gradW}$$

Calculate DW1 $\in \mathfrak{X}(\mathbb{R}^3)$

(%i76) `ldisplay(DW:list_matrix_entries(v.gradW))$`

$$DW = [y\,z^3 + y^2 z, 0, x\,y^2 + xy] \tag{\%t76}$$

(%i77) `ldisplay(DW:DW.U)$`

$$DW = U_1 \left(y\,z^3 + y^2 z\right) + U_3 \left(x\,y^2 + xy\right) \tag{\%t77}$$

5.5 Example 2.5.7.

Calculate $\nabla_V V$.

(%i78) `gradV:apply('matrix,ev(express(grad(v)),diff));`

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 2y & 0 \\ 0 & 0 & 3z^2 \end{pmatrix} \tag{gradV}$$

(%i79) `ldisplay(DV:list_matrix_entries(v.gradV))$`

$$DV = [x, 2y^3, 3z^5] \tag{\%t79}$$

(%i80) `ldisplay(DV:DV.U)$`

$$DV = 3U_3z^5 + 2U_2y^3 + U_1x \tag{\%t80}$$

(%i81) `v.ev(express(grad(norm(v))),diff);`

$$\frac{3z^8}{\sqrt{z^6+y^4+x^2}} + \frac{2y^5}{\sqrt{z^6+y^4+x^2}} + \frac{x^2}{\sqrt{z^6+y^4+x^2}} \tag{\%o81}$$

6 frames and connection forms

6.1 Definition 2.6.1.

Connection forms
If E_1, E_2, E_3 is a frame for $\mathbb{R}^3$ then define $\omega_{i\,j}(p) \in \left(T_p\mathbb{R}^3\right)^*$ by

$$\omega_{i\,j}(v) = (\nabla_v\, E_i) \cdot\, E_j(p)$$

for each $v \in\, T_p\mathbb{R}^3$. That is, $\omega_{i\,j}$ is a differential one-form on $\mathbb{R}^3$ defined by the assignment $p \to \omega_{i\,j}(p)$ for each $p \in \mathbb{R}^3$.

6.2 Proposition 2.6.2.

Properties of the covariant derivative on $\mathbb{R}^3$
Let $\{E_1, E_2, E_3\}$ be a frame on $\mathbb{R}^3$ then $\omega_{i\,j} = -\omega_{j\,i}$ and $\nabla_v\, E_i = \sum_{j=1}^{3} \omega_{i\,j}(V)\, E_j$

6.3 Example 2.6.4.

Let $A = \begin{pmatrix} \mathrm{d}x & \mathrm{d}y \\ \mathrm{d}z & z^2\,\mathrm{d}y + y^2\,\mathrm{d}z \end{pmatrix}$ and $B = \begin{pmatrix} \mathrm{d}x + \mathrm{d}y & 0 \\ z^2\,\mathrm{d}y & \mathrm{d}x + \mathrm{d}z \end{pmatrix}$

Define the space $\mathbb{R}^3$

```
(%i82) ζ:[x,y,z]$
(%i83) scalefactors(ζ)$
(%i84) init_cartan(ζ)$
(%i85) matrix_element_mult:"~"$
(%i86) ldisplay(A:matrix([dx,dy],[dz,z^2*dy+y^2*dz]))$
```

$$A = \begin{pmatrix} dx & dy \\ dz & y^2\ dz + z^2\ dy \end{pmatrix} \tag{\%t86}$$

```
(%i87) ldisplay(B:matrix([dx+dy,0],[z^2*dy,dx+dz]))$
```

$$B = \begin{pmatrix} dy + dx & 0 \\ z^2\ dy & dz + dx \end{pmatrix} \tag{\%t87}$$

Calculate $A \wedge B$

```
(%i88) A.B;
```

$$\begin{pmatrix} dx\ dy & dy\ dz - dx\ dy \\ -y^2\ z^2\ dy\ dz - dy\ dz - dx\ dz & z^2\ dy\ dz - y^2\ dx\ dz - z^2\ dx\ dy \end{pmatrix} \tag{\%o88}$$

```
(%i89) ldisplay(dA:factor(ext_diff(A)))$
```

$$dA = \begin{pmatrix} 0 & 0 \\ 0 & -2\,(z-y)\ dy\ dz \end{pmatrix} \tag{\%t89}$$

```
(%i90) ldisplay(dB:factor(ext_diff(B)))$
```

$$dB = \begin{pmatrix} 0 & 0 \\ -2z\ dy\ dz & 0 \end{pmatrix} \tag{\%t90}$$

6.4 Proposition 2.6.5.

Product rule for matrices of forms
Let A be a matrix of p-forma and B a matrix of q-forms and suppose that A, B are multipliable then

$$\mathrm{d}(A \wedge B) = \mathrm{d}A \wedge B + (-1)^p A \wedge \mathrm{d}B$$

6.5 Proposition 2.6.6.

Attitude matrix
Let A be the attitude matrix of a given frame then

$$\mathrm{d}A^T \wedge A = -A^T \wedge \mathrm{d}A, \quad \mathrm{d}A \wedge A^T = -A \wedge \mathrm{d}A^T$$

Moreover, $\mathrm{d}A = -A \wedge \mathrm{d}A^T \wedge A$

6.6 Example 2.6.8.

Following Examples 2.2.9 and 2.2.15, the **cylindrical coordinate frame** has attitude matrix:

(%i91)
```
kill(labels,t,x,y,z,r,θ)$
```

Define the space $\mathbb{R}^3$

(%i1)
```
ξ:[r,θ,z]$
```

(%i2)
```
init_cartan(ξ)$
```

(%i3)
```
matrix_element_mult:"~"$
```

(%i4)
```
A:matrix([cos(θ),sin(θ),0],[-sin(θ),cos(θ),0],[0,0,1]);
```

$$\begin{pmatrix} \cos(\theta) & \sin(\theta) & 0 \\ -\sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{A}$$

(%i5)
```
ldisplay(dA:ext_diff(A))$
```

$$dA = \begin{pmatrix} -d\theta\ \sin(\theta) & d\theta\ \cos(\theta) & 0 \\ -d\theta\ \cos(\theta) & -d\theta\ \sin(\theta) & 0 \\ 0 & 0 & 0 \end{pmatrix} \tag{\%t5}$$

(%i6)
```
ldisplay(ω:trigsimp(dA.transpose(A)))$
```

$$\omega = \begin{pmatrix} 0 & d\theta & 0 \\ -d\theta & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix} \tag{\%t6}$$

(%i7)
```
matrix_element_mult:"*"$
```

6.7 Example 2.6.9.

(%i8) `kill(labels,t,x,y,z,ρ,θ,φ)$`

Following Example 2.2.10 and 2.2.16, the **spherical coordinate frame** has attitude matrix:

(%i1) `ξ:[ρ,θ,φ]$`

(%i2) `init_cartan(ξ)$`

(%i3) `matrix_element_mult:"~"$`

(%i4) `ldisplay(A:matrix([cos(θ)*sin(φ),sin(θ)*sin(φ),cos(φ)], [cos(θ)*cos(φ),sin(θ)*cos(φ),-sin(φ)], [-sin(θ),cos(θ),0]))$`

$$A = \begin{pmatrix} \cos(\theta)\,\sin(\phi) & \sin(\theta)\,\sin(\phi) & \cos(\phi) \\ \cos(\theta)\,\cos(\phi) & \sin(\theta)\,\cos(\phi) & -\sin(\phi) \\ -\sin(\theta) & \cos(\theta) & 0 \end{pmatrix} \tag{\%t4}$$

(%i5) `ldisplay(dA:ext_diff(A))$`

$$dA = \begin{pmatrix} d\phi\,\cos(\theta)\,\cos(\phi) - d\theta\,\sin(\theta)\,\sin(\phi) & d\theta\,\cos(\theta)\,\sin(\phi) + d\phi\,\sin(\theta)\,\cos(\phi) & -d\phi\,\sin(\phi) \\ -d\phi\,\cos(\theta)\,\sin(\phi) - d\theta\,\sin(\theta)\,\cos(\phi) & d\theta\,\cos(\theta)\,\cos(\phi) - d\phi\,\sin(\theta)\,\sin(\phi) & -d\phi\,\cos(\phi) \\ -d\theta\,\cos(\theta) & -d\theta\,\sin(\theta) & 0 \end{pmatrix} \tag{\%t5}$$

(%i6) `coeff(dA,dθ);`

$$\begin{pmatrix} -\sin(\theta)\,\sin(\phi) & \cos(\theta)\,\sin(\phi) & 0 \\ -\sin(\theta)\,\cos(\phi) & \cos(\theta)\,\cos(\phi) & 0 \\ -\cos(\theta) & -\sin(\theta) & 0 \end{pmatrix} \tag{\%o6}$$

(%i7) `coeff(dA,dφ);`

$$\begin{pmatrix} \cos(\theta)\,\cos(\phi) & \sin(\theta)\,\cos(\phi) & -\sin(\phi) \\ -\cos(\theta)\,\sin(\phi) & -\sin(\theta)\,\sin(\phi) & -\cos(\phi) \\ 0 & 0 & 0 \end{pmatrix} \tag{\%o7}$$

(%i8) `ldisplay(ω:trigsimp(dA.transpose(A)))$`

$$\omega = \begin{pmatrix} 0 & d\phi & d\theta\,\sin(\phi) \\ -d\phi & 0 & d\theta\,\cos(\phi) \\ -d\theta\,\sin(\phi) & -d\theta\,\cos(\phi) & 0 \end{pmatrix} \tag{\%t8}$$

(%i9) `matrix_element_mult:"*"$`

7 coframes and Cartan Structure Equations

7.1 Definition 2.7.1.

Coframe on $\mathbb{R}^3$
Suppose $\{E_1, E_2, E_3\}$ is a frame on $\mathbb{R}^3$ then we say a set of differential one-forms $\{\theta^1, \theta_2, \theta_3\}$ on $\mathbb{R}^3$ is a **coframe** if $\theta^i(E_j) = \delta_{i\,j}$ for all i, j.

7.2 Example 2.7.2.

(%i10) `kill(labels,x,y,z)$`

Define the space $\mathbb{R}^3$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

(%i4) `U:[U_1,U_2,U_3]$`

Define array function of two arguments h

(%i5) `h[i,j]:=diff(ζ,ζ[i])|concat(d,ζ[j])$`

(%i6) `genmatrix(h,cartan_dim,cartan_dim);`

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%o6}$$

Define array function of two arguments g

(%i7) `g[i,j]:=diff(cartan_coords,cartan_coords[i])|cartan_basis[j]$`

(%i8) `genmatrix(g,cartan_dim,cartan_dim);`

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%o8}$$

Define generic vector $\mathbf{V} \in \mathfrak{X}(\mathbb{R}^3)$

(%i9) `V:[V_1,V_2,V_3]$`

(%i12) `V|dx; V|dy; V|dz;`

$$V_1 \tag{\%o10}$$

$$V_2 \tag{\%o11}$$

$$V_3 \tag{\%o12}$$

(%i13) `V:[V|dx,V|dy,V|dz];`

$$[V_1, V_2, V_3] \tag{V}$$

7.3 Proposition 2.7.3.

Components with respect to frame and coframe
If $\{E_1, E_2, E_3\}$ is a frame with coframe $\{\theta^1, \theta^2, \theta^3\}$ if $Y \in \mathfrak{X}(\mathbb{R}^3)$ and $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$ then

$$Y = \sum_{j=1}^{3} \theta^j(Y)E_j, \quad \alpha = \sum_{j=1}^{3} \alpha(E_j)\theta^j$$

7.4 Proposition 2.7.4.

Attitude of coframe
If $\{E_1, E_2, E_3\}$ is a frame with coframe $\{\theta^1, \theta^2, \theta^3\}$ and $\{U_1, U_2, U_3\}$ is the Cartesian frame with coframe $\{\mathrm{d}x^1, \mathrm{d}x^2, \mathrm{d}x^3\}$ on $\mathbb{R}^3$ then

$$E_i = \sum_j A_{i\,j} U_j \Leftrightarrow \theta^i = \sum_j A_{i\,j} \mathrm{d}x^j$$

7.5 Theorem 2.7.5.

Cartan Structure Equations for $\mathbb{R}^3$ If E_i is a frame with coframe $theta^i$ and ω is the connection form for the given frame then:

$$\mathrm{d}\theta^i = \sum_j \omega_{i\,j} \wedge \theta^j, \quad \mathrm{d}\omega_{i\,j} = \sum_k \omega_{i\,k} \wedge \omega_{k,j}$$

7.6 Example 2.7.6.

(%i14) `kill(labels,t,x,y,z,r,θ,U_1,U_2,U_3,θ_1,θ_2,θ_3,E_1,E_2,E_3)$`

(%i1) `ζ:[x,y,z]$`

(%i3)
```
assume(0≤r)$
assume(0≤θ,θ≤2*π)$
```

(%i4) `ξ:[r,θ,z]$`

Cartesian frame

(%i5) `U:[U_1,U_2,U_3]$`

Initialize cartan package

(%i6) `init_cartan(ξ)$`

(%i7) `cartan_basis;`

$$[dr, d\theta, dz] \tag{\%o7}$$

(%i8) `cartan_coords;`

$$[r, \theta, z] \tag{\%o8}$$

(%i9) `cartan_dim;`

$$3 \tag{\%o9}$$

(%i10) `extdim;`

$$3 \tag{\%o10}$$

Transformation formulas

(%i11) `ldisplay(Tr:[r*cos(θ),r*sin(θ),z])$`

$$Tr = [r\cos(\theta), r\sin(\theta), z] \tag{\%t11}$$

Jacobian matrix

(%i12) `ldisplay(J:jacobian(Tr,ξ))$`

$$J = \begin{pmatrix} \cos(\theta) & -r\sin(\theta) & 0 \\ \sin(\theta) & r\cos(\theta) & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t12}$$

Metric tensor

(%i13) `ldisplay(lg:trigsimp(transpose(J).J))$`

$$lg = \begin{pmatrix} 1 & 0 & 0 \\ 0 & r^2 & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t13}$$

Jacobian

(%i14) `ldisplay(Jdet:trigsimp(determinant(J)))$`

$$Jdet = r \tag{\%t14}$$

Initialize vect package

(%i15) `scalefactors(append([Tr],ξ))$`

(%i16) `sf;`

$$[1, r, 1] \tag{\%o16}$$

(%i17) `sfprod;`

$$r \tag{\%o17}$$

Volume

(%i18) `[dx,dy,dz]:list_matrix_entries(J.cartan_basis)$`

(%i19) `dv:trigsimp(dx~dy~dz);`

$$r\,dr\,dz\,d\theta \tag{dv}$$

(%i20) `diff(ξ,z)|(diff(ξ,θ)|(diff(ξ,r)|dv));`

$$r \tag{\%o20}$$

(%i21) `ldisplay(dζ:trigsimp(ext_diff(at(ζ,map("=",ζ,Tr)))))$`

$$d\zeta = [dr\;\cos(\theta) - r\,d\theta\;\sin(\theta), dr\;\sin(\theta) + r\,d\theta\;\cos(\theta), dz] \tag{\%t21}$$

Attitude matrix

(%i22) `ldisplay(A:apply('matrix,makelist(trigsimp(normalize(k)),k,args(transpose(J)))))$`

$$A = \begin{pmatrix} \cos(\theta) & \sin(\theta) & 0 \\ -\sin(\theta) & \cos(\theta) & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{\%t22}$$

Frame

(%i23) `E:[E_1,E_2,E_3]:trigsimp(list_matrix_entries(A.U))$`

(%i24) `map(ldisp,E)$`

$$U_2 \sin(\theta) + U_1 \cos(\theta) \tag{\%t24}$$

$$U_2 \cos(\theta) - U_1 \sin(\theta) \tag{\%t25}$$

$$U_3 \tag{\%t26}$$

Coframe

(%i27) `ldisplay(Θ:[θ_1,θ_2,θ_3]:list_matrix_entries(trigsimp(A.[dx,dy,dz])))$`

$$\Theta = [dr, r\,d\theta, dz] \tag{\%t27}$$

(%i28) `ldisplay(Θ:list_matrix_entries(trigsimp(A.dζ)))$`

$$\Theta = [dr, r\,d\theta, dz] \tag{\%t28}$$

(%i29) `ldisplay(Θ:sf*cartan_basis)$`

$$\Theta = [dr, r\,d\theta, dz] \tag{\%t29}$$

dA

(%i30) `ldisplay(dA:ext_diff(A))$`

$$dA = \begin{pmatrix} -d\theta\ \sin(\theta) & d\theta\ \cos(\theta) & 0 \\ -d\theta\ \cos(\theta) & -d\theta\ \sin(\theta) & 0 \\ 0 & 0 & 0 \end{pmatrix} \qquad (\%t30)$$

Change matrix multiplication operator

(%i31) `matrix_element_mult:"~"$`

Connection form $\omega \in \mathcal{A}^1(\mathbb{R}^3)$

(%i32) `ldisplay(ω:trigsimp(dA.transpose(A)))$`

$$\omega = \begin{pmatrix} 0 & d\theta & 0 \\ -d\theta & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix} \qquad (\%t32)$$

(%i33) `ldisplay(dω:ext_diff(ω))$`

$$d\omega = \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix} \qquad (\%t33)$$

(%i34) `trigsimp(ω.ω);`

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix} \qquad (\%o34)$$

First Cartan structure equation

(%i35) `ldisplay(dΘ:ext_diff(Θ))$`

$$d\Theta = [0, dr\ d\theta, 0] \qquad (\%t35)$$

(%i36) `list_matrix_entries(ω.Θ);`

$$[0, dr\ d\theta, 0] \qquad (\%o36)$$

Second Cartan structure equation

(%i37) `ldisplay(dω:ext_diff(ω))$`

$$d\omega = \begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix} \qquad (\%t37)$$

(%i38) `trigsimp(ω.ω);`

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix} \qquad (\%o38)$$

Restore matrix multiplication operator

(%i39) `matrix_element_mult:"*"$`

7.7 Example 2.7.7.

(%i40) `kill(labels,t,x,y,z,r,θ,U_1,U_2,U_3,θ_1,θ_2,θ_3,E_1,E_2,E_3)$`

(%i1) `kill(labels,x,y,z,r,θ)$`

(%i1) `ζ:[x,y,z]$`

(%i5)
```
assume(0≤r)$
assume(0≤θ,θ≤π)$
assume(0≤sin(θ))$
assume(0≤ϕ,ϕ≤2*π)$
```

(%i6) `ξ:[r,θ,ϕ]$`

Cartesian frame

(%i7) `U:[U_1,U_2,U_3]$`

Initialize cartan package

(%i8) `init_cartan(ξ)$`

(%i9) `cartan_basis;`

$$[dr, d\theta, d\phi] \tag{\%o9}$$

(%i10) `cartan_coords;`

$$[r, \theta, \phi] \tag{\%o10}$$

(%i11) `cartan_dim;`

$$3 \tag{\%o11}$$

(%i12) `extdim;`

$$3 \tag{\%o12}$$

Transformation formulas

(%i13) `ldisplay(Tr:[r*sin(θ)*cos(ϕ),r*sin(θ)*sin(ϕ),r*cos(θ)])$`

$$Tr = [r\,\sin(\theta)\,\cos(\phi), r\,\sin(\theta)\,\sin(\phi), r\,\cos(\theta)] \tag{\%t13}$$

Jacobian matrix

(%i14) `ldisplay(J:jacobian(Tr,ξ))$`

$$J = \begin{pmatrix} \sin(\theta)\,\cos(\phi) & r\,\cos(\theta)\,\cos(\phi) & -r\,\sin(\theta)\,\sin(\phi) \\ \sin(\theta)\,\sin(\phi) & r\,\cos(\theta)\,\sin(\phi) & r\,\sin(\theta)\,\cos(\phi) \\ \cos(\theta) & -r\,\sin(\theta) & 0 \end{pmatrix} \tag{\%t14}$$

Metric tensor

(%i15) `ldisplay(lg:trigsimp(transpose(J).J))$`

$$lg = \begin{pmatrix} 1 & 0 & 0 \\ 0 & r^2 & 0 \\ 0 & 0 & r^2 \sin(\theta)^2 \end{pmatrix} \tag{\%t15}$$

Jacobian

(%i16) `ldisplay(Jdet:trigsimp(determinant(J)))$`

$$Jdet = r^2 \sin(\theta) \tag{\%t16}$$

Initialize vect package

(%i17) `scalefactors(append([Tr],ξ))$`

(%i18) `sf;`

$$[1, r, r \sin(\theta)] \tag{\%o18}$$

(%i19) `sfprod;`

$$r^2 \sin(\theta) \tag{\%o19}$$

Volume

(%i20) `[dx,dy,dz]:list_matrix_entries(J.cartan_basis)$`

(%i21) `dv:trigsimp(dx~dy~dz);`

$$r^2 \, dr \, d\theta \, d\phi \sin(\theta) \tag{dv}$$

(%i22) `diff(ξ,ϕ)|(diff(ξ,θ)|(diff(ξ,r)|dv));`

$$r^2 \sin(\theta) \tag{\%o22}$$

(%i23) `ldisplay(dζ:trigsimp(ext_diff(at(ζ,map("=",ζ,Tr)))))$`

$d\zeta = [-r \, d\phi \sin(\theta) \sin(\phi) + dr \sin(\theta) \cos(\phi) + r \, d\theta \cos(\theta) \cos(\phi), dr \sin(\theta) \sin(\phi) + r \, d\theta \cos(\theta) \sin(\phi) + r \, d\phi \sin(\theta) \cos(\phi),$ (%t23)

Attitude matrix

(%i24) `ldisplay(A:apply('matrix,makelist(trigsimp(normalize(k)),k,args(transpose(J)))))$`

$$A = \begin{pmatrix} \sin(\theta) \cos(\phi) & \sin(\theta) \sin(\phi) & \cos(\theta) \\ \cos(\theta) \cos(\phi) & \cos(\theta) \sin(\phi) & -\sin(\theta) \\ -\sin(\phi) & \cos(\phi) & 0 \end{pmatrix} \tag{\%t24}$$

Frame

(%i25) `E:[E_1,E_2,E_3]:trigsimp(list_matrix_entries(A.U))$`

(%i26) `map(ldisp,E)$`

$$U_2 \sin(\theta)\ \sin(\phi) + U_1 \sin(\theta)\ \cos(\phi) + U_3 \cos(\theta) \tag{\%t26}$$

$$U_2 \cos(\theta)\ \sin(\phi) + U_1 \cos(\theta)\ \cos(\phi) - U_3 \sin(\theta) \tag{\%t27}$$

$$U_2 \cos(\phi) - U_1 \sin(\phi) \tag{\%t28}$$

Coframe

(%i29) `ldisplay(Θ:[θ_1,θ_2,θ_3]:list_matrix_entries(trigsimp(A.[dx,dy,dz])))$`

$$\Theta = [dr, r\ d\theta, r\ d\phi\ \sin(\theta)] \tag{\%t29}$$

(%i30) `ldisplay(Θ:list_matrix_entries(trigsimp(A.dζ)))$`

$$\Theta = [dr, r\ d\theta, r\ d\phi\ \sin(\theta)] \tag{\%t30}$$

(%i31) `ldisplay(Θ:sf*cartan_basis)$`

$$\Theta = [dr, r\ d\theta, r\ d\phi\ \sin(\theta)] \tag{\%t31}$$

dA

(%i32) `ldisplay(dA:ext_diff(A))$`

$$dA = \begin{pmatrix} d\theta\ \cos(\theta)\ \cos(\phi) - d\phi\ \sin(\theta)\ \sin(\phi) & d\theta\ \cos(\theta)\ \sin(\phi) + d\phi\ \sin(\theta)\ \cos(\phi) & -d\theta\ \sin(\theta) \\ -d\phi\ \cos(\theta)\ \sin(\phi) - d\theta\ \sin(\theta)\ \cos(\phi) & d\phi\ \cos(\theta)\ \cos(\phi) - d\theta\ \sin(\theta)\ \sin(\phi) & -d\theta\ \cos(\theta) \\ -d\phi\ \cos(\phi) & -d\phi\ \sin(\phi) & 0 \end{pmatrix} \tag{\%t32}$$

Change matrix multiplication operator

(%i33) `matrix_element_mult:"~"$`

Connection form $\omega = \mathrm{d}A \wedge A^T \in \mathcal{A}^1(\mathbb{R}^3)$

(%i34) `ldisplay(ω:trigsimp(dA.transpose(A)))$`

$$\omega = \begin{pmatrix} 0 & d\theta & d\phi\ \sin(\theta) \\ -d\theta & 0 & d\phi\ \cos(\theta) \\ -d\phi\ \sin(\theta) & -d\phi\ \cos(\theta) & 0 \end{pmatrix} \tag{\%t34}$$

First Cartan structure equation

(%i35) `ldisplay(dΘ:ext_diff(Θ))$`

$$d\Theta = [0, dr\ d\theta, dr\ d\phi\ \sin(\theta) + r\ d\theta\ d\phi\ \cos(\theta)] \tag{\%t35}$$

(%i36) `list_matrix_entries(ω.Θ);`

$$[0, dr\, d\theta, dr\, d\phi\, \sin(\theta) + r\, d\theta\, d\phi\, \cos(\theta)] \qquad (\%o36)$$

Second Cartan structure equation

(%i37) `ldisplay(dω:ext_diff(ω))$`

$$d\omega = \begin{pmatrix} 0 & 0 & d\theta\, d\phi\, \cos(\theta) \\ 0 & 0 & -d\theta\, d\phi\, \sin(\theta) \\ -d\theta\, d\phi\, \cos(\theta) & d\theta\, d\phi\, \sin(\theta) & 0 \end{pmatrix} \qquad (\%t37)$$

(%i38) `trigsimp(ω.ω);`

$$\begin{pmatrix} 0 & 0 & d\theta\, d\phi\, \cos(\theta) \\ 0 & 0 & -d\theta\, d\phi\, \sin(\theta) \\ -d\theta\, d\phi\, \cos(\theta) & d\theta\, d\phi\, \sin(\theta) & 0 \end{pmatrix} \qquad (\%o38)$$

Restore matrix multiplication operator

(%i39) `matrix_element_mult:"*"$`