Barriola-Vilenkin metric

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

(%i2) `reset()$kill(all)$`

(%i1) `derivabbrev:true$`

(%i2) `ratprint:false$`

(%i3) `fpprintprec:5$`

(%i4) `if get('draw,'version)=false then load(draw)$`

(%i5) `wxplot_size:[1024,768]$`

(%i6) `if get('optvar,'version)=false then load(optvar)$`

(%i7) `if get('rkf45,'version)=false then load(rkf45)$`

(%i8) `declare(trigsimp,evfun)$`

1 Settings

(%i12)
```
assume(r≥0)$
assume(θ≥0,θ≤π)$
assume(sin(θ)≥0)$
assume(ϕ≥0,ϕ≤2*π)$
```

(%i13) `ct_coords:[t,r,θ,ϕ]$`

(%i14) `dim:length(ct_coords)$`

(%i15) `orderless(m,c,K)$`

(%i16) `declare([c,K],constant)$`

(%i17) `assume(m>0,c>0,K>0)$`

(%i18) `params:[m=1,c=1,K=0.21]$`

(%i19) `τ:16$`

Covariant Metric Tensor

(%i20) `ldisplay(lg:matrix([-c^2,0,0,0],[0,1,0,0],[0,0,K^2*r^2,0],[0,0,0,K^2*r^2*sin(θ)^2]))$`

$$lg = \begin{pmatrix} -c^2 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & K^2\,r^2 & 0 \\ 0 & 0 & 0 & K^2\,r^2\,\sin(\theta)^2 \end{pmatrix} \tag{\%t20}$$

Contravariant Metric Tensor

(%i21) `ldisplay(ug:invert(lg))$`

$$ug = \begin{pmatrix} -\frac{1}{c^2} & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \frac{1}{K^2\,r^2} & 0 \\ 0 & 0 & 0 & \frac{1}{K^2\,r^2\,\sin(\theta)^2} \end{pmatrix} \tag{\%t21}$$

Line element

(%i22) `ldisplay(ds^2=diff(ct_coords).lg.transpose(diff(ct_coords)))$`

$$ds^2 = K^2\,r^2\sin(\theta)^2\,\mathrm{del}(\phi)^2 + K^2\,r^2\,\mathrm{del}(\theta)^2 - c^2\,\mathrm{del}(t)^2 + \mathrm{del}(r)^2 \tag{\%t22}$$

2 Using optvar

(%i23) `kill(labels)$`

(%i1) `depends(ct_coords,s)$`

Lagrangian

(%i2) `ldisplay(L:m*diff(ct_coords,s).lg.transpose(diff(ct_coords,s)))$`

$$L = m\left(K^2 r^2 \sin(\theta)^2 (\phi_s)^2 + K^2 r^2 (\theta_s)^2 - c^2 (t_s)^2 + (r_s)^2\right) \tag{\%t2}$$

Momentum Conjugate

(%i3) `ldisplay(P_t:ev(diff(L,'diff(t,s))))$`

$$P_t = -2c^2 m\ (t_s) \tag{\%t3}$$

(%i4) `linsolve(p_t=P_t,diff(t,s)),factor;`

$$\left[t_s = -\frac{p_t}{2c^2 m}\right] \tag{\%o4}$$

(%i5) `ldisplay(P_r:ev(diff(L,'diff(r,s))))$`

$$P_r = 2m\ (r_s) \tag{\%t5}$$

(%i6) `linsolve(p_r=P_r,diff(r,s)),factor;`

$$\left[r_s = \frac{p_r}{2m}\right] \tag{\%o6}$$

(%i7) `ldisplay(P_θ:ev(diff(L,'diff(θ,s))))$`

$$P_\theta = 2K^2 m\, r^2\ (\theta_s) \tag{\%t7}$$

(%i8) `linsolve(p_θ=P_θ,diff(θ,s)),factor;`

$$\left[\theta_s = \frac{p_\theta}{2K^2 m\, r^2}\right] \tag{\%o8}$$

(%i9) `ldisplay(P_ϕ:ev(diff(L,'diff(ϕ,s))))$`

$$P_\phi = 2K^2 m\, r^2 \sin(\theta)^2\ (\phi_s) \tag{\%t9}$$

(%i10) `linsolve(p_ϕ=P_ϕ,diff(ϕ,s)),factor;`

$$\left[\phi_s = \frac{p_\phi}{2K^2 m\, r^2 \sin(\theta)^2}\right] \tag{\%o10}$$

Generalized Forces

(%i11) `ldisplay(F_t:diff(L,t))$`

$$F_t = 0 \tag{\%t11}$$

(%i12) `ldisplay(F_r:factor(trigreduce(diff(L,r))))$`

$$F_r = -K^2 m r \left(\cos(2\theta)(\phi_s)^2 - (\phi_s)^2 - 2(\theta_s)^2\right) \tag{\%t12}$$

(%i13) `ldisplay(F_θ:factor(trigreduce(diff(L,θ))))$`

$$F_\theta = K^2 m\, r^2 \sin(2\theta)(\phi_s)^2 \tag{\%t13}$$

(%i14) `ldisplay(F_ϕ:diff(L,ϕ))$`

$$F_\phi = 0 \tag{\%t14}$$

Euler-Lagrange Equations

(%i15)
```
aa:el(L,ct_coords,s)$
```

(%i22)
```
bb:ev(aa,eval,diff)$
```

(%i25)
```
bb[1]:subst([k[0]=-E],-bb[1])$
bb[3]:subst([k[1]=Λ],bb[3])$
bb[7]:subst([k[4]=J],bb[7])$
```

(%i29)
```
bb[2]:lhs(bb[2])-rhs(bb[2])=0$
bb[4]:lhs(bb[4])-rhs(bb[4])=0$
bb[5]:lhs(bb[5])-rhs(bb[5])=0$
bb[6]:lhs(bb[6])-rhs(bb[6])=0$
```

Conservation Laws

(%i30)
```
map(ldisp,radcan(part(bb/m,[1,3,7])))$
```

$$K^2\,r^2\sin(\theta)^2\,(\phi_s)^2+K^2\,r^2\,(\theta_s)^2-c^2\,(t_s)^2+(r_s)^2=\frac{E}{m} \quad (\%t30)$$

$$-2c^2\,(t_s)=\frac{\Lambda}{m} \quad (\%t31)$$

$$2K^2\,r^2\sin(\theta)^2\,(\phi_s)=\frac{J}{m} \quad (\%t32)$$

Express the Energy in terms of the Angular Momentum

(%i33)
```
linsolve(eliminate(part(bb,[1,7]),[diff(ϕ,s)]),E),expand;
```

$$\left[E=K^2m\,r^2\,(\theta_s)^2+\frac{J^2}{4K^2m\,r^2\sin(\theta)^2}-c^2m\,(t_s)^2+m\,(r_s)^2\right] \quad (\%o33)$$

Equations of Motion

(%i34)
```
map(ldisp,radcan(part(bb/m,[2,4,5,6])))$
```

$$-2c^2\,(t_{ss})=0 \quad (\%t34)$$

$$-2K^2r\sin(\theta)^2\,(\phi_s)^2-2K^2r\,(\theta_s)^2+2\,(r_{ss})=0 \quad (\%t35)$$

$$-2K^2\,r^2\,\cos(\theta)\,\sin(\theta)\,(\phi_s)^2+2K^2\,r^2\,(\theta_{ss})+4K^2r\,(r_s)\,(\theta_s)=0 \quad (\%t36)$$

$$2K^2\,r^2\sin(\theta)^2\,(\phi_{ss})+\left(4K^2\,r^2\,\cos(\theta)\,\sin(\theta)\,(\theta_s)+4K^2r\,(r_s)\,\sin(\theta)^2\right)\,(\phi_s)=0 \quad (\%t37)$$

Solve for second derivative of coordinates

(%i38)
```
linsol:linsolve(part(bb,[2,4,5,6]),diff(ct_coords,s,2))$
```

(%i39)
```
map(ldisp,radcan(linsol))$
```

$$t_{ss}=0 \quad (\%t39)$$

$$r_{ss}=K^2r\sin(\theta)^2\,(\phi_s)^2+K^2r\,(\theta_s)^2 \quad (\%t40)$$

$$\theta_{ss} = \frac{r\,\cos(\theta)\,\sin(\theta)\,(\phi_s)^2 - 2\,(r_s)\,(\theta_s)}{r} \tag{\%t41}$$

$$\phi_{ss} = -\frac{(2r\,\cos(\theta)\,(\theta_s) + 2\,(r_s)\,\sin(\theta))\,(\phi_s)}{r\,\sin(\theta)} \tag{\%t42}$$

Check Conservation of Energy

(%i43) `radcan(lhs(bb[1]));`

$$K^2 m\,r^2 \sin(\theta)^2 (\phi_s)^2 + K^2 m\,r^2 (\theta_s)^2 - c^2 m\,(t_s)^2 + m\,(r_s)^2 \tag{\%o43}$$

(%i44) `subst(linsol,diff(lhs(bb[1]),s)),expand;`

$$0 \tag{\%o44}$$

Check Conservation of Λ

(%i45) `radcan(lhs(bb[3]));`

$$-2c^2 m\,(t_s) \tag{\%o45}$$

(%i46) `subst(linsol,diff(lhs(bb[3]),s));`

$$0 \tag{\%o46}$$

Check Conservation of Angular Momentum

(%i47) `radcan(lhs(bb[7]));`

$$2K^2 m\,r^2 \sin(\theta)^2\,(\phi_s) \tag{\%o47}$$

(%i48) `subst(linsol,diff(lhs(bb[7]),s)),expand;`

$$0 \tag{\%o48}$$

Legendre Transformation

(%i49)
```
kill(labels)$
```

(%i1)
```
Legendre:linsolve([p_t=P_t,p_r=P_r,p_θ=P_θ,p_ϕ=P_ϕ], ['diff(t,s),'diff(r,s),'diff(θ,s),'diff(ϕ,s)]
```

(%i2)
```
map(ldisp,radcan(Legendre))$
```

$$t_s = -\frac{p_t}{2c^2 m} \tag{\%t2}$$

$$r_s = \frac{p_r}{2m} \tag{\%t3}$$

$$\theta_s = \frac{p_\theta}{2K^2 m\, r^2} \tag{\%t4}$$

$$\phi_s = \frac{p_\phi}{2K^2 m\, r^2 \sin(\theta)^2} \tag{\%t5}$$

Hamiltonian

(%i6)
```
ldisplay(H:ev(p_t*'diff(t,s)+p_r*'diff(r,s)+ p_θ*'diff(θ,s)+p_ϕ*'diff(ϕ,s)-L,Legendre,radcan))$
```

$$H = -\frac{\left(\left(K^2\, {p_t}^2 - c^2\, K^2\, {p_r}^2\right) r^2 - c^2\, {p_\theta}^2\right) \sin(\theta)^2 - c^2\, {p_\phi}^2}{4c^2\, K^2 m\, r^2 \sin(\theta)^2} \tag{\%t6}$$

Equations of Motion

(%i7)
```
Hq:makelist(Hq[i],i,1,2*dim)$
```

(%i15)
```
Hq[1]:'diff(t,s)=diff(H,p_t)$
Hq[2]:'diff(r,s)=diff(H,p_r)$
Hq[3]:'diff(θ,s)=diff(H,p_θ)$
Hq[4]:'diff(ϕ,s)=diff(H,p_ϕ)$
Hq[5]:'diff(p_t,s)=-diff(H,t)$
Hq[6]:'diff(p_r,s)=-diff(H,r)$
Hq[7]:'diff(p_θ,s)=-diff(H,θ)$
Hq[8]:'diff(p_ϕ,s)=-diff(H,ϕ)$
```

(%i16)
```
map(ldisp,Hq:radcan(Hq))$
```

$$t_s = -\frac{p_t}{2c^2 m} \tag{\%t16}$$

$$r_s = \frac{p_r}{2m} \tag{\%t17}$$

$$\theta_s = \frac{p_\theta}{2K^2 m\, r^2} \tag{\%t18}$$

$$\phi_s = \frac{p_\phi}{2K^2 m\, r^2 \sin(\theta)^2} \tag{\%t19}$$

$${p_t}_s = 0 \tag{\%t20}$$

$${p_r}_s = \frac{{p_\theta}^2 \sin(\theta)^2 + {p_\phi}^2}{2K^2 m\, r^3 \sin(\theta)^2} \tag{\%t21}$$

$${p_\theta}_s = \frac{{p_\phi}^2 \cos(\theta)}{2K^2 m\, r^2 \sin(\theta)^3} \tag{\%t22}$$

$$p_{\phi_s} = 0 \tag{\%t23}$$

Check Conservation of Energy

(%i24) `depends([p_t,p_r,p_θ,p_ϕ],s)$`

(%i25) `subst(Hq,diff(H,s)),expand;`

$$0 \tag{\%o25}$$

Reduce Order

(%i26) `kill(labels)$`

(%i2)
```
cv_coords:[T,R,Θ,Φ]$
depends(cv_coords,s)$
```

(%i6)
```
gradef(t,s,T)$
gradef(r,s,R)$
gradef(θ,s,Θ)$
gradef(ϕ,s,Φ)$
```

Euler-Lagrange Equations

(%i7) `aa:el(L,ct_coords,s)$`

(%i14) `bb:ev(aa,eval,diff)$`

(%i17)
```
bb[1]:subst([k[0]=-E],-bb[1])$
bb[3]:subst([k[1]=Λ],bb[3])$
bb[7]:subst([k[4]=J],bb[7])$
```

(%i21)
```
bb[2]:lhs(bb[2])-rhs(bb[2])=0$
bb[4]:lhs(bb[4])-rhs(bb[4])=0$
bb[5]:lhs(bb[5])-rhs(bb[5])=0$
bb[6]:lhs(bb[6])-rhs(bb[6])=0$
```

Conservation Laws

(%i22) `ldisplay(Epm:expand(bb[1]/m))$`

$$K^2\,r^2\,\mathit{\Phi}^2\sin\left(\theta\right)^2+K^2\,r^2\,\mathit{\Theta}^2-c^2\,T^2+R^2=\frac{E}{m} \qquad (\%t22)$$

(%i23) `ldisplay(Λpm:expand(bb[3]/m))$`

$$-2c^2T=\frac{\mathit{\Lambda}}{m} \qquad (\%t23)$$

(%i24) `ldisplay(Jpm:expand(bb[7]/m))$`

$$2K^2\,r^2\,\mathit{\Phi}\sin\left(\theta\right)^2=\frac{J}{m} \qquad (\%t24)$$

Equations of Motion

(%i25) `map(ldisp,radcan(part(bb/m,[2,4,5,6])))$`

$$-2c^2\,\left(T_s\right)=0 \qquad (\%t25)$$

$$-2K^2r\,\mathit{\Phi}^2\sin\left(\theta\right)^2-2K^2r\,\mathit{\Theta}^2+2\left(R_s\right)=0 \qquad (\%t26)$$

$$-2K^2\,r^2\,\mathit{\Phi}^2\,\cos\left(\theta\right)\,\sin\left(\theta\right)+2K^2\,r^2\,\left(\mathit{\Theta}_s\right)+4K^2Rr\,\mathit{\Theta}=0 \qquad (\%t27)$$

$$\left(2K^2\,r^2\,\left(\mathit{\Phi}_s\right)+4K^2Rr\,\mathit{\Phi}\right)\,\sin\left(\theta\right)^2+4K^2\,r^2\,\mathit{\Theta}\mathit{\Phi}\,\cos\left(\theta\right)\,\sin\left(\theta\right)=0 \qquad (\%t28)$$

Solve for second derivative of coordinates

(%i29) `linsol:linsolve(part(bb,[2,4,5,6]),diff(ct_coords,s,2))$`

(%i30) `map(ldisp,radcan(linsol))$`

$$T_s = 0 \tag{\%t30}$$

$$R_s = K^2 r\, \Phi^2 \sin(\theta)^2 + K^2 r\, \Theta^2 \tag{\%t31}$$

$$\Theta_s = \frac{r\, \Phi^2\, \cos(\theta)\, \sin(\theta) - 2R\Theta}{r} \tag{\%t32}$$

$$\Phi_s = -\frac{2R\Phi\, \sin(\theta) + 2r\,\Theta\Phi\, \cos(\theta)}{r\, \sin(\theta)} \tag{\%t33}$$

Numerical solution (Lagrangian)

(%i34)
```
kill(labels)$
```

(%i8)
```
funcs:append(ct_coords,cv_coords)$ldisplay(funcs)$
initial:[0,15,π/2,π/4,2.022,-1.7,-0.1,-0.1]$ldisplay(initial)$
odes:append(cv_coords,map('rhs,linsol))$ldisplay(odes)$
interval:[s,0,τ]$ldisplay(interval)$
```

$$funcs = [t, r, \theta, \phi, T, R, \Theta, \Phi] \tag{\%t2}$$

$$initial = \left[0, 15, \frac{\pi}{2}, \frac{\pi}{4}, 2.022, -1.7, -0.1, -0.1\right] \tag{\%t4}$$

$$odes = \left[T, R, \Theta, \Phi, 0, K^2 r\, \Phi^2 \sin(\theta)^2 + K^2 r\, \Theta^2, \frac{r\, \Phi^2 \cos(\theta) \sin(\theta) - 2R\Theta}{r}, -\frac{2R\Phi \sin(\theta) + 2r\,\Theta\Phi \cos(\theta)}{r \sin(\theta)}\right] \tag{\%t6}$$

$$interval = [s, 0, 16] \tag{\%t8}$$

(%i9)
```
P:map("=",funcs,initial)$
```

(%i10)
```
lgP:lg,P,params$
```

(%i11)
```
gVV:factor(diff(ct_coords).lgP.transpose(diff(ct_coords)))$
```

(%i12)
```
gVVP:gVV,P,params;
```

$$-1.0\mathrm{del}(s)^2 \tag{gVVP}$$

(%i13)
```
rksol:rkf45(odes,funcs,initial,interval, absolute_tolerance=1E-12,report=true),params$
```

Info: rkf45:
Integration points selected:4216
Total number of iterations:4216
Bad steps corrected:1
Minimum estimated error:$1.7143 10^{-13}$
Maximum estimated error:$5.2227 10^{-13}$
Minimum integration step taken:$7.5741 10^{-4}$
Maximum integration step taken:0.025407

(%i14) wxplot2d([[discrete,map(lambda([u],part(u,[1,2])),rksol)], [discrete,map(lambda([u],part(u,[1,3]
[discrete,map(lambda([u],part(u,[1,4])),rksol)], [discrete,map(lambda([u],part(u,[1,5])),rksol)]
[style,[lines,1]],[xlabel,"s"],[ylabel,"value"], [legend,"t","r","θ","ϕ"],
[gnuplot_preamble,"set key top left"])$



(%t14)

(%i15) wxplot2d([[discrete,map(lambda([u],part(u,[1,6])),rksol)], [discrete,map(lambda([u],part(u,[1,7]
[discrete,map(lambda([u],part(u,[1,8])),rksol)], [discrete,map(lambda([u],part(u,[1,9])),rksol)]
[style,[lines,1]],[xlabel,"s"],[ylabel,"value"], [legend,"dt/ds","dr/ds","dθ/ds","dϕ/ds"],
[gnuplot_preamble,"set key bottom right"])$



(%t15)

(%i16) wxplot2d([[discrete,map(lambda([u],part(u,[2,6])),rksol)], [discrete,[part(initial,[1,5])]]],[axe
[title,"Phase Space for t"],[point_type,circle], [style,[lines,2],[points,3]],[color,magenta,red]
[xlabel,"t"],[ylabel,"dt/ds"],[legend,false])$



(%t16)

(%i17) wxplot2d([[discrete,map(lambda([u],part(u,[3,7])),rksol)], [discrete,[part(initial,[2,6])]]],[axe
[title,"Phase Space for r"],[point_type,circle], [style,[lines,2],[points,3]],[color,magenta,red]
[xlabel,"r"],[ylabel,"dr/ds"],[legend,false])$



(%t17)

(%i18) `wxplot2d([[discrete,map(lambda([u],part(u,[4,8])),rksol)], [discrete,[part(initial,[3,7])]]],[axe`
`[title,"Phase Space for θ"],[point_type,circle], [style,[lines,2],[points,3]],[color,magenta,red]`
`[xlabel,"θ"],[ylabel,"dθ/ds"],[legend,false])$`



(%t18)

(%i19) wxplot2d([[discrete,map(lambda([u],part(u,[5,9])),rksol)], [discrete,[part(initial,[4,8])]]],[axe
[title,"Phase Space for ϕ"],[point_type,circle], [style,[lines,2],[points,3]],[color,magenta,red]
[xlabel,"ϕ"],[ylabel,"dϕ/ds"],[legend,false])$



(%t19)

(%i20) draw3d(title = "Barriola-Vilenkin Geodesic", proportional_axes = xyz, axis_3d = false,
xlabel = "", ylabel = "", zlabel = "", dimensions = wxplot_size, view = [80,185],
file_name = "Barriola_Vilenkin_Geodesic1", terminal = 'pngcairo,
transform = [r*sin(θ)*cos(ϕ),r*sin(θ)*sin(ϕ),r*cos(θ),r,θ,ϕ],
color = black, point_size = 2, point_type = filled_circle, points([[0,0,0]]),
color = blue, point_size = 1, point_type = -1, points_joined = true,
points(map(lambda([u],part(u,[3,4,5])),rksol)),
color = red, point_size = 1, point_type = circle, points([part(initial,[2,3,4])])),params$

(%i21) show_image("Barriola_Vilenkin_Geodesic1.png")$



(%t21)

Check Conservation of Energy using the Numerical Data

(%i22) `P:map("=",funcs,initial)$`

(%i23) `W:lhs(Epm),P,params,numer,eval;`

$$-1.0 \tag{W}$$

(%i24) `wxplot2d([discrete,makelist([first(rkline), ev(lhs(Epm),map("=",funcs,rest(rkline)))],rkline,rks`
`[xlabel,"t"],[ylabel,"Energy/mass"],[y,W-0.001,W+0.001]),params$`



(%t24)

Check Conservation of Λ using the Numerical Data

(%i25) `W:lhs(Λpm),P,params,numer,eval;`

$$-4.044 \quad \text{(W)}$$

(%i26) `wxplot2d([discrete,makelist([first(rkline), ev(lhs(Λpm),map("=",funcs,rest(rkline)))],rkline,rks`
`[xlabel,"s"],[ylabel,"Λ/mass"],[y,W-0.001,W+0.001]),params$`



(%t26)

Check Conservation of Angular Momentum using the Numerical Data

(%i27) `W:lhs(Jpm),P,params,numer,eval;`

$$-1.9845 \quad \text{(W)}$$

(%i28) `wxplot2d([discrete,makelist([first(rkline), ev(lhs(Jpm),map("=",funcs,rest(rkline)))],rkline,rks`
`[xlabel,"s"],[ylabel,"Angular Momentum/mass"],[y,W-0.001,W+0.001]),params$`



(%t28)

Minimal Radius

(%i29) `ldisplay(r_m:lmin(map(lambda([u],part(u,3)),rksol)))$`

$$r_m = 3.8023 \tag{\%t29}$$

at proper time

(%i30) `ldisplay(s_m:assoc(r_m,map(lambda([u],part(u,[3,1])),rksol)))$`

$$s_m = 8.2566 \tag{\%t30}$$

at coordinate time

(%i31) `ldisplay(t_m:assoc(r_m,map(lambda([u],part(u,[3,2])),rksol)))$`

$$t_m = 16.695 \tag{\%t31}$$

Numerical solution (Hamiltonian)

(%i32)
```
kill(labels)$
```

Calculate the initial values

(%i1)
```
initialH:initial$
```

(%i5)
```
initialH[5]:P_t,['diff(t,s)=diff(t,s)],P,params,numer$
initialH[6]:P_r,['diff(r,s)=diff(r,s)],P,params,numer$
initialH[7]:P_θ,['diff(θ,s)=diff(θ,s)],P,params,numer$
initialH[8]:P_ϕ,['diff(ϕ,s)=diff(ϕ,s)],P,params,numer$
```

(%i11)
```
funcs:[t,r,θ,ϕ,p_t,p_r,p_θ,p_ϕ]$ldisplay(funcs)$ ldisplay(initialH)$
odes:map('rhs,Hq)$ldisplay(odes)$ ldisplay(interval)$
```

$$funcs = [t, r, \theta, \phi, p_t, p_r, p_\theta, p_\phi] \tag{\%t7}$$

$$initialH = \left[0, 15, \frac{\pi}{2}, \frac{\pi}{4}, -4.044, -3.4, -1.9845, -1.9845\right] \tag{\%t8}$$

$$odes = \left[-\frac{p_t}{2c^2 m}, \frac{p_r}{2m}, \frac{p_\theta}{2K^2 m\, r^2}, \frac{p_\phi}{2K^2 m\, r^2 \sin(\theta)^2}, 0, \frac{{p_\theta}^2 \sin(\theta)^2 + {p_\phi}^2}{2K^2 m\, r^3 \sin(\theta)^2}, \frac{{p_\phi}^2 \cos(\theta)}{2K^2 m\, r^2 \sin(\theta)^3}, 0\right] \tag{\%t10}$$

$$interval = [s, 0, 16] \tag{\%t11}$$

(%i12)
```
rksol:rkf45(odes,funcs,initialH,interval, absolute_tolerance=1E-12,report=true),params$
```

Info: rkf45:
Integration points selected:4969
Total number of iterations:4969
Bad steps corrected:1
Minimum estimated error:$4.2338 10^{-14}$
Maximum estimated error:$5.889 10^{-13}$
Minimum integration step taken:$7.2834 10^{-4}$
Maximum integration step taken:0.04262

(%i13)
```
wxplot2d([[discrete,map(lambda([u],part(u,[1,2])),rksol)], [discrete,map(lambda([u],part(u,[1,3]
[discrete,map(lambda([u],part(u,[1,4])),rksol)], [discrete,map(lambda([u],part(u,[1,5])),rksol)]
[style,[lines,1]],[xlabel,"s"],[ylabel,"value"], [legend,"t","r","θ","φ"],
[gnuplot_preamble,"set key top left"])$
```



(%t13)

(%i14) `wxplot2d([[discrete,map(lambda([u],part(u,[1,6])),rksol)], [discrete,map(lambda([u],part(u,[1,7]`
`[discrete,map(lambda([u],part(u,[1,8])),rksol)], [discrete,map(lambda([u],part(u,[1,9])),rksol)]`
`[style,[lines,1]],[xlabel,"s"],[ylabel,"value"], [legend,"p_t","p_r","p_θ","p_φ"],`
`[gnuplot_preamble,"set key top left"])$`



(%t14)

(%i15) wxplot2d([[discrete,map(lambda([u],part(u,[2,6])),rksol)], [discrete,[part(initialH,[1,5])]]],[a
[title,"Phase Space for t"],[point_type,circle], [style,[lines,2],[points,3]],[color,magenta,red]
[xlabel,"t"],[ylabel,"p_t"],[legend,false])$



(%t15)

(%i16) wxplot2d([[discrete,map(lambda([u],part(u,[3,7])),rksol)], [discrete,[part(initialH,[2,6])]]],[a
[title,"Phase Space for r"],[point_type,circle], [style,[lines,2],[points,3]],[color,magenta,red]
[xlabel,"r"],[ylabel,"p_r"],[legend,false])$



(%t16)

(%i17) wxplot2d([[discrete,map(lambda([u],part(u,[4,8])),rksol)], [discrete,[part(initialH,[3,7])]]],[a
[title,"Phase Space for θ"],[point_type,circle], [style,[lines,2],[points,3]],[color,magenta,red]
[xlabel,"θ"],[ylabel,"p_θ"],[legend,false])$



(%t17)

(%i18) wxplot2d([[discrete,map(lambda([u],part(u,[5,9])),rksol)], [discrete,[part(initialH,[4,8])]]],[a
[title,"Phase Space for ϕ"],[point_type,circle], [style,[lines,2],[points,3]],[color,magenta,red]
[xlabel,"ϕ"],[ylabel,"p_ϕ"],[legend,false])$



(%t18)

(%i19) draw3d(title = "Barriola Vilenkin Geodesic", proportional_axes = xyz, axis_3d = false,
xlabel = "", ylabel = "", zlabel = "", dimensions = wxplot_size, view = [80,185],
file_name = "Barriola_Vilenkin_Geodesic2", terminal = 'pngcairo,
transform = [r*sin(θ)*cos(ϕ),r*sin(θ)*sin(ϕ),r*cos(θ),r,θ,ϕ],
color = blue, point_size = 1, point_type = -1, points_joined = true,
points(map(lambda([u],part(u,[3,4,5])),rksol)),
color = red, point_size = 1, point_type = circle, points_joined = false,
points([part(initialH,[2,3,4])]),
color = black, point_size = 2, point_type = filled_circle, points([[0,0,0]])),params$

(%i20) show_image("Barriola_Vilenkin_Geodesic2.png")$



(%t20)

Check Conservation of Energy using the Numerical Data

(%i21) P:map("=",funcs,initialH)$

(%i22) Y:H/m,P,params,numer;

$$-1.0 \tag{Y}$$

(%i23) wxplot2d([discrete,makelist([first(rkline), ev(H/m,map("=",funcs,rest(rkline)))],rkline,rksol)],
[xlabel,"s"],[ylabel,"Energy/mass"],[y,Y-0.001,Y+0.001]),params$



(%t23)

3 Using ctensor

(%i24) `kill(labels)$`

(%i1) `if get('itensor,'version)=false then load(itensor)$`

(%i2) `imetric(g)$`

(%i3) `if get('ctensor,'version)=false then load(ctensor)$`

(%i4) `dim:length(ct_coords)$`

(%i10)
```
ctrgsimp:true$
ratchristof:true$
ratriemann:true$
rateinstein:true$
ratweyl:true$
ratfac:true$
```

(%i11) `cmetric()$`

Covariant Metric tensor

(%i12) `ishow(g([μ,ν],[])=lg)$`

$$g_{\mu\nu} = \begin{pmatrix} -c^2 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & K^2\,r^2 & 0 \\ 0 & 0 & 0 & K^2\,r^2\sin(\theta)^2 \end{pmatrix} \tag{\%t12}$$

Contravariant Metric tensor

(%i13) `ishow(g([],[μ,ν])=ug)$`

$$g^{\mu\nu} = \begin{pmatrix} -\frac{1}{c^2} & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \frac{1}{K^2\,r^2} & 0 \\ 0 & 0 & 0 & \frac{1}{K^2\,r^2\sin(\theta)^2} \end{pmatrix} \tag{\%t13}$$

Christoffel Symbol of the first kind

(%i15)
```
christof(false)$
for i thru dim do for j:i thru dim do for k thru dim do
if lcs[i,j,k]≠0 then
ishow(Γ([ct_coords[i],ct_coords[j],ct_coords[k]],[])=lcs[i,j,k])$
```

$$\Gamma_{r\theta\theta} = K^2 r \tag{\%t15}$$

$$\Gamma_{r\phi\phi} = K^2 r\sin(\theta)^2 \tag{\%t15}$$

$$\Gamma_{\theta\theta r} = -K^2 r \tag{\%t15}$$

$$\Gamma_{\theta\phi\phi} = K^2\,r^2\,\cos(\theta)\,\sin(\theta) \tag{\%t15}$$

$$\Gamma_{\phi\phi r} = -K^2 r\sin(\theta)^2 \tag{\%t15}$$

$$\Gamma_{\phi\phi\theta} = -K^2\,r^2\,\cos(\theta)\,\sin(\theta) \tag{\%t15}$$

Christoffel Symbol of the second kind

(%i17)
```
christof(false)$
for i thru dim do for j:i thru dim do for k thru dim do
if mcs[i,j,k]≠0 then
ishow(Γ([ct_coords[i],ct_coords[j]],[ct_coords[k]])=mcs[i,j,k])$
```

$$\Gamma_{r\theta}^{\theta} = \frac{1}{r} \tag{\%t17}$$

$$\Gamma_{r\phi}^{\phi} = \frac{1}{r} \tag{\%t17}$$

$$\Gamma_{\theta\theta}^{r} = -K^2 r \tag{\%t17}$$

$$\Gamma_{\theta\phi}^{\phi} = \frac{\cos(\theta)}{\sin(\theta)} \tag{\%t17}$$

$$\Gamma_{\phi\phi}^{r} = -K^2 r \sin(\theta)^2 \tag{\%t17}$$

$$\Gamma_{\phi\phi}^{\theta} = -\cos(\theta)\,\sin(\theta) \tag{\%t17}$$

Covariant Divergence

(%i18)
```
checkdiv(lg)$
```

$$0$$

$$-\frac{K^2\left(r^2\sin(\theta)^2 + r^2\right) - 2}{r}$$

$$-\frac{K^2\,r^2\,\cos(\theta)\left(\sin(\theta)^2 - 1\right)}{\sin(\theta)}$$

$$0$$

(%i20)
```
trigsimp(arrayapply(div,[2]));
trigsimp(arrayapply(div,[3]));
```

$$\frac{K^2\left(r^2\cos(\theta)^2 - 2r^2\right) + 2}{r} \tag{\%o19}$$

$$\frac{K^2\,r^2\cos(\theta)^3}{\sin(\theta)} \tag{\%o20}$$

Covariant Gradient of a scalar function

(%i21)
```
depends(ρ,r)$
```

(%i23)
```
cograd(ρ,g1)$
listarray(g1);
```

$$[0, \rho_r, 0, 0] \tag{\%o23}$$

Contravariant Gradient of a scalar function

(%i25)
```
contragrad(ρ,g2)$
listarray(g2);
```

$$[0, \rho_r, 0, 0] \tag{\%o25}$$

d'Alembertian of a scalar function

(%i26)
```
collectterms(expand(dscalar(ρ)),diff(ρ,r,2),diff(ρ,r));
```

$$\rho_{rr} + \frac{2\,(\rho_r)}{r} \tag{\%o26}$$

Riemann Tensor

(%i30)
```
riemann(false)$
lriemann(false)$
uriemann(false)$
for a thru dim do for b thru dim do
for c thru (if symmetricp(lg,dim) then b else dim) do
for d thru (if symmetricp(lg,dim) then a else dim) do
if riem[a,b,c,d]≠0 then
ishow(R([ct_coords[a],ct_coords[b],ct_coords[c]],[ct_coords[d]])=riem[a,b,c,d])$
```

$$R^{\theta}_{\phi\phi\theta} = -(K-1)\,(K+1)\,\sin(\theta)^2 \tag{\%t30}$$

(%i31)
```
for a thru dim do for b thru dim do
for c thru (if symmetricp(lg,dim) then b else dim) do
for d thru (if symmetricp(lg,dim) then a else dim) do
if lriem[a,b,c,d]≠0 then
ishow(R([ct_coords[d],ct_coords[a],ct_coords[b],ct_coords[c]],[])=lriem[a,b,c,d])$
```

$$R_{\theta\phi\phi\theta} = -(K-1)\,(K+1)\,K^2\,r^2\sin(\theta)^2 \tag{\%t31}$$

Ricci Tensor

(%i35)
```
ric:zeromatrix(dim,dim)$
ricci(false)$
uricci(false)$
for i thru dim do for j:i thru dim do
if ric[i,j]≠0 then
ishow(R([ct_coords[i],ct_coords[j]])=ric[i,j])$
```

$$R_{\theta\theta} = -(K-1)\,(K+1) \tag{\%t35}$$

$$R_{\phi\phi} = -(K-1)\,(K+1)\,\sin(\theta)^2 \tag{\%t35}$$

Scalar curvature

(%i36)
```
factor(radcan(scurvature()));
```

$$-\frac{2\,(K-1)\,(K+1)}{K^2\,r^2} \tag{\%o36}$$

Compute trace of Ricci tensor

(%i37) `factor(radcan(tracer));`

$$-\frac{2\,(K-1)\,(K+1)}{K^2\,r^2} \tag{\%o37}$$

Kretschmann invariant

(%i38) `factor(radcan(rinvariant()));`

$$\frac{4(K-1)^2\,(K+1)^2}{K^4\,r^4} \tag{\%o38}$$

Einstein Tensor

(%i42)
```
lein:zeromatrix(dim,dim)$
einstein(false)$
leinstein(false)$
for i thru dim do for j:i thru dim do
if lein[i,j]≠0 then
ishow(G([ct_coords[i],ct_coords[j]],[])=lein[i,j])$
```

$$G_{tt} = -\frac{c^2\,(K-1)\,(K+1)}{K^2\,r^2} \tag{\%t42}$$

$$G_{rr} = \frac{(K-1)\,(K+1)}{K^2\,r^2} \tag{\%t42}$$

Weyl Conformal tensor

(%i43) `kill(W)$`

(%i45)
```
weyl(false)$
for i thru dim do
for j from (if symmetricp(lg,dim) then i+1 else 1) thru dim do
for k from (if symmetricp(lg,dim) then i else 1) thru dim do
for l from (if symmetricp(lg,dim) then k+1 else 1) thru (if (symmetricp(lg,dim) and k=i)
then j else dim) do
if weyl[i,j,k,l]≠0 then
ishow(W([ct_coords[i],ct_coords[j],ct_coords[k],ct_coords[l]],[])=weyl[i,j,k,l])$
```

$$W_{trtr} = -\frac{c^2\,(K-1)\,(K+1)}{3K^2\,r^2} \tag{\%t45}$$

$$W_{t\theta t\theta} = \frac{c^2\,(K-1)\,(K+1)}{6} \tag{\%t45}$$

$$W_{t\phi t\phi} = \frac{c^2\,(K-1)\,(K+1)\,\sin(\theta)^2}{6} \tag{\%t45}$$

$$W_{r\theta r\theta} = -\frac{(K-1)\,(K+1)}{6} \tag{\%t45}$$

$$W_{r\phi r\phi} = -\frac{(K-1)\,(K+1)\,\sin(\theta)^2}{6} \tag{\%t45}$$

$$W_{\theta\phi\theta\phi} = \frac{(K-1)\ (K+1)\ K^2\, r^2 \sin(\theta)^2}{3} \tag{\%t45}$$

Geodesics

(%i46) `cgeodesic(false)$`

(%i47) `for i thru dim do ldisplay(geod[i]:radcan(geod[i]))$`

$$geod_1 = T_s \tag{\%t47}$$

$$geod_2 = -K^2 r\, \Phi^2 \sin(\theta)^2 - K^2 r\, \Theta^2 + R_s \tag{\%t48}$$

$$geod_3 = -\frac{r\, \Phi^2\, \cos(\theta)\, \sin(\theta) - r\, (\Theta_s) - 2R\Theta}{r} \tag{\%t49}$$

$$geod_4 = \frac{(r\, (\Phi_s) + 2R\Phi)\, \sin(\theta) + 2r\,\Theta\Phi\, \cos(\theta)}{r\, \sin(\theta)} \tag{\%t50}$$

Solve for second derivative of coordinates

(%i51) `Eq:makelist(Eq[i],i,1,dim)$`

(%i52) `for i thru dim do Eq[i]:first(linsolve(geod[i],diff(ct_coords[i],s,2)))$`

(%i53) `map(ldisp,radcan(Eq))$`

$$T_s = 0 \tag{\%t53}$$

$$R_s = K^2 r\, \Phi^2 \sin(\theta)^2 + K^2 r\, \Theta^2 \tag{\%t54}$$

$$\Theta_s = \frac{r\, \Phi^2\, \cos(\theta)\, \sin(\theta) - 2R\Theta}{r} \tag{\%t55}$$

$$\Phi_s = -\frac{2R\Phi\, \sin(\theta) + 2r\,\Theta\Phi\, \cos(\theta)}{r\, \sin(\theta)} \tag{\%t56}$$