Green's Theorem

Written by Daniel Volinski at danielvolinski@yahoo.es

(%i2) `info:build_info()$info@version;`

(%o2)

5.38.1

(%i2) `reset()$kill(all)$`

(%i1) `derivabbrev:true$`

(%i2) `ratprint:false$`

(%i3) `fpprintprec:5$`

(%i4) `load(linearalgebra)$`

(%i5) `if get('draw,'version)=false then load(draw)$`

(%i6) `wxplot_size:[1024,768]$`

(%i7) `if get('drawdf,'version)=false then load(drawdf)$`

(%i8) `set_draw_defaults(xtics=1,ytics=1,ztics=1,xyplane=0,nticks=100)$`

(%i9) `if get('vect,'version)=false then load(vect)$`

(%i10) `norm(u):=block(ratsimp(radcan(√(u.u))))$`

(%i11) `normalize(v):=block(v/norm(v))$`

(%i12) `angle(u,v):=block([junk:radcan(√((u.u)*(v.v)))],acos(u.v/junk))$`

(%i13) `mycross(va,vb):=[va[2]*vb[3]-va[3]*vb[2],va[3]*vb[1]-va[1]*vb[3],va[1]*vb[2]-va[2]*vb[1]]$`

(%i14) `if get('cartan,'version)=false then load(cartan)$`

(%i15) `declare(trigsimp,evfun)$`

1 Green's Theorem

Based on Michael Penn Video Green's Theorem

Let C be a positively oriented, piecewise smooth, simple closed curve bounding the region D. If P and Q have continuous partial derivatives on an open region containing D then,

$$\oint_C P\,\mathrm{d}x + Q\,\mathrm{d}y = \iint_D \left(\frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y}\right)\mathrm{d}A$$

2 Verification Example 1

Based on Michael Penn Video Verification Example 1

Calculate $\oint_C x\,\mathrm{d}x + y\,\mathrm{d}y$ where C is the line segment $(0,1) \to (0,0) \to (1,0)$, parabola $y = 1 - x^2$ $(1,0) \to (0,1)$

(%i16) `kill(labels,t,x,y,z)$`

Define the space $\mathbb{R}^2$

(%i1) `ζ:[x,y]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^2$

(%i4) `ldisplay(F:[x,y])$`

$$F = [x, y] \tag{\%t4}$$

2D Direction field

(%i5) `wxdrawdf(F,[x,-5,5],[y,-5,5])$`



(%t5)

$\nabla \times \vec{F} \in \mathbb{R}^2$

(%i6) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = 0 \tag{\%t6}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i7) `ldisplay(α:F.cartan_basis)$`

$$\alpha = y\ dy + x\ dx \tag{\%t7}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^2)$

(%i8) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = 0 \tag{\%t8}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i9) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = 2 \tag{\%t9}$$

Flux form $\beta \in \mathcal{A}^1(\mathbb{R}^2)$

(%i10) `ldisplay(β:F[1]*cartan_basis[2]-F[2]*cartan_basis[1])$`

$$\beta = x\ dy - y\ dx \tag{\%t10}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^2)$

(%i11) `ldisplay(dβ:ext_diff(β))$`

$$d\beta = 2dx\ dy \tag{\%t11}$$

(%i12) `dβ/apply("*",cartan_basis);`

$$2 \tag{\%o12}$$

End points

(%i15) `A:[0,1]$B:[0,0]$C:[1,0]$`

Curve $\vec{C}_1 \in \mathbb{R}^2$

(%i16) `ldisplay(C_1:t*B+(1-t)*A)$`

$$C_1 = [0, 1-t] \tag{\%t16}$$

Derivative of the curve $\vec{C}_1$

(%i17) `ldisplay(C\'_1:diff(C_1,t))$`

$$C'_1 = [0, -1] \tag{\%t17}$$

$\vec{F} \circ \vec{C}_1$

(%i18) `ldisplay(FoC_1:subst(map("=",ζ,C_1),F))$`

$$FoC_1 = [0, 1-t] \tag{\%t18}$$

$\vec{F} \cdot \vec{C}_1' \in \mathbb{R}$

(%i19) `ldisplay(T_1:FoC_1.C\'_1)$`

$$T_1 = t - 1 \tag{\%t19}$$

Pullback $\vec{C}_1^* \alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i20) `ldisplay(P_1:C\'_1|subst(map("=",ζ,C_1),α))$`

$$P_1 = t - 1 \tag{\%t20}$$

Line integral I_1

(%i21) `I_1:'integrate(T_1,t,0,1)$`

(%i22) `ldisplay(I_1=box(ev(I_1,integrate)))$`

$$\int_0^1 t - 1 dt = \left(-\frac{1}{2}\right) \tag{\%t22}$$

Curve $\vec{C}_2 \in \mathbb{R}^2$

(%i23) `ldisplay(C_2:t*C+(1-t)*B)$`

$$C_2 = [t, 0] \tag{\%t23}$$

Derivative of the curve $\vec{C}_2$

(%i24) `ldisplay(C\'_2:diff(C_2,t))$`

$$C'_2 = [1, 0] \tag{\%t24}$$

$\vec{F} \circ \vec{C}_2$

(%i25) `ldisplay(FoC_2:subst(map("=",ζ,C_2),F))$`

$$FoC_2 = [t, 0] \tag{\%t25}$$

$\vec{F} \cdot \vec{C}'_2 \in \mathbb{R}$

(%i26) `ldisplay(T_2:FoC_2.C\'_2)$`

$$T_2 = t \tag{\%t26}$$

Pullback $\vec{C}_2^* \alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i27) `ldisplay(P_2:C\'_2|subst(map("=",ζ,C_2),α))$`

$$P_2 = t \tag{\%t27}$$

Line integral I_2

(%i28) `I_2:'integrate(T_2,t,0,1)$`

(%i29) `ldisplay(I_2=box(ev(I_2,integrate)))$`

$$\int_0^1 t dt = \left(\frac{1}{2}\right) \tag{\%t29}$$

Curve $\vec{C}_3 \in \mathbb{R}^2$

(%i30) `ldisplay(C_3:[-t,1-t^2])$`

$$C_3 = [-t, 1 - t^2] \tag{\%t30}$$

Derivative of the curve $\vec{C}_3$

(%i31) `ldisplay(C\'_3:diff(C_3,t))$`

$$C'_3 = [-1, -2t] \tag{\%t31}$$

$\vec{F} \circ \vec{C}_3$

(%i32) `ldisplay(FoC_3:subst(map("=",ζ,C_3),F))$`

$$FoC_3 = [-t, 1 - t^2] \tag{\%t32}$$

$\vec{F} \cdot \vec{C}'_3 \in \mathbb{R}$

(%i33) `ldisplay(T_3:expand(FoC_3.C\'_3))$`

$$T_3 = 2t^3 - t \tag{\%t33}$$

Pullback $\vec{C}_3^* \alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i34) `ldisplay(P_3:C\'_3|subst(map("=",ζ,C_3),α))$`

$$P_3 = 2t^3 - t \tag{\%t34}$$

Line integral I_3

(%i35) `I_3:'integrate(T_3,t,0,1)$`

(%i36) `ldisplay(I_3=box(ev(I_3,integrate)))$`

$$\int_0^1 2t^3 - t dt = (0) \tag{\%t36}$$

Total line integral $I_1 + I_2 + I_3$

(%i37) `ldisplay(I_1+I_2+I_3=box(ev(I_1+I_2+I_3,integrate)))$`

$$\int_0^1 2t^3 - t dt + \int_0^1 t dt + \int_0^1 t - 1 dt = (0) \tag{\%t37}$$

Use Green's Theorem

(%i38)
```
I:'integrate('integrate(curlF,y,0,1-x^2),x,0,1);
```

$$0 \tag{I}$$

Graphics

(%i39)
```
wxdraw2d(proportional_axes=xy,xrange=[-0.2,1.2],yrange=[-0.2,1.2],
color=red,apply(parametric,append(C_1,[t,0,1])),
color=blue,apply(parametric,append(C_2,[t,0,1])),
color=green,apply(parametric,append(C_3,[t,-1,0])),
color=black,label(["C_1",-0.05,0.5]),
label(["C_2",0.5,-0.05],["C_3",0.65,0.65]),
color=black,label(["(0,1)",-0.05,1]),
label(["(0,0)",-0.05,-0.05],["(1,0)",1,-0.05]))$
```



(%t39)

3 Verification Example 2

Based on Michael Penn Video Verification Example 2

Verify Green's theorem where:

$$D = \left\{(x,y) \mid 1 \le x^2 + y^2 \le 4\right\} \quad , \quad \oint_C xy^2\,\mathrm{d}y - x^2y\,\mathrm{d}x$$

(%i40) `kill(labels,t,x,y,z)$`

Define the space $\mathbb{R}^2$

(%i1) `ζ:[x,y]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^2$

(%i4) `ldisplay(F:[-x^2*y,x*y^2])$`

$$F = [-x^2 y, x\, y^2] \tag{\%t4}$$

2D Direction field

(%i5) `wxdrawdf(F,[x,-5,5],[y,-5,5])$`



(%t5)

$\nabla \times \vec{F} \in \mathbb{R}^2$

(%i6) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = y^2 + x^2 \qquad (\%t6)$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i7) `ldisplay(α:F.cartan_basis)$`

$$\alpha = x\,y^2\;dy - x^2 y\;dx \qquad (\%t7)$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^2)$

(%i8) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = \left(y^2 + x^2\right)\;dx\;dy \qquad (\%t8)$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i9) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = 0 \qquad (\%t9)$$

Flux form $\beta \in \mathcal{A}^1(\mathbb{R}^2)$

(%i10) `ldisplay(β:F[1]*cartan_basis[2]-F[2]*cartan_basis[1])$`

$$\beta = -x^2 y\;dy - x\,y^2\;dx \qquad (\%t10)$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^2)$

(%i11) `ldisplay(dβ:ext_diff(β))$`

$$d\beta = 0 \qquad (\%t11)$$

(%i12) `dβ/apply("*",cartan_basis);`

$$0 \qquad (\%o12)$$

Curve $\vec{C}_1 \in \mathbb{R}^2$

(%i13) `ldisplay(C_1:[2*cos(t),2*sin(t)])$`

$$C_1 = [2\cos(t), 2\sin(t)] \tag{\%t13}$$

Derivative of the curve $\vec{C}_1$

(%i14) `ldisplay(C\'_1:diff(C_1,t))$`

$$C'_1 = [-2\sin(t), 2\cos(t)] \tag{\%t14}$$

$\vec{F} \circ \vec{C}_1$

(%i15) `ldisplay(FoC_1:subst(map("=",ζ,C_1),F))$`

$$FoC_1 = [-8\cos(t)^2 \sin(t), 8\cos(t)\sin(t)^2] \tag{\%t15}$$

$\vec{F} \cdot \vec{C}_1' \in \mathbb{R}$

(%i16) `ldisplay(T_1:FoC_1.C\'_1)$`

$$T_1 = 32\cos(t)^2 \sin(t)^2 \tag{\%t16}$$

Pullback $\vec{C}_1^* \alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i17) `ldisplay(P_1:C\'_1|subst(map("=",ζ,C_1),α))$`

$$P_1 = 32\cos(t)^2 \sin(t)^2 \tag{\%t17}$$

Line integral I_1

(%i18) `I_1:'integrate(T_1,t,0,2*π)$`

(%i19) `ldisplay(I_1=box(ev(I_1,integrate)))$`

$$32\int_0^{2\pi} \cos(t)^2 \sin(t)^2 dt = \boxed{8\pi} \tag{\%t19}$$

Curve $\vec{C}_2 \in \mathbb{R}^2$

(%i20) `ldisplay(C_2:[cos(t),sin(t)])$`

$$C_2 = [\cos(t), \sin(t)] \tag{\%t20}$$

Derivative of the curve $\vec{C}_2$

(%i21) `ldisplay(C\'_2:diff(C_2,t))$`

$$C'_2 = [-\sin(t), \cos(t)] \tag{\%t21}$$

$\vec{F} \circ \vec{C}_2$

(%i22) `ldisplay(FoC_2:subst(map("=",ζ,C_2),F))$`

$$FoC_2 = [-\cos(t)^2 \sin(t), \cos(t)\sin(t)^2] \tag{\%t22}$$

$\vec{F} \cdot \vec{C}'_2 \in \mathbb{R}$

(%i23) `ldisplay(T_2:FoC_2.C\'_2)$`

$$T_2 = 2\cos(t)^2 \sin(t)^2 \tag{\%t23}$$

Pullback $\vec{C}_2^* \alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i24) `ldisplay(P_2:C\'_2|subst(map("=",ζ,C_2),α))$`

$$P_2 = 2\cos(t)^2 \sin(t)^2 \tag{\%t24}$$

Line integral I_2

(%i25) `I_2:'integrate(T_2,t,0,2*π)$`

(%i26) `ldisplay(I_2=box(ev(I_2,integrate)))$`

$$2\int_0^{2\pi} \cos(t)^2 \sin(t)^2 dt = \left(\frac{\pi}{2}\right) \tag{\%t26}$$

Total line integral $I_2 - I_1$

(%i27) `ldisplay(I_1-I_2=box(ev(I_1-I_2,integrate)))$`

$$30\int_0^{2\pi} \cos(t)^2 \sin(t)^2 dt = \left(\frac{15\pi}{2}\right) \tag{\%t27}$$

Graphics

(%i28)
```
wxdraw2d(proportional_axes=xy,xrange=[-2.5,2.5],yrange=[-2.5,2.5],
color=red,apply(parametric,append(C_1,[t,0,2*π])),
color=blue,apply(parametric,append(C_2,[t,0,2*π])),
color=black,label(["C_1",1.5,1.5],["C_2",0.8,0.8]))$
```



(%t28)

Use Green's Theorem

Change to polar coordinates

(%i30)
```
assume(r≥0)$
assume(0≤t,t≤2*π)$
```

(%i31)
```
ξ:[r,t]$
```

(%i32)
```
ldisplay(Tr:[r*cos(t),r*sin(t)])$
```

$$Tr = [r\,\cos(t), r\,\sin(t)] \tag{\%t32}$$

(%i33)
```
ldisplay(J:jacobian(Tr,ξ))$
```

$$J = \begin{pmatrix} \cos(t) & -r\,\sin(t) \\ \sin(t) & r\,\cos(t) \end{pmatrix} \tag{\%t33}$$

(%i34)
```
ldisplay(lg:trigsimp(transpose(J).J))$
```

$$lg = \begin{pmatrix} 1 & 0 \\ 0 & r^2 \end{pmatrix} \tag{\%t34}$$

(%i35)
```
ldisplay(Jdet:trigsimp(determinant(J)))$
```

$$Jdet = r \tag{\%t35}$$

$\left(\nabla \times \vec{F}\right) \circ \vec{Tr} \in \mathbb{R}$

(%i36)
```
ldisplay(T:trigsimp(subst(map("=",ζ,Tr),curlF)))$
```

$$T = r^2 \tag{\%t36}$$

Pullback $\vec{Tr}^{\,*}\,\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^2)$

(%i37)
```
ldisplay(P:trigsimp(diff(Tr,t)|(diff(Tr,r)|subst(map("=",ζ,Tr),dα))))$
```

$$P = r^3 \tag{\%t37}$$

Surface integral

(%i38)
```
I:'integrate('integrate(T*Jdet,r,1,2),t,0,2*π)$
```

(%i39)
```
ldisplay(I=box(ev(I,integrate)))$
```

$$2\pi \int_1^2 r^3 dr = \left(\frac{15\pi}{2}\right) \tag{\%t39}$$

Clean up

(%i41)
```
forget(r≥0)$
forget(0≤t,t≤2*π)$
```

4 Two more Green's theorem examples

Based on Michael Penn Video Two more Green's theorem examples

C is a right angle triangle with vertices $(-1,2),(4,2),(4,5)$
Calculate

$$\oint_C \sin(x^2)\,\mathrm{d}x + (3x-y)\,\mathrm{d}y$$

(%i42) `kill(labels,t,x,y,z)$`

Define the space $\mathbb{R}^2$

(%i1) `ζ:[x,y]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^2$

(%i4) `ldisplay(F:[sin(x^2),3*x-y])$`

$$F = [\sin\left(x^2\right), 3x - y] \tag{\%t4}$$

2D Direction field

(%i5) `wxdrawdf(F,[x,-5,5],[y,-5,5])$`



(%t5)

$\nabla \times \vec{F} \in \mathbb{R}^2$

(%i6) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = 3 \tag{\%t6}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i7) `ldisplay(α:F.cartan_basis)$`

$$\alpha = (3x - y)\ dy + \sin\left(x^2\right)\ dx \tag{\%t7}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^2)$

(%i8) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = 3dx\ dy \tag{\%t8}$$

(%i9) `dα/apply("*",cartan_basis);`

$$3 \tag{\%o9}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i10) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = 2x\ \cos\left(x^2\right) - 1 \tag{\%t10}$$

Flux form $\beta \in \mathcal{A}^1(\mathbb{R}^2)$

(%i11) `ldisplay(β:F[1]*cartan_basis[2]-F[2]*cartan_basis[1])$`

$$\beta = \sin\left(x^2\right) dy - (3x - y)\ dx \tag{\%t11}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^2)$

(%i12) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = \left(2x\ \cos\left(x^2\right) - 1\right)\ dx\ dy \tag{\%t12}$$

(%i13) `dβ/apply("*",cartan_basis);`

$$2x\ \cos\left(x^2\right) - 1 \tag{\%o13}$$

End points

(%i16) `A:[-1,2]$B:[4,2]$C:[4,5]$`

Curve $\vec{C}_1 \in \mathbb{R}^2$

(%i17) `ldisplay(C_1:ratsimp(t*B+(1-t)*A))$`

$$C_1 = [5t-1, 2] \tag{\%t17}$$

Derivative of the curve $\vec{C}_1$

(%i18) `ldisplay(C\'_1:diff(C_1,t))$`

$$C'_1 = [5, 0] \tag{\%t18}$$

$\vec{F} \circ \vec{C}_1$

(%i19) `ldisplay(FoC_1:subst(map("=",ζ,C_1),F))$`

$$FoC_1 = [\sin\left((5t-1)^2\right), 3(5t-1)-2] \tag{\%t19}$$

$\vec{F} \cdot \vec{C}_1' \in \mathbb{R}$

(%i20) `ldisplay(T_1:expand(FoC_1.C\'_1))$`

$$T_1 = 5\sin\left(25t^2-10t+1\right) \tag{\%t20}$$

Pullback $\vec{C}_1^{*}\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i21) `ldisplay(P_1:C\'_1|subst(map("=",ζ,C_1),α))$`

$$P_1 = 5\sin\left(25t^2-10t+1\right) \tag{\%t21}$$

Line integral I_1

(%i22) `I_1:'integrate(T_1,t,0,1)$`

Curve $\vec{C}_2 \in \mathbb{R}^2$

(%i23) `ldisplay(C_2:ratsimp(t*C+(1-t)*B))$`

$$C_2 = [4, 3t + 2] \tag{\%t23}$$

Derivative of the curve $\vec{C}_2$

(%i24) `ldisplay(C\'_2:diff(C_2,t))$`

$$C'_2 = [0, 3] \tag{\%t24}$$

$\vec{F} \circ \vec{C}_2$

(%i25) `ldisplay(FoC_2:subst(map("=",ζ,C_2),F))$`

$$FoC_2 = [\sin(16), 10 - 3t] \tag{\%t25}$$

$\vec{F} \cdot \vec{C}_2' \in \mathbb{R}$

(%i26) `ldisplay(T_2:expand(FoC_2.C\'_2))$`

$$T_2 = 30 - 9t \tag{\%t26}$$

Pullback $\vec{C}_2^* \alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i27) `ldisplay(P_2:C\'_2|subst(map("=",ζ,C_2),α))$`

$$P_2 = 30 - 9t \tag{\%t27}$$

Line integral I_2

(%i28) `I_2:'integrate(T_2,t,0,1)$`

(%i29) `ldisplay(I_2=box(ev(I_2,integrate)))$`

$$\int_0^1 30 - 9t dt = \left(\frac{51}{2}\right) \tag{\%t29}$$

Curve $\vec{C}_3 \in \mathbb{R}^2$

(%i30)
```
ldisplay(C_3:ratsimp(t*A+(1-t)*C))$
```

$$C_3 = [4 - 5t, 5 - 3t] \quad (\%t30)$$

Derivative of the curve $\vec{C}_3$

(%i31)
```
ldisplay(C\'_3:diff(C_3,t))$
```

$$C'_3 = [-5, -3] \quad (\%t31)$$

$\vec{F} \circ \vec{C}_3$

(%i32)
```
ldisplay(FoC_3:ratsimp(subst(map("=",ζ,C_3),F)))$
```

$$FoC_3 = \left[\sin\left(25t^2 - 40t + 16\right), 7 - 12t\right] \quad (\%t32)$$

$\vec{F} \cdot \vec{C}'_3 \in \mathbb{R}$

(%i33)
```
ldisplay(T_3:expand(FoC_3.C\'_3))$
```

$$T_3 = -5\sin\left(25t^2 - 40t + 16\right) + 36t - 21 \quad (\%t33)$$

Pullback $\vec{C}_3^* \alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i34)
```
ldisplay(P_3:C\'_3|subst(map("=",ζ,C_3),α))$
```

$$P_3 = -5\sin\left(25t^2 - 40t + 16\right) + 36t - 21 \quad (\%t34)$$

Line integral I_3

(%i35)
```
I_3:'integrate(T_3,t,0,1)$
```

Total line integral $I_1 + I_2 + I_3$

(%i36)
```
ldisplay(I_1+I_2+I_3=box(ev(I_1+I_2+I_3,integrate,ratsimp)))$
```

$$5\int_0^1 \sin\left(25t^2 - 10t + 1\right)dt + \int_0^1 -5\sin\left(25t^2 - 40t + 16\right) + 36t - 21dt + \int_0^1 30 - 9tdt = \left(\frac{45}{2}\right) \quad (\%t36)$$

Use Green's Theorem

(%i37) rhs(first(solve(eliminate(map("=",ζ,C_3),[t]),y)));

$$\frac{3x+13}{5} \tag{\%o37}$$

(%i38) I:'integrate('integrate(curlF,y,2,%),x,-1,4)$

(%i39) ldisplay(I=box(ev(I,integrate)))$

$$3\int_{-1}^{4}\frac{3x+13}{5}-2dx = \boxed{\frac{45}{2}} \tag{\%t39}$$

Graphics

(%i40) wxdraw2d(proportional_axes=xy,xrange=[-2,5],yrange=[1,6],
color=red,apply(parametric,append(C_1,[t,0,1])),
color=blue,apply(parametric,append(C_2,[t,0,1])),
color=green,apply(parametric,append(C_3,[t,0,1])),
color=black,label(["C_1",1.5,1.9]), label(["C_2",4.1,3.5],["C_3",1.9,3.8]),
color=black,label(["(-1,2)",-1,1.9]), label(["(4,2)",4,1.9],["(4,5)",4,5.1]))$



(%t40)

Find area of ellipse $\frac{x^2}{a^2}+\frac{y^2}{b^2}=1$

(%i41) `kill(labels,a,b)$`

(%i1) `assume(a>0,b>0)$`

(%i2) `declare([a,b],constant)$`

(%i3) `E:x^2/a^2+y^2/b^2=1$`

(%i4) `sol:solve(E,y);`

$$\left[y=-\frac{b\sqrt{a^2-x^2}}{a}, y=\frac{b\sqrt{a^2-x^2}}{a}\right] \tag{sol}$$

(%i5)
```
I:'integrate('integrate(1,y,0,rhs(sol[2])),x,-a,a)+
'integrate('integrate(1,y,rhs(sol[1]),0),x,-a,a)$
```

(%i6) `ldisplay(I=box(ev(I,integrate)))$`

$$\frac{2b\int_{-a}^{a}\sqrt{a^2-x^2}dx}{a}=(\pi a b) \tag{\%t6}$$

Vector field $\vec{F}_1 \in \mathbb{R}^2$

(%i7) `ldisplay(F_1:[0,x])$`

$$F_1=[0,x] \tag{\%t7}$$

$\nabla \times \vec{F}_1 \in \mathbb{R}^2$

(%i8) `ldisplay(curlF_1:ev(express(curl(F_1)),diff))$`

$$curlF_1=1 \tag{\%t8}$$

Vector field $\vec{F}_2 \in \mathbb{R}^2$

(%i9) `ldisplay(F_2:[-y,0])$`

$$F_2=[-y,0] \tag{\%t9}$$

$\nabla \times \vec{F}_2 \in \mathbb{R}^2$

(%i10) `ldisplay(curlF_2:ev(express(curl(F_2)),diff))$`

$$curlF_2=1 \tag{\%t10}$$

Vector field $\vec{F}_3 \in \mathbb{R}^2$

(%i11) `ldisplay(F_3:`$\frac{1}{2}$`*(F_1+F_2))$`

$$F_3 = \left[-\frac{y}{2}, \frac{x}{2}\right] \tag{\%t11}$$

$\nabla \times \vec{F}_3 \in \mathbb{R}^2$

(%i12) `ldisplay(curlF_3:ev(express(curl(F_3)),diff))$`

$$curlF_3 = 1 \tag{\%t12}$$

Work form $\alpha_3 \in \mathcal{A}^1(\mathbb{R}^2)$

(%i13) `ldisplay(`α`_3:F_3.cartan_basis)$`

$$\alpha_3 = \frac{x\,dy}{2} - \frac{y\,dx}{2} \tag{\%t13}$$

$\mathrm{d}\alpha_3 \in \mathcal{A}^2(\mathbb{R}^2)$

(%i14) `ldisplay(d`α`_3:ext_diff(`α`_3))$`

$$d\alpha_3 = dx\,dy \tag{\%t14}$$

(%i15) `d`α`_3/apply("*",cartan_basis);`

$$1 \tag{\%o15}$$

$\nabla \cdot \vec{F}_3 \in \mathbb{R}$

(%i16) `ldisplay(divF_3:ev(express(div(F_3)),diff))$`

$$divF_3 = 0 \tag{\%t16}$$

Flux form $\beta_3 \in \mathcal{A}^1(\mathbb{R}^2)$

(%i17) `ldisplay(`β`_3:first(F_3)*cartan_basis[2]-second(F_3)*cartan_basis[1])$`

$$\beta_3 = -\frac{y\,dy}{2} - \frac{x\,dx}{2} \tag{\%t17}$$

$\mathrm{d}\beta_3 \in \mathcal{A}^2(\mathbb{R}^2)$

(%i18) `ldisplay(d`β`_3:edit(ext_diff(`β`_3)))$`

$$d\beta_3 = 0 \tag{\%t18}$$

(%i19) `d`β`_3/apply("*",cartan_basis);`

$$0 \tag{\%o19}$$

$$A(D) = \iint_D \mathrm{d}A = \frac{1}{2} \oint_C -y\,\mathrm{d}x + x\,\mathrm{d}y$$

Parametrize the ellipse

```
(%i20) assume(0≤t,t≤2*π)$
(%i21) ldisplay(r:[a*cos(t),b*sin(t)])$
```

$$r = [a\,\cos(t), b\,\sin(t)] \tag{\%t21}$$

Verify this is an ellipse

```
(%i22) is(trigsimp(subst(map("=",ζ,r),E)));
```

$$\text{true} \tag{\%o22}$$

Derivative of the curve $\vec{r}$

```
(%i23) ldisplay(r\':diff(r,t))$
```

$$r' = [-a\,\sin(t), b\,\cos(t)] \tag{\%t23}$$

$\vec{F}_3 \circ \vec{r}$

```
(%i24) ldisplay(For_3:subst(map("=",ζ,r),F_3))$
```

$$For_3 = \left[-\frac{b\,\sin(t)}{2}, \frac{a\,\cos(t)}{2}\right] \tag{\%t24}$$

$\vec{F}_3 \cdot \vec{r}' \in \mathbb{R}$

```
(%i25) ldisplay(T_3:trigsimp(For_3.r\'))$
```

$$T_3 = \frac{ab}{2} \tag{\%t25}$$

Pullback $\vec{r}^{*}\alpha_3 \in \mathcal{A}^1(\mathbb{R}^2)$

```
(%i26) ldisplay(P_3:trigsimp(r\'|subst(map("=",ζ,r),α_3)))$
```

$$P_3 = \frac{ab}{2} \tag{\%t26}$$

Line integral I_3

```
(%i27) ldisplay(I_3:box('integrate(T_3,t,0,2*π)))$
```

$$I_3 = \boxed{\pi ab} \tag{\%t27}$$

Clean up

```
(%i29) forget(a>0,b>0)$
       forget(0≤t,t≤2*π)$
```

5 Vector forms of Green's Theorem

Based on Michael Penn Video Vector forms of Green's Theorem

(%i30) `kill(labels,t,x,y,z,P,Q)$`

Define the space $\mathbb{R}^3$

(%i1) `ζ:[x,y,z]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^2$

(%i4)
```
ldisplay(F:[x^2*y,x*z,z^3])$
```

$$F = [x^2 y, xz, z^3] \tag{\%t4}$$

3D Direction field

(%i6)
```
/* vector origins are (x,y,z)| x,y=1,...,5 */
coord:setify(makelist(k,k,0,5))$
points3d:listify(cartesian_product(coord,coord,coord))$
```

(%i8)
```
/* compute vectors at the given points */
define(vf3d(x,y,z),vector(ζ,F))$
vect3:makelist(vf3d(k[1],k[2],k[3]),k,points3d)$
```

(%i9)
```
wxdraw3d([head_length=0.1,color=blue,head_angle=25,unit_vectors=true],vect3)$
```



(%t9)

$\nabla \times \vec{F} \in \mathbb{R}^3$

(%i10) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = [-x, 0, z - x^2] \tag{\%t10}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^3)$

(%i11) `ldisplay(α:F.cartan_basis)$`

$$\alpha = z^3\, dz + xz\, dy + x^2 y\, dx \tag{\%t11}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^3)$

(%i12) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = \left(z - x^2\right)\, dx\, dy - x\, dy\, dz \tag{\%t12}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i13) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = 3z^2 + 2xy \tag{\%t13}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i14)
```
ldisplay(β:F[1]*cartan_basis[2]~cartan_basis[3]+
F[2]*cartan_basis[3]~cartan_basis[1]+
F[3]*cartan_basis[1]~cartan_basis[2])$
```

$$\beta = x^2 y\, dy\, dz - xz\, dx\, dz + z^3\, dx\, dy \tag{\%t14}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^3)$

(%i15) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = \left(3z^2 + 2xy\right)\, dx\, dy\, dz \tag{\%t15}$$

(%i16) `dβ/apply("*",cartan_basis);`

$$3z^2 + 2xy \tag{\%o16}$$

Based on Opentextbc Website 41 Green's Theorem

Based on Openstax Website 6.4 Green's Theorem

Circulation Form of Green's Theorem

$$\oint_C \vec{F}\,\mathrm{d}\vec{r} = \oint_C P\,\mathrm{d}x + Q\,\mathrm{d}y = \iint_D (Q_x - P_y)\,\mathrm{d}A$$

Flux Form of Green's Theorem

$$\oint_C \vec{F}\cdot\vec{N}\,\mathrm{d}s = \iint_D (P_x + Q_y)\,\mathrm{d}A$$

Define the space $\mathbb{R}^2$

(%i17) `ζ:[x,y]$`

(%i18) `scalefactors(ζ)$`

(%i19) `init_cartan(ζ)$`

Vector field $\vec{F} \in \mathbb{R}^2$

(%i20) `F:[P,Q]$`

(%i21) `depends(F,ζ)$`

Version 1 of Green's theorem (Circulation Form)

$$\oint_C \vec{F}\,\mathrm{d}\vec{r} = \iint_D \left(\nabla \times \vec{F}\right)\,\mathrm{d}A$$

$\nabla \times \vec{F} \in \mathbb{R}^2$

(%i22) `ldisplay(curlF:ev(express(curl(F)),diff))$`

$$curlF = Q_x - P_y \tag{\%t22}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i23) `ldisplay(α:F.cartan_basis)$`

$$\alpha = Q\;dy + P\;dx \tag{\%t23}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^2)$

(%i24) `ldisplay(dα:edit(ext_diff(α)))$`

$$d\alpha = \left(Q_x - P_y\right)\;dx\;dy \tag{\%t24}$$

Version 2 of Green's theorem (Flux Form)

$$\oint_C \left(\vec{F}\cdot\hat{n}\right)\,\mathrm{d}s = \iint_D \left(\nabla \cdot \vec{F}\right)\,\mathrm{d}A$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

(%i25) `ldisplay(divF:ev(express(div(F)),diff))$`

$$divF = Q_y + P_x \tag{\%t25}$$

Flux form $\beta \in \mathcal{A}^2(\mathbb{R}^2)$

(%i26) `ldisplay(β:F[1]*cartan_basis[2]-F[2]*cartan_basis[1])$`

$$\beta = P\;dy - Q\;dx \tag{\%t26}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^2)$

(%i27) `ldisplay(dβ:edit(ext_diff(β)))$`

$$d\beta = \left(Q_y + P_x\right)\;dx\;dy \tag{\%t27}$$

(%i28) `dβ/apply("*",cartan_basis);`

$$Q_y + P_x \tag{\%o28}$$

6 When Green's theorem doesn't apply

Based on Michael Penn Video When Green's theorem doesn't apply

Find all possible values of $\oint_C \vec{F} \cdot \mathrm{d}\vec{r}$ where C satisfies all the conditions.

(%i29) `kill(labels,t,x,y)$`

Define the space $\mathbb{R}^2$

(%i1) `ζ:[x,y]$`

(%i2) `scalefactors(ζ)$`

(%i3) `init_cartan(ζ)$`

Parameters

(%i4) `assume(a>0)$`

(%i5) `declare(a,constant)$`

(%i6) `params:[a=1]$`

Vector field $\vec{F} \in \mathbb{R}^2$

(%i7) `ldisplay(F:1/(x^2+y^2)*[y,-x])$`

$$F = \left[\frac{y}{y^2+x^2}, -\frac{x}{y^2+x^2}\right] \tag{\%t7}$$

2D Direction field

(%i8) `wxdrawdf(F,[x,-5,5],[y,-5,5])$`



(%t8)

$\nabla \times \vec{F} \in \mathbb{R}^2$

```
(%i9) ldisplay(curlF:ratsimp(ev(express(curl(F)),diff)))$
```

$$curlF = 0 \tag{\%t9}$$

Potential

```
(%i10) ldisplay(ϕ:potential(F))$
```

$$\phi = \operatorname{atan}\left(\frac{x}{y}\right) \tag{\%t10}$$

Work form $\alpha \in \mathcal{A}^1(\mathbb{R}^2)$

```
(%i11) ldisplay(α:factor(F.cartan_basis))$
```

$$\alpha = -\frac{x\,dy - y\,dx}{y^2 + x^2} \tag{\%t11}$$

$\mathrm{d}\alpha \in \mathcal{A}^2(\mathbb{R}^2)$

```
(%i12) ldisplay(dα:ratsimp(edit(ext_diff(α))))$
```

$$d\alpha = 0 \tag{\%t12}$$

$\nabla \cdot \vec{F} \in \mathbb{R}$

```
(%i13) ldisplay(divF:ev(express(div(F)),diff))$
```

$$divF = 0 \tag{\%t13}$$

Flux form $\beta \in \mathcal{A}^1(\mathbb{R}^2)$

```
(%i14) ldisplay(β:factor(F[1]*cartan_basis[2]-F[2]*cartan_basis[1]))$
```

$$\beta = \frac{y\,dy + x\,dx}{y^2 + x^2} \tag{\%t14}$$

$\mathrm{d}\beta \in \mathcal{A}^2(\mathbb{R}^2)$

```
(%i15) ldisplay(dβ:edit(ext_diff(β)))$
```

$$d\beta = 0 \tag{\%t15}$$

Curve $\vec{r}_a \in \mathbb{R}^2$

(%i16) ldisplay(r_a:[a*cos(t),a*sin(t)])$

$$r_a = [a\,\cos(t), a\,\sin(t)] \quad (\%t16)$$

Derivative of the curve $\vec{r}_a$

(%i17) ldisplay(r\'_a:diff(r_a,t))$

$$r'_a = [-a\,\sin(t), a\,\cos(t)] \quad (\%t17)$$

$\vec{F} \circ \vec{r}_a$

(%i18) ldisplay(For_a:trigsimp(subst(map("=",ζ,r_a),F)))$

$$For_a = \left[\frac{\sin(t)}{a}, -\frac{\cos(t)}{a}\right] \quad (\%t18)$$

$\vec{F} \cdot \vec{r}'_a \in \mathbb{R}$

(%i19) ldisplay(T_a:trigsimp(For_a.r\'_a))$

$$T_a = -1 \quad (\%t19)$$

Pullback $\vec{r}_a^* \alpha \in \mathcal{A}^1(\mathbb{R}^2)$

(%i20) ldisplay(P_a:trigsimp(r\'_a|subst(map("=",ζ,r_a),α)))$

$$P_a = -1 \quad (\%t20)$$

Line integral I_a

(%i21) ldisplay(I_a:'integrate(T_a,t,0,2*π))$

$$I_a = -2\pi \quad (\%t21)$$

Clean up

(%i22) forget(a>0)$